
第1章 Maximaの処理原理について

この章で解説する事:

• Maxiaの順序

• Maximaの文脈

• 宣言と属性

• 演算子

• 規則と式の並び

• 評価の方法

• 式の評価

• LISPに関連する函数



2 第 1章 Maximaの処理原理について

1.1 順序について

数学には順序という考えがあります.順序は集合 Sの二つの元に対して成立する関係 (二項関係
と呼びます)の一つです.例えば,集合を実数とすると,大小関係 ≥ は順序関係の一例です.この大
小関係を一般化したものです.
ここで,集合 S の二項間の関係 ≥order が以下の性質を満す時に順序と呼び,順序の入った集合

S の事を順序集合と呼びます.
順序の定義¶ ³

反射律: x ≥order x

対称律: x ≥order y かつ y ≥order x ⇒ x = y

推移律: x ≥order y かつ y ≥order z ⇒ x ≥order zµ ´
集合 Sの順序 ≥order で,集合の任意の元 x, y に対し, x ≥order y ,或いは y ≥order x の何れか

が成立する場合, 順序 ≥order の事を全順序, 集合 S の事を全順序集合と呼びます.
例えば,集合を実数 R とすると,二項関係 ≥ は全順序, 集合 R が全順序集合となります.但し,

二項関係を包含関係 ⊇ とした場合,包含関係 ⊇ は集合 R の順序になりますが,全順序にはなりま
せん.何故なら,偶数全体の集合Aと奇数全体の集合 B を考えるとどうでしょうか？集合Aと Bの
間には包含関係がありませんね.全順序となる為には,包含関係が無ければなりません.従って,全
順序にはなりません.
数式処理の場合,集合は数,変数や函数で構築された式の集合を考えます.式は和や積の様に可換

な演算で結合された数,変数や函数の為,数と変数には変数の方が大きいと順序付け,変数よりも函
数が大きいと順序付け,変数同士にも順序を入れると,後は項の順序が残ります.

1変数の場合,単項式 xm と xn との順序を羃の次数 m と n の大小関係 ≥ で決めれば良いで
しょう.では,より一般の多変数の場合はどうすれば良いでしょうか？この場合,変数の間に順序を
入れ,その変数の順序で変数を入れ換えた項の次数リストで比較する方法があります.
例えば, x,y,z の 3変数多項式環 K[x, y, z] で,変数間の順番をアルファベット順で並べ, y z の様

に変数が抜けている場合,抜けている変数の次数を 0 とすれば,項は xi1yi2zi3 の形式になります.
これから長さ 3の次数のリスト (i1, i2, i3) が得られ,このリストと項は一対一に対応します.従っ
て, n 変数 x1, · · · , xn の単項式 A と B が与えられた時, n 個の変数 x1, · · · , xn の並びを固定しま

す.次に, 単項式 A と B の中の変数 xi が項に存在しない場合, x0
i を挿入します.こうする事で,

単項式 A と B を表現する長さが n の次数リスト (α1, · · · , αn) と (β1, · · · , βn) が一意に決ります.
後はこれらの次数リストに対して順序を入れれば良い事になります. ここで,項の順序としては,二
つの項 xα , xβ に対して, xα > xβ であれば,xγ を両辺にかけた場合でも, xα+γ > xβ+γ を満すも

のの方が都合が良いですね. この項に対する性質 x > y ⇔ x · a > y · a を満す順序の事を項順序と
呼びます.
では,先程の例の二つの次数リスト x2 y2 z(= (2, 2, 1)) と x y2 z3(= (1, 2, 3)) が得られた時,順序

はどの様に入れれば良いでしょうか.単純に x の多項式と見れば, x2 y2 z の方が大きく, 次に Z の

多項式と見れば, x y2 z3 の方が大きくなります. ところが, x = y = z として x の多項式として変

形すると, x y2 z3 の方が次数が大きくなります.この場合は,次数の大きさも含めて考えた方が良
さそうですね.この様に多項式の場合,項の順序には色々な考え方があります. 次の小節では代表的

な項の順序について説明しましょう.

1.1.1 色々な順序

項順序として代表的なものに,辞書式順序,斉次辞書式順序,逆辞書式順序,逆斉次辞書式順序,これ
らの順序に,変数の重みを加味したもの等があります. ここでは,基本的な辞書式順序,斉次辞書式順
序,逆辞書式順序と斉次逆辞書式順序について簡単に説明しましょう.尚,変数の並び順を x1, · · ·xn

とし, xa1
1 , · · ·xan

n の次数リストを a = (a1, · · · , an) xb1
1 , · · ·xbn

n の次数リストを b = (b1, · · · , bn) と
します.更に, |a| = a1 + · · · + an で次数リスト a に含まれる次数の総和を表記します.又,順序の
定義に含まれる > は通常の整数環 Z の元に対する大小関係です.

辞書式順序 >lex
¶ ³

a >lex b ⇔ a1 = b1 · · · ai = bi かつ ai+1 > bi+1µ ´
この関係で定められる順序を辞書式順序と呼びます.この順序を示す記号として, ここでは >lex

を使います.
辞書式順序では二つの項を比較した時に,左側の変数の次数が大きなものが大きな項となります.
例えば, x2 y2 z と x y2 z3 を各々表現する整数リスト (2, 2, 1) と (1, 2, 3) に対しては,先頭の 2

と 1 を比較した段階で, 2 > 1 から (2, 2, 1) >lex (1, 2, 3) , 即ち, x2 y2 z >lex x y2 z3 となります.

斉次辞書式順序 >glex
¶ ³

a >glex b ⇔

{
|a| > |b|または
a1 = b1, · · · , ai = bi, ai+1 > bi+1µ ´

この関係で定められる順序を斉次辞書式順序と呼びます.この順序を示す記号として,ここでは
>glex を使います.
この斉次辞書式順序の場合,総次数で最初に項を比較し,総次数が一致すれば, 今度は項を辞書式

順序で比較する二段方式となっています.
例えば, x2 y2 z と x y2 z3 を各々表現する整数リスト (2, 2, 1) と (1, 2, 3) に対し, |x2y2z| = 5 と

|xy2z3| = 6 となるので, (1, 2, 3) >glex (2, 2, 1) , 即ち, x y2 z3 >glex x2 y2 z となり,辞書式順序と
は逆の結果になります.

逆辞書式順序 >revlex
¶ ³

a >revlex b ⇔ an = bn · · · ai = bi かつ ai−1 < bi−1µ ´
この関係で定められる順序を逆辞書式順序と呼びます. この順序を示す記号として, ここでは

>revlex を使います.辞書式との違いは,先ず,調べる方向が逆で, 次数の小さい方を取る点で,逆に
なっています.
具体的には,二つの項 x3y2z3 と xy2z3 が与えられた場合、各々の項を表現する整数の列 (3, 2, 3)

と (1, 2, 3)が得られます.逆辞書式順序では,この列の後から比較を行います.すると,後から先に移動

して,リストの先頭の 3 と 1 に到達すると,定義から (1, 2, 3) >revlex (3, 2, 3) , 即ち, xy2z3 >revlex

x3y2z3 を得ます.

斉次逆辞書式順序 >grevlex
¶ ³

a >grevlex b ⇔

{
|a| > |b|または
an = bn, · · · , an = bn, ai−1 < bi−1µ ´

この関係で定められる順序を斉次逆辞書式順序と呼びます.この順序を示す記号として,ここで
は >grevlex を使います.
この順序は項の総次数で最初に項を比較し,総次数が等しければ,逆辞書式順序で項の順序を決

める二段方式のものです.
x2y2z と xy2z3 の場合、総次数が各々,5 と 6 になるので, xy2z3 >grevlex x2y2z となります.因

に,逆辞書式の場合は, 次数リストの右端の zの次数から見るので, x2y2z >revlex xy2z3 となり,斉
次逆辞書式順序とは逆の結果となります.
これらの項順序は多項式の計算で非常に重要です.特にGroebner基底の計算では, 項順序を用い

た計算を行います.この Groebner基底は項順序に対しては一意に決まりますが,順序を替えると,
普通は別の基底が得られます. この事もあって為,順序は新しい数式処理システムでは目的に応じ
て様々な順序が扱える様になっています.
尚,代数学の様々な定理は,対象に順序を入れる事で証明出来ものが多くあります. その意味で

も,項順序は非常に重要です.

1.2 多項式の表現

ここでは多項式の表現について考えてみましょう.
変数に順序を入れ,その順序で並べた項に対しても順序を入れました.これで項を順番に並べて

しまえば,与えられた数式の表記に対応する前置表現により,式が一意に決定します.但し,ここで
の前置表現は与えられた式そのものを単純に置換えるだけで, 関係 ∼ で同値な式が全て同じ前置
表現で置換えられる訳ではありません. その為,多項式は予め展開しておき,項毎に簡易化を行って
おく必要があります. この簡易化を行えば,少なくとも,変形操作による同値関係 ∼に関しては前
置表現による式と本来の式が一対一に対応します.
とは言え,内挿表現は無駄が多い表現です.これをより簡単にしたものに,正準表現というものが

あります.この正準表現を解説する為に,多項式 3x2 −1 を使って正準表現を構成を説明しましょう.
先ず,与式が何の変数の多項式であるかが重要ですね.その重要なものを内挿表現から前置表現

に変換する時と同じ様に先頭に変数 x を置きましょう. 次に,この式は 3x2 + (−1)x0 と同値にな

ります.するとこの式は x の多項式としては 2次の項と 0次の項があり,2次の項の係数は 3 で,0
次の係数の項は −1 ですね.そこで (x (2 3) (0 − 1)) と記述すればどうでしょう. 前置表現であれ
ば, (+ (∗ 3 (ˆ x 2)) − 1) となりますが,随分とすっきりしましたね.更に,内部の括弧は実は羃と
係数の対と決まっているので, (x 2 3 0 − 1) で十分ですね.すると,この表現 (x 2 3 0 − 1) と多
項式 3x2 + (−1)x0 は一対一で対応します.この様に,式と表現の関係が一対一になる表現を正準表
現と呼びます.

この方法を単変数多項式に適用すれば,以下の正準表現が得られます.

単変数多項式の正準表現¶ ³
(〈変数 〉 〈次数1〉 〈係数1〉 〈次数2〉 〈係数2〉 · · ·)µ ´
これで,一変数の場合は片付きました.では一般の多変数多項式の場合はどうでしょうか？実際,

多変数多項式 K[x1, · · · , xn] に対しても,同様の手法で正準表現が構成出来ます.但し,多変数多項
式の場合で重要な事は,項に順序を入れる事です.そこで,辞書順序 >lex を多項式環 K[x1, · · · , xn]
に入れてみましょう.
先ず,変数の並びを x1, · · · , xn で固定し,項 xα1

1 · · ·xαn
n を次数リスト (α1, · · · , αn) で表現します.

二つの項 xα1
1 · · ·xαn

n と xβ1
1 · · ·xβn

n の比較は,次数リストを用いて行います. この際,リストの左端か
ら順番に比較しますが, α1 = β1, · · · , αi−1 = βi−1 で, αi > βi の場合, xα1

1 · · ·xαn
n >lex xβ1

1 · · ·xβn
n

となります.
次に,多項式

∑
a(i1,···,in)x

i1
1 · · ·xin

n が与えられると,最初に x1 の多項式と看倣して,一変数の場
合と同様の考え方で表現を構成します.先ず,式に含まれる変数が x1 だけの場合は,一変数の方法
で正準表現が得られます.もし,x1 以外の変数が存在する場合は,最初に x1 を変数とする多項式と

看倣して式を纏めます.すると, x1 の各項の係数は 高々n − 1 変数の多項式 (∈ K[x1, · · · , xn−1])
となります. 今度は各係数に対し,同様の考え方で x2 の多項式表現を構成します. 以降,係数に対
して帰納的に処理を行う事で,以下の様な正準表現が得られます.

多変数多項式の正準表現¶ ³
(〈変数 〉 〈次数1〉 〈係数多項式の正準表現1〉 〈次数2〉 〈係数多項式の正準表現2〉 · · ·)µ ´
具体的に多項式環 Z[x, y] の元 y x + 2 x y3 − 3 に対し, 順序を辞書式順序 >lex で考えましょう.

尚,変数の並びは x, y とします. この場合,最初に与式を x の多項式と看倣し,変数 x で式を纏め

ましょう. その結果 (2 y3 + y) x− 3 が得られます.そこで,第一段目は (x 1 2 y3 + y 0 − 3) となり
ます.次に係数の処理に移ります.この場合, リストの係数各成分の処理を行います.そこで,係数を
x の次の変数 y の多項式として書き直します.すると,第二成分の 2 y3 + y が (y 3 2 1 1) で置換え
られるので,最終的に (x 1 (y 3 2 1 1) 0 − 3) が y x + 2 x y3 − 3 の正準表現として得られます.
この様に,順序を決めていれば,正準表現が得られますが,変数の並びや順序を変更すると同じ多

項式でも表現が一般的に異なります.

1.3 Maximaに導入された順序

1.3.1 Maximaの変数と順序

Maximaで用いられる変数や項の順序について簡単に解説します.順序の一般的な話は前節の 1.1
節を参照して下さい.
先ず,Maximaで変数として扱える文字はアルファベットの小文字と大文字, それと記号 等の幾

つかの記号,そして, alphabetic属性が与えられた文字です. 又,変数の先頭以外ならば 0から 9の
数も利用可能です.

例えば,記号@や~はそのままでは変数として使えませんが, alphabeticとして declare函数で宣言
すれば使えます.以下に,@と~を変数名に用いる例を示しておきます.

(%i1) @x12;

Incorrect syntax: x12 is not an infix operator

@x12;

^

(%i1) ~123x;

Incorrect syntax: 123 is not an infix operator

~123x

^

(%i1) declare(@,alphabetic,~,alphabetic);

(%o1) done

(%i2) @y+@z+@z*4-~12x;

(%o2) - ~12x + 5 @z + @y

但し,Maximaの演算子として用いられている記号の様に alphabeticとして宣言しても使えない
ものもあります.

Maximaで使える変数の集合をAmとします. すると,この集合Amには順序が入っています.こ
の順序を以降 >m と記述します.
順序 >m によりMaximaの変数集合 Am に入る順序を以下に示します.

Maximaの変数順序¶ ³
宣言された主変数 >m ordergreatの第一引数 >m · · · >m

ordergreatの最後の引数 >m alphabetic宣言した文字1 >m · · · >m

alphabetic宣言した文字n >m 頭文字が Zの変数 >m · · · >m

頭文字が Aの変数 >m 頭文字が zの変数 >m · · · >m

頭文字が aの変数 >m orderlessの最後の引数 >m · · · >m

orderlessの第一引数 >m 宣言されたスカラー >m 宣言された定数 >m

Maximaの定数 >m 9 >m · · · >m

0µ ´
先ず,アルファベットに関しては,大文字が小文字よりも大きく,Zが Aよりも大きくなっており,

通常のアルファベットと逆の順番になっています.次に,alphabetic属性を与えた文字は,アルファ
ベットの大文字の Zよりも大きく,ordergreat函数の最後の引数よりも小さな順番となります.

0から 9迄の数値に対しては >m は通常の大小関係 > と同値です. 基本的に ordergreat函数や
orderless函数で入れた順序に無関係な文字や記号同士では ASCIIコードの大きい方が順位が高く
なります.これは,文字の大小関係で内部の LISPの great函数を用いて判別する為です.
変数が一文字であれば >m による比較は分かり易いものです.しかし,変数は通常, abc の様に複

数の文字で構成されます.この複数の文字に対しても, 順序 >m で順序付ける事が可能です.この考

え方を簡単に説明しましょう. 二つの変数 x1x2 · · ·xn と y1y2 · · · ym が与えられた時, n = m でな

ければ,短い変数の側に空白を入れて等しい長さにしたものを考えます. この時, i = 1, · · · , k− 1 で
は, xi = yi , k 番目の文字 xk と yk で初めて異なるとします.すると,文字 xk と yy の順序は >m

の順序で決める事が出来ます.そこで,全体もその順序に従うものとしましょう.すると, xk >m yk

であれば, x1x2 · · ·xn >m y1y2 · · · ym となります.
具体的には,二つの変数 abc と aaz が与えられた場合,頭は同じ文字 a となる為,二番目以降の

文字で大小関係を定めます.すると,二番目の文字で b >m a となる為, abc >m aaz となります.
こうする事で一般の変数に対しても >m で順序を入れる事が出来ます.
Maximaの変数で最も順位が高いものが,主変数 (mainvar) として宣言された変数です. CRE表

現への変換等の多くの函数は主変数を用いますが,式の中でmainvarとして宣言された変数が存在
しない場合,式中の変数で順序 >m による最高位の変数が主変数として用いられます.
これで変数の順序が与えられました.次に,Maximaの変数の集合Amの元の積で構成される項の

順序について説明しましょう.
先ず,Maximaの順序 >m は辞書式順序と呼ばれる順序です. この順序による項の比較の方法を

説明しましょう.先ず,項を構成する変数を >m の順序に従って大きなものから並べます.例えば,
項の変数が x1 から x9 の場合, x9 >m x8 >m · · · >m x1 となるので, 変数の並びは x9, · · · , x1
となります.Maximaではこの項の変数順序の入れ替えは自動的に実行します.次に,二つの項が与
えられて順序 >m を用いて大きさを比較する場合,変数の並びから各項の次数リストを構築しま
す.この次数リストは >m によって二つの項を構成する変数を並び換え,各変数に対応する次数を
左から順に並べたリストです.ここで,項にある変数が抜けている場合には,次数 0を入れます.例
えば,二つの項 x1x22 x83 と x1x22 x3x9 が与えられた場合,二つの項を構成する変数 x9, · · · , x1
の順序に従って変数の並び換えを行います.その結果, x1x22 x83 は,x83 x22 x1 に,x1x22 x3x9 は
x9 x3x22 x1 になります.それから,一方の項のみに現れる変数があれば, その変数が現れていない
項では 0の羃で置換します.すると, x83 x22 x1 は x9と x3が抜けているので x90 x83 x30 x22 x1
になり, x9 x3x22 x1 は x8 が無いので, x9x80 x3x22 x1 とします.この様に正規化してから次数を
左から順に並べたリストを構築します. その結果,5成分の次数リスト (0, 3, 0, 2, 1) と (1, 0, 1, 2, 1)
が得られます. 次に,項の大きさの比較では,これらのリストの先頭から通常の大小関係 > を使っ

て決めます.この場合,(0, 3, 0, 2, 1) の先頭の値が 0であるのに対し, (1, 0, 1, 2, 1) の先頭が 1となる
ので (1, 0, 1, 2, 1) の方が大となります. この次数リストの大小関係をそのまま項の大小関係としま
す.従って, x1x22 x3x9 >m x1 x22 x83 となります.
この順序 >m は x >m y であれば,任意の 0と異なる項 aに対して, ax >m ay が成立する事が

判ります.この性質を満す順序の事を項順序と呼びます.
Maximaでは,この項順序>mに従って入力された多項式の項の変数と項の順番の並び換えを自

動的に行っています.

(%i16) expr1:x1*x2^2*x8^3+x1*x2^2*x3*x9;

2 2 3

(%o16) x1 x2 x3 x9 + x1 x2 x8

(%i17) expr2:x1*x2^2*x3*x9+x1*x2^2*x8^3;

2 2 3

(%o17) x1 x2 x3 x9 + x1 x2 x8

(%i18) :lisp $expr1;

((MPLUS SIMP)((MTIMES SIMP) $X1 ((MEXPT SIMP) $X2 2)((MEXPT SIMP) $X8 3))

((MTIMES SIMP) $X1 ((MEXPT SIMP) $X2 2) $X3 $X9))

(%i18) :lisp $expr2;

((MPLUS SIMP)((MTIMES SIMP) $X1 ((MEXPT SIMP) $X2 2)((MEXPT SIMP) $X8 3))

((MTIMES SIMP) $X1 ((MEXPT SIMP) $X2 2) $X3 $X9))

(%i18) :lisp (equal $expr1 $expr2)

T

この例では,同値な式の項の順番を変更して Maximaに入力しています. しかし,Maximaが返
す式は順番が変更されて同じ式で返されています. 又,:lisp $expr1と:lisp $expr2で表示させた式
expr1と expr2の内部表現も同じものになっています.尚,:lispは後に続く文字列を LISPに直接渡
して結果をMaximaに戻す函数です.
この内部表現で注目して頂きたいのは,順序 >m で小さなものが,リストの左側に置かれ,大きな

もの程,右側に置かれる事です.この内部表現に対して, 式の表示は項順序の大きなものから左側に
並べられます.但し,項の変数順序に関しては逆で,左側に小さなものが表示されます.この辺は,数
式の通常の書き方に準拠したものと言えるでしょう.

1.3.2 変数順序の変更

Maximaでは順序 >m を局所的に変更する事が可能です. 尚,前述の様に,新たに記号をアルファ
ベットとして宣言する事で,新たに順序に記号が加えられますが,これも既存の枠に新しい記号を
嵌め込む以上のものではありません.
この変数順序を変更する函数には ordergreat函数と orderless函数があります.

Maximaの順序変更を行なう函数¶ ³
ordergreat( 〈v1〉, · · · , 〈vn〉 )
orderless( 〈v1〉, · · · , 〈vn〉 )
unorder()µ ´

ordergreat函数によって,引数左端の変数 〈v1〉 が最大で,右側に行くに従って小さくなり, 〈vn〉
が最小となる順序が新たにMaximaの順序 >m に組込まれます.orderless函数の場合も同様に,左
端の 〈v1〉 が最小で,右に行くに従って大きくなり, 〈vn〉 が最大になる順序がMaximaの順序 >m

に組込まれます.
この ordergreat函数や orderless函数でMaximaに入れた順序は unorder(); を入力する事で全

て解除出来ます.
尚,ordergreat函数や orderless函数で入れた変数順序は unorder函数で無効にしない限り,これ

らの函数を用いて新しい変数順序を入れる事は出来ません.

(%i13) ordergreat(c,b);

(%o13) done

(%i14) ordergreat(b,z);

Reordering is not allowed.

-- an error. Quitting. To debug this try debugmode(true);

(%i15) unorder();

(%o15) [b, c]

この例で示す様に,c >m bという順序を入れましたが,その次に,b >m zという変数順序を入れ
ようとするとエラーが出ています.因に,ordergreat(b,z)の代わりに ordergreat(z,b)でもエラーが
出ます.そこで, unorder() を入力する事で ordergreat(c,b)で入れた変数順序を解除しています.

ordergreat 函数や orderless 函数で入れた変数順序で,Maxima 全体の変数順序の変更を伴なう
為,ordergreat函数や orderless函数で入れた変数順序を解除しない限り, 更新を認めない方法を採
用しているお陰で,色々と順序の追加を行なって互いに矛盾した順序が出来上る恐れがない分,安
全と言えます.

1.3.3 順序に関連する函数

Maximaの項順序は,変数に辞書の逆順で順序が入っていますが,項順序自体は辞書式順序と呼ば
れる順序が入っています.
そこで今度はMaximaで変数順序や項順序がどの様に入っているか,実際に調べてみましょう.

二つの与えられた式の順序を調べる函数としてMaximaには ordergreatp函数と orderlessp函数の
二つの述語函数があります.

順序を確認する述語函数¶ ³
述語函数 trueとなる条件
ordergreatp(〈式1〉 ,〈式2〉 ) 〈式1〉が 〈式2〉よりも大となる場合
orderlessp(〈式1〉 ,〈式2〉 ) 〈式1〉が 〈式2〉よりも小となる場合µ ´

ordergreatp函数と orderlessp函数は,与えられた二項に対し,Maximaの項順序 >m による判定

結果を返す述語函数です.尚,ordergreatp函数と orderlessp函数は LISPの great函数を用いて変数
の比較を行っています.

(%i33) ordergreatp(abc,a);

(%o33) true

(%i34) ordergreatp(abc,ax);

(%o34) false

(%i35) ordergreatp(x^2,y^2);

(%o35) false

(%i36) ordergreatp(z^2,y^2);

(%o36) true

(%i37) ordergreatp(z,y^2);

(%o37) true

(%i38) ordergreatp(z^3,z^2);

(%o38) true

(%i39) ordergreatp(z^2*x*y^2,z^2*x*t^3);

(%o39) true

最初の例では,変数 abcと変数 aの順序を比較しています.Maximaの項順序 >mでは, アルファ
ベットに関しては逆アルファベット順で大きさが決ります. 変数の比較では,左端の文字から両者
を比較し,最初に大きなものがあった方を大とします.この例では,abcと aでは頭は同じですが,a
には他の文字が無い為, abc >m aとなります.次の,abcと axの場合,頭の aは同じでも,bと xで
は xの方が大となる為,ax >m abcとなります.
以降,項に対する比較となります.項順序に関しては,zの方が yよりも大きい為, 次数とは無関係

に zの方が y2 よりも大になります.同じ変数の場合は次数の大きな方が大になり,z2xy2 と z2xt3

の場合は,頭から比較して,yの方が tよりも大になる為, z2xy2 の方が z2xt3 よりも大になります.
この事から,Maximaの変数順序は斉次ではなく,辞書式順序に基く順序である事が理解出来るかと
思います.
この様にMaximaでは変数の順序を ordergreat函数や orderless函数で多少変更する事が可能で

すが,基本的な順序は辞書式順序のみです

1.3.4 函数を含めた順序

ここまでは多項式の項順序について述べましたが,Maximaでは通常の多項式に加えて expや sin
等のMaxima組込の初等函数にも順序が入ります.
この場合,同じ引数の函数の比較では函数名で>mによる順序が入ります. 次に,函数と他の変数

では,函数に含まれる変数に主変数があれば,函数が変数よりも上位になりますが,函数に主変数が
含まれずに比較する変数が主変数の場合,変数の方が上位になります.その為,函数の引数が比較す
る項を構成する変数よりも順位の低い変数であれば,項の方が順位が高くなります.
ところで, 利用者定義の函数の場合, 一度 Maxima 側で値を解釈する為に ordergreatp 函数や

orderlessp函数の値はその状況に応じて変化します.
以下にその例を示します.

(%i77) neko(x):=if x<0 then x^2 else cos(x)^3;

2 3

(%o77) neko(x) := if x < 0 then x else cos (x)

(%i78) assume(p0>0);

(%o78) [p0 > 0]

(%i79) ordergreatp(cos(p0),neko(p0));

(%o79) false

(%i80) assume(p1<0);

(%o80) [p1 < 0]

(%i81) ordergreatp(cos(p1),neko(p1));

(%o81) true

(%i82) ordergreatp(’neko(x),atan(x));

(%o82) true

(%i83) ordergreatp(neko(x),atan(x));

Maxima was unable to evaluate the predicate:

x < 0

#0: neko(x=x)

-- an error. Quitting. To debug this try debugmode(true);

(%i84)

この例で示す様に利用者函数に単引用符を付けなければMaximaで解釈が実行され, その結果で
ordergreatp函数の結果が決ります.これに対して,単引用符を付けて名詞型で比較すると単純に函
数名で比較されます.この様に初等関数や利用者定義函数が名詞型の場合は引数も含めた函数名で
変数順序 >m による比較が実行されます.

1.4 文脈の概要

Maximaには文脈 (context)があります. この文脈とは要するに,問題を考える上での様々な仮定
や事実の積重ねです. 式の計算や簡易化といった式の処理は,この文脈を利用して行います.
例えば,非常に簡単な例ですが,4 の平方根は 2 になりますが, x2 はどうでしょうか？ x は正か負

か,ひょっとすると複素数かもしれません. この様に
√

x2 を考えるだけでも, x に関して様々な情

報が必要になります.先ず,実数であれば |x| となりますね.更に,x ≥ 0であれば答は x になります.
この処理では,以下の x に対する仮定を用いています.

• x は実数である.

• x ≥ 0 である.

文脈とは,この様な問題を考える上で必要となる情報の蓄積の事です.
Maximaには複数の文脈を持たせる事が可能です.更に,文脈には階層構造もあります. 先ず,文

脈 globalが最上位の文脈です.Maximaを立ち上げた時には既に文脈 globalの子文脈 initialがあり
ます.デフォルトの文脈はこの initialです.

Maximaの起動時に存在する文脈¶ ³
文脈名 概要

initial Maximaを立ち上げた時点で利用されるデフォルトの文脈名.
global Maximaの文脈全体の親文脈.µ ´
文脈には様々な事実や仮定を蓄積して行く事になりますが,その操作を行う函数を以下に纏めて

おきます.
文脈上の述語に関連する函数¶ ³

assume( 〈式1〉, 〈式2〉, · · · )
forget( 〈述語1〉, · · · , 〈述語n〉 )
forget( [〈述語1〉, · · · , 〈述語n〉] )
facts( 〈事項 〉 )
facts( 〈文脈 〉 )
facts()µ ´

Maximaに事実や仮定を教える函数は assume函数です.様々な事実や仮定は,Maximaで述語と
呼ばれるMaximaの式で表現されます. 述語は評価を行う事で trueか falseが返却される論理式,
例えば, x > 0 や, x < 1 and x > 0 の様な二項関係に論理積 and, 論理和 or,否定 not等の論理演
算子が作用したMaximaの式です.
この述語は assume函数によって現行の文脈上に設定されます.これに対し,教えた事実や仮定を

忘れさせる,即ち,文脈に設定した述語を消去する場合,forget函数を用います.そして,或る事項に
関して,どの様な述語が含まれているかを調べたければ,facts函数 を用います.facts函数は文脈を
指定した場合,その文脈に含まれる述語を返します. 何も指定しない facts()を入力した場合,現行
の文脈が保持する述語を全て表示します.

Maximaでは文脈を複数持たせる事が可能です.この事は,文脈 Aでは x > 0を仮定して述語を
蓄積し,文脈 Bでは x < 0を仮定して述語を文脈Aとは独立して蓄積させる事が可能です.Maxima
では,文脈を新たに生成したり,既存の文脈の間に親子関係を入れる等の操作が行えます.
以下に文脈操作函数の一覧を示します.

文脈操作函数¶ ³
activate( 〈文脈1〉, · · · )
deactivate( 〈文脈1〉, · · · )
killcontext( 〈文脈1〉, · · · )
newcontext( 〈文脈 〉 )
supcontext( 〈親文脈 〉, 〈子文脈 〉 )µ ´
先ず,既存の文脈を有効にする場合,activate函数を用い,逆に無効にする場合は decativate函数

を用います.
新しい文脈の生成は newcontext函数を用い, 既存の文脈の親文脈を生成する場合は supcontext

函数を用います.
文脈の削除は killcontext函数で行いますが, 利用中の文脈や文脈が有効 (activate)の場合は削除

が出来ません.
デフォルトの文脈は initialですが,文脈を生成して行くと,現在利用中の文脈が何で,どの様な文

脈があるのか判らなくなるかもしれません.Maximaの大域変数 contextに現在,利用中の文脈名が
割当てられ,大域変数 contextsにMaximaの中にある全ての文脈名のリストが割当てられています.

文脈に関連する大域変数¶ ³
変数名 初期値 概要

context initial 現行の文脈名が割当てられた変数

contexts [initial,global] Maxima内部の文脈名リストが割当てられた変数µ ´
先ず,どの様な文脈がMaximaに存在するかは contexts; と入力すれば文脈の一覧がリストの形

式で表示されます.例えば,Maximaを立ち上げた時点では, 文脈 initialと文脈 globalの二種類の文
脈がある事が判り, context; と入力すると,文脈 initialが現行の文脈,即ち,assume函数等で設定し
た述語が文脈 initialに蓄積される事が判ります.

(%i1) contexts;

(%o1) [initial, global]

(%i2) context;

(%o2) initial

複数の文脈が存在する場合に,文脈を切替えたければ,大域変数 context に文脈名を直接指定し
ます.

では,以下に文脈の使い方の実例を示しましょう.

(%i1) contexts;

(%o1) [initial, global]

(%i2) context;

(%o2) initial

(%i3) newcontext(mike);

(%o3) mike

(%i4) supcontext(neko,mike);

(%o4) neko

(%i5) context;

(%o5) neko

この例では,最初に立ち上げた時点でMaximaが持っている文脈を contexts函数で表示していま
す.次に context函数を使って最初に用いている文脈が文脈 initial である事を示しています.それ
から,newcontext函数を使って新しい文脈mikeを生成しています.それから文脈mikeの親文脈と
なる文脈 nekoを今度は supcontext函数を使って生成しています.この supcontext函数には子文脈
に既存の文脈を指定しなければなりません.
次に,assume函数を用いて文脈上に述語を設定し, その述語を用いて式を簡易化する例と文脈の

切り替えによる効果を見ましょう.

(%i6) assume(y>0);

(%o6) [y > 0]

(%i7) assume(x>0,z<0);

(%o7) [x > 0, z < 0]

(%i8) facts();

(%o8) [y > 0, x > 0, 0 > z]

(%i9) sqrt(x^2);

(%o9) x

(%i10) context:initial;

(%o10) initial

(%i11) sqrt(x^2);

(%o11) abs(x)

(%i12) facts();

(%o12) []

(%i13) activate(neko);

(%o13) done

(%i14) context;

(%o14) initial

(%i15) sqrt(x^2);

(%o15) x

(%i16) facts();

(%o16) []

(%i17) deactivate(neko);

(%o17) done

(%i18) sqrt(x^2);

(%o18) abs(x)

(%i19) killcontext(neko);

(%o19) done

(%i20) contexts;

(%o20) [mike, initial, global]

先程の文脈の定義に続いて,文脈 nekoで assum函数を使って変数 x,y,zに対して仮定, 即ち,述
語を設定します.利用している文脈で設定した述語全体は facts(); で表示が出来ます.文脈 nekoで
abs(xˆ2)を実行すると,述語 x > 0より xが返されます.次に,文脈の切替は変数 contextに文脈名を
割当てる事で行います.ここの例では,文脈 nekoから文脈 initalへと現行の文脈を context:initial;

で変更しています.それから, sqrt(xˆ2) を実行すると abs(x)が返されます.これは何故でしょうか？

文脈には x > 0 という情報が無いからです.実際, facts(); を実行すると, 文脈 initialには何も述

語が無い事が分ります.ここで,文脈 nekoの述語を文脈 initialで利用したければ, actvate(neko);
を実行します. その結果,sqrt(xˆ2)の結果が xになります.但し,文脈 initialの内容には変化があり
ません.不要になった文脈は deactivate函数で無効にしたり, killcontext函数で削除する事が可能
です.
次に,大域変数 featuresにはMaximaに組込まれた属性のリストが割当てられています. Maxima

の式に featuresのリストに含まれる属性を持たせたければ, declare函数を用います.
例えばbirthdayを整数として宣言する事で,整数属性を持たせたければ, declare(birthday,integer);

とします.逆に式 pが属性 qを持つかどうかを調べる場合は featurep函数を使って, featurep(p,q);
で調べます.
尚,declare函数による宣言は,assume函数と違い大域的なものになります. その一方で,featurep

函数は declare函数を用いた文脈上でしか使えません. 詳細は,1.5.1節を参照して下さい.

(%i1) newcontext("mike");

(%o1) mike

(%i2) supcontext("neko","mike");

(%o2) neko

(%i3) context:mike;

(%o3) mike

(%i4) declare(bb,lassociative);

(%o4) done

(%i5) assume(x>0);

(%o5) [x > 0]

(%i6) facts();

(%o6) [kind(bb, lassociative), x > 0]

(%i7) bb(bb(a,b),bb(c,d));

(%o7) bb(bb(bb(a, b), c), d)

(%i8) sqrt(x^2);

(%o8) x

(%i9) context:initial;

(%o9) initial

(%i10) bb(bb(a,b),bb(c,d));

(%o10) bb(bb(bb(a, b), c), d)

(%i11) aa(aa(a,b),aa(c,d));

(%o11) aa(aa(a, b), aa(c, d))

(%i12) facts();

(%o12) [kind(kron_delta, symmetric)]

(%i13) sqrt(x^2);

(%o13) abs(x)

この例では,文脈mikeと nekoを生成し, context:mike; で文脈をデフォルトの文脈 initialから文
脈 mikeに切替えています.そこで,bbに左結合律が成立する事を宣言し, x > 0 も assume函数で
文脈mikeに入れています.facts函数で文脈に蓄積した述語を確認し,bb(bb(a,b),bb(c,d))が左結合
律によって,bb(bb(bb(a,b),c),d) に自動的に変形される事と,sqrt(xˆ2)が xに簡易化される事を確
認しています. それから, context:initial; でデフォルトの文脈 initialに移動します.
ここで,declare函数による宣言は assume函数による述語とは違い,大域的な影響を及ぼす為,bb

に関しては左結合律が成立します.その一方で,x > 0は文脈 mike上の仮定の為,文脈 initには影
響を与えません.その為,sqrt(xˆ2)は abs(x)になります.更に,bbに関しては facts()には表われま
せん.
最後に,文脈で多く設定されるものの一つが変数の正値性です.この正値性の設定では通常,assume

函数を用いて設定を行いますが,大域変数を用いて assume函数による設定を省く事も可能です.

変数の正値性に関連する大域変数¶ ³
変数名 初期値 概要

assume pos false assume pos predで指定した述語函数が trueとなる対象に対し, 正
値であると仮定.µ ´

先ず,大域変数 assume posは,大域変数 assume pos predと対で用います. 大域変数 assume pos
は,大域変数 assume pos predで指定した述語函数が trueとなるMaximaの式に対して,その正値
性を設定します.但し, assume posが trueで assume pos predをデフォルトの falseのままにして
いる場合,symbolp函数が trueとなるMaximaの式に対し,その正値性を設定します. 尚,assume函
数による設定がある場合,その assume函数の設定が大域変数 assume posよりも優先されます.

(%i13) declare(aa,even);

(%o13) done

(%i14) featurep(aa,even);

(%o14) true

(%i15) assume_pos_pred:lambda([x],featurep(x,even));

(%o15) lambda([x], featurep(x, even))

(%i16) assume_pos:true;

(%o16) true

(%i17) sqrt(aa^2);

(%o17) aa

(%i18) sqrt(bb^2);

(%o18) abs(bb)

この例では変数 aaに偶数として属性を与えています.実際,featurep函数による検査では trueに
なります.ここで,大域変数 assume pos predに featurep函数による検査を行う函数を割当てます.
それから,大域変数 assume posを trueに設定すると, 偶数としての属性を持つ変数 aaには正値
性が付与されます.その為,sqrt(aaˆ2)は aaになりますが,偶数としての属性を持たない変数 bbは
abs(bb)となります.

1.5 属性の宣言と属性値

Maximaではアトムや式に属性や属性値を付加する事が出来ます. この処理を行う代表的な函数
に declare函数と put函数があります.

declare函数はMaximaに予め設定された属性をアトムや式が持つ事を宣言する事に用います.
尚,declareで宣言する事で式に与えられる属性は,基本的に数値であれば, 偶奇性,函数の場合は線
形性といった数学上の性質が主です.

declare函数はMaximaのアトムや式に属性を与える事で式の変形を促進する作用もあります.
例えば,函数 f(x) が linear,即ち,線形写像であると宣言していれば, Maximaは f(x + y) が与え
られると f(x) + f(y) に自動的に簡易化します.

(%i3) declare(f1,linear);

(%o3) done

(%i4) f1(x+y);

(%o4) f1(y) + f1(x)

又,利用者が必要な属性を新たにMaximaに付加し,その属性を持つアトムや式に対して様々な
函数を作成する事も可能になります.
これに対し,put函数はアトムに対して用いるもので,利用者の指定する属性に対応する値をアト

ムに設定する事が出来ます.
尚,put函数で与えた属性値の取出しは,get函数で属性値を設定したアトムと値を取出したい属性

を指定する事で行います.更に,アトムに指定した属性とその値の削除は rem函数や remove函数
で行います.

1.5.1 declare函数

declare函数はアトムや式に対し,Maxima上で定義された属性を付加する函数です.

declare函数の構文¶ ³
declare(〈a1〉, 〈属性1〉, 〈a2〉, 〈属性2〉, · · ·)
declare(〈a〉, [〈属性1〉, · · · , 〈属性n〉)
declare([〈a1〉, · · · , 〈am〉], [〈属性1〉, · · · , , 〈属性n〉)µ ´

declare函数による宣言で 〈ai〉 に 〈属性i〉 が付加されます.基本的に,declare函数の引数は属性
を付加されるものと属性の対になります.更に,属性がリストで与えられる場合, 〈ai〉 は属性リスト
に含まれる全ての属性を持つ事になります.
この事を以下の例で説明しましょう.

(%i10) declare(a1,[integer,odd]);

(%o10) done

(%i11) declare([b1,c1,d1],[integer,odd]);

(%o11) done

(%i12) featurep(d1,odd);

(%o12) true

(%i13) featurep(c1,integer);

(%o13) true

最初の例では変数 a1が整数で,奇数となる事を宣言しています. 次の例では変数 b1,c1と d1が
整数で奇数となる事を宣言しています.この様に, 複数の属性を設定する場合は属性をリストで与
え,複数の式にある属性を与える場合には,式をリストで指定します.
但し,複数の属性を一つの式に設定する場合,設定する属性は互いに無矛盾なものでなければな

りません.例えば,偶数,且つ奇数であると宣言する事は出来ません.

(%i11) declare(n1,odd);

(%o11) done

(%i12) declare(n1,even);

Inconsistent Declaration: declare(n1,even)

-- an error. Quitting. To debug this try debugmode(true);

(%i13) declare(n2,integer);

(%o13) done

(%i14) declare(n2,even);

(%o14) done

この例では,最初に奇数として宣言してから偶数と宣言すると,偶数且つ奇数となる数が存在しな
い為に declare函数はエラーを返します.但し,整数である事と偶数である事は矛盾しない為,declare
函数で別個に宣言しても問題はありません.

declare函数による宣言の影響は,文脈で利用される assume函数による設定と異なり, 特定の文脈
上に留まらずに大域的に影響を及ぼします.その一方で, facts(); で表示される内容は declare函

数を用いた文脈上でのみ設定されています.その為,文脈Aで宣言した内容を文脈B上で facts();

と入力して確認する事は出来ませんが,宣言した事実を利用する事は可能です.
declare函数はMaximaが持っている属性を与えるだけの函数ではありません.利用者がMaxima

に新たな属性を与える為に利用する事も可能です.
属性を追加する場合¶ ³

declare(〈新属性 〉,feature)µ ´
Maximaに属性を新規に追加する場合,第一引数に属性名を設定し,第二引数を feature にします.

この宣言を行うと,大域変数 featuresに割当てられたリストに 〈新属性 〉 が追加されます.尚,式が
ある属性を持つかどうかは述語函数の featurep函数を使って調べられます.

述語函数 featurep¶ ³
featurep(〈アトム 〉, 〈属性 〉)µ ´
featurep函数は,〈アトム 〉 が 〈属性 〉 を持つかどうかを調べる函数です.ここで, 〈属性 〉 は大域

変数 featuresに割当てられたリストの成分でなければなりません.
ここでは新しい属性として四元数 (quaternion)を追加してみましょう.

(%i8) declare(quaternion,feature);

(%o8) done

(%i9) features;

(%o9) [integer, noninteger, even, odd, rational, irrational,

real, imaginary, complex, analytic, increasing, decreasing,

oddfun, evenfun, posfun, commutative, lassociative,

rassociative, symmetric, antisymmetric, quaternion]

(%i10) declare(q1,quaternion);

(%o10) done

(%i11) featurep(q1,quaternion);

(%o11) true

この例では, declare(quaternion,feature); で quaternionを featuresに追加しています.それから,

declare(q1,quaternion); で q1が quaternionであると宣言しています. 最後に,featurep函数で q1
が quaternionである事を確認しています.
これで,四元数 quaternionという属性を入れる事が出来ました.後は四元数を扱えるMaximaの

函数を定義すれば良いのです.
では,以降の小節では,Maximaに予め用意された属性について順番に解説しましょう.

1.5.2 Maximaに用意された属性

最初に物事の基本となるアトムに対して与えられる属性について説明します. declare函数を使
う事で,スカラー,定数,主変数といった属性,更には,変数として利用可能な文字を宣言する事が出
来ます.

アトムの属性¶ ³
declare(〈a〉 ,scalar) 〈a〉 をスカラーとして宣言
declare(〈a〉 ,nonscalar) 〈a〉 をスカラーではないと宣言
declare(〈a〉 ,nonarray) 〈a〉 を配列ではないと宣言
declare(〈a〉 ,constant) アトム 〈a〉 を定数として宣言
declare(〈a〉 ,mainvar) 変数 〈a〉 を主変数として宣言
declare(〈a〉 ,alphabetic) 〈a〉 をアルファベットとして宣言
declare(〈a〉 ,special) 〈a〉 を special変数として宣言µ ´

最初のスカラーの宣言は行列やベクトルの演算で特に影響します.
主変数の宣言はMaximaの多項式の変換等で大きな影響が出ます.通常,Maximaの変数順序>m

より式に含まれる変数の中で最大の変数を主変数とします. そして, 自動的に主変数を筆頭に順
序 >m に沿って項と変数を並び換えます. 但し,この主変数は局所的なものです.これに対し,変数
に対して mainvar宣言を行うと,宣言された変数は順序 >m で最高位の変数になります.その為,
mainvar属性を持った変数は常にその変数が含まれる式の中では主変数として処理されます.
その為,mainvar属性を持った変数が存在すると,式の内部表現にも影響が及ぶ為に, 式の表示に

限らず,様々な処理も異なる事があります.

(%i1) f1:(x+y)^4;

4

(%o1) (y + x)

(%i2) f1,expand;

4 3 2 2 3 4

(%o2) y + 4 x y + 6 x y + 4 x y + x

(%i3) f1,declare(x,mainvar),expand;

4 3 2 2 3 4

(%o3) x + 4 y x + 6 y x + 4 y x + y

(%i4) ans1:%o2$

(%i5) ans2:%o3$

(%i6) :lisp $ans1

((MPLUS SIMP) ((MEXPT SIMP) $X 4) ((MTIMES SIMP) 4 ((MEXPT SIMP) $X 3) $Y)

((MTIMES SIMP) 6 ((MEXPT SIMP) $X 2) ((MEXPT SIMP) $Y 2))

((MTIMES SIMP) 4 $X ((MEXPT SIMP) $Y 3)) ((MEXPT SIMP) $Y 4))

(%i6) :lisp $ans2

((MPLUS SIMP) ((MEXPT SIMP) $Y 4) ((MTIMES SIMP) 4 ((MEXPT SIMP) $Y 3) $X)

((MTIMES SIMP) 6 ((MEXPT SIMP) $Y 2) ((MEXPT SIMP) $X 2))

((MTIMES SIMP) 4 $Y ((MEXPT SIMP) $X 3)) ((MEXPT SIMP) $X 4))

(%i6) ans1+ans2;

4 3 2 2 3 4 3 2 2 3 4

(%o6) y + 4 x y + 6 x y + 4 x y + 2 x + 4 y x + 6 y x + 4 y x + y

(%i7) ev(%,simp);

4 3 2 2 3 4

(%o7) 2 x + 8 y x + 12 y x + 8 y x + 2 y

この例では,最初に多項式 (x + y)4 を展開しています.ここでは,デフォルトの順序 >m で展開

される為, y の多項式として展開されます.それから, x を主変数として宣言して展開した為,多項
式 f1は x の多項式として展開されています.ここで, f1,declare(x,mainvar),expand は ev函数に
よる評価の方法で,f1 を展開する際に x を主変数として評価しています.この ev函数内部での宣
言は ev函数による式の評価以外には影響を与えません.詳細は 1.8.1節を参照して下さい.

ここで主変数を宣言しなかった場合とした場合では,同じ式の表示の順序が異なっています.こ
れは内部表現自体が異っている事を意味します.実際,mainvar宣言をしなかった場合 (ans1)の内部
表現とmainvar宣言を行った場合 (ans2)の内部表現は ans1が変数 Y で式を纏めているのに対し

て ans2が変数X で纏めている点で異っています.これだけであれば,単純に式の表現の違いだけで
済みますが,ans1+ans2の和を計算させても,単純に ans1と an2を繋げて ans1の末端と ans2の先
頭の x4を足し合せただけの結果を返します.この様な場合,ev函数を用いて式を再評価する必要が
あります.実際,ev函数を用いて簡易化させる事で本来の結果が得られています.

alphabeticで引数の文字をMaximaのアルファベットとして利用可能にします. 尚,Maximaで
変数等で利用可能な文字は,デフォルトで aから z迄の小文字と大文字, それに%と.を加えたもの
です.それ以外の文字を変数で利用したければ,この alphabeticで宣言を行う必要があります.
最後に,specialで変数を special変数として宣言します.この宣言はあまり表に出る事はありませ

ん.この変数は函数の最適化等でも用いられますが,変数に属性を指定する事で,変数に割当てられ
る値に制限を加える事も可能になります. この方法は,define variable函数 で解説します.
次に,declare函数で宣言可能な数値属性を以下に示します.

数値属性¶ ³
declare(〈a〉 ,integer) 〈a〉 を整数として宣言
declare(〈a〉 ,noninteger) 〈a〉 を非整数として宣言
declare(〈a〉 ,even) 〈a〉 を偶数として宣言
declare(〈a〉 ,odd) 〈a〉 を奇数として宣言
declare(〈a〉 ,rational) 〈a〉 を有理数として宣言
declare(〈a〉 ,irrational) 〈a〉 を無理数として宣言
declare(〈a〉 ,real) 〈a〉 を実数として宣言
declare(〈a〉 ,imaginary) 〈a〉 を純虚数として宣言
declare(〈a〉 ,complex) 〈a〉 を複素数として宣言µ ´
数値属性には整数,非整数,偶数や奇数,有理数や無理数,実数,純虚数に複素数があります.これ

らの宣言を行う事でMaximaの函数で動作が異なるものも多くあります.又,-1の整数による羃乗
や三角函数では引数に含まれている整数が偶数であるか,或いは奇数であるかによって式自体が自
動的に簡易化される事もあります.
次に,函数に対する属性について説明しましょう.函数に関する属性としては函数と積や和といっ

たMaximaの演算子や引数の並び替えといった作用に関連した属性, 函数の単調増加といった函数
自体の特徴に関連する属性に大きく二つに分けられます.

函数属性 (作用に関して)¶ ³
declare(〈f〉 ,additive) 函数 〈f〉 の加法性
declare(〈f〉 ,multiplicative) 函数 〈f〉 の乗法性
declare(〈f〉 ,outative) 函数 〈f〉 と積の可換性
declare(〈f〉 ,linear) 函数 〈f〉 の線形性
declare(〈f〉 ,commutative) 函数 〈f〉 の対称性
declare(〈f〉 ,symmetric) 函数 〈f〉 の対称性
declare(〈f〉 ,lassociative) 函数 〈f〉 の左分配律
declare(〈f〉 ,rassociative) 函数 〈f〉 の右分配律µ ´
函数 f に対する additive属性の宣言は,函数 f の第一変数,即ち,左端の変数に対する加法性を

宣言します.即ち, f(x1 + y1, · · ·) は f(x1, · · ·) + f(y1, · · ·) と同値になります.尚,この宣言は式の
内部表現に依存する為,sum函数に対しては効果がありません.
函数 f に対するmultiplicative属性の宣言は,函数 f の第一変数に対する乗法性を与えます.即

ち, f(x1 ∗ y1, · · ·) は f(x1, · · ·) ∗ f(y1, · · ·) と同値になります.尚,multiplicative属性は,additive属
性と同様,式の内部表現に依存する為,product函数に対して無効になります.

outative属性の宣言では,函数 fに可換積*との交換性を付与します.即ち, f(a ∗ x1, · · ·) は a ∗
f(x1, · · ·) と同値になります.但し,外に出される a はアトムに限定されます.
デフォルトで sum函数,integrate函数と limit函数 の名詞型が outative属性を持つものとして

宣言されています.
linear属性は二つの属性 additiveと outativeの両方を持たせたものに相当し,更に, 函数 f の第

一変数に対して線形性を宣言します.即ち, f(x1 + y1, · · ·) は f(x1, · · ·) + f(y1, · · ·) に同値になり,
定数 a に対し, f(a ∗ x1, · · ·) は a ∗ f(x1, · · ·) に同値になります.
函数 f に対する commutative属性の宣言は,函数 f の引数の順番を入換えても結果が変化しな

い事,即ち,変数順序の可換性を意味します.又,commutative属性は symmetric属性,即ち,対称性
と同値です.ここで commutative属性は演算の可換性に関連し,symmetric属性は函数の引数の対
称性に対処します.但し,これらの基本的な動作は同じです.
ここで,一度引数の置換を行うと符号が反転する函数,即ち, f(x, y) = −f(y, x) を満す場合を f

は歪対称性があると呼びます.Maximaでは属性 antisymmetricを持つ事を意味します.
属性 antisymmetricで函数 f は丕対称として宣言されます.これは一度函数 f の引数の順番を

入れ換えたものが元の値を-1倍したものになる性質です. 例えば, f(x, y, · · ·) = −f(y, x, · · ·) とな
る性質です.

lassociative属性で函数 f の左結合律性を宣言します. f(f(a, b), f(c, d)) が f(f(f(a, b), c), d) と
同値である事を意味しますが, ここで f を演算子 ∗ で置換えると, (a ∗ b) ∗ (c ∗ d) = ((a ∗ b) ∗ c) ∗ d

を意味し, これは左側からの結合律を満すものになっています.
rassociative属性の宣言により函数 f は右結合律律を満す函数となります. 即ち, f(f(a, b), f(c, d))

は f(a, f(b, f(c, d)))に簡易化されます. これも,函数 f を演算子 ∗で置換えると, ((a∗b)∗(c∗d)) =
a ∗ (b ∗ (c ∗ d)) に簡易化される事を意味し,これは右側からの結合律を満すものになっています.

declare函数を用いた函数の属性としては,増加,減少,奇函数,偶関数,正値函数, 解析的函数等の
性質を指定する事も可能です.

函数の属性¶ ³
declare(〈f〉 ,analytic) 函数 〈f〉 を解析的函数として宣言
declare(〈f〉 ,increasing) 函数 〈f〉 を増加函数として宣言
declare(〈f〉 ,decreasing) 函数 〈f〉 を減少函数として宣言
declare(〈f〉 ,oddfun) 函数 〈f〉 を奇函数として宣言
declare(〈f〉 ,evenfun) 函数 〈f〉 を偶函数として宣言
declare(〈f〉 ,posfun) 函数 〈f〉 を正値函数として宣言
declare(〈f〉 ,noun) 函数 〈f〉 を名詞型函数として宣言µ ´
最後の重要な属性に evfun属性と evflag属性があります.これらの属性を持つ函数や大域変数は

ev函数を用いた評価で利用されます.これらの属性は全て declare函数を用いて設定する事が可能
です.

ev函数で用いる evfun,evflag属性¶ ³
declare(〈f〉 ,evfun) 函数 〈f〉 に evfun属性を付加
declare(〈a〉 ,evflag) 大域変数 〈a〉 に evflag属性を付加µ ´
詳しくは,1.8.1節で述べますが,ev函数では引数として与えられた evflag属性を持つ大域変数は,ev

函数内部でのみ trueが設定され,又,evfun属性を持つ函数を ev函数の引数として与えると,evflag
属性を持った引数と,その函数を与式に適用して処理を行います.利用者が定義した大域変数や函
数に対して declare函数を用いて宣言する事で,ev函数の引数として与えられる様になります.

1.5.3 put函数による属性値の指定

Maximaではアトムに対して様々な属性値の指定が行えます.declare函数による宣言の他に,put
函数でアトムと属性に対応する属性値を指定し,get函数でアトムと属性を指定して対応する属性値
を取出す事が出来ます.

属性に設定に関連する函数¶ ³
put(〈アトム 〉 ,〈属性値 〉, 〈属性 〉 )
qput(〈アトム 〉 ,〈属性値 〉, 〈属性 〉 )
get(〈アトム 〉 ,〈属性 〉 )µ ´

put函数は 〈アトム 〉 に 〈属性 〉 で指定した 〈属性値 〉 を加えます.これは利用者がアトムに任
意の属性を与える事が可能になります.この設定した属性値は get函数で取出せます.

put函数による属性の設定では属性を幾つも指定可能で,利用者独自の指定も出来ます. 但し,ア
トムと属性に対応する属性値は一つです.アトムに設定された属性は properties函数で確認出来ま
す.属性値の取出しは get函数を用いて,アトムと属性を指定すると得られます.

qput函数は put函数に似ていますが,引数の評価を行わない点で put函数と異なります.
get函数は, 〈アトム 〉 に設定された属性で, 〈属性 〉 に対応する属性値を取出す函数です.属性値

自体は put函数で与えます.

1.5.4 変数宣言

Maximaで値を割当てた変数は基本的に大域変数になります.この変数は通常は型も属性も何も
ありません.この大域変数に特定の型を設定したり,初期値を割当てる事も出来ます.
先ず,変数に予め型を指定する函数としてmode declare函数とmodedeclare函数があります.但

し,mode declare函数とmodedeclare函数はMaxima内部では同じ函数です.この変数の宣言に加
えて,初期値を設定する函数に define variable函数があります.

変数宣言を行う函数¶ ³
mode declare(〈変数1〉, 〈型1〉, · · ·, 〈変数n〉, 〈型n〉,)
modedeclare(〈函数1〉, 〈型1〉, · · ·, 〈函数n〉, 〈型n〉,)
mode identity( 〈引数1〉, 〈引数2〉 )
define variable (〈変数名 〉, 〈初期値 〉,〈型 〉)µ ´

mode declare函数は 〈変数i〉 に 〈函数i〉 や 〈型i〉 を設定します.この函数は translate函数で変
換する利用者定義の函数で用いる変数の宣言で利用されます.ここで mode declare函数で指定し
た型は変数のmode属性に割当てられます.

大域変数のmode属性¶ ³
型 mode declare函数で利用可能な表記 概要

float float,real,floatp,flonum,floatnum 浮動小数点

fixnum fixp,fixnum,integer 整数

rational rational,rat 有理数

number number,bignum,big 任意精度の数

complex complex 複素数

boolean boolean,bool Boolean
list list,listp リスト

any any,none,any check 任意の型µ ´
以下に簡単な例を示します.

(%i28) mode_declare(x1,integer);

(%o28) [x1]

(%i29) :lisp (get ’$x1 ’mode)

$FIXNUM

(%i29) mode_declare(x2,rat);

(%o29) [x2]

(%i30) :lisp (get ’$x2 ’mode)

$RATIONAL

(%i30) mode_declare(x2,rational);

(%o30) [x2]

(%i31) :lisp (get ’$x2 ’mode)

$RATIONAL

この例では変数 x1 を整数型,x2 と x3 を有理数型として宣言しています. これらの値は各変数
の mode 属性に蓄えられています. この属性は LISP の get 函数を用いて取出せます. この様
に,mode declare函数は基本的に型の指定を行う函数です. mode declare函数で指定した変数と型
はmode identity函数を用いて検証が出来ます.

(%i9) mode_identity(integer,x1);

(%o9) 128

(%i10) x1:256.988;

(%o10) 256.988

(%i11) mode_identity(integer,x1);

Warning: x1 was declared mode fixnum, has value: 256.988

(%o11) 256.988

(%i12) mode_identity(float,x1);

(%o12) 256.988

(%i13) :lisp (get ’$x1 ’mode);

$FIXNUM

この様に,mode declareで宣言した型以外の値を与える事は可能で, mode identityも変数に割当
てられた型と異なると文句を言うだけです. mode identityは変数のmode属性と指定された属性
が同じものかどうかを検証するのではなく,単純に変数に割当てられた型がmode identityの引数
で指定された型と同じものかどうかを検証する函数で,正しい場合には変数の値を返し,異なる場
合には警告するだけです.

型の検証に関連する大域変数¶ ³
大域変数 初期値 概要

mode checkp true 定数変数の型を検証するかどうか制御

mode check errorp false 型エラー処理の制御

mode check warnp true 型エラーの表示の制御µ ´
大域変数 mode checkpが trueの場合,mode declare函数で宣言する変数に割当てられた値の型

と,新たに宣言する型が矛盾しないか検証し,矛盾する場合はエラーを返します.

(%i17) x0:1.0$

(%i18) mode_declare(x0,integer);

Warning: x0 was declared mode fixnum, has value: 1.0

(%o18) [x0]

(%i19) mode_checkp:false$

(%i20) mode_declare(x0,integer);

(%o20) [x0]

大域変数mode check errorpが trueであれば,mode declare函数で値が割当てられた変数に対し,
異なる型で変数の宣言を行う場合にエラーを出力します.
大域変数 mode check warnpが trueの場合,mode identity函数は変数に割当てられた値の型と

指定した型が異なる場合に警告を出します.
define variable函数は変数の宣言に加え,初期値と型の設定が行える函数です. 但し,この函数は

属性を上手く使う事で,変数に値を割当てる際に,その変数の利用者が設定した条件に適合するか
どうかを検証する事も行える様に出来ます.
この define variable函数の処理手順を以下に示しておきます.

• mode declare函数を用いて変数型を宣言します.

• declare函数を用いて変数が special属性を持つと宣言します.

• 型が anyでなければ,属性 assign に属性値 assign-mode-checkを設定します. この属性値が
設定されていない変数に対して,Maximaは変数への値の割当の際に, 値の検証を行いません.
値の検証を行う為には大域変数の value check属性に述語函数を割当てておく必要がありま
す.この設定では qput函数が有効です.

• 変数に値が割当てられていなければ,指定した初期値を設定します. もしも値が既に割当てら
れていれば,define variable函数は与えられた初期値を割当てずに,そのままの値にしておき
ます.

割当ての際に,変数値の検証を行う方法は,qput函数を用いて宣言した大域変数の value check属
性に変数値を検証する述語函数名を割当ておきます.以下に,動作例を示しておきましょう.

(%i1) ptest(y):=if not primep(y) then error(y,"is not prime!!")$

(%i2) define_variable(tama,5,integer)$

(%i3) qput(tama,ptest,value_check)$

(%i4) tama;

(%o4) 5

(%i5) tama:15;

15 is not prime!!

#0: ptest(y=15)

-- an error. Quitting. To debug this try debugmode(true);

(%i6) :lisp (get ’$tama ’assign)

ASSIGN-MODE-CHECK

(%i6) define_variable(mike,5,any)$

(%i7) properties(mike);

(%o7) [value, special]

(%i8) qput(mike,ptest,value_check)$

(%i9) properties(mike);

(%o9) [value, [user properties, value_check], special]

(%i10) mike:15;

(%o10) 15

(%i10) :lisp (get ’$mike ’assign)

NIL

(%i10) :lisp (put ’$mike ’assign-mode-check ’assign)

ASSIGN-MODE-CHECK

(%i10) mike:15;

15 is not prime!!

#0: ptest(y=15)

-- an error. Quitting. To debug this try debugmode(true);

この例では,最初に素数でなければエラーを返す述語函数ptestを定義し,それから, define variable
函数で大域変数 tamaに初期値 5を与えて整数変数として宣言します.それから,qput函数を用い
て check value属性に ptestを与えます.この時, 変数mikeに 15を設定しようとすれば,15は素数
ではないのでエラーになっています.
ここまでの動作をもう少し詳しく解説しましょう.先ず,define variable函数による宣言で,変数

の型を integerと宣言しています.define variable函数は, 変数の型を any以外に指定していれば,
宣言する大域変数の assign属性に内部函数の assign-mode-check函数を割当てます.この内部函数
は変数の value check属性に設定された函数を用いて変数の評価を実行する函数です.次に,この
value check属性の設定で qput函数を用いています.ここでの例では,大域変数 tamaの value check
属性に ptest函数を割当てていますね.すると,大域変数 tamaに値を割当てる時に,assign-mode-
check函数が,大域変数 tamaの check value属性に割当てられた述語函数で割当てようとする値を
評価します.この例では,ptest函数に 15が引渡されます.この例では 15が素数でない為に falseと
なり,変数に 15が割当てられません.
次の例では, 大域変数 mike の型を any とした場合です. この場合, 変数 mike の value check

属性に値を設定しても, 先程の様に mike:15 を実行した場合にエラーも何も出ません. この場合,
大域変数 mike の assign 属性に assign-mode-check が設定されていない為です. これは LISP で
(get ’$mike ’assign) を実行すれば, NILが返されるので判ります.

そこで, :lisp (put ’$mike ’assign-mode-check ’assign) として見ましょう.これによって,属性 as-
signの値として assign-mode-checkを設定されます.すると,割当の際に検証が実行される様になり
ます.

1.5.5 atvalue函数による函数値の設定

Maximaでは属性は色々な所で用いられています.atvalue函数等を用いる事で, 函数の値の設定
も属性で与える事が出来ます.

atvalue函数¶ ³
atvalue(〈式 〉 ,〈点リスト 〉 , 〈境界値 〉 )
at(〈式 〉 ,〈点リスト 〉 )µ ´

atvalue函数による境界値の設定では,函数や式に対して点をリストで指定し, そのリストに対応
する境界値を設定します.
この atvalue函数で値を指定すると,atvalue属性に座標と函数値が対応付けられ, 〈式 〉 の prop-

ertiesリストに atvalue属性が追加されます.
尚,引数の 〈式 〉 には函数 f(〈v1〉 ,· · ·, 〈vn〉 ) やその微分が指定可能ですが,その場合には変数の

従属性が明確になる様に指定しなければなりません.
〈リスト 〉 は 〈変数1〉 =〈式 〉 の様に変数と値を演算子=で繋ぎ合せた式のリストになります.
atvalue(f(x),x=xˆ2+x+1,0)とした場合,x=xˆ2+x+1の左辺は函数 f(x) で用いた疑似変数であ

り,=も方程式で用いる演算子=とは意味が異なっています. その為,xˆ2+x+1で用いている変数 x
とは関係ありません.この事から, x2 + 1 = 0 の解 x = ±i を函数 f に代入しても函数 f は零にな

らない事に注意して下さい.函数 f が零になるのは f(x2 + x + 1) の時です.

(%i1) atvalue(f(x),x=x^2+x+1,0);

(%o1) 0

(%i2) f(x^2+1);

2

(%o2) f(x + 1)

(%i3) f(x^2+x+1);

(%o3) 0

atvalue函数で設定された値を printprops函数を使って表示する際に,函数の疑似変数を表記す
る為に,記号@1,@2,· · · が用いられます.

(%i35) atvalue(h(x,y,z),[x=1,y=0,z=0],10);

(%o35) 10

(%i36) atvalue(diff(h(x,y,w),w),[x=1,y=0,w=0],0);

(%o36) 0

(%i37) printprops(h,atvalue);

!

d !

--- (h(@1, @2, @3))! = 0

d@3 !

!@1 = 1, @2 = 0, @3 = 0

h(1, 0, 0) = 10

但し,atvalue函数で設定した函数の値は get函数で取出せず,削除も rem函数では出来ません.

(%i1) put(f,C-inf,type);

(%o1) C - inf

(%i2) atvalue(f(x),x=0,0);

(%o2) 0

(%i3) properties(f);

(%o3) [atvalue, [user properties, type]]

(%i4) get(f,type);

(%o4) C - inf

(%i5) rem(f,atvalue);

(%o5) false

(%i6) remove(f,atvalue);

(%o6) done

(%i7) properties(f);

(%o7) [[user properties, type]]

しかし,atvalueで設定した属性値は他のものと同様に printprops函数で函数名と属性を指定す
る事で表示されます.

(%i19) atvalue(f(x),x=0,0);

(%o19) 0

(%i20) atvalue(g(x),x=0,1);

(%o20) 1

(%i21) atvalue(g(x),x=1,2);

(%o21) 2

(%i22) printprops(all,atvalue);

f(0) = 0

g(0) = 1

g(1) = 2

(%o22) done

次の at函数は atvalue函数と対で用います. 〈式 〉 を atvalue函数に対して与えられるものと同
じ方程式のリスト,方程式中に含まれる変数に対し, atvalue函数で指定した値を入れて評価します.
部分式が atvalue函数で指定された値を持たない為に,評価出来ない場合,その部分式に作用する

at函数の名詞型が返されます.又,その部分式は二次元的書式で表示されます.

勾配と従属性の設定¶ ³
gradef(〈函数名 〉 (〈変数1〉 ,· · ·, 〈変数m〉 ),〈式1〉 ,· · ·, 〈式n〉 )
gradef(〈アトム 〉 ,〈変数 〉 , 〈式 〉 )
depends(〈函数 〉 ,〈変数1〉 ,· · ·, 〈函数n〉 ,〈変数n〉 )µ ´

gradef函数は 〈函数 〉 の n個の引数に対する微分を d〈f〉
dxi

= 〈式i〉 で定めます. 尚,gradef函数
で勾配を定義すると,大域変数 gradefs に函数名が蓄えられます.更に,指定した函数には gradef属
性が付与されます.
引数が変数の総数 m · n 個の勾配 nよりも少ない場合,最初の 〈函数 〉 の i番目の引数が参照さ

れます.xi は函数定義で用いる疑似変数と同類で,函数 〈函数 〉 の i 番目の変数を指定する為に用

います.
最初の引数を除く全ての gradef函数の引数は 〈式 〉 が定義された函数ならばその函数が呼出さ

れて結果が用いられます.勾配は函数が第一階微分を除いて正確に判らない場合で,より高階の微
分を得たい時に必要となります.

gradef(〈アトム 〉 ,〈変数 〉 ,〈式 〉 ) で変数 〈変数 〉 による 〈函数 〉 の微分が 〈式 〉 となる事を宣
言します.この場合,大域変数 gradefsには登録されず,属性も atomgrad属性になります.
この時, depends(〈f〉 ,〈x〉 ) も実行されるので, depends函数によって属性 dependency が付加さ

れ, 同時に大域変数 dependencies にも 〈函数 〉 が追加されます.
gradef函数はMaximaの既に定義された函数の微分を再定義する事にも使えますが, 添字された

函数に対して gradef函数は使えません.
depends函数で,函数の変数に対する従属を宣言します.例えば, depends(f,x) で函数 fが変数 x

に従属する性質を付加します.尚,depends函数による従属性の宣言にて,函数自体を予め定義する
必要はありません.

(%i41) depends(neko,[tama,mike]);

(%o41) [neko(tama, mike)]

(%i42) diff(neko,tama);

dneko

(%o42) -----

dtama

(%i43) diff(diff(neko,tama),tama);

2

d neko

(%o43) ------

2

dtama

(%i44) depends([rat1,rat2],[cheese,milk]);

(%o44) [rat1(cheese, milk), rat2(cheese, milk)]

(%i45) depends([rat1,rat2],[cheese,milk],neko,[tama,mike]);

(%o45) [rat1(cheese, milk), rat2(cheese, milk), neko(tama, mike)]

勿論,depends函数を実行していなければ,diff函数を実行した時点で 0になります. depends函数
で nekoが変数 tamaと mikeに従属する函数と宣言した為に,微分を行っても零になりません.最
初の例では函数 nekoが 1成分しかない為に,リストの大括弧を外しています.
函数の変数に対する従属性は大域変数 dependenciesに登録された函数の情報から調べられます.

gradefと dependsに関連する大域変数¶ ³
変数名 初期値 概要

gradefs [] 勾配を定義された函数名リスト

dependencies [] 従属性を宣言された函数名リストµ ´
gradef函数で勾配を定義すると,大域変数 gradefsに函数名が蓄えられます.
大域変数 dependenciesには,dependsや gradefで指定された函数名が蓄えられます.
これらの変数は共にデフォルト値が空のリスト []です.

(%i4) gradef(f(x,y),y,x);

(%o4) f(x, y)

(%i5) gradefs;

(%o5) [f(x, y)]

(%i6) diff(f(x,y),x);

(%o6) y

(%i7) diff(f(x,y),y);

(%o7) x

(%i8) dependencies;

(%o8) []

(%i9) depends(g,x,y,z);

(%o9) [g(x), y(z)]

(%i10) dependencies;

(%o10) [g(x), y(z)]

(%i11) gradefs;

(%o11) [f(x, y)]

(%i12) gradef(h,x,x^2);

(%o12) h

(%i13) dependencies;

(%o13) [g(x), y(z), h(x)]

この例で示す様に,gradefや depends函数を用いる事で,gradefsや dependenciesに函数名が蓄積
されます. 尚,gradef を用いて dependencies に函数を追加する為には, gradef(〈 アトム 〉 ,〈 変数 〉
,〈式 〉 ) の形式に限定されます.

1.5.6 属性と属性値の削除

属性値の削除は rem函数や remove函数で行えます.この場合はアトムと属性を指定すると, ア
トムに設定した属性と属性値が一緒に削除されます.

属性や属性値を削除する函数¶ ³
rem(〈アトム 〉 ,〈属性 〉 )
remove(〈アトム1〉 ,〈属性1〉 ,· · ·, 〈アトムn〉 ,〈属性n〉 )
remove([〈アトム1〉 ,· · ·,〈アトムm〉 ], [〈属性1〉 ,· · ·,〈属性n〉 ])
remove (”〈アトム 〉 ”,operator)
remove(〈アトム 〉 ,transfun)
remove (all,〈属性 〉 )µ ´

rem函数は 〈アトム 〉 から 〈属性 〉 で指定した属性とその属性値の両方を削除します.
これに対し,remove函数は変数や関数に関連した属性の一部や全てを削除します. この属性はシ

ステム側が定義したものでも利用者が与えたものや function, mode declareでも構いません.
remove(〈アトム1〉 ,〈属性1〉,· · ·, 〈アトムn〉 ,〈属性n〉 ) で, 〈属性i〉 を 〈アトムi〉 から削除します.
ここで指定するアトムと属性は各々が対応するリストでも構いません.
属性に operatorや opを指定した場合,declare等で宣言した prefix(前置式), infix(内挿式),nary(内

挿式),postfix(後置式),matchfixや nofix(無演算子) といった演算子の属性が削除されます.尚,演算
子には必ず二重引用符で括る必要がある事に注意して下さい.
属性が transfunの場合,translate函数で変換されたLISP函数が削除されます.この後は, Maxima

版の函数が用いられます.
引数にアトムではなく allを指定した場合,指定された属性を持つアトム全てから, その属性が削

除されます.
尚,remove函数は与えられた属性が存在しない時でもエラーを返しません.返却値は常に done

です.

1.5.7 属性の表示

属性の表示に関連する函数としては,properties函数,propvars函数と printprops函数があります.

属性の表示に関連する函数と大域変数¶ ³
properties(〈アトム 〉 )
propvars(〈属性 〉 )
propsµ ´

properties函数は 〈アトム 〉 に関連する全ての属性を記載したリストを生成します.
以下の例では put函数でアトムに属性を指定し,properties函数でアトムに指定した属性に何が

あるかを確認し,指定した属性に対応する属性値を取出しています.

(%i37) put(Mike,"2004/07/4",birthday);

(%o37) 2005/07/4

(%i38) put(Mike,"10[Kg]",Weight);

(%o38) 10[Kg]

(%i39) put(Mike,"White-Black-Red",Color);

(%o39) White-Black-Red

(%i40) properties(Mike);

(%o40) [[user properties, Color, Weight, birthday]]

(%i41) get(Mike,Color);

(%o41) White-Black-Red

propvarsは大域変数 propsのリストに含まれる成分から, 〈属性 〉 を持つアトムのリストを生成
します.例えば, propvars(atvalue) で atvalue 函数で値が設定されたアトムのリストを生成します.

(%i23) atvalue(f(x),x=0,0);

(%o23) 0

(%i24) atvalue(g(x),x=1,0);

(%o24) 0

(%i25) propvars(atvalue);

(%o25) [f, g]

大域変数 propsには declare函数,atvalue函数やmatchdeclares函数等で属性が指定されたアト
ムが追加されたリストが割当てられています.

(%i1) props;

(%o1) [nset, kron_delta, dva, %n, %pw, %f, %f1, l%, solvep, %r, p, %cf,

algebraicp, hicoef, genpol, clist, unsum, prodflip, prodgunch, produ, nusum,

funcsolve, dimsum, ratsolve, prodshift, rforn, rform, nusuml, funcsol,

desolve, eliminate, bestlength, trylength, sin, cos, sinh, cosh, list2,

trigonometricp, trigsimp, trigsimp3, trigsimp1, improve, listoftrigsq,

specialunion, update, expnlength, argslength, pt, yp, yold, %q%, ynew, method,

%f%, %g%, msg1, msg2, intfactor, odeindex, singsolve, ode2, ode2a, ode1a,

desimp, pr2, ftest, solve1, linear2, solvelnr, separable, integfactor, exact,

solvehom, solvebernoulli, genhom, hom2, cc2, exact2, xcc2, varp, reduce, nlx,

nly, nlxy, pttest, euler2, bessel2, ic1, bc2, ic2, noteqn, boundtest, failure,

adjoint, invert]

(%i2) properties(invert);

(%o2) [transfun, transfun]

(%i3) properties(failure);

(%o3) [transfun, transfun]

(%i4) properties(kron_delta);

(%o4) [symmetric, database info, kind(kron_delta, symmetric), rule]

(%i5) propvars(rule);

(%o5) [kron_delta, sin, cos, sinh, cosh]

この例で示す様に,大域変数 propsには属性が設定されたMaximaの函数や変数が登録されてい
ます.

printprops函数¶ ³
printprops(〈アトム 〉 ,〈属性 〉 )
printprops([〈アトム1〉 ,· · ·,〈アトムn〉 ], 〈属性 〉 )
printprops(all,〈属性 〉 )µ ´

printprops函数は 〈アトム 〉 と 〈属性 〉 に対応する属性値を表示する函数です.〈アトム 〉 のリス
トも指定可能ですが, 〈属性 〉 は一つだけです.
尚,printpropsで表示可能な属性は以下のものに限定されます.

printpropsで表示可能な属性¶ ³
属性名 概要

atvalue 函数値属性.atvalue函数で与えられます.
atomgrad 勾配属性.gradef函数で与えられます.
gradef 勾配属性.gradef函数で与えられます.
matchdeclare 並びの照合変数属性.matchdeclare函数で与えられます.µ ´
第一引数に allを指定すると,指定した属性を持つ全てのアトムと値が表示されます.

(%i30) matchdeclare([_a,_b],true);

(%o30) done

(%i31) printprops(all,matchdeclare);

(%o31) [true(_b), true(_a)]

1.6 演算子

1.6.1 Maximaの演算子について

Maximaには数学で用いられる算術演算子,変数に式を割当てる演算子等の様々な演算子があり
ます.演算子は利用者が新規に定義する事も可能で,既存の演算子の属性を変更する事も容易に行
えます.

Maximaの演算子には前置表現,内挿式, 後置表現,無引数の 5種類の演算子の属性があります.
前置式は 1引数の演算子で,演算子が引数の前に置かれる書式になります.
例えば,数学の微分演算子 d

dx の様な書式になります.次の内挿表現の演算子は + の様に二つの
引数を演算子の左右に置く演算子です.そして,後置式の演算子は 3! の様に演算子の後に引数を置
く演算子で,最後の無引数の演算子は引数を取らない演算子になります.

Maximaでは演算子 (或いは函数)をその引数の数に応じて前置,内挿,後置表現演算子として指
定したり,既存の演算子の属性を変更する事が可能です.演算子の属性はMaximaの属性を利用し
て与えています.この節では演算子の属性に限定して解説しますが,属性の詳細に関しては,1.5節を
参照して下さい.
最初に簡単な例として,適当な文字列を演算子として宣言してみましょう.

(%i25) prefix("mike");

(%o25) mike

(%i26) mike neko;

(%o26) mike neko

(%i27) infix(":/")$

(%i28) x :/ mike y;

(%o28) x :/ mike y

この例では,函数としての実体が無い為に評価はこれ以上行われていません. この宣言の後に演
算子の定義を行う事や函数の定義の後に演算子としての宣言を行う事も出来ます.

(%i29) mike x:=2*x+1;

(%o29) mike x := 2 x + 1

(%i30) x :/ y := (x+sin(x))/y;

sin(x) + x

(%o30) x :/ y := ----------

y

(%i31) pochi(x,y):=x^y;

y

(%o31) pochi(x, y) := x

(%i32) nary("pochi");

(%o32) pochi

(%i33) mike 3;

(%o33) 7

(%i34) 5 :/6;

sin(5) + 5

(%o34) ----------

6

(%i35) 4 pochi 2;

(%o35) 16

この例では函数 (mikeや pochi)の定義を行っています.この定義で用いている:=も内挿表現の演
算子の一つです.この様にMaximaでの演算子は引数を括弧で括る必要がない函数とも看倣せます.

1.6.2 演算子の束縛力

普通,数学の演算子には順位があります.例えば,1+a*bˆ2*c-dが与えられたとしましょう.通常は
この式を (1+a*((bˆ2))*c))-dの様に演算子と被演算子と結び付きを演算子の強さで解釈していま
すね.Maximaの場合,この演算子の強さを束縛力 bp (Binding Power) と呼び,200までの整数で表
現しています.更に,演算子と被演算子の配置関係から束縛力は二つの属性,即ち,左束縛力 lbp(Left
Binding Power) と右束縛力 rbp(Right Binding Power) に分けて指定出来る演算子もあり,Maxima
では左右の束縛力を指定可能な演算子が大多数です.
この束縛力は演算子の属性に応じて利用者が自由に設定する事が出来ます. 因に,Maximaの組

込演算子の属性 lbpと rbpの値は,nparse.lispで設定されています. 例えば,和演算子+は両方共に
100,可換積演算子*は lbpのみ 120,羃演算子は lbpが 140で rbpが 139,最後に差の演算子は lbpが
100で rbpが 134となっています.左右の束縛力の値が大きなもの程,演算子と被演算子の結び付き
が強くなります.例えば,可換積が 120 ですが,一方で羃は 140となっています.これは羃の方が被
演算子を引き付ける力が大きな事を意味し,それ故に,bˆ2*cは (bˆ2)*cと解釈される事になります.
但し,演算子の束縛力を指定せずにしておく事も可能です.特に束縛力を指定しない場合,デフォ

ルトで 180が設定されます.この事を実際に確認してみましょう.

(%i1) prefix("tama");

(%o1) tama

(%i2) :lisp (get ’$tama ’lbp);

NIL

(%i2) :lisp (get ’$tama ’rbp);

180

この例では前置表現の演算子 tamaを定義し,その左右の束縛力を取出しています. この束縛力
の設定はMaximaでは LISPの属性リストを用いています.その為,get函数を用いて調べられます.
この例では,演算子 tamaが前置表現の為,左側に演算子は不要となるので左束縛力を設定する必要
はありません.但し,右束縛力は必要なので 180が設定されている事が判ります.
次に,後置式表現の演算子mikeを定義して束縛力を調べてみましょう.

(%i4) postfix("mike");

(%o4) mike

(%i5) :lisp (get ’$mike ’lbp);

180

(%i5) :lisp (get ’$mike ’rbp);

NIL

この場合は前置式演算子とは束縛力が逆になっている事が判りますね.
この束縛力は演算子以外の右小括弧 (や左小括弧)といった記号にも設定されています. ここで,

小括弧の束縛力は最大で 200となっています.その為,処理に不安がある場合は, 小括弧でひと纏め
にすると演算子の影響は括弧の中に及ばない事になります.
この束縛力で重要な事として,演算子の左右の結合律にも影響する事が挙げられます.

(%i5) infix("><",100,120);

(%o5) ><

(%i6) (a >< b):=a^b;

b

(%o6) (a >< b) := a

(%i7) a><b><c;

b c

(%o7) (a )

(%i8) infix("><",120,100);

(%o8) ><

(%i9) a><b><c;

c

b

(%o9) a

この例では演算子><を最初に左束縛力を 100,右束縛力を 120で定義しています. その為,a><b><c
は右束縛力の方が強い為に,(a><b)><cで処理されます. ところが,次に左束縛力を 120,右束縛力
を 100と逆にした為,演算子と左側の被演算子との繋がりが強くなった為に,a><b><cは a><(b><c)

で処理されます. その為,左右の結合律に関しては演算子の左右の束縛力の差も考慮しなければな
りません.
又,束縛力は内挿式や前置式の演算子本体の定義でも重要です.函数として定義する場合,その時

に用いる演算子:=の束縛力がある為です.この演算子:=の束縛力は左が 180.右が 20となっていま
す.その為,演算子の束縛力が弱い,即ち,演算子の右束縛力が 180よりも小さい場合には演算子:=
の側に被演算子が引き寄せられる事になります. この場合,先程述べた様に小括弧で被演算子と一
緒に演算子を括ります.この小括弧は最大値の 200が左右の束縛力として設定されているからです.
その為,Maximaの式を記述する際に,演算子の束縛力に疑問があれば,式を小括弧で一纏めにする
と良いでしょう.

1.6.3 演算子の型

演算子の被演算子と結果に型があります.この型の指定は,演算子の宣言を行う函数で行えます
が,この時,演算子の属性として指定されています. 先ず,nary函数とmatchfix函数で宣言される演
算子の場合,被演算子の型は, argpos属性に指定し,返却値の型は pos属性に設定します.その他の
函数では,左右の被演算子を独立して宣言します.実際,左被演算子と右被演算子は lpos(Left part
of speech)属性と rpos(Right part of speech)属性, そして結果は pos(Part of speech)で指定しま
す.その指定可能な型を以下に示します.

演算子の型¶ ³
型 概要 属性値

expr 数式全般 algebraic
clause 論理式全般 logical
any 任意のMaximaの式 untypedµ ´
最初に exprはMaximaで扱う数式全般になります.これに対し clauseは評価を行う事で,trueや

falseを返す述語や論理式です.anyは,これらの式を含むMaximaの任意の式となります.
尚,expr,clause,anyには属性として englishがあり,その属性値は各々algebraic,logical,untypedと

なっています.

(%i1) :lisp (get ’$expr ’english);

algebraic

(%i1) :lisp (get ’$clause ’english);

logical

(%i1) :lisp (get ’$any ’english);

untyped

演算子宣言の函数で演算子を宣言する際に型を指定しなければ,左右の被演算子の型と返却値の
型として$anyが自動的に設定されます.尚,これらの属性は LISPの属性リストを用いて実現され
ている為,LISPの put函数でも強引に設定が出来ます.

(%i4) prefix("mike");

(%o4) mike

(%i5) :lisp (get ’$mike ’pos)

$ANY

(%i5) :lisp (put ’$mike ’$clause ’pos)

$CLAUSE

(%i5) :lisp (get ’$mike ’pos)

$CLAUSE

(%i5) mike a := freeof(a,x);

(%o5) mike a := freeof(a, x)

(%i6) if mike (x^2+1) then print("test1");

test1

(%o6) test1

(%i7) :lisp (put ’$mike ’$expr ’pos)

$EXPR

(%i7) if mike (x^2+1) then print("test1");

Incorrect syntax: Found algebraic expression where logical

expression expected if mike (x^2+1) then

^

この例では前置式演算子mikeを定義しています.デフォルトでは$anyになっているものを LISP
の函数 putを使って属性 posの値を clauseに変更します.次にmike本体を定義しています.if文で
はmikeは trueか falseの何れかを返す述語函数となっているので問題ありませんが,次に LISPの
put函数を使って,今度は型を expr型にしています. ここで,ifは述語を要求するので,式 expr型と
は異なります,その為,エラーになっています.

1.6.4 演算子の属性を宣言する函数

ここでは最初に演算子の属性を宣言する函数から述べましょう. Maximaでは利用者が函数やア
トムを演算子として利用する事が可能です.特に,適当な文字列を演算子の属性を宣言する函数を
用いて属性を与えてしまえば,それだけで演算子として利用が出来ます.更に,値もきちんと函数と
して定義しなくても, atvalueの様な属性で与える事も可能です.以下に組合せを計算する演算子 C
の定義の様子を示します.

(%i62) nary("C");

(%o62) C

(%i63) m C n:= m!/(n!*(m-n)!);

m!

(%o63) m C n := -----------

n! (m - n)!

(%i64) 5 C 3;

(%o64) 10

演算子の属性を宣言する函数には以下のものがあります.

演算子の属性を宣言する函数¶ ³
infix infix(a) 文字列 aを内挿表現の演算子として宣言
infix infix(a,lbp,rbp) 内挿表現の演算子 aに左右の束縛力を含めて宣

言

infix infix(a,lbp,rbp,lpos,rpos,pos) 内挿表現の演算子 aに左右の束縛力と型を含め
て宣言

nary nary(a) 文字列 aを内挿表現の演算子として宣言
nary nary(a,bp) 内挿表現の演算子 aに束縛力を含めて宣言
nary nary(a,bp,argpos,pos) 内挿表現の演算子 aに束縛力と型を含めて宣言
nofix nofix(a) 文字列 aを無引数の演算子として宣言
nofix nofix(a,pos) 無引数の演算子 aに出力の型を含めて宣言
postfix postfix(a) 文字列 aを後置表現の演算子として宣言
postfix postfix(a,lbp) 後置表現の演算子 aに左束縛力を含めて宣言
postfix postfix(a,lbp,rpos,pos) 後置表現の演算子 aに左束縛力と型を含めて宣

言

prefix prefix(a) 文字列 aを前置表現の演算子として宣言
prefix prefix(a,rbp) 前置表現の演算子 aに右束縛力を含めて宣言
prefix prefix(a,rbp,rpos,pos) 前置表現の演算子 aに右束縛力と型を含めて宣

言

matchfix matchfix(a,b) 変数を文字列 aと文字列 bで挟む演算子を宣言
matchfix matchfix(a,b,argpos,pos) 引数の型と結果の型を含めて宣言µ ´
この表で,lbpと rbpが左束縛力と右束縛力,lposと rposが左右の被演算子の型,posが返却値の

型です. 又,matchfix 函数と infix 函数では左右に分けずに演算子の束縛力を bp, 被演算子の型を
argposとしています.
最初の infixは内挿 (infix)表現の演算子の宣言を行います.内挿表現の演算子は二つの引数を持

つ函数で, a + b の和の演算子 + の様に引数が演算子の前後に置かれるものです.nary函数も内挿
表現の演算子の宣言を行います.

nary函数も,函数を内挿表現演算子として宣言する事が出来ます.この nary函数で宣言可能な演
算子は左右の束縛力が同じでも構わないものに限定されます.とは言え, デフォルトでは全て 180
が設定されるので,左右の束縛力を調整する必要が無ければ, nary函数の方が変数が少ないので使
い易いかもしれません.

nofix函数は無引数演算子の宣言を行います.無引数演算子は演算子が何等の引数を持たない事
を明示する為に使われます.

postfix函数は後置表記演算子の宣言を行います.後置表記の演算子は一つの引数のみを持ち, 3!
の様にその引数は演算子の前に置かれます.

prefix函数は前置表記演算子の宣言を行います.前置表記の演算子は一個の引数のみを持ち,その
引数は演算子の直後に置かれるものです.

matchfix函数は任意個数の引数を二つの文字列で囲む演算子を宣言します.

%i5) matchfix("@-","-@");

(%o5) @-

(%i6) @- a,b,c,d,e,f -@:=a*b*c+d*e^f;

f

(%o6) @-a, b, c, d, e, f-@ := a b c + d e

(%i7) @- 1,2,3,4,5,6 -@;

(%o7) 62506

(%i8) dispfun("@-");

f

(%t8) @-a, b, c, d, e, f-@ := a b c + d e

(%o8) done

尚,dispfun函数を使って matchfixで定義した演算子に割当てた利用者定義函数の内容を見る場
合,演算子の先頭の文字列だけを二重引用符で括ったものを dispfun函数の引数とします.
上記の函数で設定した属性は kill函数や remove函数で削除する事が可能です.尚,removeの場合

は属性のみを削除しますが,kill函数の場合は演算子として定義した文字列そのものを削除してしま
います.

(%i10) nary("tama");

(%o10) tama

(%i11) a tama b:=a+b^2;

2

(%o11) a tama b := a + b

(%i12) properties("tama");

(%o12) [function, operator, noun]

(%i13) remove("tama",op);

(%o13) done

(%i14) properties("tama");

(%o14) []

(%i15) prefix("mike");

(%o15) mike

(%i16) mike x:=x!+1;

(%o16) mike x := x! + 1

(%i17) kill("mike");

(%o17) done

(%i18) properties("mike");

(%o18) []

この例では,中置演算子 tamaを定義し,属性を削除する remove函数で演算子 tamaを削除して
います.ここで,tamaの属性は properties函数を用いて調べられます. 演算子の属性を removeを実
行された時点で tamaの属性全てが削除されています. 又,演算子mikeに対しては kill函数を使っ
て演算子mike全てを削除しています. この様に remove函数と kill函数で削除する事が可能です.
尚,この例では remove函数は演算子の属性を指定した為,副作用として演算子全体が削除されて

しまいましたが,propertiesで調べた特定の属性だけを削除する函数です.

(%i19) nary("tama");

(%o19) tama

(%i20) a tama b:=a+b^2;

2

(%o20) a tama b := a + b

(%i21) remove("tama",function);

(%o21) done

(%i22) properties("tama");

(%o22) [operator, noun]

(%i23) 3 tama 4;

(%o23) 3 tama 4

この例では,函数を削除しただけで演算子としての属性は残しています.その為, 3 tama 4; と

入力してもエラーにはなりません.
演算子定義函数で定められる属性¶ ³

演算子 左束縛力 右束縛力 左変数型 右変数型 返却値型

postfix 180 any any
prefix 180 any any
infix 180 180 any any
nofix any
nary any any any
matchfix any any anyµ ´
基本的に,変数型と返却型は$anyに設定されます.尚,nary函数とmatchfix函数で宣言された演

算子の被演算子型は他の函数と異なり,左右の変数型属性 lposや rposではなく,argpos属性で指定
し,その属性値は$anyです.

1.6.5 数式演算子

Maximaには数式に関連する二項演算子として以下のものがデフォルトで定義されています.
二項数式演算子¶ ³

+ a + b aと bの和
− a − b aと bの差
∗ a ∗ b aと bの可換積
/ a / b aの bによる商
∗∗ a ** b 羃 ab

ˆ a ˆ b 羃 ab.a**bと同じ
. a . b aと bの非可換積

ˆˆ a ˆˆ b aと bの非可換積の羃乗µ ´
これらの演算子の持つ属性を以下に示しておきます.

二項数式演算子の持つ属性¶ ³
演算子 左束縛力 右束縛力 左変数型 右変数型 返却値の型

+ 100 100 不要 expr expr
− 100 134 不要 expr expr
∗ 120 不要 expr 不要 expr
/ 120 120 expr expr expr
ˆ 140 139 expr expr expr
∗∗ 140 139 expr expr expr
. 130 129 expr expr expr

ˆˆ 140 139 expr expr exprµ ´
これらの演算子は通常の数値計算で利用される演算子と殆ど同じ表記になります. 但し,Maxima

の内部では−,/の扱いは表示された式とは別の表現となっています.例えば,x−yは x+(−1)∗y,x/y

は x ∗ y−1となっています.この内部表現を強制的に本来の形に変換する函数 dispform もあります.
又,非可換積 .と非可換積の羃ˆˆはMaxima独特の演算子で,関連する非可換積の羃ˆˆと通常の可
換積の羃ˆは別物です.例えば,aˆˆ3は a . a . aを意味し,aˆ3は a*a*aを意味します.尚,可換積*と
非可換積.,そして,それらに対応する羃乗ˆとˆˆは混在しても構いません. Maximaのデフォルトの
表示では,ˆは x3 に似せて上付きで表示され, 演算子ˆˆは x〈n〉 の様に 〈〉で括られて表示されます.

(%i1) a^^b;

<b>

(%o1) a

尚,非可換積.と非可換羃ˆˆは行列演算で特に利用します.この場合, 非可換積が通常の行列の積
を意味し,可換積は同じ大きさの行列に対し,成分毎の積に対応します.

(%i54) A:matrix([1,2],[3,4]);

[ 1 2 ]

(%o54) [ ]

[ 3 4 ]

(%i55) B:matrix([2,1],[4,3]);

[ 2 1 ]

(%o55) [ ]

[ 4 3 ]

(%i56) A*B;

[ 2 2 ]

(%o56) [ ]

[ 12 12 ]

(%i57) A.B;

[ 10 7 ]

(%o57) [ ]

[ 22 15 ]

Maximaでは可換積の羃ˆと非可換積の羃ˆˆの計算結果が長くて式が表示し切れない場合,記号
exptと ncexptが各々の演算の表記で利用されます.
非可換積.と非可換羃ˆˆにはその挙動を制御する様々な大域変数が存在します.以下に,代表的な

大域変数を示しておきます.
非可換積演算子に関連する大域変数¶ ³

変数名 初期値 概要

dot0nscsimp true trueの場合,零と非スカラの項の非可換積を可換積に変換します.

dot0simp true trueの場合,零のスカラ項の非可換積を可換積に変換します.
dot1simp true trueの場合,1との他の項との非可換積を可換積に変換します.
dotassoc true trueの場合, (a.b).c と a.(b.c) を a.b.cに変換します.即ち,結合

律を式に対して自動的に適用します.
dotconstrules true trueの場合,定数と項の非可換積が可換積で置換されます. この

大域変数は dotocimo,dotonscsimp,dot1simpも影響を与えます.
dotdistrib false trueの場合,a.(b + c)を a.b + a.cに置換します. 即ち,分配律を自

動的に適用します.
dotexptsimp true trueの場合,複数の同じ式による非可換積を非可換積の羃乗に変

換します.
dotident 1 非可換羃の 0乗で返される値を設定します.
dotscrules false trueの場合,群環の元 aとスカラー bとの非可換積を可換積に変

換します.µ ´

尚,ここでのスカラーと群環の元との非可換積に関する大域変数は declare函数で scalar属性を
与えた式に対して制御が効きますが,1,2等の数値に対しては,これらの大域変数とは無関係に可換
積で置換されます.

(%i6) 2 . 3 . x . y;

(%o6) 6 (x . y)

尚,非可換積をMaximaで記述する場合,a . bの様に空白を空けるべきです. これは引数が数値
の場合は必須です.例えば,1.2は 1と 2の非可換積でなく, 浮動点小数 1.2を意味し,1. 2や 1 .2の
様に半端に空行を入れてしまうとMaximaの式としての意味を持ちません.
大域変数 dotassocを falseにすると,複数の非可換積で勝手に括弧を外されない為, 非可換積に

対する規則を定義した場合に,その適用が容易になります.

後置式の数式演算子¶ ³
! n! nが整数の場合,nの階乗を計算.一般の場合 Γ(x + 1)
!! n!! nが奇数 (偶数)ならば,n以下の奇数 (偶数)の積µ ´
後置式演算子は引数を一つ取ります.Maximaでは階乗に関連する ! と !! が後置式の演算子とな

ります.但し, !! は正確には演算子ではありませんが,演算子の属性を持たせた函数です.
先ず, n! は n が整数の場合に n の階乗を計算します.一般の数値に対しては Γ (n + 1) の計算を

行います. ここで, Γ (n + 1) の Γ は Γ 函数と呼ばれる函数で, Γ(x) =
∫ ∞
0

tx−1e−tdt で定義され

ています.
n!! は少し変った函数で,整数 n が奇数の場合, n 以下の全ての奇数の積を計算し, n が偶数の場

合は同様に n 以下の全ての偶数の積を計算します. その為, n! = n!!(n − 1)!! が成立します.この

函数は,
∏entier(n

2 )
i=0 (n − i) の計算を行っています. この n!! の計算では, n! と違い整数以外の数値

に対して Γ 函数を用いるものではありません.

(%i6) 10!;

(%o6) 3628800

(%i7) 10!!;

(%o7) 3840

(%i8) 9!!;

(%o8) 945

(%i9) 10!!*9!!;

(%o9) 3628800

これらの演算子の属性を以下に纏めておきます.

後置式表現演算子の属性¶ ³
演算子 左束縛力 右束縛力 左変数型 右変数型 返却値型

! 160 expr expr
!! 160µ ´

1.6.6 論理演算子

Maximaの論理演算子は,式や述語を被演算子とし,trueや falseといった値を返すMaximaの式,
即ち,述語を構成する演算子です.否定の not,論理和の orや論理積の andを除くと基本的に二項間
関係を表すものとなります.

論理演算子¶ ³
not not a 述語 aを否定
and a and b 述語 aと bの論理積
or a or b 述語 aと bの論理和
= a = b aと bが等しい
# a # b aと bが等しくない

>= a >= b aは b以上
> a > b aは bよりも大

<= a <= b aは b以下
< a < b aは bより小µ ´

先ず,Cの&&や——に相当するものが,andと orになります. 二項間の関係を示す演算子は,Max-
ima独特の#と=を除くと,Cや FORTRANで利用されるものとの違いはありません.
ここで,注意が必要なのは演算子=です.Maximaでは変数への値の割当では:を用い,等しい事を

示す関係子に=を用います.Cの==に相当します. この割当の演算子と等号は混同し易いので注意
して下さい.

論理演算子の属性値¶ ³
演算子 左束縛力 右束縛力 左変数型 右変数型 返却値の型

not 70 clause clause clause
and 65 clause clause
or 60 clause clause
= 80 80 expr expr clause
# 80 80 expr expr clause
>= 80 80 expr expr clause
> 80 80 expr expr clause
<= 80 80 expr expr clause
< 80 80 expr expr clauseµ ´

1.6.7 割当の演算子

割当の二項演算子¶ ³
: a : b aに bの値を割当てる
:: a :: b aに bの値を割当てる

::= a ::= b 本体が bのマクロ aを定義
:= a:=b 本体が bの函数 aを定義µ ´
通常の変数に何等かの値を割当てる場合には:を利用します.ここで,演算子=は前述の様に割当

ではなく,二つの式が等しい事を示す演算子となるので意味が異なります.
Cで割当に用いられる演算子:=はMaximaで函数の定義で用います.

割当の二項演算子の属性¶ ³
演算子 左束縛力 右束縛力 左変数型 右変数型 返却値の型

: 180 20 any any any
:: 180 20 any any any
::= 180 20 any any any
:= 180 20 any any anyµ ´
尚,割当の演算子の左束縛力は演算子を宣言する函数で宣言した演算子の束縛力と同じ大きさで

すが,infix函数等で右束縛力を 180よりも小さくすると,割当の演算子に吸い取られてしまうので
演算子の定義で,演算子と被演算子を含む左辺全体を小括弧括る等の注意が必要になります.

(%i7) infix("tama",111,111)$

(%i8) x tama y:= x+y*2;

Improper function definition:

y

-- an error. Quitting. To debug this try debugmode(true);

(%i9) (x tama y):= x+y*2;

(%o9) (x tama y) := x + y 2

(%i10) 2 tama z;

(%o10) 2 z + 2

この例では,左右の束縛力を 11に設定した為,最初に演算子を定義した時に, 仮の被演算子 yが
演算子:=に引き寄せられてしまい,その結果エラーになっています. そこで,x tama y全体を小括
弧で括ります.この括弧の束縛力は 200の為,:=の束縛力はそこで遮断され,演算子 tamaが無事に
定義出来ます.

1.6.8 その他の演算子

Maximaでは式を構成する多くが内部的には演算子としての扱いになっています. 特に束縛力は
式の評価やプログラム作製で大きく影響します.

記号の属性¶ ³
演算子 左束縛力 左変数型 返却値型

] 5
[ 200 any any
) 5
( 200
’ 190
” 190
, 10 any anyµ ´
先ず,大括弧や小括弧にも束縛力があります.この束縛力を上手く使う事で, block文内部のMaxima

の式や被演算子をグループ化する事を可能にしています.
更に,if文や do文, そして関連する語句にも同様の属性があります.

if文に関連する演算子¶ ³
演算子 左束縛力 右束縛力 右変数型 返却値型

if 45 clause any
then 5 25
else 5 25
elseif 5 45 clause anyµ ´

do文に関連する演算子¶ ³
演算子 左束縛力 右束縛力 右変数型 返却値型

for 25 200 any any
from 25 95 any any
step 25 95 expr any
next 25 45 any any
thru 25 95 expr any
unless 25 45 clause any
while 25 45 clause any
do 25 25 any anyµ ´

1.7 規則と式の並びについて

Maximaは単純に式を纏めたり展開する事で式を変形するだけではなく,函数や変数に規則を与
える事で式の置換操作を行う事が可能です.
例えば, tan (x) と sin(x)

cos(x) が同値であるという事も規則のひとつです.
Maximaにはデフォルトで幾つかの規則が設定されています.これらは主に三角函数に関するもの

です.規則名だけを知りたい場合,大域変数の rulesに規則名のリストが設定されているので, rules;
と入力してみて下さい.すると,その時点でMaximaに追加されている規則名一覧のリストが表示
されます.

(%i1) rules;

(%o1) [trigrule0, trigrule1, trigrule2, trigrule3, trigrule4,

htrigrule1,htrigrule2, htrigrule3, htrigrule4]

ここで表示される規則の具体的な内容はどの様なものなのでしょうか？それを表示する函数とし

て disprule函数や letrules函数が存在します.

規則の表示を行う函数¶ ³
disprule(〈規則1〉 ,〈規則2〉 ,· · · )
disprule(all)
letrules(〈規則パッケージ 〉 )
letrules()µ ´

disprule函数や letrules函数はMaximaに設定した規則の表示を行う函数です. disprule函数は
defrule函数,tellsimp函数,tellsimpafter函数で利用者が設定した規則の詳細を defmatch函数によっ
て定義された並びの名前込みで表示します.特に, 引数を allとするとMaximaが持つ全ての規則を
表示します.
これに対して letrules函数は let函数で定義した規則を表示する函数です.引数として指定した

〈規則パッケージ 〉 に含まれる規則の詳細を表示します. ここで,引数を指定しない letrules()の場
合,大域変数 current let rule packageに割当てられた規則パッケージに含まれる規則を表示します.
尚,この大域変数の初期値は default let rule package です.
では,実際に disprule(all); を実行してみましょう.



1.7. 規則と式の並びについて 51

(%i10) disprule(all);

sin(a)

(%t10) trigrule0 : tan(a) -> ------

cos(a)

sin(a)

(%t11) trigrule1 : tan(a) -> ------

cos(a)

1

(%t12) trigrule2 : sec(a) -> ------

cos(a)

1

(%t13) trigrule3 : csc(a) -> ------

sin(a)

cos(a)

(%t14) trigrule4 : cot(a) -> ------

sin(a)

sinh(a)

(%t15) htrigrule1 : tanh(a) -> -------

cosh(a)

1

(%t16) htrigrule2 : sech(a) -> -------

cosh(a)

1

(%t17) htrigrule3 : csch(a) -> -------

sinh(a)

cosh(a)

(%t18) htrigrule4 : coth(a) -> -------

sinh(a)

disprule函数を使って表示される規則は,先ず,左端が規則名,->の左側が適用される函数,右側が
適用される函数が置換される函数となります. 例えば,最初の trigrule0は tan (x) を sin(x)

cos(x) で置換

する規則です.

ところで,Maximaに tan(x); と入力しても,直ちに変換される訳ではありません.実際の利用で
は函数 tanは式の中にあり,その変数も xの様な単純なものであるとは限りません.その為,与えら
れた式の並び方 (パターンとも呼びます) を指定し,その並び方に対して規則を当て嵌める処理を実
行しなければなりません. この処理を並びの照合 (パターンマッチング) と呼びます.
ここで trigrule0は単純に tan (x) を sin(x)

cos(x) で置換する事を意味しますが,この置換の為には,置
換を行う函数を必要とします. disprule函数で表示される規則は apply1,apply2,applyb1函数を用
います.

apply函数族¶ ³
apply1(〈式 〉 ,〈規則1〉 , · · · ,〈規則n〉 )
apply2(〈式 〉 ,〈規則1〉 , · · ·,〈規則n〉 )
applyb1(〈式 〉 ,〈規則1〉 , · · ·,〈規則n〉 )µ ´

apply函数族は基本的に defrule函数で与えられた規則を式に適用させる函数ですが, apply1函数
と apply2函数が式の木構造の上側から作用するのに対し,applyb1函数が木構造の下側 (Bottom)
から作用させます.
最初の apply1函数は与えられた 〈式 〉 に対して 〈規則1〉 を最初に作用させます.この時, 〈式 〉

の木構造の根側から大域変数maxapplydepthで指定される深さ迄の全ての部分式に 〈規則1〉 を適
用させます.これによって得られた 〈式2〉 に対し,次の 〈規則2〉 を同様に適用します.以降,帰納的
に各部分式に作用させ, 〈規則n〉 を全ての部分式に作用させて終了します.

apply2函数は 〈規則1〉 が 〈式 〉 の部分式で失敗すると, 〈規則2〉 を適用する点で apply1函数
とは異なります. 大域変数maxapplydepthで指定された深度以下の全ての部分式で失敗した時に
限って, 全ての規則が次の部分式に繰返し適用されます.もし,規則の一つが成功すれば, その同じ
部分式が 〈規則1〉 で再実行されます.

applyb1函数は apply1函数と似ていいますが,apply1函数が 〈式 〉 の木構造に対し,上から下へ
と作用して行くのに対して applyb1函数は, 〈式 〉 の最下層の部分式から作用させ,規則の適合に失
敗すると, もう一つ上の階層の部分式に帰納的に作用させます.
尚,apply1函数,apply2函数と applyb1函数は無制限に階層構造の全ての部分式に規則を適用し

ません.大域変数 maxapplydepthで apply1函数と apply2函数が規則を適用する階層の深さを指
定し,大域変数maxapplyheightで applyb1函数が到達する階層の高さを指定します.但し,これら
のデフォルト値は 10000の為,通常の利用では殆ど無制限と言っても構わないでしょう.
では,apply1函数を使って,函数 tanを含む式にMaximaに組込の規則 trigrule0を適用してみま

しょう.この規則は式中の tan (x) に規則 trigrule0 を適用する事で sin(x)
cos(x) に置換します.この時,函

数 tan の変数は xである必要はありません.一般的な式でも構いません.

(%i19) tan(x);

(%o19) tan(x)

(%i20) apply1(tan(x),trigrule0);

sin(x)

(%o20) ------

cos(x)

(%i21) apply1(tan(a1*x+y+b1),trigrule0);

sin(y + a1 x + b1)

(%o21) ------------------

cos(y + a1 x + b1)

1.7.1 規則の定義と適用

Maximaの規則を利用者が定義する事も可能です.この場合,規則の定義は defrule函数や let函
数で行います.
ここでは最初に defrule函数による規則の定義方法を示しましょう.

defrule函数¶ ³
defrule(〈規則名 〉,〈並び 〉 , 〈置換 〉 )µ ´

defrule函数は与えられた 〈並び 〉 を指定した 〈置換 〉 で置換える規則の定義と規則の名付けを
行う函数です.
この場合 ,〈規則名 〉 で指定された規則が apply函数族の函数の一つで式に適用されると, 〈並び 〉

に適合する全ての部分式が 〈置換 〉 で指定した値で置換されます.尚,照合に失敗すると元の式を返
却します.
次に,defrule函数を使って前置式演算子 dfxに適当な規則を入れてみましょう.

(%i1) prefix("dfx");

(%o1) dfx

(%i2) defrule(chain1,dfx(a.b),dfx(a).b+a.dfx(b));

(%o2) chain1 : dfx (a . b) -> dfx a . b + a . dfx b

(%i3) apply1(dfx(a.b),chain1);

(%o3) dfx a . b + a . dfx b

(%i4) apply1(dfx(x.y),chain1);

(%o4) dfx (x . y)

最初にdfxが前置式演算子である事をprefix函数で宣言します.次に,dfxがdfx(a.b) → dfx(a).b+a.dfx(b))
となる規則 chain1を defrule函数で定めます.
それから apply1函数を用いて dfx(a.b)に規則 chain1を適用すると期待通りの結果を得ます.とこ

ろが,dfx(x.y)を計算させると残念な事に,dfx(x.y)のままです. これは何が悪かったのでしょうか？
この例では,式の並びを定義する際に用いた変数に対し,その変数が満すべき条件を定義してい

ないからです.Maximaでは変数が満すべき述語が与えられなければ, defrule函数や let函数で用い
られた変数の並びに対してのみ規則が適用される仕様となっています.従って,より一般的な規則
を定義したければ,defrule函数や let函数で規則を定義する前に,予め 〈並び 〉 で用いる変数が満す
べき条件を付加する必要があります.この処理を行うのがmatchdeclare函数です.

matchdeclare函数¶ ³
matchdeclare(〈変数 〉 ,〈述語函数 〉 , · · ·)
matchdeclare([〈変数1〉 ,· · ·,〈変数1〉 ], 〈述語函数 〉 ,· · ·)µ ´

matchdeclare函数は引数として 〈変数 〉と 〈述語函数 〉の変数と述語函数名の対を取ります.従っ
て,引数は常に偶数個になります.ここで指定する述語函数は,matchdeclareで宣言する変数を与え
ると,trueか falseの何れかが返却される函数です.述語函数として,特に指定する函数が存在しない
場合は true,その他には,Maxima組込の述語函数,lambda函数や block函数を含まない利用者定義
函数が使えます.
例えば, matchdeclare(q,freeof(x,%e)) とすれば,変数 qは xと%eを含まない全ての式に適合し

ます.又,述語には複数の引数を持つ函数で,最後の引数に matchdeclare函数で宣言する変数が入
るものを与える事も可能です.
尚,matchdeclare函数によって 〈変数 〉にはmatchdeclare属性が付加されます.この属性は print-

props函数で参照する事が可能です.
matchdeclare函数に与える変数と述語函数の対は,後で行う規則の定義で,変数が満すべき条件

を述語函数が与え,この述語が trueとなる変数に対してのみ defrule等で定めた規則が適用されま
す.その為,無条件に規則を適用したければ,述語函数を trueとします.又,複数の変数が同じ述語
を満す場合 に 〈変数 〉 の個所を 〈変数リスト 〉 で置換える事も可能です.
では,matchdeclare 函数で並び変数を予め定義し, それから defrule 函数で規則を定義してみま

しょう.

(%i1) prefix("dfx");

(%o1) dfx

(%i2) matchdeclare([_a,_b],true);

(%o2) done

(%i3) defrule(chain1,dfx(_a._b),dfx(_a)._b+_a.dfx(_b));

(%o3) chain1 : dfx (_a . _b) -> dfx _a . _b + _a . dfx _b

(%i4) apply1(dfx(a.b),chain1);

(%o4) dfx a . b + a . dfx b

(%i5) apply1(dfx(x.y),chain1);

(%o5) dfx x . y + x . dfx y

この例では,変数 aと bを常に trueとし,これらの変数に対して規則をあてはめています.その
為,一般の変数に対しても同様に規則の適用が実行されます.
但し,matchdeclareで宣言した変数の可換積 ∗ や和 + を取る場合には注意が必要です.

(%i7) matchdeclare([_c,_d],true);

(%o7) done

(%i8) defrule(chain1,dfx(_c*_d),dfx(_c)*_d+_c*dfx(_d));

_d _c partitions ‘product’

(%o8) chain1 : dfx (_c _d) -> _c dfx _d + dfx _c _d

この例では,defruleで規則を定義すると,_d _c partitions ‘product’ と表示されています.こ
の調子で apply1函数を実行するとまともな処理が出来ずに, Maximaが落ちてしまいます.

matchdeclare函数で宣言した変数の可換積*や羃ˆを含む式に対しては, let函数を用いる事が出
来ますが,上記の例の様に函数の引数に可換積*や羃ˆを含ませる場合には,let函数は有効ではありま
せん.寧ろ,非可換積を用いるか, 適当な演算子を宣言して defrule函数を用いる方が良いでしょう.

let函数¶ ³
let(〈項 〉 ,〈式 〉 ,〈述語 〉 , 〈変数1〉 ,· · ·,〈変数n〉 , 〈パッケージ名 〉)
let(〈項 〉 ,〈式 〉 ,〈述語 〉 , 〈変数 〉 ,· · ·,〈変数n〉 )
let(〈項 〉 ,〈式 〉 )µ ´

let函数は 〈述語 〉 が trueの場合, 〈項 〉 を 〈式 〉 で置換する規則を定義する函数です.
この 〈項 〉 には,アトム,sin(x)や f(x,y)の様な函数の可換積 ∗, 商/や羃ˆを含む項となります.但

し,負の羃を用いる場合には大域変数 letrat を trueにする必要があります.
〈式1〉 に含まれる 〈変数i〉 と対応する 〈述語 〉 を省く場合には,それらの変数がmatchdeclare函

数によって予め trueであると宣言されていなければなりません.
let函数の引数の末尾に 〈パッケージ名 〉 を追加すれば,定義した置換規則を指定したパッケージ

に追加します.未設定の場合,自動的に current let rule packageに割当てられたパッケージに追加
されます.
これらの置換函数は一度に幾つかの規則の組合せを用いて作用させられます. 各々の規則の組合

わせは任意数の let函数で操作された任意の数の規則を含む事が可能で,利用者が与えた名前で参
照されます.

let函数で定義した規則を式に適用する場合は,letsimp函数を用います.

letsimp函数¶ ³
letsimp(〈式 〉 ,〈規則パッケージ名1〉 ,· · ·, 〈規則パッケージ名n〉 )
letsimp(〈式 〉 ,〈規則パッケージ名 〉 )
letsimp(〈式 〉 )µ ´

letsimp函数は 〈式 〉 が指定した規則パッケージに含まれる規則の適用を続けて,式の変化が無
くなるまで規則の適用を続けます.
尚,規則パッケージの指定が無い場合,current let rule packageに割当てられた規則パッケージが

利用されます.
パッケージを複数指定した場合, 〈 式 〉 には左端のパッケージから順番に適用されます. 例え

ば, letsimp(expr,package1,package2) を実行すると, 最初に letsimp(expr,package1) を実行し, 次
に,letsimp(%,package2) を実行したものと同じ結果が得られます.
ここで規則パッケージを指定して current let rule packageが切替えられる事はありません.

では簡単に let函数と letsimp函数の動作を確認しておきましょう.

(%i1) matchdeclare([_a,_b],true);

(%o1) done

(%i2) let(tama(_a)^2-1,tama(2*_a));

2

(%o2) tama (_a) - 1 --> tama(2 _a)

(%i3) letsimp(tama(x)^2);

2

(%o3) tama (x)

(%i4) let(tama(_a)^2,tama(2*_a)+1);

2

(%o4) tama (_a) --> tama(2 _a) + 1

(%i5) letsimp(tama(x)^2);

(%o5) tama(2 x) + 1

この例では,sin 函数の倍角公式を模擬したものです.
最初に, let(tama( aˆ2-1,tama(2* a)) としている為,letsimp函数による規則の適用に失敗している
事に注意して下さい.let函数では置換される並びは項に限定されます.

tellsimp一家¶ ³
tellsimp(〈並び 〉 ,〈置換 〉 )
tellsimpafter(〈並び 〉 ,〈置換 〉 )µ ´

tellsimp函数と tellsimpafter函数は各々規則に沿った置換を実行します.これらの函数は共に似
ていますが,ここでの置換は,規則を適用する前に置換を行うか,規則を適用した後で置換を行うか
の違いがあります.
最初の tellsimp函数は,新しい情報を古い情報の前に置いて,組込の簡易化規則よりも前に適用

します.この tellsimp函数は簡易化が実行される前に式を改変する事が必要な場合に用います.こ
こでの 〈並び 〉 には和,積, 単変数や数値は使えません.置換には規則名のリストを指定し,defrule
函数, defmatch函数,tellsimp函数や tellsimpafter函数で追加されたものです.

tellsimpafter函数は tellsimp函数と同様に 〈並び 〉 に対する 〈置換 〉 を定義します.ここで指定
する 〈並び 〉 はMaxima組込みの簡易化規則の適用後に用いるものになります. この 〈並び 〉 に
は単変数や数を除く任意の式が設定出来ます.

defmatch函数¶ ³
defmatch(〈函数名 〉 ,〈並び 〉 , 〈助変数1〉 ,· · ·,〈助変数n〉 )µ ´

defmatch函数は n+1個の引数を持ち,特定の並びに適合するかどうか式を検査する函数を 〈函
数名 〉 で指定した名前で生成する函数です.

defmatch函数の 〈並び 〉 は変数と 〈助変数1〉,· · ·,〈助変数n〉 を含む式になります.ここで変数

が既に matchdeclare函数で宣言されていたとしても, defmatch函数の 〈助変数i〉 の引数として
defmatch函数に与えます.
函数の最初の引数は並びに対して照合する式で,他の n個の引数は式中の実際の値, 並びの中で

変数として置かれるものです.
defmatchで構成された函数は照合に成功すると 〈助変数i〉 = 適合する変数のリスト を返しま

す. ここで,照合に失敗すれば falseを返します.
次の例では,与えられた函数の線形性を調べる函数 linearの定義を行い,函数 linear を実行して

います.

(%i2) defmatch(linear,a*x+b,x)

(%i3) linear(3*z+(y+1)*z+y^2,z);

(%o3) false

(%i4) linear(a*z+b,z);

(%o4) [x = z]

(%i5) nonzeroandfreeof(x,e):=if e#0 and freeof(x,e)

then true else false

(%i6) matchdeclare(a,nonzeroandfreeof(x),b,freeof(x))

(%i7) linear(3*z+(y+1)*z+y^2,z);

(%o7) false

(%i8) defmatch(linear,a*x+b,x)

(%i9) linear(3*z+(y+1)*z+y^2,z);

2

(%o9) [b = y , a = y + 4, x = z]

最初に defmatchで linearを定義し,次に式 3*z+(y+1)*z+y^ 2が変数 z の一次式であるかどう
か検証しています.この例では falseが返却されています. これは最初の変数 aと bに関して何等
の情報も無い為,単純にアトム aとアトム bを持たない一次式と判断されて falseとなっています.
次の a*z+bの場合,並び変数 x に対応するものが zであると判断した為に [x = z]が返却されてい
ます.
そこで,並び変数 aと bの情報を追加します.この例では,最初に is(e#0 and freeof(x,e))と同値

な函数 nonzeroandfreeofを定義します.それから並び変数 aと bに対して,共に 0ではなく,変数 x
も含まない式であるとmatchdeclare函数を用いて宣言しています.この宣言の後に defmatch函数
で函数 linearを再定義します. この再定義を行わないと,並び変数 aと bに関する情報は更新され
ません.これによって並び変数 aと bが変数 xから独立したもので,変数 xに関して線形であれば,
対応する成分リストを返す函数 linearが定義されます.
そこで, linear(3*z+(y+1)*z+yˆ2,z) を実行すると,linearで与式と並びの式 a*x+bを比較し,今

度は [b=yˆ2, a=y+4, x=z]を返します.

1.7.2 規則の削除

規則の削除を行う函数¶ ³
remlet(〈項 〉 )
remlet(〈項 〉 ,〈パッケージ名 〉 )
remlet(all)
remlet()
remrule(〈函数 〉 ,〈規則 〉 )
remrule (all)µ ´
これらの函数は全て let函数で定義された置換規則 〈項 〉 → 〈式 〉 を削除します. 〈パッケージ

名 〉 が与えられると指定された規則パッケージから規則を削除します.
それに対し,引数を指定しない remlet() や引数が allの remlet(all) は規則パッケージから代入

規則の全てを削除します.
ここで,規則パッケージ名が,例えば, relmet(all,〈パッケージ名 〉) で与えられていれば,指定され

た規則パッケージも削除されます.もし,構築した規則が古い規則を上書きしたものであれば,remlet
で新しい規則を削除すれば,古い規則が復活します.

remrule函数は 〈規則 〉 で指定した規則を 〈函数 〉 から削除します.この 〈規則 〉 は,defrule函
数,defmatch函数, tellsimp函数や tellsimpafter函数で設定したものです.remrule函数も引数が all
の場合は全ての規則を削除します.

1.7.3 規則に関連する大域変数

規則に関連する大域変数¶ ³
変数名 初期値 概要

maxapplydepth 10000 apply1と apply2が停止する階層
maxapplyheight 10000 applyb1が停止する階層
current let rule package default let rule package 利用規則パッケージ名

letrat false letsimpの動作に影響
let rule packages default let rule package 規則パッケージのリストµ ´
大域変数maxapplyheightは apply1函数,apply2函数や applyb1函数が停止する与式の階層構造

の最高位となります.ここで,maximaの式には LISPの S式の様な階層構造があります.apply1等
の函数はmaxapplyheightよりも低い個所,即ち,木構造の根本側に作用し,それよりも高ければ作
用しません (木構造の根元側を最低辺としてみます). 但し,デフォルト値の 10000は式の殆ど全て
と言える程の高さになるでしょう.
大域変数 current let rule packageに let rule packageで利用した規則パッケージ名が設定されま

す.let命令で定義したどの様な規則パッケージ名も, この変数に再設定して構いません.letパッケー
ジに関連するどの様な函数でも, current letl rule packageが何時でも使えます.

大域変数 letratが falseであれば,letsimp函数が式の分子と分母を各々別に簡易化して結果を返
します.但し, n!/n を (n − 1)! にする様な置換は出来ません. この様な置換を行う為には,大域変
数 letratを trueに設定しておかなければなりません. すると,分子,分母の商は要求の通りに簡易
化されます.
大域変数 let rule packageの値は全ての利用者定義の規則パッケージに特殊なパッケージを加え

たもののリストとなります. ここで,default let rule packageは利用者が特に規則パッケージを指
定しない場合に用いられる規則パッケージの名前です.

1.8 評価

1.8.1 ev函数

この節ではMaximaで最も重要な函数の一つである ev函数を中心に解説します.
ev函数¶ ³

ev(〈式 〉, 〈引数1〉, · · · , 〈引数n〉 )
〈式 〉, 〈引数1〉,· · ·, 〈引数n〉µ ´

ev函数は与式 〈式 〉 の評価を実行しますが,その際に,評価を行う為の局所的な環境を設定し,そ
の環境で式の評価を実行します. この環境は 〈引数1〉, · · · , 〈引数n〉 で設定されるもので,Maxima
の様々な大域変数を trueにしたり,式の変数に値を割当てる事,更には,式を評価する函数 (evfun)
を指定する事で設定されるものです.
尚,Maximaの最上層のみで ev()を外した表記も可能です.

(%i1) ev((x+1)^4,expand);

4 3 2

(%o1) x + 4 x + 6 x + 4 x + 1

(%i2) (x+1)^4,expand;

4 3 2

(%o2) x + 4 x + 6 x + 4 x + 1

(%i3) x.y.z;

(%o3) x . y . z

(%i4) (x.y).z,dotassoc:false;

(%o4) (x . y) . z

(%i5) (x.y).z;

(%o5) x . y . z

(%i6) x^2+2*x+1,factor;

2

(%o6) (x + 1)

(%i7) x^2/(y+1)+2*x/(y^2-1)+1,ratsimp;

2 2 2

y + x y - x + 2 x - 1

(%o7) ------------------------

2

y - 1

この例では,最初に (x + 1)4 の展開を行います.最初の書き方が ev函数の正規の記述方法です
が,Maximaの最上層に限定して (x+1)ˆ4,expandの様に ev()を省略出来ます.この表記が使えない

のは block文内部や lambda函数内部で,Maximaの最上層,即ち,通常の入力プロンプトが出てい
る場合やMaximaのバッチファイル内部にはこの表記が利用出来ます.
次に,大域変数 dotassocの値を変更して式の評価を実行しています.ここで,非可換積の結合律を

制御する大域変数 dotassocはデフォルトで trueとなっている為, (x . y) . zは x . y . zに自動的
に評価されます.ここで,ev函数で評価を行う際に,一時的に大域変数 dotassocを falseにしていま
す.ev函数の評価と大域変数の値が無関係な事に注目して下さい.
最後に evfun属性を持つ函数である factor函数と ratsimp函数を使って評価しています.ここで

evfun属性を持たない函数を用いた場合,式が評価されずにそのままの式が返されます.
更に,ev函数では式に含まれる変数に値を一時的に割当てて式を評価する事も可能です.この評

価は方程式の解を解いた後に,その結果を使って他の式の評価する場合や解の検証で便利です.

(%i29) solve([x^2-y^2+x*y-1,x+y-3],[x,y]);

sqrt(41) - 9 sqrt(41) - 3

(%o29) [[x = - ------------, y = ------------],

2 2

sqrt(41) + 9 sqrt(41) + 3

[x = ------------, y = - ------------]]

2 2

(%i30) x*y,%[1];

(sqrt(41) - 9) (sqrt(41) - 3)

(%o30) - -----------------------------

4

この例では,最初に連立方程式 x2 − y2 + xy − 1 = 0, x + y − 3 = 0を解き,その結果を用いて式
xy を評価しています.これは ev函数の性質で,引数に論理演算子=を含む式で,左辺が通常の変数
であれば,左辺の変数に右辺の式が割当てられます.
ここで,具体的に ev函数で利用可能な引数を示しましょう.

ev函数で利用可能な引数¶ ³
evflag属性を持つアトム evflag属性を持つアトムに trueを一時的に設定.
evfun属性を持つ函数 evfun属性を持つ函数を,引数の左側から順番に式に適用.
expand 大域変数 expopに大域変数maxposexの値,大域変数 exponに

大域変数maxnegex の値を各々割当てて式を展開.
expand(〈整数1〉,〈整数2〉) 大域変数 maxposexに 〈整数1〉 ,maxnegexに 〈整数2〉 を各々

設定し式を展開.
eval 式を評価.
noeval 式を無評価.
nouns 式を名詞型で評価.
numer 大域変数 numerと大域変数 floatを trueにして式を評価
risch risch積分を実行
diff 式中の名詞型の微分を評価.
derivlist(〈x1〉, · · · , 〈xn〉) 名詞型の微分で, 〈x1〉, · · · , 〈xn〉 による微分のみを評価.
変数への式の割当 大域変数や式中の変数に割当と式の評価を実行.
local(〈x1〉, · · · , 〈xn〉) ev内部で用いる局所変数 〈x1〉, · · · , 〈xn〉 を宣言.
detout 逆行列の計算で行列式を行列の外に出す.µ ´

ev函数では,evflag属性を持つ引数を trueに設定したり,大域変数に値を設定する事で式の評価
の為に局所的な環境設定を行い,それから,与式に代入を実行し, evfun属性を持つ函数や,大域変数
の値によるMaximaの自動変換によって式を評価します.
先ず,evflag属性を持つ大域変数はデフォルトで以下のものがあります.

evflag属性をデフォルトで持つ大域変数¶ ³
float,pred,simp,numer, detout, exponentialize, demoivre,

keepfloat, listarith, trigexpand, simpsum, algebraic,

ratalgdenom, factorflag, %emode, logarc, lognumer,

radexpand, ratsimpexpons, ratmx, ratfac, infeval, %enumer,

programmode, lognegint, logabs, letrat, halfangles,

exptisolate, isolate_wrt_times, sumexpand, cauchysum,

numer_pbranch, m1pbranch, dotscrules,logexpandµ ´
大域変数に evflag属性を持たせる為には, declare函数を用います.更に,ある大域変数が evflag

属性を持つかどうかは,properties函数を使って調べられます.
evflag属性を持つ大域変数を定義し,ev函数によって評価する例を以下に示します.

(%i1) declare(tama,evflag);

(%o1) done

(%i2) tama:false;

(%o2) false

(%i3) ev(’(if tama=true then print("nekoneko") else print("1234")));

1234

(%o3) 1234

(%i4) ev(’(if tama=true then print("nekoneko") else print("1234")),tama);

nekoneko

(%o4) nekoneko

(%i5) properties(tama);

(%o5) [value, evflag]

(%i6) :lisp (get ’$tama ’evflag)

T

この例では,変数 tamaに evflag属性を declare函数を用いて持たせ,それから, if文で構成され
た式の評価を行っています.最初の例では tamaには falseを設定している為,評価式では tamaが
falseの場合の処理が実行されています.次に,ev函数の引数として tamaを与えると,tamaに evflag
属性を持たせている為,自動的に値に trueが設定され,その値を用いて式が評価され,結果として
tamaが trueの場合の処理が実行されている事が判ります.次に,properties函数を用いて属性を調
べています.LISPで調べる場合には,内部表現に対して get函数で evflag属性を取出します.evflag
属性があれば Tが返却され,そうでない場合には NILが返却される事で判別出来ます.

evflag属性に似た属性に evfun属性があります. これは函数に対する属性で,ev函数の引数とし
て利用可能な函数である事を意味します. デフォルトで evfun属性を持つ函数を以下に示しておき
ます.

evfun属性を持つ函数¶ ³
factor,trigexpand,trigreduce,bfloat, ratsimp,ratexpand,

radcan,logcontract,rectform,polarform

µ ´
evflag属性と同様に,declare函数を用いて函数に evfun属性を持たせる事が出来ます. ここでは

簡単な函数を定義して,実際にその作用を観察してみましょう.

(%i1) mike(z):=diff(z,x,2);

(%o1) mike(z) := diff(z, x, 2)

(%i2) properties(mike);

(%o2) [function]

(%i3) x^2,mike;

2

(%o3) x

(%i4) declare(mike,evfun);

(%o4) done

(%i5) properties(mike);

(%o5) [evfun, function, noun]

(%i6) x^2,mike;

(%o6) 2

(%i7) :lisp (get ’$mike ’evfun)

T

この例では,変数 xで二階微分を行う函数mikeを定義しています.この函数は evfun属性を最初
は持っていない為に ev函数による評価が出来ません.ところが,declare函数で evfun属性を付加す
れば,ev函数による評価では mikeが実行され,結果として二階微分が得られます.函数が evfun属
性を持つかどうかの判定は,properties函数や, LISPの get函数を用いて行えます.この点も evflag
属性と同様です.

evfun属性を持つ函数が複数 ev函数に与えられた場合,この evfun函数の作用の順番は ev函数
の左端の evfun函数から順番に式に作用させます.

2

(%o29) tst(z) := expand(z )

(%i30) declare(tst,evfun);

(%o30) done

(%i31) (x+1)^2,tst,factor;

4

(%o31) (x + 1)

(%i32) (x+1)^2,factor,tst;

4 3 2

(%o32) x + 4 x + 6 x + 4 x + 1

(%i33) tst(factor(x+1)^2);

4 3 2

(%o33) x + 4 x + 6 x + 4 x + 1

(%i34) factor(tst((x+1)^2));

4

(%o34) (x + 1)

この例で示す様に,ev函数の引数として,evfun属性を持つ factorと利用者定義函数の tstを与え
ています.先ずはじめに ev函数の引数として左から tst,factorの順に与えた為,factor(tst(与式))を
処理しています.次に,factor,tstの順で ev函数の引数として引き渡した場合,式は展開されていま
す.即ち,tst(factor(与式))で処理したからです.

expand を指定すると, 大域変数 maxposex と maxnegex に設定された値を大域変数 expop と
expon に各々割当てて式の展開を行います. 尚,大域変数 expopと exponは (x + 1)3の様な冪乗で
自動展開する次数を指定する大域変数です.
大域変数maxposexとmaxnegexは expand函数で展開する冪乗の最大と最小次数を設定する大

域変数です.デフォルトで大域変数maxposexとmaxnegexの値が 1000の為,羃の最大次数が 1000

以下であれば式の展開を自動実行します. expand(〈整数1〉,〈整数2〉)の場合は大域変数 maxposex
に 〈整数1〉 ,maxnegexに 〈整数2〉 を各々設定して与式を展開します.

(%i1) (x+2)^1001,expand;

1001

(%o1) (x + 2)

(%i2) (x+2)^2/(x+1)^3,expand(2,3);

2

x 4 x 4

(%o2) ------------------- + ------------------- + -------------------

3 2 3 2 3 2

x + 3 x + 3 x + 1 x + 3 x + 3 x + 1 x + 3 x + 3 x + 1

(%i3) (x+2)^2/(x+1)^3,expand(2,2);

2

x 4 x 4

(%o3) -------- + -------- + --------

3 3 3

(x + 1) (x + 1) (x + 1)

最初の例では,羃の次数が 1001と maxposexの値 1000を越えている為に展開を行いません.次
の例では,expopに 2,exponに 3を指定していますが,展開する式の羃が正の羃の次数の絶対値が 2,
負の羃の絶対値が 3の為,展開を実行しています. しかし,expopに 2,exponに 2を指定すると,指定
した値が負の羃の次数の絶対値よりも小さい為に,負の冪乗の部分式の展開のみが実行されません.

nounの場合は与式を評価しません.
numerの場合,大域変数 numerと大域変数 floatを trueにして式の評価を実行します.

(%i45) sin(%pi/10);

%pi

(%o45) sin(---)

10

(%i46) sin(%pi/10),numer;

(%o46) .3090169943749474

(%i47) 2*%e*x+%pi/4,numer;

(%o47) 5.43656365691809 x + .7853981633974483

(%i48) 2*%e^x+%pi/4,numer;

x

(%o48) 2 %e + .7853981633974483

尚,式の中に冪乗でない定数%eが存在する場合は,大域変数%enumer を trueにして式の評価を
行っています.但し,%eの羃の場合は%eを浮動小数点に変換しません.

rischは integrate函数に Risch積分を実行させます.これは,integrate函数内部で,rischが指定さ
れる場合, rischint函数を用い,それ以外は sinit函数を用いる様になっている為です.

derivlistは函数に似た構文を持っています.
derivlist¶ ³

derivlist(〈変数1〉 ,· · ·,〈変数k〉 )µ ´
ev函数は derivlistで指定した変数 〈変数1〉 ,· · ·, 〈変数k〉 を含む名詞型の微分を評価します.

(%i9) a1:’diff(’diff(x^2+2*x*y^2+y^4,x),y);

2

d 4 2 2

(%o9) ----- (y + 2 x y + x )

dx dy

(%i10) a1,diff;

(%o10) 4 y

(%i11) a1,derivlist(x);

d 2

(%o11) -- (2 y + 2 x)

dy

(%i12) a1,derivlist(y);

d 3

(%o12) -- (4 y + 4 x y)

dx

この derivlist引数に似たものに local引数があります.この localも derivlistに似た構文を持ち
ます,

local¶ ³
local(〈変数1〉 ,· · ·,〈変数k〉 )µ ´
こちらは ev函数内部で利用する局所変数の宣言に用いられます.但し,ev函数では後述の式への

割当がある為,localでなければ局所変数が扱えない訳ではありません.

(%i16) solve(x^3+2*x-b,x),local(b),b=3;

sqrt(11) %i + 1 sqrt(11) %i - 1

(%o16) [x = - ---------------, x = ---------------, x = 1]

2 2

(%i17) solve(x^3+2*x-b,x),b=3;

sqrt(11) %i + 1 sqrt(11) %i - 1

(%o17) [x = - ---------------, x = ---------------, x = 1]

2 2



1.8. 評価 67

式の割当は,x=1や x:1等によって,評価する式の変数に値を割当てる事で与式の評価を実行する
ものです.

(%i31) ev(sin(x),x=1);

(%o31) sin(1)

(%i32) ev(sin(x),x=1,float);

(%o32) sin(1)

(%i33) ev(sin(x),x=1,bfloat);

(%o33) 8.414709848078965B-1

(%i34) ev(sin(x),x=1);

(%o34) sin(1)

(%i35) ev(sin(x),x=1,bfloat);

(%o35) 8.414709848078965B-1

(%i36) ev(sin(x),x:%pi/4,bfloat);

(%o36) 7.071067811865475B-1

(%i37) ev(sin(sqrt(x^2+y^2)),[x:%pi/4,y=1]);

2

%pi

(%o37) sin(sqrt(---- + 1))

16

この例では,sin(x)の変数 xに 1や π/4等を割当てています.割当は演算子=でも演算子:でも構
いません.複数の変数に一度に変数を割当てる場合には,リストで与えます.この際に,演算子=と:
が混在していても問題はありません.
この評価方法は,algsys函数や solve函数等の方程式を解く函数と評価したい式を組合せると強力

です.

(%i42) algsys([x^5-x^3+5],[x]);

(%o42) [[x = - 1.53955007256894], [x =

- 1.183445980013718 %i - .4590933961159689],

[x = 1.183445980013718 %i - .4590933961159689],

[x = 1.228868436016586 - .7109481105485196 %i],

[x = .7109481105485196 %i + 1.228868436016586]]

(%i43) map(lambda([z],ev(realpart(x^2),z)),%);

(%o43) [2.37021442594703, - 1.189777641253336,

- 1.189777641253336, 1.004670417145341, 1.004670417145341]

この例では,方程式 x5 − x3 + 5 = 0 の数値近似解を求め,その近似解を二乗したものの実部を求
めています.

detoutは大域変数detoutと似た結果が得られます.detoutを指定した場合,大域変数のdoallmxops
と doscmxopsを falseにし,大域変数の detoutを trueにした場合と同じ結果が得られます.

(%i7) A:matrix([1,2,3],[4,3,1],[2,4,1]);

[ 1 2 3 ]

[ ]

(%o7) [ 4 3 1 ]

[ ]

[ 2 4 1 ]

(%i8) A^^(-1),detout;

[ - 1 10 - 7 ]

[ ]

[ - 2 - 5 11 ]

[ ]

[ 10 0 - 5 ]

(%o8) -----------------

25

1.8.2 評価に関連する函数

評価に関連する述語函数¶ ³
equal(〈式1〉,〈式2〉 )
is(〈述語 〉 )µ ´

equal 函数は is 函数と一緒に使われ, 〈 式1〉 と 〈 式2〉 が ratsimp 函数で指定された全ての可
能な変数値に対して等しければ true, 等しくない場合には false を返します.x が不定元であって
も is(equal((x+1)ˆ2,xˆ2+2*x+1))は trueを返しますが,is((x+1)ˆ2=xˆ2+2*x+1)は falseを返し
ます.

is(rat(0)=0)は falseですが,is(equal(rat(0),0))は trueとなる事に注意して下さい.もし,equalで
判別出来ない場合,同値だが簡易化された形式で返されますが,=を使っていれば常に trueか false
が返されます. 式中の全ての変数は実数値であると予め仮定しています.
尚, ev(〈式 〉,pred) は is(〈式 〉) と同じ事を実行しています.

(c1) is(x^2 >= 2*x-1);

(d1) true

(c2) assume(a>1);

(d2) done

(c3) is(log(log(a+1)+1)>0 and a^2+1>2*a);

(d3) true

is函数は 〈述語 〉 が,Maximaの文脈や宣言等に含まれている事象に適合するかどうかを判定し
ます.is函数が trueと返すのは, 〈述語 〉 に含まれる変数に関して,全ての値で述語が trueとなる場
合で,そうでない場合は false を返します.それ以外は大域変数 prederror の設定に依存します. つ
まり,大域変数 prederrorが trueの場合,is函数はエラーを出力を行い,falase であれば unknownを
返します.

函数評価に関連する函数¶ ³
’〈式 〉
”〈式 〉
eval(〈式 〉 )µ ´
単引用符’はMaximaによる評価を防止します.例えば,’(f(x))とする事でMaximaに式 f(x)を評

価しない事を報せます.一方で’f(x)は xに函数 fを作用させ,その名詞型で返す形になります.この
際,fの評価は行なわれませんが xは評価されてしまいます.
二つの単引用符”は特殊な評価を行います.例えば,”%o4で%i4を再評価します. 又,”f(x)は函数

fを xに作用させたものを動詞型で返します.

(%i65) test:2*%pi;

(%o65) 2 %pi

(%i66) sin(test);

(%o66) 0

(%i67) test:%pi/4;

%pi

(%o67) ---

4

(%i68) ’’%i66;

1/2

2

(%o68) ----

2

(%i69) ’’sin(test);

(%o69) .7071067811865475

eval函数は 〈式 〉 の評価を行います.LISPの eval函数と同じ働きをします.

1.9 LISPに関係する関数

1.9.1 MaximaとLISP

Maximaは Common Lispと呼ばれる LISPの一方言で記述されています. LISPは函数型と呼
ばれるプログラム言語で,プログラムは様々な函数を定義し,それらを組合せて行く作業とも言え
ます.因みに,Cや FORTRANは手続型と呼ばれます.

Maximaはこの LISPの上で動作する環境ですが,Maxima自体は PASCAL風の処理言語を持っ
ており,構文的にも LISPを意識する事は単純な利用では殆どありません.
但し,Maximaで酷いエラーを出すと LISPのデバッガに落ちる事があります. LISPのデバッガ

からの抜け方は,Maximaを実装した LISPによって微妙に異なりますが,CLISPの場合は :q と入
力してみて下さい.するとMaximaに戻ります.

LISPの特徴は言語仕様が非常に柔軟な点です.LISPにはアトムと呼ばれる変数があり,それらを
空行で区切って小括弧 ()で括ったリストと呼ばれるデータが, 最も基本的なデータとなります.こ
のリストはリストのリストといったものも許容します.このアトムとリスト等で構成されたデータ
を S式と呼びます.因に,LISPのプログラム自体も S式です.その為,LISPの函数でプログラムを操
作する事も容易に行えます.
その為,Maximaのプログラムでは足りない所を LISPで代用する事も多くあります. 又,Maxima

から LISPを直接利用する事も可能です.この場合はMaximaの to lisp 函数を利用します.Maxima
で to lisp(); と入力すると, 裏方の LISPが表に出ます.これで LISPを使って遊べます.この状態

からMaximaに戻りたければ, (to-maxima) と入力します.すると通常のMaximaに戻ります.

(%i1) to_lisp();

type (to-maxima) to restart, ($quit) to quit Maxima.

Maxima> (setq $a ’1)

1

Maxima> (to-Maxima)

returning to Maxima

(%o1) true

(%i2) a;

(%o2) 1

この例では to lisp(); でLISPに入って,アトム$aに 1を割当てています. それから (to-maxima)
でMaximaに戻っています.to lisp函数の返却値は trueです.最後に a; を入力すると,LISPで$aに
割当てた値 1が返却されます.これはMaximaでのアトムの内部表現では$aの様に先頭に$が付く
からです.
この例で注目して頂きたい事は,Maximaで表示されているものと LISP側で見たものと様子が

違う事です.即ち,Maximaで扱うデータ,更には函数それ自体も LISP側では別の表記方法があり
ます.この LISP側での表現を単純に内部表現と呼んでいます.この内部表現を通常の処理で気にす
る事は殆どありませんが, 細かな処理を行う必要が出た時点で初めて意識する事になります.

尚,Maximaの函数名で先頭に?が付いているものが幾つか存在します.この様な函数は?を外した
部分は LISPの函数で,Maximaから裏の LISPで処理させて,その結果をMaxima側に持って来る
函数です.?は通常の LISP の函数にも適応可能で?を頭に付けた函数は,Maxima内部では?を外し
て,LISPの函数として処理されます.この様にMaximaが介在する為,?を用いて LISPの函数を利
用する場合には,引数はMaximaで見えているものを設定します.
尚,?と函数の間に空行を入れるとMaximaのオンラインマニュアルを呼出そうとするので,注意

が必要です.
?と似た函数に:lisp函数があります. こちらは,直接 LISPの S式をMaxima側から入力し,LISP

に評価させた値を得る為の函数です.?演算子との違いは,?演算子では引数が LISPの函数で,その引
数はMaxima に準じたものとなりますが,:lispの場合はより一般的な LISPの S式となります. そ
の為,引数もMaximaの内部表現そのものとなります.

(%i26) a:x+y+z;

(%o26) z + y + x

(%i27) :lisp $a;

((MPLUS SIMP) $X $Y $Z)

(%i27) :lisp (car $a)

(MPLUS SIMP)

(%i27) ?car(a);

(%o27) ("+", simp)

上記の例では変数 aに x+y+zを割当てていますが,$aが変数 aの内部変数名となります. :lisp $a;
でこの変数に割当てた値を参照していますが, 返却値は内部表現そのものです.この様な変数の参
照は?では出来ません.
更に, :lisp (car $a); の値は内部表現そのものですが, ?car(a); の値はそれをMaximaで解釈し

た (”+”, simp)となっており, 引数に$が付いていない事と:lispの結果に%oラベルが無い事に注目
して下さい.
この様に,?は入力と出力にMaximaが介在する為に,入出力をする度にMaximaの評価を受ける

事になりますが,:lispの場合は直接操作となります. :lispはMaximaで内部表現を確認する必要が
ある場合に特に便利な函数です.
尚,Maximaは LISPで記述されている為,そのデータだけではなく,全てが LISPの S式で,LISP

の函数で処理が可能となります.この事を利用してMaximaの機能を拡張する事が容易に行えます.

Maximaの函数を LISPから利用する場合,mfuncall函数を用います. この場合,引数はMaxima
の函数名の頭に’$を付け,その後に引数を並べます.

MAXIMA> (mfuncall ’$diff ’$x ’$x 1)

1

この例では,変数 xを diff函数を使って xで微分するものです.この様に, 引数は全て内部表現に
対応したものでなければなりません. 従って,Maxima上で引数を割当ててある状態でなければ,使
い難いものです. 現実的には,Maxima函数の虫取りや動作の確認で用いる程度でしょう.

第2章 Maximaのデータとその操作

この章で解説する事:

• 数値

• 多項式

• 式について

• 代入操作

• 式の展開と簡易化

• リスト

• 配列

• 行列

2.1 数値

2.1.1 Maximaで扱える数値について

Maximaで扱える数値には,整数,有理数,浮動小数点と複素数があります. その他に代数的整数
もありますが,代数的整数の場合はその最小多項式を用いる為, 多項式の節で説明します.
整数値の入力は C等と同じ入力になります.Cとの違いはメモリ制約を除けば桁数に限界はあり

ません.有理数は 128/8989 の様に,演算子/の両辺に整数を置いたもので表現されます.この有理
数に関しても整数と同様にメモリの制約を除くと分母,分子の桁数に上限はありません.
尚,整数と有理数の演算を行うと,基本的に有理数になります.但し,分母を消去出来る様な計算

を整数や有理数で行うと,結果は整数になります. 尚,整数には実際は通常の整数の fixnum型と可
変桁の整数の bignum型の二種類があります. 整数の場合,この二種類の違いを認識する必要は特
にありません.

Maximaの浮動小数点には float型と bigfloat型の二種類があります.float型は 16桁の倍精度で
1.2や 1.3e-4の様に入力が出来ます.bigfloat型は任意精度の実数で大域変数 fpprecで指定した桁
数の実数です.この fpprecの値を大きくすると精度も当然高くなります. ここで計算精度がそれな
りに必要でも,結果の表示には少ししか必要がない場合, 大域変数 fpprintprecに大域変数 fpprecと
は別の表示の桁数を設定すれば,大域変数 fpprintprecに設定した桁数で bigfloat型の数値が表示さ
れます.float型から bigfloat型への変換は bigfloat函数で行います.
浮動小数点と整数や有理数の演算は,型の変更を実行しない限り浮動小数点となります.又,bigfloat

型と float型を混在して計算する場合はエラーになります. その為に変換函数による処理が必要に
なります.
複素数は実部と%iをかけた虚部との和で指定されます.例えば,方程式 x2 − 4x + 13 = 0 の根

はMaximaでは 2+3*%iと 2-3*%iで表現されます. ここで複素数の実部は realpart函数,虚部は
imagpart函数で取り出せます.複素数は整数,有理数,浮動小数点の自然な拡張となっている為,実
部と虚部は,これらの数で表現されています.

2.1.2 数値に関連する大域変数

数値に関連する大域変数¶ ³
変数名 初期値 取り得る値 概要

domain real [real,complex] 多項式の係数環を指定

float2bf false [true,false] float → bigfloatの際の警告の有無を指定
fpprec 16 正整数 bigfloat型の桁数を指定
fpprintprec 0 正整数 bigfloat型の表示桁数を指定
m1pbranch false [ture,false] -1の原始 n乗根自動変換の有無を指定
radexpand true [true,false] 根号の外に出すかどうかを指定µ ´
大域変数 domainは函数等の動作に影響を与えます.domainはデフォルトで realが設定されてい

ます.これはMaximaが主に実数上で処理を行う事を意味しています. これに対して値を complex

にすると,Maximaで処理する世界は複素数の世界となる事を意味します.
更に,domainを complex,大域変数のm1pbranchを trueにした場合,-1の n乗根は原始 n乗根に

自動的に変換されます.
大域変数 float2bfは,falseの場合に bfloat函数で浮動小数点数を bigfloat型の数値に変換する時

点で計算精度が落るとの警告メッセージを表示させます.
大域変数 fpprecは bigfloat型数値の桁数を定めます.即ち,fpprecを正整数 nに設定すると bigfloat

型は n桁になります.
大域変数 radeexpandは式

√
a2b の様に根号の外に出せる因子が式に含まれる場合,trueであれ

ば,その様な因子を根号の外に自動的に出させる大域変数です.

2.1.3 Maximaの定数

Maximaの定数¶ ³
%e 自然底 e
%gamma Eulerの定数
%phi 1+

√
5

2

%pi 円周率 π

false Bool代数の定数.偽 (LISPの nil)
true Bool代数の定数.真 (LISPの t)
inf 正の実無限大.
minf 負の実無限大.
infinity 複素無限大,任意の偏角で無限大
zeroa 0+.limit函数で利用
zerob 0−.limit函数で利用µ ´

Maximaには幾つかの定数があり,大雑把に 3種類に分類出来ます.
最初の一つは%piの様な通常の定数,もうひとつは,tや nilといった真偽値, 最後にMaximaが用

いる infの様な定数です.更に,zeroaや zerobは limit函数だけで用いる定数です.

(%i55) limit(1/x,x,zeroa);

(%o55) inf

(%i56) limit(1/x,x,zerob);

(%o56) minf

但し,limit(1/(x-1),x,1,’plus)の意味で limit(1/(x-1),x,1+’zeroa)の様な使い方は出来ません.
尚,inf や minf といった極限に関連する数を含む式の評価は limit 函数で行えます. この場合

は,limit( 〈式 〉 )で式の評価が行えます.



76 第 2章 Maximaのデータとその操作

2.1.4 数値に関連する函数

数値に関連する述語函数¶ ³
函数名 trueとなる条件
numberp 整数,有理数,浮動小数や可変長実数
bfloatp bigfloat型の数値
floatnump 浮動小数点数

integerp 整数

evenp 偶数

oddp 奇数

constantp 定数の場合µ ´
これらの述語函数は全て引数が一つです.

maxとmin¶ ³
max(〈実数値1〉, 〈実数値2〉, · · ·) 〈実数値1〉, 〈実数値2〉, · · ·)の最大値
min(〈実数値1〉, 〈実数値2〉, · · ·) 〈実数値1〉, 〈実数値2〉, · · ·)の最小値µ ´

Maximaは与えられた実数列の最大値をmax函数,最小値をmin函数で求めます.

数値に関連する函数¶ ³
函数名 変換前 変換後

bfloat 全ての数 → bigfloat型
isqrt 整数 → 与えられた整数の平方根を越えない整数

fix 実数 → 与えられた実数を越えない最大の整数

entier 実数 → 与えられた実数を越えない最大の整数

random 数値 → 0から与えられた数値-1迄の間の乱数
cabs 式 → 与式の複素数としての絶対値

realpart 式 → 与式の実部

imagpart 式 → 与式の虚部

cargs 式 → 与式の偏角

sqrt 式 → 与式の平方根µ ´
最初の bfloatは全ての数と数値函数を bigfloat型に変換します.

isqrt函数は引数 〈整数 〉 の絶対値の平方根を越えない整数値を返します.

(%i50) isqrt(-3);

(%o50) 1

(%i51) isqrt(-4);

(%o51) 2

(%i52) isqrt(10);

(%o52) 3

(%i53) isqrt(-10);

(%o53) 3

尚,isqrt函数は引数が整数で無ければ,入力式をそのまま返却します.
fix函数と entirer函数は 〈数値 〉 が実数の場合, 〈数値 〉 を越えない最大の整数 nを返す函数で,

両者の違いは全くありません.

(%i42) fix(10);

(%o42) 10

(%i43) fix(-10);

(%o43) - 10

(%i44) fix(10.5);

(%o44) 10

(%i45) fix(-10.5);

(%o45) - 11

(%i46) entier(10);

(%o46) 10

(%i47) entier(-10);

(%o47) - 10

(%i48) entier(10.5);

(%o48) 10

(%i49) entier(-10.5);

(%o49) - 11

尚,この例の様に,絶対値で越えない数を返すのではないので注意が必要です.
random函数は,引数として 〈数値 〉 を取り,0から 〈数値 〉 − 1 の間の整数乱数を返す函数です.
cabs函数,realpart函数と imagpart函数は各々複素数の絶対値,実部と虚部を返す函数です.一般

の式に対しては,展開した式の%iを含まない部分式を実部,%iを含む式を虚部とします.
carg函数は与えられた複素数の偏角 θ を π ≥ θ > −π の範囲で返す函数です.
sqrtは 〈式 〉 が実数の場合はその平方根を返します.尚,Maxima内部では 〈式 〉ˆ (1/2) で表現さ

れています.

LISP由来の数値函数¶ ³
?round 〈数値 〉の最も近い整数
?truncate 〈数値 〉の小数点以下を切り捨てµ ´
これらの数値函数は LISPの函数をそのまま利用します.その為,先頭に?が付きます.
?round函数は 〈数値 〉 を最も近い整数に丸めます.ここで,引数は float型であって,bigfloat型で

はありません.
?truncate函数は float型の 〈数値 〉 を引数とし,小数点以下を切り捨てます.

2.2 多項式

この節では,Maximaの多項式の表現について述べます.

2.2.1 多項式の一般表現

Maximaの多項式は通常の Cや FORTRANで記述する xˆ2+3*x*z+4や x**2+3*x*z+4の様
な書式で入力されます.ところが,内部表現と表示は入力式そのままではありません.大域変数の設
定に従って簡易化を行ない,Maximaの順序 >m に従って項内部の変数や項自体の並び換えを実行

したものが返されます.
以下に,入力式と内部式の例を示します.

(%i28) a:x+y+z;

(%o28) z + y + x

(%i29) :lisp $a;

((MPLUS SIMP) $X $Y $Z)

(%i29) b:z+x+y;

(%o29) z + y + x

(%i30) :lisp $b;

((MPLUS SIMP) $X $Y $Z)

(%i31) c:(1+2)*x+3*y+(2+1-2)*z-z;

(%o31) 3 y + 3 x

(%i32) :lisp $c;

((MPLUS SIMP) ((MTIMES SIMP) 3 $X) ((MTIMES SIMP) 3 $Y))

(%i33) a1*x+a2*x;

(%o33) a2 x + a1 x

(%i34) d:x1^2*x8^2*x3;

2 2

(%o34) x1 x3 x8

(%i35) :lisp $d;

((MTIMES SIMP) ((MEXPT SIMP) $X1 2) $X3 ((MEXPT SIMP) $X8 2))

この例では多項式 x+y+zと (1+2)*x+3*y+(2+1-2)*z-zの処理を示しています. 最初に変数 a
に x+y+zを割当てています. :lisp $a; で変数 aの内部表現を参照すると,((MPLUS SIMP) $X $Y
$Z)が返却されます.この S式の頭にある (MPLUS SIMP)は式の主演算子が和 + である事を示
し,その後に被演算子となる項が並んでいます.この様にMaximaの多項式の内部表現でも先頭に
演算子が置かれる前置式表現になっています.尚,項の並びは同じ式であれば項や変数の順番を変
更しても同じ結果になります.この例で示す様に x+y+zでも z+x+yと入力しても同じです. これ
はMaximaの変数順序 >m に従って与えられた多項式の項を構成する変数や項を並べ替えている

からです.

この順序 >m は基本的に逆アルファベット順で変数を並べる順序です. 順序一般に関しては 1.1
節,順序 >m に関しては,1.3 節を参照して下さい.多項式の内部表現では,項に対して,その項を構
成する変数を順序 >m に対して小さいもの順に並べ,式を構成する項に対して,順序 >m で小さな

項で並べます.最初の例の式 x+y+zの内部表現リストは (+ x y z)となります.これは,項順序が z
>m y >m xとなる為で,項を順序 >m に従って小さいものから順に並べたからです.次に,単項式
x1ˆ2 x8ˆ2x x3の場合, 単項式は x1ˆ2 x3 x8ˆ2に並べ替えられています. これは,x8 >m x3 >m x1
となる為にMaxima内部で変数を x1,x3,x8の順に並べ替えた為です.更に,x*4の様な変数と数値
の場合,数値が >m の下位に来る為に,4*xになります.
以下に,多項式と単項式の一般表現について纏めておきましょう.

多項式と単項式の一般表現¶ ³
多項式 ((mplus simp) 項1 · · · 項m) 項m >m · · · >m 項1

単項式 ((mtimes simp) 数値 変数1の羃 · · · 変数nの羃) 変数n >m · · · >m 変数1µ ´
式の表示では,この内部表現を基に式の表示を行います.多項式の場合,大きな項から順番に表示

されます.その為,x+y+zと入力しても z+y+xと表示されます.ところが項に関しては,大きな変
数からではなく,小さい順に並べたままで表示されます. その為,x1ˆ2*x8ˆ2*x3は変数を順序 >m

に従って並べ替えられた x1ˆ2*x3*x8ˆ2で表示されます.
では x+y,x-y,x*y,x/y,xˆy についても :lisp函数を使って,その内部表現を調べてみましょう.

(%i33) t0:x+y;

(%o34) y + x

(%i34) :lisp $t0;

((MPLUS SIMP) $X $Y)

(%i35) t1:x-y;

(%o35) x - y

(%i35) :lisp $t1;

((MPLUS SIMP) $X ((MTIMES SIMP) -1 $Y))

(%i36) t2:x*y;

(%o36) x y

(%i37) :lisp $t2;

((MTIMES SIMP) $X $Y)

(%i38) t2:x/y;

x

(%o38) -

y

(%i39) :lisp $t3;

((MTIMES SIMP) $X ((MEXPT SIMP) $Y -1))

(%i40) t4:x^y;

y

(%o40) x

(%i41) :lisp $t4;

((MEXPT SIMP) $X $Y)

Maximaでは x-yが x+(-y),x/yが xˆ(-y)に置換されている事が判ります.この様にMaximaの
内部では可換積や可換積の羃と和を用いて多項式が表現されます. この様に変換しておく事で,差
の処理や商の処理の手間を減らす事が出来るのです.
この一般表現に対し,Maximaには内部表現をより簡潔にし,係数を有理数に変換した正準有理式

表現 (Cannonical Rational Expressions,略して CRE表現)もあります.

2.2.2 多項式のCRE表現

CRE表現は factor函数や ratsimp函数等々で内部的に利用されるもので,利用者はこの表現を
あまり意識する必要はありません. この CRE表現は本質的に展開された多項式や有理式函数に適
したリストによる表現の一つです.
最初に 1変数多項式 〈係数1〉〈変数 〉〈 次数1〉 + 〈係数2〉〈変数 〉〈 次数2〉 + · · · の CRE表現は以下の

書式で与えられます.
単変数多項式の CRE表現¶ ³

(〈変数 〉 〈次数1〉 係数1 〈次数2〉 係数2 · · ·)µ ´
尚,この CRE表現では,次数に関し, 〈次数1〉 > · · · > 〈次数2〉 > · · · を満しています.
有理数係数の 1変数多項式とその CRE表現は項順序 >m により一対一に対応します. これを多

項式 3x2 − 1 を使って説明しましょう.この多項式はMaxima内部では, 3x2 + (−1)x0 で表現され

ます.これを λ 式風に考えれば, λ x ·
(
3x2 + (−1)x0

)
となりますね.以上から,係数と次数の対か

ら構成されるリストは ((2 3) (0 -1))になります.ここでは予め変数が xであるとして処理していた
ので,変数の情報を落していても問題はありません. しかし,一般的には変数を明確にしておくべき
です.そこで,リストの先頭に xを入れてみましょう.すると (x (2 3) (0 -1))となります.更に,リ
スト中の小括弧を外して (x 2 3 0 -1)としても 3x2 + (−1)x0 の復元には問題ありません.
多変数多項式の場合も同様に CRE表現で書換える事が出来ます.但し,1変数の場合の様に平坦

なリストで CRE表現は表現されず,CRE表現のリストによる複合リストになります.多変数の場
合で重要な事は変数の間に順序を入れる事です.Maximaの場合, 辞書式順序を基礎にした順序 >m

が入っています.この順序の詳細は 1.3節を参照して下さい.
Maximaのデフォルトの順序 >m の変数順序の要点だけを述べると, アルファベットの大文字が

小文字よりも大きく,zが一番大きく aが一番小さいという逆アルファベット順で順番が付けられて
います.変数が二文字以上の場合,先頭の文字から順番に比較します.もし,先頭の文字が等しけれ
ば次の文字で比較します.途中でMaximaの変数順序 >m で大小関係がつくと,その順序で変数に
順番が入ります. 例えば,xxzと xxyの場合は頭の二つが xなので順序がまだ決まりませんが,最後
の zと y に関しては,z >m yとなるので,xxz >m xxyとなります.
さて,多変数多項式のCRE表現に戻りますが,この場合は,再帰的な考えで処理します. ここでは

例として 2xy +x−3 で考えてみましょう.先ず,この多項式を x の多項式と看倣すと (2y +1)x−3



82 第 2章 Maximaのデータとその操作

となるので,CRE表現の第一段目は (x 1 2y + 1 0 -3) となりますね.但し,第二成分の 2y + 1 は
CRE表現ではないので,これを CRE表現に変換した (y 1 2 0 1)で置換える必要があります.結局,
(x 1 (y 1 2 0 1) 0 -3)が求めるCRE表現となります. 次に, y の多項式として考えると, 2xy +x−3
なので,中間的には (y 1 2x 0 x − 3)中の x の式を CRE表現に置換えると, 最終的に,(y 1 (x 1 2)
0 (x 1 1 0 -3))が得られます.
この様に多変数の場合,CRE表現は順序をあやふやにしていると,表現が一意に定まるとは限り

ません.変数に順序を入れて大きな変数順に式を括れば,一意に定まります. Maximaでは主変数と
して宣言された変数が式に含まれている場合は,その主変数の多項式と看倣し,それ以外の変数に対
してはMaximaの変数順序 >m を用いて順序を入れます.式に主変数として宣言された変数が存在
しなければ,順序 >m に対して最高位の変数の式中の変数を主変数としてCRE表現を構築します.
ここで,二つの CRE表現が与えられた時に,それらの変数順序が異なれば,処理は非常に厄介な

事になるのが判るかと思います.その為,CRE表現を用いる函数に関しては, 後の事も考えて主変数
の設定を行う必要があります.
又,CRE表現の変数には,通常の算術演算子 (+,−, ∗, /) や整数羃 (ˆ)を持たないものを与えます.

その為, y ˆ 2 の様な式は変数に使えませんが, log(x) や cos(x + 1) の様な函数は使えます.
Maximaの変数順序は基本的に辞書式順序です.尚,局所的な変数の順序の変更が ratvars函数等

で指定出来ます.

2.2.3 一般表現とCRE表現への変換を行う函数

Maximaには与えられた式を一般表現から CRE表現,CRE表現から一般表現に変換する函数が
用意されています.

一般表現と CRE表現への変換に関連する函数¶ ³
rat(〈式 〉, 〈変数1〉, · · · , 〈変数n〉 ) 一般表現 ⇒ CRE表現
ratdisrep(〈式 〉 ) CRE表現 ⇒ 一般表現

totaldisrep(〈式 〉 ) CRE表現 ⇒ 一般表現µ ´
先ず,最初の rat函数は与えられた 〈式 〉 を展開し,浮動小数点を ratepsilonで指定された許容範

囲以内の有理数に変換し,全ての項を共通の分母で纏め,分子と分母の最大公約因子を除去します.
ここで変数の順序は無指定であれば, Maximaの順序 >mに従いますが,ravtars函数を用いて導入
された順序があれば, その順序を用います.rat函数は +,−, ∗, /と羃乗の他の函数を一般的には簡
易化しません.CRE表現でのアトムは一般形式のものと異なります.従って,rat(x)-xは 0と異なる
内部表現の rat(0)で計算されます.

(%i1) exp1:rat(x)-x;

(%o1)/R/ 0

(%i2) :lisp $exp1;

((MRAT SIMP ($X) (X13157)) 0 . 1)

(%i2) exp0:0;

(%o2) 0

(%i3) :lisp $exp0;

0

この rat函数には ratfac,ratprint,keepfloatといった直接動作に関連する大域変数があります.
ratdisrep函数は CRE表現から一般表現に引数を変換します.
totaldisrep函数は CRE表現から一般表現に 〈式 〉の全ての部分式を変換します.

2.2.4 有理式のCRE表現

有理式は多項式の分数ですが,有理式の CRE表現の表現は,多項式の分母と分子に共通因子が無
く,分母の筆頭項 (leading term)の係数を正にしたものとなります.
以下に実例を示しましょう.

(%i1) r1: rat((y-1)/((y-x)*z^2+1));

y - 1

(%o1)/R/ --------------

2

(y - x) z + 1

(%i2) r2: rat((y-1)/((x-y)*z^2+1));

y - 1

(%o2)/R/ - --------------

2

(y - x) z - 1

(%i3) r3:rat((y-1)/(-(y-z)*x^2+1));

y - 1

(%o3)/R/ ---------------

2 2

x z - x y + 1

(%i4) :lisp $r3;

((MRAT SIMP $(X $Y $Z) (X13180 Y13181 Z13182)) (Y13181 1 1 0 -1)

Z13182 1 (X13180 2 1) 0 (Y13181 1 (X13180 2 -1) 0 1))

(%i4) t3:(y-1)/(-(y-z)*x^2+1);

y - 1

(%o4) --------------

2

x (z - y) + 1

(%i5) :lisp $t3;

((MTIMES SIMP)((MPLUS SIMP) -1 $Y)

((MEXPT SIMP)

((MPLUS SIMP) 1

((MTIMES SIMP)((MEXPT SIMP) $X 2)((MPLUS SIMP)

((MTIMES SIMP) -1 $Y) $Z)))

-1))

(%i5)

この例では変数順序 >m が逆アルファベット順の為,変数の順序は z >m y >m xになります.そ
の為,zが主変数となり,式は zの多項式として纏められます.最初の二つの例では, (x-y)*zˆ2は,y
>m xとなる為に, -(y-x)*zˆ2に並び替えられてしまいます.この際に,最も順序の高い項となる y*
zˆ2の係数を正にする為に,必要に応じて-1がかけられています.この例で示す様に,式が CRE表
現,或いは CRE表現の部分式を含む場合,記号/R/が行ラベルに続きます.

:lisp $r3; で CRE表現の内部表現を表示しています.先頭のMRATで CRE表現である事を示
し,その後のリストで有理式の変数が X,Y,Z,これらの変数に対応する内部変数が X13180,Y13181
と Z1382である事を示しています.ここで内部変数は LISPの gensym函数で生成されるもので,後
ろの番号は生成順に付けられるものです. 尚,有理式の変数はMaximaの showvar函数を使って取
り出す事が可能ですが, それらの変数に対応する内部変数はMaximaの内部変数の genvarに登録
され,参照する場合には :lisp genvar の様にします.
又,変数リストの順版は順序>mに対して小さい順に左から並ぶ為,左端が最小で右端が主変数と

なります.その為,この式ではZが主変数となります.この様な変数リストを含むリスト (MAT cdots)
の後に分子となる式が来ます.この式は先頭が Y13181の為,変数 Yの多項式で,後の 1 1 0 -1か

ら次数が 1でその係数が 1の項と, 次数が 0で係数が-1である事が判ります.最後の副リストが分
母の式となり,先頭が Z13182となっているので,主変数 Zの多項式である事が判ります. 以降,分
子の時と同じ様に読めば良いのです.但し,式はMaximaの変数順序 >m に従って読む必要があり

ます.
この例では Z >m Y >m X の順なので,変数 Zが式中にあれば, Zの多項式として表現し,Zがな

くて変数YがあればYの多項式,そして,変数 Zと Yの両方が無ければ変数Xの多項式と,帰納的
に解釈します.
これに対し,同じ多項式の一般表現を最後に示しますが,非常に長いものになっている事が判り

ます.
尚,CRE表現で分母が整数の場合 (CRE表現では,浮動小数点は有理数で近似されます),CRE表

現の内部表現は少し変化します.

(%i4) r1:rat((x-1)/5);

x - 1

(%o4)/R/ -----

5

(%i5) :lisp $r1;

((MRAT SIMP $(X) (X13157)) (X13157 1 1 0 -1) . 5)

この例で示す様に分子は x-1ですが,分子と分母の間に記号.が入っています. CRE表現では分
子と分母の間に,分母が整数の時に限って記号.を入れています. これは CRE表現の生成で LISP
の cons函数が用いられている事を示しています. また変数Xの後に数値が入っていますが,これは
LISP内部の処理で変数 Xに対応するシンボルを生成した際に割当てられた通し番号です.
拡張 CRE表現は taylor級数を表現する為に用いられています.有理式の表記は正の整数では無

く,正か負の有理数となる様に拡張されており,係数はそれ自身, 多項式と云うよりは,上で記述し
た様に有理式となっています. これらは内部的に再帰的な多項式形式によって表現され,多項式形
式は CRE表現に類似していますが,より一般化したものです.尚,切り捨てられる次数の様な情報
も追加されています.
式を表示する際に,拡張 CRE表現の場合は記号/T/が式の行ラベルに続きます.

(%i1) t1:taylor(exp(x),x,0,5);

2 3 4 5

x x x x

(%o1)/T/ 1 + x + -- + -- + -- + --- + . . .

2 6 24 120

(%i2) :lisp $t1;

((MRAT SIMP (((MEXPT SIMP) $%E $X) $X) (%e^x13162 X13163)

($(X ((5 . 1)) 0 NIL X13163 . 2)) TRUNC)

PS (X13163 . 2) ((5 . 1)) ((0 . 1) 1 . 1)

((1 . 1) 1 . 1) ((2 . 1) 1 . 2)

((3 . 1) 1 . 6) ((4 . 1) 1 . 24) ((5 . 1) 1 . 120))

雰囲気は CRE表現に近いものですが,係数と羃のリストの書式が consで結合されたリストであ
り,他に,リストの第一成分に Taylor展開の様々な情報が追加されている事が判ります.

CRE表現変換に関連する大域変数¶ ³
変数名 初期値 概要

keepfloat false 実係数の有理数への近似を制御

ratepsilon 2.0E-8 実係数の有理数近似する際の誤差を指定

ratalgdenom true 代数的整数を分母とする項の有理化を制御

ratprint true CRE表現変換時のメッセージを制御µ ´
大域変数 keepfloatが trueであれば,浮動小数点を含む式が CRE表現に変換される際に, 浮動小

数点が有理数に近似変換される事を防ぎます.尚,浮動小数点が有理数に近似される際に生じる誤
差は ratepsilonで制御されます.
大域変数 ratepsilonは式を CRE表現に変換する際に,係数を有理数に変換する時に用いられる

許容範囲となります.大域変数 ratepsilonよりも小さな浮動小数点は無視されます.浮動小数点を
有理数に変換したくなければ,大域変数 keepfloatを trueに設定します.

(%i30) ratepsilon;

(%o30) 2.0e-8

(%i31) ratsimp((1+2.0e-8)*x);

rat replaced 1.00000002 by 1//1 = 1.0

(%o31) x

(%i32) ratsimp((1+2.0e-7)*x);

rat replaced 1.0000002 by 5000001//5000000 = 1.0000002

5000001 x

(%o32) ---------

5000000

この例で示す様に,ratsimp函数を作用させた場合に大域変数 ratepsilonよりも小さな数が無視さ
れ,浮動小数点が有理数に変換されている事が判ります。
大域変数 ratalgdenomは式中に代数的整数を項の分母として持つ式に対し,大域変数 ratalgdenom

が trueの場合に,その分母を有理化します.これを実行する為には, 大域変数 algebraicを trueに設
定して,式を CRE表現に変換しておく必要があります.

(%i16) algebraic:true;

(%o16) true

(%i17) ratalgdenom:true;

(%o17) true

(%i18) rat(1/sqrt(2)*x^2+1);

2

sqrt(2) x + 2

(%o18)/R/ --------------

2

(%i19) ratalgdenom:false;

(%o19) false

(%i20) rat(1/sqrt(2)*x^2+1);

2

x + sqrt(2)

(%o20)/R/ ------------

sqrt(2)

この例で示す様に,大域変数 algebraicと大域変数 ratalgdenomを同時に trueにすると, 分母に√
2 を持つ式の分母が有理化されている事に注意して下さい.
大域変数 ratprintが falseであれば,浮動小数点の有理数への変換を報せるメッセージ出力を抑

制します.
CRE表現の式の処理に関連する大域変数¶ ³

変数名 初期値 概要

ratdenomdivide true 分子の項の分離を制御

ratexpand false CRE表現の展開を制御
ratfac false CRE表現式の因子分解を制御
ratsimpexpons false ratsimpの自動実行を制御
ratwtlvl false 近似の際の切捨てを制御

ratweights [] 重みのリスト

rootsconmode true rootscontract函数を制御µ ´
大域変数 ratdenomdivideが falseであれば,ratexpand函数を作用させた式に対して, 分子の項を

分離する事を抑制します.
大域変数 ratexpandが trueであれば,それらが一般形式に変換されるか表示された時に CRE式

が展開されます.
大域変数 ratfacが trueであれば,CRE有理式に対して部分的に因子分解された形式で出力しま

す.有理的操作の間,factor函数を実際に呼ばずに,式を可能な限り因子分解します.これでメモリ空
間を節約し,計算時間を幾らかを節約します.有理式の分子と分母は互いに素とします.
例えば,rat((xˆ2 -1)ˆ4/(x+1)ˆ2)は (x-1)ˆ4*(x+1)ˆ2になりますが, 各部分の因子は互いに素と

は限りません.
大域変数 ratfacと大域変数 ratweightsの手法は互換性が無いので,両者を同時に使っていけま

せん.
大域変数 ratsimpexponsが trueであれば,簡易化中に式の羃に対し,自動的に ratsimp 函数が実

行されます.
大域変数 ratwtlvlは ratweight函数を用いた式を纏める際に,CRE表現の切捨ての制御で用いら

れます.尚,デフォルト値が falseであれば,切捨ては生じません.
大域変数 ratweights は ratweight 函数で設定される指定された重みのリストです. 大域変数

ratweightsや ratweight()函数でそのリストが見られます.
大域変数 rootsconmodeは rootscontract函数の挙動を定めます.大域変数 rootsconmodeが false

ならば,rootscontract函数は有理数次数の分母が同じ次数の羃だけを纏めます. trueの場合,次数

の分母が割切れる羃だけを纏めます allの場合,全ての有理次数の分母の LCMを取って纏めます.

式 rootsconmode rootscontractの結果

xˆ(1/2) ∗ yˆ(3/2) false (x ∗ yˆ3)ˆ(1/2)

xˆ(1/2) ∗ yˆ(1/4) false xˆ(1/2) ∗ yˆ(1/4)

xˆ(1/2) ∗ yˆ(1/4) true (x ∗ yˆ(1/2))ˆ(1/2)

xˆ(1/2) ∗ yˆ(1/3) true xˆ(1/2) ∗ yˆ(1/3)

xˆ(1/2) ∗ yˆ(1/4) all (xˆ2 ∗ y)ˆ(1/4)

xˆ(1/2) ∗ yˆ(1/3) all (xˆ3 ∗ yˆ2)ˆ(1/6)

2.2.5 係数体について

Maximaでは多項式環の係数体として有理数 Q に純虚数 %i を付加した体 Q[%i]/〈%i2 + 1〉 が
デフォルトです.
その他の係数体をMaximaでは扱う事が可能です.先ず,pを正素数とする場合に, 多項式の係数

体を Zp[%i]/〈%i2 + 1〉とする事も可能です.
更に,Maximaの係数体Q[%i]/〈%i2 + 1〉に代数的整数を追加した体も扱えます.ここで代数的整

数とは整数係数の 1変数多項式で,最高次数の項の係数が 1となるmonicな多項式の解となる数の
事です. より一般的に整数係数の 1変数多項式の解となる数は代数的数と呼ばれます. 但し,Maxima
では代数的数は扱えません.
代数的整数はその定義から多項式と密接に関連する数です.aが代数的整数である為には,aを解

とする多項式が定義から必ず存在します.その aを解とする多項式の中で,最小次数の多項式を aの
最小多項式と呼びます.代数的整数の例として, 純虚数 i や

√
2 等が挙げられます.これらの最小多

項式は各々 x2 + 1 と x2 − 2 になります.
この係数体に関連するMaximaの大域変数を以下に示しておきます.

環に関連する大域変数¶ ³
変数名 初期値 概要

modulus false 剰余 pを設定

algebraic false 代数的整数の自動簡易化を制御µ ´
大域変数modulusに正素数 p を設定すると,多項式のCRE表現への変換の際に, p の剰余で計算

されます.即ち,CRE表現の係数体はQ[%i]/〈%i2 +1〉 から Zp[%i]/〈%i2 +1〉 になります.この大域
変数はmod函数にも影響を与えます.但し,mod函数の第二引数を与えない場合, 大域変数modulus
の値が用いられる為です.又,大域変数modulusに正素数以外の値を設定する事も可能です.
大域変数modulusに正素数 pを設定した場合 p/2よりも大きな整数全てを考えなくても良くなり

ます.例えば, pとして 5を採った場合,整数の剰余は {0, 1, 2, 3, 4}となりますが,3 ≡ −2(mod 5),4 ≡
−1(mod 5) となります.実際, 3− (−2) = 4− (−1) = 5(mod 5) となるので, 絶対値では {0, 1, 2} だ
けを考えれば良い事になります.この事を利用すれば, 計算をより簡単に行う事が可能になります.
次の大域変数 algebraicはMaxima上で代数的整数の簡易化を行う場合,必ず trueにしなければ

なりません.この大域変数だけでは,自動的に代数的整数が処理される訳ではありませんが,代数的



2.2. 多項式 89

整数を処理する函数や他の大域変数では,この algebraic が trueとなっている事が前提となってい
るものがある為,代数的整数を扱う場合には, trueに設定するのが良いでしょう.ここで,代数的整
数に関連する大域変数としては, ratalgdenomがあります.これは代数的整数を分母とする項の有
理化を制御するものです.これらの変数は ratexpand函数,ratsimp函数等の CRE表現の式を扱う
函数に大きく影響します.その他の函数では,gcd函数も影響を受けます.

factor函数の様に最小多項式を与える事で,多項式の処理を代数的整数を付加した係数体上で処
理が行える函数もあります.例えば,factor函数の場合,最小多項式が x2 − 2 となる代数的整数 a

を用いて式を分解する場合,最小多項式に,その代数的整数を代入した形で factor函数に引き渡し
ます.

(%i1) factor(x^4-4,a^2-2);

2

(%o1) (x - a) (x + a) (x + 2)

この例では,多項式 x2 − 4 を環 Q[
√

2][x] 上で因子分解を行っています.

2.2.6 tellrat函数

より徹底して代数的整数を利用する場合は tellrat函数を用います.この tellrat函数は最小多項
式や等式に tellrat属性を設定するもので,ratsimp函数や ratexpand函数で処理を行う際に大域変
数 algebraicが trueとなっていれば,その属性が処理に反映されます.
この tellrat函数は次の写像をMaximaに組込むものです.

tellrat(式) : Maximaの係数環→ Maximaの係数環/〈式 〉

但し,tellrat属性は CRE表現を扱う ratexpand函数や ratsimp函数で処理を行う時のみに反映
されます.

tellrat函数¶ ³
tellrat(〈monicな多項式 〉)
tellrat(〈等式 〉)
tellrat()µ ´
引数として与えられる式は主変数に対して monicな多項式に限定されます.これは tellrat函数

が代数的整数をMaximaに与える事を目的としているからです. 又,等式として与える場合に等式
の左辺は代数的整数の羃乗で係数が 1のものに限定されます.この時,左辺の変数が主変数と看倣
され,右辺を左辺に移した式を最小多項式とする代数的整数が与えられます.従って,a2 = c3 − 2 を
tellrat函数に与えた場合, 主変数が a となり,この a がMaximaに追加される代数的整数となり,
その最小多項式が a2 − c3 + 2 で与えられます.ここで a2 − c3 + 2 を引数として与えると,主変数
はMaximaの項順序 >m から c となる為に代数的整数は c となる事に注意して下さい.
この tellrat函数は任意個の因子を取る事が可能です.
tellrat()で,Maximaに設定された代数的整数を表現する最小多項式のリストを返します.

(%i22) tellrat(x^2+x+1);

2

(%o22) [x + x + 1]

(%i23) tellrat(y^3+y^2+y+1);

3 2 2

(%o23) [y + y + y + 1, x + x + 1]

(%i24) tellrat();

3 2 2

(%o24) [y + y + y + 1, x + x + 1]

(%i25)

tellratで最小多項式を導入しても即座には反映されません.先ず,代数的整数の簡易化を行う為
に大域変数 algebraicを trueに設定していなければなりません. それから ratsimp等の CRE表現
が扱える函数で処理を行う必要があります.
例として,tellrat(aˆ2-2,bˆ2=cˆ4)と入力した場合を示します.

(%i11) tellrat(a^2-2,b^2=c^4);

2 4 2

(%o11) [b - c , a - 2]

(%i12) (a+2)^4,algebraic,expand;

4 3 2

(%o12) a + 8 a + 24 a + 32 a + 16

(%i13) (a+2)^4,algebraic,expand,ratsimp;

(%o13) 48 a + 68

(%i14) (b+c)^3,algebraic,expand,ratsimp;

5 4 3 2

(%o14) 3 c + b c + c + 3 b c

(%i15) (b+c)^3,expand,ratsimp;

3 2 2 3

(%o15) c + 3 b c + 3 b c + b

(%i16) algebraic:true;

(%o16) true

(%i17) ratexpand((b+c)^3);

5 4 3 2

(%o17) 3 c + b c + c + 3 b c

(%i18) expand((b+c)^3);

3 2 2 3

(%o18) c + 3 b c + 3 b c + b

この例で示す様に,tellratで設定した属性は大域変数 algebraicを trueに設定した環境下で,ratsimp
函数や ratexpand函数等の CRE表現を内部で用いる函数で処理する時に反映されます.
尚,この例での (a+2)ˆ4,algebraic,expand,ratsimpといった表記は,ev函数の表記方法の一つで,

ev((a+2)ˆ4,algebraic,expand,ratsimp)と同値です.
次に, tellrat(aˆ2-2,bˆ2=cˆ4); で代数的整数 bを追加した為, ratexpand((b+c)ˆ3)で,bˆ2が cˆ4

で置換されている事に注意して下さい. この様に,多変数の多項式を用いて代数的整数を入れる場
合,通常はMaximaの項順序 >m の影響を受ける為,等式の右辺に代数的整数の項を置きます. 例
えば,a=aˆ2+cˆ3や aˆ2=cˆ3-aの様にします.
ここで,tellrat函数を用いて多項式を被約する際に零因子で分母の有理化を試みない様に注意し

なければなりません.
例えば,tellrat(wˆ3-1);algebraic:true;rat(1/(wˆ2-w)) は零による割算になります. このエラーは

ratalgdenom:falseに設定する事で回避出来ます.
tellrat函数で設定した属性は tellrat()で見る事が出来ます.この場合,引数は特に設定する必要

がありません.
untellrat函数¶ ³

untellrat(〈x〉 )µ ´
tellrat函数で設定した属性は untellrat函数を使えば削除出来ます. この場合,untellratで代数的

整数を直接指定します.

(%i36) tellrat(a^2-2,b^3-c^2);

2 3 2

(%o36) [c - b , a - 2]

(%i37) tellrat();

2 3 2

(%o37) [c - b , a - 2]

(%i38) untellrat(a);

2 3

(%o38) [c - b ]

(%i39) untellrat(b);

2 3

(%o39) [c - b ]

(%i40) untellrat(c);

(%o40) []

(%i41) tellrat(a^2-2,b^3=c^2);

3 2 2

(%o41) [b - c , a - 2]

(%i42) untellrat(c);

3 2 2

(%o42) [b - c , a - 2]

(%i43) untellrat(b);

2

(%o43)

次に,複数の変数に対して tellratと untelratを実行した例を示します.

(%i21) tellrat(x^2+1,y^2+1);

2 2

(%o21) [y + 1, x + 1]

(%i22) ev(rat(x^3+1+y^3+y),algebraic);

(%o22)/R/ - x + 1

(%i23) untellrat(y);

2

(%o23) [x + 1]

(%i24) ev(rat(x^3+1+y^3+y),algebraic);

3

(%o24)/R/ y + y - x + 1

この例では変数 x,yが x2 + 1 = 0と y2 + 1 = 0を満す代数的整数と設定しています. この様に複
数の変数に対して整係数多項式を tellratに入力する事も可能で, evでも的確に評価されています.
次に untellrat(y)で yに関してのみ,tellratで設定した性質 (yは y2 + 1 = 0を満す代数的整数)

である事を除去しています.その後には,xに関する性質だけで評価が行われています.
尚,tellrat函数で入力可能な多項式は主変数に関して monic(最高次項の係数が 1) なものでなけ

ればならなりません.又,多変数の場合,untellrat函数は主変数 (maimvar)に対して行います.

(%i15) tellrat(x+y+z*y1);

Minimal polynomial must be monic

-- an error. Quitting. To debug this try debugmode(true);

(%i16) tellrat(x+y+z+1);

(%o16) [z + y + x + 1]

(%i17) untellrat(y);

(%o17) [z + y + x + 1]

(%i18) untellrat(z);

(%o18) []

(%i19) tellrat(2*x+y+z+1);

(%o19) [z + y + 2 x + 1]

(%i20) untellrat(z);

(%o20) []

この例で示す様に,x+y+z*y+1に関しては主変数が zで係数が yとなる為にエラーになります.
但し,2*x+y+z+1の様に主変数 zがmonicでありさえすれば問題はありません.untellrat函数が主
変数のみに使える事も上の例から分ります.

2.2.7 Horner表記

多項式の表記でHorner則に基く式の表記方法があります.これはX の多項式が与えられた場合,
変数X の次数の高い順番に項を並べます.例えば,与えられた多項式が anXn + an−1X

n−1 + · · ·+
a1X + a0 とすると, X ((· · · (anX + an−1) + · · ·) + a1) + a0 の様に帰納的に変数 X の積で式を纏

める方法です.この Horner則を用いると,式の積の回数を減らす事が出来るので,複雑な多項式の
数値計算を高速化する場合には非常に有効な手段の一つです.

horner表記に変換する函数¶ ³
horner(〈式 〉, 〈主変数 〉)
horner(〈式 〉)µ ´
主変数を指定した場合には,その主変数を用いてHorner則を適用します. 式の主変数を指定しな

い場合,Maximaの変数順序>mに従って,>mで最大の変数を主変数とし,その変数に対して与式に
Horner則を適用します.

(%i3) expr:(x+2*y)^5,expand;

5 4 2 3 3 2 4 5

(%o3) 32 y + 80 x y + 80 x y + 40 x y + 10 x y + x

(%i4) horner(expr,x);

5 4 3 2

(%o4) 32 y + x (80 y + x (80 y + x (40 y + x (10 y + x))))

(%i5) horner(expr);

2 3 4 5

(%o5) y (y (y (y (32 y + 80 x) + 80 x ) + 40 x ) + 10 x ) + x

尚,主変数として函数を指定する事も可能です.

(%i12) neko:(sin(x)+2*y)^5,expand$

(%i13) horner(neko,sin(x));

5 4 3

(%o13) 32 y + sin(x) (80 y + sin(x) (80 y

2

+ sin(x) (40 y + sin(x) (10 y + sin(x)))))

2.2.8 多項式に関する函数

多項式に関連する述語函数¶ ³
函数 trueを返す条件
ratnump(〈式 〉 ) 〈式 〉が有理式の場合
ratp(〈式 〉) 〈式 〉が拡張 CRE表現の場合µ ´

ratnump函数は 〈式 〉が 〈有理数式 〉であれば true,それ以外は falseを返します.
ratp函数は 〈式 〉が CRE表現,或いは拡張 CRE表現であれば trueで, それ以外は falseを返し

ます.
多項式の係数を取出す函数¶ ³

coeff(〈式 〉, 〈変数 〉, 〈次数 〉 )
ratcoef(〈式1〉, 〈式2〉, 〈n〉 )
ratcoef(〈式1〉, 〈式2〉 )
bothcoef(〈式 〉, 〈変数 〉 )µ ´

coeff函数は,〈式 〉に含まれる項 〈変数 〉〈 次数 〉の係数を求めます. 〈次数 〉を省略すると次数は 1
が設定されます. 〈変数 〉はアトムか真の部分式です. 具体的には,x,sin(x),a[i + 1],x + y等です.
尚,真の部分式の場合,(x+ y)が式の中に現れていなければならなりません. ここで 〈変数 〉〈 次数 〉

の項を正確に求める為には, 式の展開や因子分解が必要な場合もあります.何故なら,coeff函数では
自動的に式の展開や因子分解が実行されないからです.

(%i1) coeff(2*a*tan(x)+tan(x)+b=5*tan(x)+3,tan(x));

(%o1) 2 a + 1 = 5

(%i2) coeff(y+x*%e**x+1,x,0);

(%o2) y + 1

ratcoefは 〈式1 に含まれる項 〈式2〉〈n〉 の係数を返します. 〈n〉が 1の場合は 〈n〉が省略出来ま
す. 尚,返却値には,〈式2〉に含まれる変数を函数の変数としても含まないものです.
この様な係数が存在しない場合は零を返します.ratcoefは展開等を行って式を簡易化するので,

単純に 〈式2〉〈n〉 の係数を返す coefと異った答を返します.
例えば,ratcoeff((x+1)/y+x,x)は (y+1)/yを返却しますが,coeffは 1を返します.
ratcoef(〈式1〉, 〈式2〉 ,0)は, 〈式1〉から 〈式2〉を含まない項の和を返します. その為,〈式2〉が負

の羃の項に含まれれば,ratcoefは使ってはいけません.〈式1〉が有理的に簡易化されていれば,係数
は予期した様に現れないかもしれません.

bothcoef函数は二成分のリストを返し,このリストの第一成分が 〈式 〉中の 〈変数 〉の係数 (式が
CRE表現であれば ratcoef,それ以外は coeff函数で見つけたもの)となります. 第二成分が 〈式 〉
の残りとなります.即ち,[a,b]が返却値であれば,〈式 〉 = a ∗ 〈変数 〉 + bとなります.

項数や次数を返す函数¶ ³
nterms(〈式 〉 )
powers(〈式 〉 ,〈変数 〉 )µ ´

nterms函数は 〈式 〉 を展開した時の項数を返却します.この函数は多項式以外の函数を含む通常
の式でも利用可能ですが,函数は引数の形式を問わず一つで数えられます.

(%i26) nterms((x+1)^2);

(%o26) 3

(%i27) nterms(sin(x+1)^2);

(%o27) 1

(%i28) nterms((sin(x+1)+1)^3);

(%o28) 4

(%i29) nterms((sin((x+1)^10)+1)^3);

(%o29) 4

この例で示す様に,sin函数の引数がどの様な式であっても,一つで数えられている事に注意して
下さい.
次の powers函数は 〈式 〉 に現われる 〈変数 〉 の次数リストを返します.この函数を利用する為に

は,予め, load(powers); で函数の読込を行う必要があります.

多項式の項の次数を返す函数¶ ³
hipow(〈多項式 〉, 〈変数 〉)
lopow(〈多項式 〉, 〈変数 〉)µ ´

hipow函数は 〈多項式 〉に含まれる 〈変数 〉の項の最高次数を返します.尚,hipowでは式の展開
を自分で実行しない為,予め式を展開しておく必要があります.以下の例では (x + 1)4 を展開せず
に hipowを用いた結果と expandで展開した式に対して hipowを用いた結果を示しています.

(%i5) hipow((x+1)^4,x);

(%o5) 1

(%i6) hipow(expand((x+1)^4),x);

(%o6) 4

lopow函数は 〈多項式 〉の部分式 〈変数 〉の次数で, 明示的に現われものの中で最も低い次数を
返します.

多項式の変数に関する函数¶ ³
ratvars(〈変数1〉, · · · , 〈変数n〉 )
ratweight(〈変数1〉, 〈重み1〉, · · · , 〈変数n〉, 〈重みn〉)
showratvars(〈式 〉 )µ ´

ratvars函数は与えられた変数リストに含まれる変数に沿った順序をMaximaに入れる函数です.
変数リストは n個の変数引数で構成し,ratvars函数を実行した後, リスト中の一番右側の 〈変数n〉
が有理式にあれば, その変数を有理式の主変数になります.又,他の変数の順序はリストの右から左
への順番に従います.ある変数が ratvarsリストから抜けていれば, その変数は一番左側の 〈変数1〉
よりも低い順序が付けられます.

ratvars函数の引数は変数,或いは sin(x)の様な非有理的函数の何れかでなければなりません.尚,
大域変数 ratvarsは,この函数に与えられた引数のリストとなります.

(%i26) ratvars(x,y,z);

(%o26) [x, y, z]

(%i27) rat(x+y+z);

(%o27)/R/ z + y + x

(%i28) rat(a+x+y+z);

(%o28)/R/ z + y + x + a

(%i29) ratvars(z,y,x);

(%o29) [z, y, x]

(%i30) rat(a+x+y+z);

(%o30)/R/ x + y + z + a

ratweight函数は,〈重みi〉を 〈変数i〉に割当てます.その重みが大域変数 ratwtlvlの値に従って項
が 0で置き換えられます. 項の重みは項の中の変数の羃を重みに掛けた積の和となります. 例えば,
3 · v2

1 · v2 の重みは 2 ·w1 + w2 となります.この切捨ては,式の CRE表現の積や羃乗を実行する時
のみに生じます.
尚,ratfac函数と ratweight函数の手法は互換性がないので両方同時に使えません.
showratvars函数は 〈式 〉の大域変数 ratvarsのリストを返します.

(%i30) exp:x^2+y^2+z^3;

3 2 2

(%o30) z + y + x

(%i31) showratvars(exp);

(%o31) [x, y, z]

多項式を纏める函数¶ ³
factcomb(〈式 〉 )
fasttimes(〈多項式1〉, 〈多項式2〉n )
rootscontract(〈式 〉 )µ ´

factcomb函数は 〈式 〉中に現われる階乗の係数を階乗それ自体に置換して纏めます.即ち, (n+1)∗n!
を (n + 1)! にする事です.ここで, 大域変数 sumsplitfactが falseに設定されていれば,minfactorial
が factcombの後に適用されます.

fasttimes函数は多項式の積に対する特殊なアルゴリズムを用いて, 〈多項式1〉と 〈多項式2〉の積
を計算します. これらの多項式は,多変数で各次数に対して係数が 0でない項が多く,両者共に殆ど
同じ大きされなければ効果があまり出ません.

nとmを多項式の次数とすると,古典的な積では n ∗mのオーダーで計算を行いますが,fasttimes
函数を用いると max(n,m)1.585 のオーダーになります.

rootscontract函数は有理数次数の羃同士の積を大域変数 rootsmodeの値に従って纏めます.例
えば,rootsmodeが trueの場合,xˆ(1/2)*yˆ(3/2)の様な根同士の積を sqrt(x*yˆ3)に纏めます.
大域変数 radexpandが trueで大域変数 domainが realであれば,rootscontract函数は absを sqrt

に変換します.即ち,abs(x)*sqrt(y)を sqrt(xˆ2*y)に変換します.
rootscontract函数は logcontract函数と似た手法で ratsimp函数を用います.

有理式に関連する函数¶ ³
combine(〈式 〉 )
denom(〈有理式 〉 )
num(〈有理式 〉 )
ratdenom(〈有理式 〉 )
ratnumer(〈有理式 〉 )
ratdiff(〈有理式 〉, 〈変数 〉 )µ ´

combine函数は 〈式 〉に含まれる和の部分式を同じ分母で纏めて一つの項にします.
denom函数は 〈有理式 〉の分母 (DENOMinator)を返します. 尚,有理式が通常の多項式であれ

ば,1を返します.

(%i40) denom((x^2+1)/(y^2+1)/2);

2

(%o40) 2 (y + 1)

(%i41) denom(x^2+1);

(%o41) 1

(%i42) denom(1/2*x^2+1/2);

(%o42) 1

(%i43) denom((x^2+1)/2);

(%o43) 2

num函数は有理式の分子 (NUMerator)を返します.
ratdenom函数は 〈有理式 〉の分母を計算します. 〈有理式 〉が一般形式で,結果も一般形式のも

のが必要ならば, denom函数を用いましょう.
ratnumer函数は 〈有理式 〉の分子を取り出します. 一般形式の 〈有理式 〉 に対して,CRE表現の

結果が不要であれば, num函数を使いましょう.
ratdiff函数は 〈有理式 〉の微分を 〈変数 〉で行います.有理式に対しては diff函数よりも処理が

速く,計算結果は CRE表現になります.尚,ratdiff函数は因子分解された CRE表現には使ってはい
けません. 因子分解された式では通常の diff函数を使いましょう.

終結式に関連する函数¶ ³
resultant(〈多項式1〉, 〈多項式2〉, 〈変数 〉 )
bezout(〈多項式1〉, 〈多項式2〉, 〈変数 〉 )
eliminate([〈方程式1〉, 〈方程式2〉, · · · , 〈方程式n〉], [〈変数1〉, 〈変数2〉, · · · , 〈変数k〉])µ ´

bezout函数は, 〈多項式1〉 と 〈多項式2〉 に対して, 〈変数 〉 を主変数とした場合のある係数行列
を返します. この係数行列の行列式を取ると終結式に等しくなるものです.
この行列 bezout(f,g,x)の determinantが多項式 f と gの終結式になります. 従って,bezout函数

と determinant函数と組合せれば resultant函数の代替になります.
resultant函数は二つの多項式 〈多項式1〉と 〈多項式2〉の終結式を計算し,指定した 〈変数 〉 を消

去します.
ここで,多項式 f と gの解を各々αi,βj とすると,多項式 f と gの終結式 res(f, g)は以下の式と

等しくなる事が知られています.

res(f, g) = an
mbm

n

∏
1≤i≤m,1≤j≤n

(αi − βj)

その為,終結式は 〈多項式1〉と 〈多項式2〉が共通の定数の因子を持つ場合に限って零になる多項
式となる事が判ります.
終結式の計算方法は,〈多項式1〉 と 〈多項式2〉 を 〈変数 〉 の多項式と看倣した場合の係数から構

成される行列の行列式から計算出来ます. 行列の大きさは, 〈多項式1〉 と 〈多項式p〉 の次数を各々
m,n とすると, m + n 次の正方行列となりますが,bezoutでは行列操作によって,それよりも小さ
な行列が得られる場合には,その行列を表示します.
具体的には,多項式 f と g を次のものとします.

f =
m∑

i=0

aix
i

g =
n∑

i=0

bix
i

すると,

resultant(f, g, x) = det



am am−1 · · · · · · a1 a0 · · · 0
...

. . . . . . . . . . . .
...

0 · · · am am−1 · · · · · · a1 a0

bn bn−1 · · · · · · b1 b0 · · · 0
...

. . . . . . . . . . . .
...

0 · · · bn bn−1 · · · · · · b1 b0


で終結式は計算出来ます.尚,〈多項式1〉, 〈多項式2〉が因子分解可能であれば resultant函数を呼

び出す前に factor函数を呼出すと良いでしょう.
eliminate函数は与えられた方程式,或いは零と等しいと仮定した式から続けて終結式を取る事

で,指定された変数の消去を行います. eliminate函数に引渡した k個の 〈変数1〉, · · · , 〈変数k〉を消
去した n-k個の式のリストを返します. 最初の 〈変数1〉は消去されて n-1個の式を生成し, 〈変数2〉
以降も同様です.

k=nの場合,結果リストは k個の 〈変数1〉, · · · , 〈変数k〉 を持たない一つの式となります.そして
最後の変数に対応する終結式を解く為に solve函数を呼出します.

(%i1) exp1:2*x^2+y*x+z;

2

(%o1) z + x y + 2 x

(%i2) exp2:3*x+5*y-z-1;

(%o2) - z + 5 y + 3 x - 1

(%i3) exp3:z^2+x-y^2+5;

2 2

(%o3) z - y + x + 5

(%i4) eliminate([exp3,exp2,exp1],[y,z]);

8 7 6 5

(%o4) [7425 x - 1170 x + 1299 x + 12076 x

4 3 2

+ 22887 x - 5154 x - 1291 x

+ 7688 x + 15376]

(%i5) eliminate([x+y=2,2*x+3*y-5=0],[x,y]);

(%o5) [1]

(%i6) eliminate([x+y=2,2*x+3*y-5=0],[x]);

(%o6) [y - 1]

(%i7) eliminate([x+y=2,2*x+3*y+5=0],[x]);

(%o7) [y + 9]

(%i8) eliminate([x+y=2,2*x+3*y+5=0],[x,y]);

(%o8) [- 9]

終結式のアルゴリスムを指定する大域変数¶ ³
変数名 初期値 可能な値

resultant subres [subres,mod, red]µ ´
大域変数 resultantは同名の resultant函数による終結式の計算で用いるアルゴリズムを設定し

ます.指定可能なアルゴリズムを以下に示しておきます.

デフォルト subres

モジュラー終結式アルゴリズム mod

縮約 prs red

殆どの問題では subres が最適です. 単変数の大きな次数や 2 変数問題では mod がより良いで
しょう.

共通因子を求める函数¶ ³
gcd(〈式1〉, 〈式2〉, 〈変数1〉, · · ·)
gcde(〈多項式1〉, 〈多項式2〉, 〈変数 〉)
gcdex(〈多項式1〉, 〈多項式2〉)
gcfactor(〈Gauss整数 〉)
gfactor(〈多項式 〉)
gfactorsum(〈多項式 〉)
ezgcd(〈多項式1〉, 〈多項式2〉, · · ·)
content(〈多項式 〉, 〈変数1〉, · · · , 〈変数n〉 )µ ´

gcd 函数は与えられた多項式の最大公約因子を計算します. この gcd 函数は多くの函数, 例え
ば,ratsimp函数や factor函数等でも利用されています.

gcd函数に直接影響を及ぼす大域変数として,同名の大域変数 gcdがあります. この大域変数 gcd
は gcd函数で用いるアルゴリズムを決定する変数です.

大域変数 gcdに設定可能な値¶ ³
値 概要

subres 副終結式を利用 (デフォルト値)
ez ezgcd函数を利用
eez eez gcdを利用
red 被約

spmod 剰余

false gcd函数は常に 1を返却µ ´
代数的整数を扱う場合,例えば,gcd(xˆ2-2*sqrt(2)*x+2,x-sqrt(2))のGCDを計算する為には,大

域変数の algebraicが trueであり,大域変数 gcdが ezと false以外の値でなければなりません.
同次多項式に対しては,gcd:subresを用いる事を推奨します.

gcdex函数は 3個の多項式を成分とするリスト [a,b,c]を返します. 多項式 cが引数の 〈多項式1〉
と 〈多項式2〉 の最大公約因子で,a,bは共に c = a ∗ 〈多項式1〉 + b ∗ 〈多項式2〉 を満す多項式とな
ります.この函数が用いているアルゴリズムは Euclidの互除法に基づくものです.
尚,多項式が単変数の場合は 〈変数 〉を指定する必要はありませんが, 多変数の場合,多項式を 〈変

数 〉で指定した単変数の多項式と看倣して GCDを計算します.
何故,多変数多項式が相手の場合に,変数の指定を行う必要があるかと言えば, 多変数多項式の場

合は,二つの多項式の最大公約因子が存在するとは限らないからです.
数学的には,最大公約因子は二つの多項式が生成するイデアルの生成元となります. 通常,多項式

の係数が実数や複素数で,考えている多項式が単変数のみ,即ち, 多項式環 k[x]であれば,任意のイ
デアルは単項イデアル,即ち,一つの多項式だけで生成されるので,この場合は最大公約因子が必ず
存在します.
その為,多変数の場合には二つの多項式の主変数として変数を一つ選択する必要があります.但

し,その選択が妥当でなければ,gcdexは適当な答を返すだけです.

(%i16) gcdex(x^2+1,x^3+4);

2

x + 4 x - 1 x + 4

(%o16)/R/ [- ------------, -----, 1]

17 17

(%i18) gcdex(x*(y+1),y^2-1,x);

1

(%o18)/R/ [0, ------, 1]

2

y - 1

(%i19) gcdex(x*(y+1),y^2-1,y);

(%o19)/R/ [1, 0, x y + x]

ここで,最後の多変数の例で,gcdex(x*(y+1),yˆ2-1,x)の結果で,GCDとして 1を返している事に
注意して下さい.この場合,多項式環K(y)[x]で処理を行っているので,共通の因子として期待され
る y+1にはなりません. ここで,K(y)[x]は xを主変数とした xと yの多項式環,つまり,xの多項式
で,その係数が体K 上の yの多項式となるものとして,xと yの多項式環K[x, y]を見直したもので
す.一般的に可換環 Kが UFD(Unique Factorized Domain: 一意分解整域)であれば,K[x]も UFD
になる事が知られています.その為,Eculidの互除法が利用可能になるので,gcdexは必ず結果を返
します.

gcdex(x*(y+1),yˆ2-1,y); とすれば,多項式環K(x)[y]の話になるので 1ではなく xy + x になり

ます.但し,この返却値が良いものとは言い難いものがあります.
gcfactor函数は Gauss整数上で 〈Gauss整数 〉の因子分解を行います. 尚,Gauss整数とは,複素

数 a + biで,aと bが整数になります.因子は aと bを非負とする事で正規化されています.

(%i56) gcfactor(5*%i+1);

(%o56) (1 + %i) (3 + 2 %i)

(%i57) gcfactor(2);

2

gfactor函数はGauss整数上で 〈多項式 〉の因子分解を行ないます. これは factor(exp,aˆ2+1)と
同様の結果を返します.

(%i3) gfactor(x^4-1);

(%o3) (x - 1) (x + 1) (x - %i) (x + %i)

(%i4) factor(x^4-1,a^2+1);

(%o4) (x - 1) (x + 1) (x - a) (x + a)

(%i5)

この例で,factorを用いたものでは方程式 x2 +1 = 0の解となる代数的整数 a(= i)を用いて x4−1
を因子分解しています.

gfactorsum函数は factorsumに似ていますが,factorの代りに gfactorが適用されます.

(%i58) gfactorsum(x^2+1);

(%o58) (x - %i) (x + %i)

(%i59) factor(x^2+1);

2

(%o59) x + 1

ezgcd函数は,第一成分が全ての多項式の GCD,残りの元が GCDで割った値を成分に持つリス
トを返します.この ezgcd函数では ezgcdアルゴリズムが常用されています.

content函数は二成分のリストを返し,このリストの第一成分が, 〈変数1〉を 〈多項式 〉の主変数
とした場合の各係数の最大公約因子,第二成分を第一成分で多項式を割ったmonicな多項式となり
ます.

(%i43) content(2*x*y+4*x^2*y^2,y);

2

(%o43) [2 x, 2 x y + y]

因子分解を行う函数¶ ³
factor(〈式 〉 )
factor(〈式 〉, 〈p〉 )
factorsum(〈式 〉 )
sqfr (〈式 〉 )
factorout(〈式 〉, 〈変数1〉, 〈変数2〉, · · · )
nthroot(〈多項式 〉, 〈n〉 )µ ´

factor函数は 〈式 〉を整数上で既約因子に分解します. factor(〈式 〉, 〈p〉)の場合,最小多項式が
〈p〉 となる代数的数 α を付加した体 Q[α] 上で式の因子分解を行います.尚,factor函数には動作に
影響を与える大域変数が存在します.この大域変数に関しては,factorに影響を与える大域変数を参
照して下さい.

factorsum函数はグループ単位で 〈式 〉の因子分解を試みます. このグループの項はそれらの和が因
子分解可能なものです. expand((x+y)ˆ2+(z+w)ˆ2)の結果は復元可能ですが, expand((x+1)ˆ2+(x+y)ˆ2)
の結果は共通項が存在する為に復元出来ません.

sqfr函数は factor函数と似ていますが,多項式因子は無平方 (square-free)となります.ここで多項
式因子 f が無平方であるとは,定数と異なる多項式 gで,g2が f を割切る様なものが存在しない場合

です.特に 1変数多項式の場合,f と d
dxf が共通の零点を持ちません. そこで,多項式式 aの無平方因

子分解は, 多項式 aの分解
∏n

i=1 ai
iで,その各因子 aiは無平方であり, i 6= jであれば gcd(ai, aj)=1

を満すものです.Maximaの sqfrは与えられた式に対して,この無平方因子分解を計算します.
では,無平方因子分解 sqfrと因子分解 factorとの違いを 4x4 + 4x3 − 3x2 − 4x− 1で比較して確

認しましょう.

(%i44) sqfr(4*x^4+4*x^3-3*x^2-4*x-1);

2 2

(%o44) (2 x + 1) (x - 1)

(%i45) factor(4*x^4+4*x^3-3*x^2-4*x-1);

2

(%o45) (x - 1) (x + 1) (2 x + 1)

この例で示す様に,factor函数は式を徹底して分解しているのに対し,sqfr函数は程々で止めてい
る事です.
この様に,多項式を無平方因子分解は通常の因子分解程の手間をかけません. 更に,有理多項式の

積分計算では,無平方な因子に分母を分解して,各因子を分母に持つ式に変形して積分を行うアル
ゴリズムもあります. この無平方分解は幅広く利用されています.

factorout函数は 〈式 〉を f(〈変数1〉, 〈変数2〉, · · ·) ∗ gの形式の項の和に書換えます. ここで, g は

factoroutの引数の各変数を含まない式の積で, f は因子分解されたものになります.
nthroot函数は与えられた整数係数の 〈多項式 〉が, ある整数係数の多項式を 〈正整数 〉で羃乗し

たものであれば, その多項式を返します.もし,その様な多項式が存在しなければ,エラーメッセー
ジが表示されます.
この函数は factor函数や sqfr函数よりも処理が遥かに速いものです.

(%i22) nthroot(x^2+2*x+1,2);

(%o22) x + 1

(%i23) nthroot(x^3+3*x^2+3*x+1,2);

Not an nth power

-- an error. Quitting. To debug this try debugmode(true);

(%i24) nthroot(1-3*x+3*x^2-x^3,3);

(%o24) 1 - x

因子分解に関連する大域変数¶ ³
変数名 初期値 概要

dontfactor [] factor函数で因子分解しない式のリスト
facexpand true factor函数で返される因子を制御
factorflag false 式に含まれる整数の因数分解を制御

berlefact true 因子分解のアルゴリズムを制御

intfaclim 1000 factorで用いる最大の約数の設定
newfac false 因子分解ルーチンの選択

savefactors false 式の因子の保存を制御µ ´
大域変数 dontfactorに factor函数で因子分解しない式を登録します. 尚,ratvars函数で入れた変

数の順序に対し,大域変数 dontfactorに割当てられたリストに含まれる変数よりも小さな変数を持
つ式の因子分解は実行されません.
大域変数 facexpandは factor函数で返された既約因子が展開された形式 (デフォルト)か, 再帰

的 (通常の CRE)形式であるかを制御します.
大域変数 factorflagが falseであれば有理式に含まれる整数の因数分解を抑制します.
大域変数 berlefactが falseであれば,factor函数で Kronecherの因子分解アルゴリズムが利用さ

れ,それ以外ではデフォルトの Berlekampアルゴリズムが使われます.
大域変数 intfaclimは大きな整数の因子分解を行う時に試す最大の約数を指定します. falseを指

定した場合,即ち,利用者が factor函数を明示的に呼び出す場合や, 整数が fixnum,つまり,1機械語
長に適合する場合,整数の完全な因数分解が試みられます.intfaclimの設定は factorの内部呼び出
しで用いられます.
大域変数 intfaclim は大きな整数の因数分解に長時間を費すのを防ぐ為に再設定しても構いま

せん.
大域変数 newfacが trueであれば,factorは新しい因子分解ルーチンを用います.
大域変数 savefactorsが trueであれば,幾つかの同じ因子を含む後の式の展開の処理速度向上の

為,式の各因子がある函数で保存されます.

剰余を計算する函数¶ ³
divide(〈多項式1〉, 〈多項式2〉, 〈変数1〉, · · · , 〈変数n〉 )
quotient(〈多項式1〉, 〈多項式2〉, 〈変数1〉, · · ·)
remainder(〈多項式1〉, 〈多項式2〉, 〈変数1〉, · · ·)
mod(〈多項式 〉)
mod(〈多項式 〉, 〈整数 〉)µ ´

divide函数は 〈多項式2〉による 〈多項式1〉 の商と剰余を計算します.各多項式は 〈変数n〉を主変
数とし, その他の変数は ratvars函数に現れるものとします.結果はリストで返却され, 第一成分が
商,第二成分が剰余となります.

(%i1) divide(x+y,x-y,x);

(%o1) [1, 2 y]

(%i2) divide(x+y,x-y);

(%o2) [- 1, 2 x]

quotient函数は 〈多項式2〉による 〈多項式1〉の割り算の商を計算します.
remainder函数は 〈多項式2〉による 〈多項式1〉の剰余を計算します.
mod函数は 〈多項式 〉を大域変数modulusで指定した値に対する剰余を計算します.
尚,大域変数 modulusはデフォルトでは falseの為,mod(〈多項式 〉)を実行するとエラーになり

ます.
大域変数modulusの値と無関係に,多項式の剰余を計算したければ, mod(〈多項式 〉, 〈整数 〉 ) で

〈多項式 〉の 〈整数 〉 による剰余を計算します. 尚,大域変数modulusの値は変更されません.

CRE表現の簡易化に関連する函数¶ ³
ratexpand(〈式 〉 )
fullratsimp(〈式 〉, 〈変数1〉, · · · , 〈変数n〉 )
fullratsimp(〈式 〉n )
ratsimp(〈式 〉 )
ratsimp(〈式 〉, 〈変数1〉, · · · , 〈変数n〉 )µ ´

ratexpand函数は和の積や指数の和を掛け,共通の分子で因子を纏め, 分子と分母の共通約数を通
分し,分子を分母によって割られた項へと分割して 〈式 〉の展開を行います.これは 〈式 〉を CRE
表現に変換し,それからを一般形式に戻しています.
その為,ratexpandに影響を与える大域変数には ratexpand,ratdenomdivideや keepfloatといっ

た CRE表現への変換に影響を与える大域変数があります.
fullratsimp函数は 〈式 〉に非有理式が含まれていれば, 普通,簡易化した結果を返す際に,やや非

力な非有理的 (一般的)簡易化に続いて ratsimpを呼出します.時には,その様な呼出しが一回以上
必要であるかもしれません. fullratsimpは,この操作を簡易にしたものです.

fullratsimp函数は非有理的簡易化に続けて ratsimpを式に変化が生じなくなる迄適用します.
例えば,式 exp:(xˆ(a/2)+1)ˆ2*(xˆ(a/2)-1)ˆ2/(xˆa-1)に対し, ratsimp(exp)によって

(xˆ(2*a)-2*xˆa+1)/(xˆa-1)が得られ,fullratsimp(exp)を実行すれば xˆa-1が得られます.
ratsimp函数は,非有理的函数に対して,有理的に 〈式 〉とその部分式の全てを引数も含めて,ra-

texpand の様に簡易化します. 結果は二つの多項式の商として, 再帰的な形式で返されます. 即
ち,主変数の係数は他の変数の多項式となっており,その係数もまた変数の順序に沿って,主変数の
次に順序の高い変数の多項式の係数と纏められています.変数は ratexpandの様に非有理的函数
(例えば,sinx2 + 1 を含みますが,ratsimpで非有理的函数に対する引数は有理的に簡易化されます.
ratsimpは ratexpandに影響を与える幾つかの大域変数の影響を受ける事に注意して下さい.
尚,ratsimp(〈式 〉, 〈変数1〉, · · · , 〈変数n〉)で,ratvarsに変数の 〈変数1〉, · · ·を設定した場合と同様

に,この変数の並びの順序で有理的簡易化を行います.

2.3 式について

Maximaでは殆どのものが式 (Expression)になります.例えば,128, 1 + 2 + 3や sin (x)の様に,
数値,変数といった複数のMaximaのアトムを演算子や函数で結び付けたものです.但し,式は文脈
で設定された各種の仮定や大域変数の指定に従って解釈されます.その為,1+2+3を入力すると 6
が返されますが,a:sin(x)の場合は xにどの様な値が割当てられているか,或いは関連する大域変数
の設定によって結果が異なります.
複数の式をコンマで区切って並べたものの全体を小括弧 () で括ったものが式の列となります.

尚,Maxima言語で生成した函数も実体は式の列です.その為, 利用者定義函数もリスト処理函数に
よる処理の対象となります. これはMaximaが動作する LISP側でもMaxima言語による函数を操
作出来る事を意味しています.その為,少しでも効率の良いプログラムを記述したければ, LISP函
数を記述する事になります.

Maximaは LISPで記述されています.その為,Maximaのデータや式等は LISP内部では別の表
現を持っています.ここでは簡単にMaximaの内部表現を示しましょう.

2.3.1 変数や文字列の内部表現

Maximaの内部表現：変数¶ ³
Maxima 内部表現

a $ a
?a a
”a” & a
‘a ((mquote) $a)µ ´

変数は内部で先頭に$を付けたものです.逆に,Maximaで?が先頭に付くものは内部では?を外し
たものになります.その為,LISPの函数をMaximaから実行させる場合,先頭に?を付けてやれば使
える事になります.但し,引数等はMaximaを通して与えられる為,Maximaの函数の様に引数を与
える必要があります.

2.3.2 二項演算の内部表現

Maximaの内部表現：数値演算¶ ³
Maximaの式 内部表現

x+y ((mplus simp) $x $y)
x-y ((mplus simp) $x ((mminus simp) $y))
x*y ((mtimes simp) $x $y)
x/y ((mquotient simp) $x $y)
x .y ((mnctimes simp) $x $y)
xˆ 2又は x**2 ((mexpt simp) $x 2)
xˆ ˆ 2 ((mncexpt simp) $x 2)µ ´
基本的に 変数1 演算 · · · 変数n は内部では ((演算名) 変数1 · · · 変数n) で表現されています.

更に演算名は全て mから開始している事に注意して下さい.mを先頭に付けてMaximaの函数と
LISPの函数を区別しているのです.
更に,差 x-yは実質的に x+(-y)で内部では表現されている事に注意して下さい.
ここで x-yの表現を調べてみましょう.使う函数は LISPの基本の基本,carと cdrです. Maxima

から LISP函数を利用する方法には,演算子?を使ってMaximaの函数として LISP の函数を利用す
る方法と,:lisp函数を使って直接 LISPの函数を入れる方法があります. ここでは,演算子?を使う方
法で調べてみましょう.

(%i74) ?car(x-y);

(%o74) ("+", simp)

(%i75) ?cdr(x-y);

(%o75) (x, - y)

演算子?を用いて LISPの函数を実行すると,その結果はMaximaで解釈された結果が帰って来
ます.この例からも,式 x-yがMaxima内部では x+(-y)になっている事が分るかと思います.
次に,論理演算の一覧を示しますが,内容的には数値演算のものと同様です.



108 第 2章 Maximaのデータとその操作

Maximaの内部表現：論理演算¶ ³
Maximaの式 内部表現

not a ((mnot simp) $a)
a or b ((mor simp) $a $b)
a and b ((mand simp) $a $b)
a = b ((mequal simp) $a $b)
a > b ((mgreater simp) $a $b)
a >= b ((mgeqp simp) $a $b)
a < b ((mlessp simp) $a $b)
a <= b ((mleqp simp) $a $b)
a # b ((mnotequal simp) $a $b)µ ´

2.3.3 割当の内部表現

次に,Maximaで行う割当にも同様の内部表現があります.

Maximaの内部表現：割当¶ ³
Maximaの式 内部表現

a:b ((msetq simp) $a $b)
a::b ((mset simp) $a $b)
a(x):=f ((mdefine simp) (($a) $x) $f)µ ´
最初の,a:bは内部では ((msetq simp) $a $b)と表現されていますが, 余計な括弧と simpを外し

てしまえば,(msetq $a $b)となり, LISPの (setq a b)に対応する事が分りますね.
函数定義を行う演算子:=も LISPの (defun a(x) f)に似た並びで内部表現されている事が判り

ます.

2.3.4 Maximaの函数の内部表現

MaximaXの内部表現：函数¶ ³
Maximaの式 内部表現

a(x) (($a simp) $x)
sin(x) ((%sin simp) $x)
diff(y,x) (($diff simp) $y $x 1)
diff(y,x,2,z,1) (($diff simp) $y $x 2 $z 1)
‘diff(y.x) ((%derivative simp) $y $x 1)
integrate(a,b,c,d) (($integrate simp) $a $b $c $d)
‘integrate(a,b,c,d) ((%integrate simp) $a $b $c $d)µ ´

函数名には基本的に$が先頭に付きますが,単引用符’が先頭に付いた場合は%で置換えられます.
この%が先頭に付いた函数はMaximaで自動的に解釈される事はありません.
微分を行う diff函数の例に示す様に一階の微分を行う場合,階数の 1を省略しても構いませんが

内部的には補完されている事が分りますね.

(%i93) ?car(’diff(x,y));

(%o93) (derivative, simp)

(%i94) ?cdr(’diff(x,y));

(%o94) (x, y, 1)

2.3.5 配列とリストの内部表現

Maximaの内部表現：函数¶ ³
Maximaの式 内部表現

a[1,2] (($a simp array) 1 2)
a[1,2](x) ((mqapply simp) (($a simp array) 1 2) $x)
[a,b, c] ((mlist simp) $a $b $c)µ ´

Maxima内部の配列は,多少勝手が違います.内部表現の第一成分は最初に$が付けられた配列名
があり,その後に毎度の simpと最後の arrayでデータが配列である事を示しています.

Maximaのリストも LISPのリストとは別の構造で,他の演算子等と同様に通常のリストに (mlist
simp)が先頭に置かれ,変数には何時もの$が先頭に付加されています.

2.3.6 Maximaの制御文の内部表現

Maximaの内部表現：制御文¶ ³
if a then b ((mcond simp) $a $b t nil)
if a then b else c ((mcond simp) $a $b t $c)
for i:a thru b step c unless q do f(i) ((mdo simp) $i $a $c nil $b $q (($f simp) $i))
for 1:a next n unless q do f(i) ((mdo simp) $i $a nil $n nil $q (($f simp) $i))
for i in l do f(i) ((mdoin simp) $i $l nil nil nil nil (($f simp) $i))
block([l1,l2],s1,s2) ((mprog simp) ((mlist simp) $l1 $l2) $s1 $s2)
block(s1,s2) ((mprog simp) $s1 $s2)µ ´
この表に示す様に,Maximaの制御文も内部表現では LISPの S式となっています. Maximaは結

局 S式で全てが記述されており,その S式を裏の LISPが解釈して結果を表に出しているのが実状
です.
以上からも,この内部表現を利用して新規にMaximaの制御文を LISPで構築する事が可能な事

が分るかと思います.

2.3.7 表示式と内部表現

入力式とその式の内部表現は,入力式の演算子が全て内部表現では前置式に変換されています.そ
れ以外にも細かな書式の違いがあります.
この内部表現への変換で変更される式を具体的に示しておきましょう.

式の表現の違い¶ ³
入力 part inpart
a − b; a − b a + (−1)b
a/b; a

b a b−1

sqrt(x); sqrt(x) x1/2

x ∗ 4/3; 4x
3

4
3x

exp(x) %ex %exµ ´
先ず,入力列がMaximaに入力する式で,part列がMaximaで表示される式,最後の inpart列が

内部表現に対応する式となっています.その為,これらの式に対して, 後述の part函数と inpart函
数では結果が異なる事になります.
尚,Maximaには内部表現を part列で示すMaximaで表示される式に適合する様に式を変換する

函数として dispform函数があります.

dispform函数¶ ³
dispform(〈式 〉 )
dispform(〈式 〉 ,all)µ ´

dispform函数は 〈式 〉 の内部表現から構築される式から式に含まれる演算から構成される式の
前置式表現 (外部表現と呼んでいます)に変換します.具体的にはMaximaは入力された式を内部
表現に変換する際に,x-yを x+(-y), x/yを x*yˆ(-1)等に変換します.dispformは内部表現が表現す
る式に含まれる全ての演算子から,Maximaに用意された演算子を用いて,式に一対一に対応する前
置式表現に作り直します.

(%i23) exp1:x-y;

(%o23) x - y

(%i24) exp2:dispform(exp1);

(%o24) x - y

(%i25) :lisp $exp1

((MPLUS SIMP) $X ((MTIMES SIMP) -1 $Y))

(%i25) :lisp $exp2

((MPLUS SIMP) ((MMINUS) $Y) $X)

(%i25) exp1:x/y;

x

(%o25) -

y

(%i26) exp2:dispform(exp1);

x

(%o26) -

y

(%i27) :lisp $exp1;

((MTIMES SIMP) $X ((MEXPT SIMP) $Y -1))

(%i27) :lisp $exp2;

((MQUOTIENT SIMP) $X $Y)

この様に,内部表現で自動的に変換されていた演算子が内部表現で表現される本来の演算子に置
換えられています.この dispform函数は与えられた CRE表現もこの様な外部表現に変換します.
但し,dispform(〈式 〉)では,式に含まれた函数の引数は内部形式のままです.この様に,一切の例

外を認めずに式全てを外部表現に変換する為, allオプションを用いて,dispform(〈式 〉 ,all)で変換
を行います.

(%i40) expr1:f(sqrt(y/x));

y

(%o40) f(sqrt(-))

x

(%i41) expr2:dispform(expr1);

y

(%o41) f(sqrt(-))

x

(%i42) expr3:dispform(expr1,all);

y

(%o42) f(sqrt(-))

x

(%i43) :lisp $expr1

(($F SIMP)

((MEXPT SIMP) ((MTIMES SIMP) ((MEXPT SIMP) $X -1) $Y) ((RAT SIMP) 1 2)))

(%i43) :lisp $expr2

(($F SIMP)

((MEXPT SIMP) ((MTIMES SIMP) ((MEXPT SIMP) $X -1) $Y) ((RAT SIMP) 1 2)))

(%i43) :lisp $expr3

(($F SIMP) ((%SQRT) ((MQUOTIENT) $Y $X)))

この例では,f
(√

y
x

)
を内部表現,dispformを作用させたもの,dispformに allオプション付きで作

用させたものの結果を示しています. 表示は全て同じものですが,内部表現では,allオプションを付
けたもの以外は全て同じで,allオプションを付けたものは,式全体が外部表現になっています.

2.3.8 変数

Maxima で利用可能な変数名は, 基本的にアルファベットの大文字や小文字, 幾つかの記号と
alphabetic属性を与えた文字です. 0から 9の数値の様に変数の先頭以外ならば利用可能な文字も
あります.この変数に関しては,逆アルファベット順を基礎とした変数順序 >m が組込まれていま

す. 詳細は,1.3節を参照して下さい.
Maximaには式に指定した変数が含まれているかどうかを判別する述語函数 freeof があります.

変数の述語函数¶ ³
freeof(〈変数1〉 ,· · ·,〈変数n〉 , 〈式 〉 ) 〈変数i〉が 〈式 〉に存在しない場合
lfreeof([〈変数1〉 ,· · ·,〈変数n〉], 〈式 〉 ) 〈変数i〉が 〈式 〉に存在しない場合µ ´
この freeof函数は 〈変数i〉 が 〈式 〉 の中に現われなければ trueを返し,それ以外は falseを返し

ます.ここで, 〈変数i〉 は変数アトム,添字付けられた名前,函数,或いは二重引用符””で括られた演
算子が扱えます.
尚,freeof函数は,sum函数や product函数の内部で利用される疑似変数に対して適用する事は出

来ません.疑似変数を 〈変数i〉 に指定した場合, 式に疑似変数が含まれていても,trueを返します.
lfreeof函数も動作は freeofと殆ど同じものです.但し,変数をリストで与える点と演算子が扱え

ない点で異なります.
次に,Maxima上で値が割当てられた変数は大域変数 valuesに登録されます.

Maximaに含まれる変数一覧¶ ³
大域変数名 初期値 概要

values [] 利用者設定のアトム変数の一覧µ ´
この大域変数 valuesには,Maximaの大域変数を除いて,演算子:,演算子::や函数的な束縛のある

変数名を含むリストが割当てられます.

(%i1) a:1;

(%o1) 1

(%i2) b::sin(a);

(%o2) sin(1)

(%i3) x;

(%o3) x

(%i4) values;

(%o4) [a, b]

この例では aと bには値を割当てていますが,単純に xは入力しただけです. すると,大域変数
valuesには値が束縛された変数 a,bが登録されています.
大域変数 valuesに登録された変数は,次の remvalue函数を用いてMaximaから削除する事が出

来ます.

変数値の削除を行う函数¶ ³
remvalue(〈変数1〉 ,〈変数2〉 ,· · ·)
remvalue (all)µ ´

remvalue函数は指定した大域変数 valuesの変数リストに登録された変数をMaxima から削除し
ます.変数は添字されていても構いません.

remvalue(all)で,全ての利用者変数が削除されます.

(%i10) b1:sin(y);

(%o10) sin(y)

(%i11) x:1;

(%o11) 1

(%i12) c1:b1*x;

(%o12) sin(y)

(%i13) values;

(%o13) [b1, c1, x]

(%i14) remvalue(x);

(%o14) [x]

(%i15) c1;

(%o15) sin(y)

(%i16) values;

(%o16) [b1, c1]

(%i17) remvalue(all);

(%o17) [b1, c1]

(%i18) values;

(%o18) []

この例では,変数 b1,x,c1に値を割当て,最初に変数 xを削除し,最後に allで全ての変数を削除し
ています.変数 c1の割当では変数 xを利用していますが,入力直後に評価された値が変数 c1に割
当てられている為,変数 xを削除しても問題はありません.

Maximaには式中の変数を取出したりする函数が幾つか用意されています.
式中の変数を扱う函数¶ ³

args(〈式 〉 )
listofvars(〈式 〉 )µ ´

args函数は 〈式 〉 が函数の場合,函数の引数を返します. 但し,多項式の CRE表現から変数を取
出す showratvars函数とは異なり,内部表現の先頭の成分をMaximaのリストに変換したもの,具
体的には, substpart(”[”,〈式 〉,0) と同じ結果となります. その為,複雑な式になると部分式のリス
トが返される事になります.

(%i21) args(sin(x));

(%o21) [x]

(%i22) args(sin(x+y));

(%o22) [y + x]

(%i23) expand((sin(x)+y)^2);

2 2

(%o23) y + 2 sin(x) y + sin (x)

(%i24) args(%);

2 2

(%o24) [y , 2 sin(x) y, sin (x)]

尚,args函数と substpart函数の両方は大域変数 inflag の設定に依存します.
listofvars函数は 〈式 〉 の変数リストを生成します.ここで, 大域変数 listconstvarsが trueであれ

ば,〈式 〉 にMaximaの数学定数%e,%pi,%iや定数として宣言した変数で含まれているものがあれ
ば, listofvars函数が返すリストに,これらの定数が含められます.但し,デフォルトの falseの場合,
定数を除外したリストを返却します.

listofvarsの動作を制御する大域変数¶ ³
変数名 初期値 概要

listconstvars false Maximaの定数を与式の変数リストに加えるかどうかを制御
listdummyvars true 疑似変数を与式の変数リストに加えるかどうかを制御µ ´
大域変数 listconstvarsが trueであれば,定数として宣言した変数とMaximaの数学定数%e,%pi,%i

が式に含まれていると,listofvars函数はこれらの定数も変数として加えたリストを返します.
大域変数 listconstvarsが falseの場合,数学定数と定数として宣言された変数は除外され,listvars

が返すリストには含まれません.

(%i6) listofvars(x^2*y+aa+%e);

(%o6) [x, y]

(%i7) listconstvars:true;

(%o7) true

(%i8) listofvars(x^2*y+aa+%e);

(%o8) [%e, aa, x, y]

大域変数 listdummyvarsが falseであれば,式の疑似変数は listofvars函数が出力するリストの中
に含まれません.尚,疑似変数は sum函数や product函数等の添字や極限変数や定積分変として利
用される変数です.

2.3.9 部分式に分解する函数

Maximaには与えられた式を部分式に分解する函数があります.先ず,isolate函数と disolate函数
は指定した変数を含む式と含まない式に分解して表示します.

指定した変数を分離して式を表示する函数¶ ³
isolate(〈式 〉 ,〈変数 〉 )
disolate(〈式 〉 〈変数1〉 , · · ·,〈変数n〉 )µ ´

isolate函数は 〈式 〉 から 〈変数 〉 との積を持つ部分式と持たない部分式に分けて表示します.〈変
数 〉 を持たない部分式は中間ラベルで置換えられ,式全体は中間ラベルと 〈変数 〉 の項の和で表現
されます.尚,isolate函数は式の展開を行いません.単純に,与えられた式を指定された変数を持つ
項と持たない項に分けて表示するだけの函数です.
尚,大域変数 isolate wrt timesが falseの場合,〈式 〉 は 〈変数 〉 との積を持たない部分式と 〈変

数 〉 との積を持つ部分式に分解して表示します.
大域変数 isolate wrt timesが trueの場合,isolateは更に 〈式 〉 を分解し,項も 〈変数 〉 の羃とそ

れ以外の変数との積に分解して表示します.

(%i12) isolate_wrt_times:false;

(%o12) false

(%i13) exp1:expand((1+a+x)^2);

2 2

(%o13) x + 2 a x + 2 x + a + 2 a + 1

(%i14) isolate(exp1,x);

2

(%t14) a + 2 a + 1

2

(%o14) x + 2 a x + 2 x + %t14

(%i15) isolate_wrt_times:true;

(%o15) true

(%i16) isolate(exp1,x);

(%t16) 2 a

2

(%o16) x + %t16 x + 2 x + %t14

(%i17) isolate((1+a+x)^2,x);

(%t17) a + 1



116 第 2章 Maximaのデータとその操作

2

(%o17) (x + %t17)

disolate函数は isolate(〈式 〉 ,〈変数 〉)に似ていますが,こちらでは利用者が一つ以上の変数を同
時に孤立させて表示する事が出来ます

isolate函数に影響を与える大域変数¶ ³
変数名 初期値 概要

exptisolate false 指数項を範囲に含めるかを決定

isolate wrt times false 表示を制御µ ´
大域変数 exptisolateが trueであれば, isolate(〈式 〉 ,〈変数 〉)の実行で 〈変数 〉 に含まれる (%e

の様な)アトムの指数項に対しても調べます.
大域変数 isolate wrt timesが falseの場合,isolate函数は指定した変数を含まない項と含む項に

分けて表示を行います.
trueの場合,更に式を分解し,指定した変数を含む項も,指定した積を除く項と指定した変数の項

の積に分解して表示を行います.

(%i17) eq1:expand((a+b+x)^2);

2 2 2

(%o17) x + 2 b x + 2 a x + b + 2 a b + a

(%i18) isolate_wrt_times;

(%o18) false

(%i19) exp1:expand((a+b+x)^2);

2 2 2

(%o19) x + 2 b x + 2 a x + b + 2 a b + a

(%i20) isolate_wrt_times;

(%o20) false

(%i21) isolate(exp1,x);

2 2

(%t21) b + 2 a b + a

2

(%o21) x + 2 b x + 2 a x + %t21

(%i22) isolate_wrt_times:true;

(%o22) true

(%i23) isolate(exp1,x);

(%t23) 2 a

(%t24) 2 b

2

(%o24) x + %t24 x + %t23 x + %t21

isolate函数では分離して表示するだけでしたが,指定した変数を含む部分と含まない部分に分解
する函数 partitionと純虚数を含む項と含まない項に分解する函数 rectformがあります.

部分式に分解する函数¶ ³
partition(〈式 〉 ,〈変数 〉 )
rectform(〈式 〉 )µ ´

partition函数は与えられた 〈式 〉 を分解し,二つの部分式を成分とするリストを返します.これ
らの部分式は 〈式 〉 の第一層に属するもので,第一成分が 〈変数 〉 を含まない部分式,第二成分が
〈変数 〉 を含む部分式となります.

(%i89) part(x+1,0);

(%o89) +

(%i90) partition(x+1,x);

(%o90) [1, x]

(%i91) part((x+1)*y,0);

(%o91) *

(%i92) partition((x+1)*y,x);

(%o92) [y, x + 1]

(%i93) part([x+1,y],0);

(%o93) [

(%i94) partition([x+1,y],x);

(%o94) [[y], [x + 1]]

rectform函数は与式を a+b*%iの形式で返します.〈式 〉 が複素数の場合,aと bは実数になりま
すが,そうでない場合,%iを持たない部分と%i で括られた式に分解し,内部の項順序に従って出力
される為,正確に a+b*%iの形式にはなりません.

(%i12) rectform((x+%i)^3);

3 2

(%o12) x + %i (3 x - 1) - 3 x

(%i13) rectform((10+%i)^3);

(%o13) 299 %i + 970

2.3.10 部分式を扱う函数

Maximaの式には函数や演算子を節とする木構造表現が表にあり,それを内部で表現する内部表
現の二つがあります.式の木構造では,演算子や函数の情報は比較的分かり易いものですが,内部表
現は式の前置式表現を基にしたものの為, 判り難いものとなっています.Maximaの式を操作する函
数には,式の木構造を基に式を操作する函数と,式の内部表現を直接操作する函数の二種類があり
ます.
先ず,部分式を取出す函数には,内部表現を利用する inpart函数,木構造表現を用いる part函数

の二種類があります.
部分式を取出す函数¶ ³

inpart(〈式 〉 ,〈整数1〉 , · · ·,〈整数k〉 )
inpart(〈式 〉 , [〈整数1〉 ,· · ·,〈整数k〉] )
part(〈式 〉 ,〈整数1〉 , · · ·,〈整数k〉 )
part(〈式 〉 , [〈整数1〉 ,· · ·,〈整数k〉] )
pickapart(〈式 〉 ,〈整数 〉 )µ ´

inpart函数は 〈式 〉 の内部表現に直接作用する函数で, 〈整数1〉 ,· · ·,〈整数k〉 で指定された 〈式 〉
の部分式を取出す函数です.この inpart函数は,式の内部表現に直接作用するので,その分,処理が
速くなります.

part函数は inpart函数に似ていますが, 〈式 〉 の木構造表現に対応する部分式を取出します.
部分式の取出し方は,最初に 〈式 〉 から 〈整数1〉 で指定される部分式を取出します.次に,取出し

た部分式から 〈整数2〉 で指定される成分を取出します.以降同様に 〈整数k−1〉 で指定された部分式
から 〈整数k〉 で指定される成分を取出して,この部分式を結果として返します.

(%i15) part((x+1)^3+2,1);

3

(%o15) (x + 1)

(%i16) part((x+1)^3+2,1,1);

(%o16) x + 1

(%i17) part((x+1)^3+2,1,1,1);

(%o17) x

尚,[〈整数1〉, · · · , 〈整数n〉]の様にリストで指定する事も可能ですが,この場合は,第一層の 〈整数1〉
で指定された部分式,以降, 〈整数n〉で指定される部分式が取出され,これらの部分式に part(〈式 〉,0)
で得られる主演算子を作用させた式が返されます.

(%i72) expr:x+y+sin(x^2+2*x+1)+cos(z/w);

z 2

(%o72) cos(-) + y + sin(x + 2 x + 1) + x

w

(%i73) inpart(expr,[2,4]);

z 2

(%o73) cos(-) + sin(x + 2 x + 1)

w

(%i74) part(expr,[1,4]);

z

(%o74) cos(-) + x

w

(%i75) expr2:x*y*z*sin(x^2+1);

2

(%o75) x sin(x + 1) y z

(%i76) inpart(expr2,[1,4]);

(%o76) x z

(%i77) part(expr2,[1,2]);

2

(%o77) x sin(x + 1)

(%i78) inpart(expr,0);

(%o78) +

(%i79) inpart(expr2,0);

(%o79) *

この例で示す様に,リストで指定した場合には部分式を抜き出して主演算子を作用させたものが
返されています.その為,和や積,一つだけの被演算子を取る演算子 (unary)としての演算子-,差と
商を扱う場合,部分式の順序に注意が必要になります.

part函数に関連する大域変数¶ ³
変数名 初期値 概要

piece inpart/part函数で取出した部分式を一時保存
partswitch false inpart/part函数のエラーメッセージを制御µ ´
大域変数 pieceには inpart函数や part函数を用いて取出した最新の部分式が保存されます.
大域変数 partswitchが trueに設定していると,式に指定した成分が存在しない場合に inpart函

数と part函数は endを返します.falseの場合は,エラーメッセージを返します.
pickapart函数は 〈整数 〉 で指定された式の階層に含まれる全ての部分式を%tラベルに割当て,

ラベルを用いた式に 〈式 〉 を変換します. 階層指定は part函数と同様で,表示された形式に対して
指定を行います. pickapart函数は大きな式を扱う際に,part函数を使わずに部分式に変数を自動的
に割当てる事にも使えます.

(%i49) exp:(x+1)^3;

3

(%o49) (x + 1)

(%i50) pickapart(exp,1);

3

(%o50) %t48

(%i51) exp2:expand((x+1)^3);

3 2

(%o51) x + 3 x + 3 x + 1

(%i52) pickapart(exp2,1);

3

(%t52) x

2

(%t53) 3 x

(%t54) 3 x

(%o54) %t54 + %t53 + %t52 + 1

2.3.11 総和と積 ∑
と

∏¶ ³
product(〈式 〉 ,〈添字変数 〉 , 〈下限 〉 ,〈上限 〉 )
sum(〈式 〉 ,〈添字変数 〉 , 〈下限 〉 ,〈上限 〉 )µ ´

product函数は 〈添字変数 〉 の 〈下限 〉 から 〈上限 〉 迄の 〈式 〉 の値の積を与えます. 〈上限 〉 が
〈下限 〉 より小になると空の積となり, この場合,product函数はエラー出力ではなく 1を返します.
評価は sum函数と似ていますが,総和函数 sumとは異なる点は,productの簡易化は無い事です.

product函数を制御する大域変数¶ ³
変数名 初期値 概要

prodhack false product函数の簡易化を制御µ ´
大域変数 prodhackが trueであれば,product(f(i),i,3,1)は 1/f(2)となります. これは a > bの場

合,product(f(i),i,a,b) = 1/product(f(i),i,b+1,a-1)とする為です.尚,prodhackがデフォルトの false



2.3. 式について 121

で,a < bの場合, product(f(i),i,a,b)はエラーになります.

(%i12) prodhack:true;

(%o12) true

(%i13) product(f(x),x,5,1);

1

(%o13) --------------

f(2) f(3) f(4)

sum函数は 〈添字変数 〉 を 〈下限 〉 から 〈上限 〉 迄の値の 〈式 〉 の和を取ります. 〈上限 〉 と 〈下
限 〉 が整数で異なっていれば, 和の各々の項は評価されて互いに加えられます.

sum函数を制御する大域変数¶ ³
変数名 初期値 概要

simpsum false sum函数の簡易化を制御
sumhack false 上限と下限の入換えの制御

sumexpand false 総和の積の処理を制御

cauchysum fale Cauchy積の適用を制御
genindex i sumや productで利用される疑似変数名を設定
gensumnum false 疑似変数の番号µ ´
大域変数 simpsumが trueの場合,sum函数を自動的に簡易化します.

(%i33) simpsum;

(%o33) false

(%i34) sum(x^n,n,0,m);

m

====

\ n

(%o34) > x

/

====

n = 0

(%i35) simpsum:true;

(%o35) true

(%i36) sum(x^n,n,0,m);

m + 1

x - 1

(%o36) ----------

x - 1

(%i37) sum(x^n,n,0,inf);

Is abs(x) - 1 positive, negative, or zero?

pos;

(%o37) inf

(%i38) sum(x^n,n,0,inf);

Is abs(x) - 1 positive, negative, or zero?

neg;

1

(%o38) -----

1 - x

この様に simpsum が false の場合,sum(xˆn,n,0,m) の簡易化は実行されませんが, simpsum が
trueとなると,自動的に簡易化が実行されます.sum(xˆn,n,0,inf)の場合, xの絶対値から 1を引い
たものが正か負か零であるかを利用者が指定する事で簡易化が行えます.
大域変数 sumhack が true の場合,sum(f(x),x,a,b) の a,b が a > b の場合, sum(f(x),x,a,b) を-

sum(f(x),x,b-1,a+1)で置換えます.但し,sumhackが falseの場合はエラーになります.

(%i12) sumhack:true$

(%i13) sum(f(i),i,3,1);

(%o13) -f(2)

(%i14) sum(f(i),i,5,1);

(%o14) -f(4)-f(3)-f(2)

(%i15) sumhack:false;

(%o15) false

(%i16) sum(f(i),i,5,1);

Lower bound to sum: 5

is greater than the upper bound: 1

-- an error. Quitting. To debug this try debugmode(true);

大域変数 sumexpandが trueの場合,sum函数同士の積を纏めて sum函数の入れ子にします.

(%i1) sum(f(x),x,0,m)*sum(g(x),x,0,n);

m n

==== ====

\ \

(%o1) ( > f(x)) > g(x)

/ /

==== ====

x = 0 x = 0

(%i2) sumexpand:true;

(%o2) true

(%i3) sum(f(x),x,0,m)*sum(g(x),x,0,n);

m n

==== ====

\ \

(%o3) > > f(i1) g(i2)

/ /

==== ====

i1 = 0 i2 = 0

但し,この処理は infやminfを含む総和に対して利用する事は危険です.
大域変数 cauchysumは大域変数 sumexpandと対で用います.大域変数 sumexpandと cauchysum

の両方が trueの場合,総和の掛算を纏める際に,通常の積ではなく Cauchy積が利用されます.

(%i1) sumexpand:true$

(%i2) cauchysum:true$

(%i3) sum(f(x),x,0,m)*sum(g(x),x,0,n);

m n

==== ====

\ \

(%o3) > > f(i1) g(i2)

/ /

==== ====

i1 = 0 i2 = 0

(%i4) sum(f(x),x,0,inf)*sum(g(x),x,0,n);

inf n

==== ====

\ \

(%o4) > > f(i3) g(i4)

/ /

==== ====

i3 = 0 i4 = 0

(%i5) sum(f(x),x,0,inf)*sum(g(x),x,0,inf);

inf i5

==== ====

\ \

(%o5) > > g(i5 - i6) f(i6)

/ /

==== ====

i5 = 0 i6 = 0

大域変数 genindexと大域変数 gensumnumは sum函数内部の疑似変数を生成する為に用いられ
る大域変数です.大域変数 genindexには疑似変数のアルファベットが, 大域変数 gensumnumには
疑似変数の番号が各々設定されています.ここで, 大域変数 gensumnumは sum函数内部で疑似変
数を生成する度に,一つづつ増加します.

(%i1) sumexpand:true$

(%i2) gensumnum;

(%o2) 0

(%i3) sum(f(x),x,0,m)*sum(g(x),x,0,n);

m n

==== ====

\ \

(%o3) > > f(i1) g(i2)

/ /

==== ====

i1 = 0 i2 = 0

(%i4) gensumnum;

(%o4) 2

この例では,大域変数 sumexpandを trueにした為,総和の積が纏められてしまい, その結果,二
つの疑似変数 i1と i2が新たに生成されています.この時,gensumnumは最初が 0で,それから二つ
の疑似変数を生成した為,2になっています.

2.3.12 式の最適化

式の最適化を行う函数¶ ³
optimize(〈式 〉 )µ ´

optimize函数は 〈式 〉 をより効率的に計算出来るMaximaの式を返却します.optimizeは式の展
開を行うものではありませんが,式に含まれる共通部分式を内部変数で置換えて,効率的に計算出
来る様な式に変換します. この際に,block文が用いられます.但し,共通部分式が無い場合は, 〈式 〉
をそのまま返却します.

(%i40) optimize((x+1)^3+1/(x+1)^2+exp((x+1)^2));

2 %2 3 1

(%o40) block([%1, %2], %1 : x + 1, %2 : %1 , %e + %1 + --)

%2

(%i41) ans1:solve( x^4+x^3+3*x-1=0,x);

sqrt(5) sqrt(25 - sqrt(5)) 1

(%o41) [x = - ------- - ------------------ - -,

4 1/4 4

2 sqrt(2) 5

sqrt(5) sqrt(25 - sqrt(5)) 1

x = - ------- + ------------------ - -,

4 1/4 4

2 sqrt(2) 5

sqrt(sqrt(5) + 25) %i sqrt(5) 1

x = - --------------------- + ------- - -,

1/4 4 4

2 sqrt(2) 5

sqrt(sqrt(5) + 25) %i sqrt(5) 1

x = --------------------- + ------- - -]

1/4 4 4

2 sqrt(2) 5

(%i42) optimize(ans1);

1 1

(%o42) block([%1, %2, %3, %4, %5, %6, %7], %1 : -------, %2 : ----,

sqrt(2) 1/4

5

%3 %3

%3 : sqrt(5), %4 : sqrt(25 - %3), %5 : - --, %6 : --, %7 : sqrt(%3 + 25),

4 4

%1 %2 %4 1 %1 %2 %4 1 %1 %2 %7 %i 1

[x = %5 - -------- - -, x = %5 + -------- - -, x = - ----------- + %6 - -,

2 4 2 4 2 4

%1 %2 %7 %i 1

x = ----------- + %6 - -])

2 4

optimize函数に影響する大域変数¶ ³
変数名 初期値 概要

otimprefix % optimize函数で生成される記号で利用µ ´
大域変数 otimprefix に設定される文字は optimize 函数で生成された記号に使用される前置詞

です.

2.4 代入操作

多項式の計算で,方程式を求めた結果を早速,式に代入したい事があります. この場合,規則によ
る代入や,直接変数に対して x:aの様に割当てを行う事もありますが,これらとは別に subst等の代
入用の函数を用いる方法があります.
この節では最初に通常の代入函数について解説します.ここで通常とは,Maxima での式の表現に

注意を払わなくても済む函数の事です.例えば,式中の変数 xに 2 を代入する様な函数です.Maxima
には式の表現から部分式や演算子を指定して入れ換える函数があります.この函数は与式から部分
式を取出す part函数や inpart函数 に対応するものです.

2.4.1 通常の代入函数

subst函数¶ ³
subst(〈式1〉, 〈式2〉, 〈式3〉)µ ´

subst函数は 〈式3〉 中の 〈式2〉 を 〈式1〉 で置換します.
〈式1〉 と 〈式2〉 は二重引用符”で括られた式の演算子や関数名,或いは,〈式2〉 は アトムや 〈式3〉

に完全に含まれる部分式でなければなりません.
例えば,x+y + z は 2∗ (x+y + z)∗w に完全に含まれる部分式になりますが, x+y は部分式にな

りません.これは式の木構造を考えると明瞭になります. 部分式は,式の木構造を考えた時に,各階
層が構成する式を取出したものになるからです.ここで,〈式2〉 が 〈式3〉 の部分式でない場合,subst
ではなく,式の階層を直接指定出来る substpart函数や ratsubst 函数を使いましょう.

subst函数では, 〈式2〉 が 式a/式b の様に割算を伴う場合,subst(〈式1〉 ∗ 〈式b〉, 〈式a〉, 〈式3〉)が使
えます.又, 〈式2〉 が 〈式a〉1/〈 式b〉 の形式であれば, subst(〈式1〉ˆ 〈式b〉, 〈式a〉, 〈式3〉) が使えます.

substに影響を与える大域変数¶ ³
変数名 初期値 概要

exptsust false 指数函数の代入操作を制御

opsubst true 演算子に代入する事を抑制µ ´
大域変数 exptsubstが trueであれば%eˆ(a*x)の%eˆxを yで置換える操作が可能になります.
大域変数 opsubstが falseであれば,subst函数は式に含まれる演算子に対して代入を行いません.
例えば, (opsust:true,subst(xˆ2,r,r+r[0])) と (opsubst:false,subst(xˆ2,r,r+r[0])) を実行すると,大

域変数 opsusbtが trueであれば,subst函数は与式 r+r[0]の全ての rに xˆ2を代入しますが,大域変
数 opsubstを falseにすると左側の rのみに代入操作が行われ,r[0]の rには xˆ2が代入されません.

(%i63) (opsubst:true,subst(x^2,r,r+r[0]));

2 2

(%o63) x + (x )

0

(%i64) (opsubst:false,subst(x^2,r,r+r[0]));

2

(%o64) x + r

0

ratsubst函数¶ ³
ratsubst (〈式1〉, 〈式2〉, 〈式3〉)µ ´

ratsubst函数は langle式3〉 に含まれる 〈式2〉 に 〈式1〉 を代入します. 〈式2〉 は和,積,羃等の演
算子でも構いません.subst函数が代入を行う個所で ratsubst函数は式が何を意味するかを判って
います.その為,subst(a,x+y,x+y+z)は x+y+zを返しますが, ratsubst函数は z+aを返します.

ratsubstに影響を与える大域変数¶ ³
変数名 初期値 概要

radsubstflag false 羃乗の扱いを制御µ ´
大域変数 radsubstflag(ratsubstflagではない事に注意)が trueの場合,ratsubst函数を使って項の

sqrtや羃で式の入れ換えが可能となります.
例えば,aを xˆ(1/3)とした場合,xは aˆ3に等しくなりますが, この様に,式 xの xˆ(1/3)を aで

置換える事は通常出来ません. 大域変数 radsubstflagが trueの場合,この様な代入が行えます.

(%i5) radsubstflag:true;

(%o5) true

(%i6) ratsubst(a,x^(1/3),x);

3

(%o6) a

fullratsubst函数¶ ³
fullratsubst(〈式1〉, 〈式2〉, 〈式3〉 )µ ´

fullratsubst函数は ratsubst函数と同じですが,結果が変化しなくなる迄,自分自身を再帰的に呼
出します.この函数は,式の置き換えや置き換えられた式が一つ又はそれ以上の変数を共通に持つ
場合に便利です.fullratsubst函数は lratsubst函数と同じ引数の書き方が出来ます.最初の引数は単
一の代入方程式かその様な方程式のリストで,第二の引数は仮定された式となります.

lratsubst函数¶ ³
lratsubst(〈リスト 〉, 〈多項式 〉 )µ ´

lratsubst函数は subst(〈方程式のリスト 〉, 〈式 〉 )に似ていますが,rstsubst函数が subst函数の
代りに使われる点で異なります. lratsubst函数の最初の因子は方程式か方程式のリストで,subst函

数から得られる書式と同一のものでなければなりません.代入は方程式のリストの左から右の順で
処理します.

(%i1) load ("lrats")$

(%i2) subst ([a = b, c = d], a + c);

(%o2) d + b

(%i3) lratsubst([a^2=b,b=c^2,c^3=d], a^2+b+c^3);

2

(%o3) d + 2 c

(%i4) subst([b=c^2,a-2=b,c^3=d], a^2+b+c^3);

2 2

(%o4) d + c + a

(%i4) lratsubst([b=c^2,a-2=b,c^3=d], a^2+b+c^3);

2 2

(%o4) d + c + b + 4 b + 4

sublis函数¶ ³
sublis(〈リスト 〉, 〈式 〉 )µ ´

sublis函数は 〈式 〉に 〈リスト 〉で指定した複数の代入を並行して行ないます.〈リスト 〉には,a=b
の形で式を記述します.演算子=の左辺の aが 〈式 〉に含まれるアトムや函数名を指定し, 右辺の b
に置換える値や式を設定します.

(%i23) sublis([sin=cos,x=2*theta+1],sin(x-1)^2);

2

(%o23) cos (2 theta)

(%i24) sublis([sin=cos,cos=sin],cos(x)^2+sin(x+1)^3);

3 2

(%o24) cos (x + 1) + sin (x)

尚,sublis([sin=cos,cos=sin],cos(x)^2+sin(x+1)^3)の様な入れ換えの指定では,〈リスト 〉
に含まれる式の代入を順番に行うのではなく同時に行う為,cosと sinが入れ換えられている事に注
意して下さい.
大域変数 sublis apply lambda が sublis函数を実行した後の簡易化を制御します.

2.4.2 substpart函数と substinpart函数

substpart 函数と substinpart 函数は与式の部分式を取出す part 函数と函数に各々対応する代
入函数です.part 函数は式の木構造に対して部分式を取出す函数で, inpart 函数は Maxima の式

の内部表現に対して部分式を取出す函数となっています. substpart函数と substinpart函数の対
応もこれと同様です.その為,大域変数 inflag を trueに設定して part/substpart函数を呼出す事
は,inpart/substinpart函数を呼出す事と同じ事になります.

substpartと substinpart函数¶ ³
substpart(〈式1〉, 〈式2〉, 〈正整数1〉, · · · , 〈正整数n〉)
substinpart(〈式1〉, 〈式2〉, 〈正整数 〉, · · ·)µ ´

substpart函数は 〈式2〉から part関数の様に 〈正整数1〉, · · · , 〈正整数n〉でで抜出した部分式に
〈式1〈を代入します. 〈式1〉 に演算子を入れる場合には,二重引用符で substpart(”+”,a*b,0)の様
に括る必要があります.
具体的な例で説明しましょう.式 1

x3+3x2+1 の成分の入れ換えを行いましょう.この式の構造は,図
2.1に示すものになります.

+

(0)

(2,0)

/

1
(1)

* (2,2,0)

3
(2,2,1)

(2,1,0)^

3
(2,1,2)

x
(2,1,1)

1
(2,3)

x
(2,2,2,1)

(2,2,2)^

2
(2,2,2,2)

図 2.1: 1
x3+3x2+1 の構造

substpart函数はこの構造に従って入れ換えを行います.その為,演算子や式で成分を置換える事
も可能です.

(%i7) 1/(x^3+3*x^2+1);

1

(%o7) -------------

3 2

x + 3 x + 1

(%i8) substpart(4,%,2,1,2);

1

(%o8) -------------

4 2

x + 3 x + 1

(%i9) substpart(1,%,2,2,2,2);

1

(%o9) ------------

4

x + 3 x + 1

(%i10) substpart(x,%,1);

x

(%o10) ------------

4

x + 3 x + 1

(%i11) substpart("^",%,0);

4

x + 3 x + 1

(%o11) x

(%i12) substpart(sin(x),%,1);

4

x + 3 x + 1

(%o12) sin(x)

(%i13) substpart(y,%,2);

y

(%o13) sin (x)

substinpart函数は substpart函数に似ていますが,substpart函数と違い, 式の内部表現に対して
作用します.

2.5 式の展開と簡易化

2.5.1 自動展開を行う大域変数

Maximaでは x+xの様な簡単な入力に対して,入力式をMaximaの内部表現に変換する際に式
の簡易化を実行しますが,式の展開迄は通常実行しません.ところが,大域変数の設定を変更する事
で,Maximaは与式が多項式や指数関数を含む場合に与式を自動的に展開したり,簡易化を行う様に
なります.

自動展開に関連する大域変数¶ ³
変数名 初期値 概要

demoivre false 指数函数の表示を制御

%emode true 指数関数の簡易化を制御

%enumer false 自然底の浮動小数点への自動変換

expon 0 自動展開される負の羃の次数を設定

expop 0 自動展開される正の羃の次数を設定µ ´
大域変数demoivreを trueで,与式%eˆ(a+b*%i)のbが実数の場合,与式は%eˆa*(cos(b)+%i*sin(b))

に自動的に展開されます.

(%i18) exp(a+b*%i);

%i b + a

(%o18) %e

(%i19) demoivre:true;

(%o19) true

(%i20) exp(a+b*%i);

a

(%o20) %e (%i sin(b) + cos(b))

大域変数%emodeが trueであれば,%eˆ(%pi*%i*x)が次の様に簡易化されます.

• xが整数,或いは 1/2,1/3,1/4や 1/6と整数の積であれば, cos(%pi*x)+%i*sin(%pi*x)とな
ります.

• その他の数値の場合,%eˆ(%pi*%i*y)となります. ここで yは x-2*k,kは abs(y) < 1 となる
整数です.

大域変数%emodeが falseの場合,%eˆ(%pi*%i*x)の特殊な簡易化は何も実行されません.

(%i25) %emode:true$

(%i26) exp(%pi*%i/2);

(%o26) %i

(%i27) %emode:false;

(%o27) false

(%i28) exp(%pi*%i/2);

%i %pi

------

2

(%o28) %e

大域変数%enumerが trueであれば,%eは 2.718 · · · に変換されます.%eˆx の指数が整数の場合
に限り,この変換が行なわれます.
大域変数 exponは expand函数とは別個に,Maximaが自動的に展開する式に含まれる負の羃の

次数を定めます.同様に,大域変数 expopは自動的に展開される正の最も高い次数を定めます.
大域変数 exponと expopは初期値が 0に設定されている為,(x+1)ˆ0の様に羃の次数が零であれ

ば自動的に 1に変換されます.大域変数 expopを例えば 4に変更すると, 羃の次数が 0以上,4以下
であればMaximaは羃を自動的に展開します.又,exponを 4にすると,負の羃の次数の絶対値が 0
以上,4以下であれば自動的に羃を展開します. 以下に簡単な例を示しましょう.

(%i38) expon:4;

(%o38) 4

(%i39) (x+1)^(-3);

1

(%o39) -------------------

3 2

x + 3 x + 3 x + 1

(%i40) (x+1)^(-5);

1

(%o40) --------

5

(x + 1)

(%i41) expop:4;

(%o41) 4

(%i42) (x+1)^4;

4 3 2

(%o42) x + 4 x + 6 x + 4 x + 1

(%i43) (x+1)^5;

5

(%o43) (x + 1)

2.5.2 簡易化に関連する函数

述語函数の unknown函数¶ ³
函数名 trueを返す条件
unknown(〈式 〉 ) 〈式 〉が演算子を一つ持ち, 簡易化函数が不明の場合µ ´

〈式 〉が一つの演算子を持ち,その簡易化函数が分らない場合は true を返します.

2.5.3 指数函数の展開に関連する函数

指数函数の表示¶ ³
demoivre(〈式 〉 )
exponentialize(〈式 〉 )µ ´

demoivre函数は大域変数 demoivreの設定や ev函数による式の再評価なしで変換を行います.
exponentialize函数は 〈式 〉を指数函数形式に変換します.

2.5.4 式の展開に関連する函数

expand函数¶ ³
expand(〈式 〉 ,〈p〉 ,〈n〉 )
expand(〈式 〉 )
expan(〈式 〉,p,n)
expandwrt(〈式 〉 ,〈変数1〉 ,· · ·, 〈変数n〉 )
expandwrt factored(〈式 〉 ,〈変数1〉 , · · ·,〈変数n〉 )µ ´

expand函数は和の積や指数函数内の和を展開し,有理式の分子を各々の項に分離し, 可換積と非
可換積の両方の積を 〈式 〉 の全ての階層で加法に対して分配します.
尚,多項式に対してはより効率的なアルゴリズムを用いる ratexpandを通常用いるべきです.
大域変数の maxnegexと maxposexはMaximaが展開する式の負と正の羃の次数の最大値を設

定します.
expand函数を制御する大域変数¶ ³

変数名 初期値 概要

maxnegex 1000 expandで展開される負の羃の次数
maxpogex 1000 expandで展開される正の羃の次数µ ´
大域変数maxnegexは expand函数で展開される絶対値が最大となる負の羃の次数です. 正の羃

の最大次数はmaxposexです.

大域変数maxposexは expand函数で展開される最大の正の羃の次数です.尚,負の羃の最大次数
はmaxnegexです.

expan函数は,expan(〈式 〉,p,n)の場合,pを大域変数 maxposex, nを大域変数 maxnegexに割当
てて 〈式 〉 の展開を行います.

expandwrt函数は 〈変数1〉 ,· · ·,〈変数n〉 に対し,〈式 〉 を展開します.〈変数i〉 を含む全ての積は明
示的に現れます.返される形式は 〈変数i〉 を持つ式の和の積を持たないものとなります.〈変数i〉 は
変数,演算子や式でも構いません.
デフォルトで分母は展開されませんが,大域変数の expandwrt denomでこれは制御が出来ます.

この函数を使う為には予め load(stopex); で読込を実行します.
expandwrt factored函数は expndwrtに似ていますが,幾分違った式の積を扱います.expand factored

は要求される展開を処理しますが, 引数リストの中の変数に含まれる 〈式 〉 の因子に対してのみ処
理を行います.予め, load(stopex) で読込を実行する必要があります.

演算子の分配に関連する函数¶ ³
distrib(〈式 〉)
multthru(〈式1〉 ,〈式2〉 )
multthru(〈式 〉 )µ ´

distrib函数は可換積演算子 ∗ に対し和 + を分配します.expandとの違いは,式の最上層のみで
作用する点です.又,multthruとも最上層の全ての和を展開する点でも異なります.

multthru函数は 〈式 〉 の部分展開を行います.即ち, 〈式 〉 が f1 ∗ f2 ∗ · · · ∗ fn の形式で,各因子
の中で, 羃乗でない 〈式 〉 中で最も左側の因子を fi とすると, 〈式 〉 の fi 以外の因子と fi の項と

の積の和に分解します. 例えば, (x + 1)2 ∗ (z + 1) ∗ (y + 1) の場合,最も左側の因子 y + 1 で式が展
開され, (x + 1)2 ∗ (y + 1) ∗ z + (x + 1)2 ∗ (y + 1) となります.

multthru(〈 式1〉 ,〈 式2〉 ) の場合, 〈 式2〉 の各項を 〈 式1〉 倍にします. つまり,multthru(〈 式1〉
*〈式2〉) と同値です.
尚,〈式2〉 には方程式を指定出来ます.この場合,演算子=の二つの被演算子に 〈式1〉 との積が返

されます.
尚,multthruは羃乗された和の展開は行いません.この函数は和に対する可換, 或いは,非可換積

の分配に関して最も速いものです.

(%i18) multthru((x+1)^2*(z+1));

2 2

(%o18) (x + 1) z + (x + 1)

(%i19) multthru((x+1)^2*(y+1)^2*(z+1)^2,z+1);

2 2 2 2 2 2

(%o19) (x + 1) (y + 1) z (z + 1) + (x + 1) (y + 1) (z + 1)

(%i20) multthru((x+1)^2*(y+1)^2*(z+1)^2,x+1);

2 2 2 2 2 2

(%o20) x (x + 1) (y + 1) (z + 1) + (x + 1) (y + 1) (z + 1)

(%i21) multthru((x+1)^2*(z+1)*(y+1));

2 2

(%o21) (x + 1) (y + 1) z + (x + 1) (y + 1)

(%i22) multthru((x+1)^2*(y+1)^2*(z+1)^2,x^2+1=0);

2 2 2 2 2 2 2

(%o22) x (x + 1) (y + 1) (z + 1) + (x + 1) (y + 1) (z + 1) = 0

distrib,multthru,expandの比較¶ ³
distrib((a+b)*(c+d)) ⇒ a*c + a*d + b*c + b*d
expand((a+b)*(c+d)) ⇒ a*c + a*d + b*c + b*d
multthru ((a+b)*(c+d)) ⇒ (a + b)*c + (a + b)*d
distrib (1/((a+b)*(c+d))) ⇒ 1/((a+b) * (c+d))
expand(1/((a+b)*(c+d)),1,0) ⇒ 1/(a*c + a*d + b*c + b*d)
multthru(1/((a+b)*(c+d)),1,0) ⇒ 1/((a+b)*(c+d)),1,0)µ ´

2.5.5 sum函数の簡易化に関連する函数

sumcontract¶ ³
sumcontract(〈式 〉 )µ ´

sumcontract函数は上限と下限の差が定数となる加法の全ての総和を結合します. 結果は,各総和
の集合に対して,全ての適切な外の項を加えて一つの総和にしたものを含む式になります.suncontract
は全ての互換な総和を結合し,可能であれば, 総和の一つから添字の一つを用います.sumcontract
を実行する前に intosum( 〈式 〉 )の実行が必要かもしれません.

(%i18) sum(1/n^2,n,1,m)+sum(1/n^3,n,1,m);

m m

==== ====

\ 1 \ 1

(%o18) > -- + > --

/ 2 / 3

==== n ==== n

n = 1 n = 1

(%i19) sumcontract(%);

m

====

\ 1 1

(%o19) > (-- + --)

/ 2 3

==== n n

n = 1

intosum函数¶ ³
intosum(〈式 〉 )µ ´
総和の乗法がなされる全ての物を取り,それらを総和の内部に置きます. 添字が式の外側で用い

られていれば,この函数は sumcontractに対して実行するのと同様に適切な添字を探そうとします.
これは本質的に総和の outative属性の観念の逆になりますが,この属性を取り除かずに素通りする
だけである事に注意して下さい.

intosum函数を用いる前に scanmap(multthru,〈式 〉 )を実行しなければならない場合もあります.

2.5.6 簡易化を行う函数

radcan函数¶ ³
radcan(〈式 〉 )µ ´

radcan函数の引数 〈式 〉は,対数函数,指数函数と羃乗根を含んでいても構いません. 〈式 〉 をあ
る変数順序に対する CRE表現に変換し,簡易化を行います. 特定の変数順序に対し,CRE表現は一
意に定まります (従って,CRE表現は式の正準表現になります).その為,radcanを用いた簡易化も一
意に定まります. 但し,radcanは時間を多く消費します.これは因子分解と指数の部分分数展開を基
本とした簡易化の為,式の成分の間の関係を探索する為です.

scsimp函数¶ ³
scsimp(〈式 〉 ,〈規則1〉 , · · ·,〈規則n〉 )µ ´

scsimp(=Sequential Comparative SIMPlification)は,式 (その最初の引数), 同一性や規則 (その他
の引数)の集合を取って簡易化を試みます. より小さな式が得られると,その処理が繰返されます.
そうでなければ, 全ての簡易化が試みられた後に,もとの式が返却されます.

2.5.7 簡易化に関する補助的函数

askinteger函数¶ ³
askinteger(〈式 〉 ,〈オプション引数 〉 )
askinteger(〈式 〉 )µ ´

〈式 〉 は任意の有効な Maximaの式で, 〈オプション引数 〉 は even(偶数),odd(奇数),integer(整
数)の何れか一つで,省略された場合は内部で integerが設定されます. この函数はMaximaに蓄え
られた情報から 〈式 〉 が even,odd, 或いは integerであるかを決定しようとします.Maximaに蓄え
られた情報では不十分な場合,利用者に質問して,Maximaに情報を蓄えます.

(%i16) aa:1;

(%o16) 1

(%i17) askinteger(aa);

(%o17) yes

(%i18) askinteger(aa,odd);

(%o18) yes

(%i19) askinteger(aa,even);

(%o19) no

(%i20) askinteger(yy);

Is yy an integer?

yes;

(%o20) yes

(%i21) askinteger(yy,odd);

Is yy an odd number?

no;

(%o21) no

(%i22) askinteger(yy,even);

(%o22) yes

(%i23) askinteger(zz,even);

Is zz an even number?

no;

(%o23) no

(%i24) askinteger(zz,odd);

Is zz an odd number?

no;

(%o24) no

(%i25) askinteger(zz,integer);

Is zz an integer?

yes;

(%o25) yes

(%i26) askinteger(zz+yy+aa,integer);

(%o26) yes

(%i27) askinteger(zz,integer);

(%o27) yes

(%i28) askinteger(zz,even);

Is zz an even number?

yes;

(%i29) askinteger(zz*2+aa,even);

(%o29) no

ここでの例で示す様に yyが integerであると指定すると,それから yyは integer となります.更
に,yyが oddであると宣言すれば,自動的に evenになります. 但し,zzが oddでなく,evenではない
と宣言しても,askinteger(zz)は no とはならずに尋ねて来ます.ここで,yesとすれば,それまで入力
した oddでも evenでもない事が消去されます.askintegerは zz+yy+aaや 2*zz+aaの様な式に対
しても,それ以前の入力情報から,整数,奇数か偶数であるかを判断します.

asksign函数¶ ³
asksign (〈式 〉 )µ ´

asksign函数は 〈式 〉 が,正,負,或いは零であるかを決定します. この際に,Maximaに蓄えられ
た情報をもとに決定しようとしますが, 情報が不十分で決定出来なければ,その演繹を完遂する為
に必要な質問を利用者に対して行います.
利用者の答はMaximaに記録されます.
asksignが尋ねる値は pos(正値),neg(負値),zero(零)の何れか一つです.

numerval函数¶ ³
numerval(〈変数1〉 ,〈式1〉 , · · ·,〈変数n〉 ,〈式n〉 )µ ´

numerval函数は 〈変数i〉 を 〈式i〉 の数値変数として宣言し,〈式i〉 は大域変数 numerが trueで
あれば, 任意の式に現われる変数に対して評価と代入が行われます.

2.6 リスト

2.6.1 Maximaのリスト

Maximaにはリストと呼ばれるデータ型があります.これは配列に似たデータ構造ですが,より
柔軟で視覚的にも把握し易い構造を持っており,多くの数式処理で採用されています.

Maximaのリストは [1, 2, 7, x+y]の様に,各成分をコンマ,で区切った列を大括弧 []で括った

ものです.この様に,LISPのリストの様に成分を単純に空行で区切るのではない事に注意をして下
さい.勿論,[1,2,[3,4],[4,[5]]]の様にリストを入れ子にしても構いません.
ここでMaximaは LISPで記述されたシステムの為,Maximaの式やプログラムも内部では LISP

の S式と呼ばれるリストで表現されています.更に,Maximaの演算子が前置式でなくても内部では
前置表現になっています.その為,Maxima内部表現では演算子の部分が 0,それ以降の成分に 1, 2, · · ·
と番号が振られた階層構造を持っています.
この事から一見してリスト処理専用の函数であっても,Maximaの殆どの式で利用可能な事が多

くあります.尚,Maximaの内部表現に関しては 2.3節を参照して下さい.

2.6.2 リスト処理に関連する大域変数

リストに関連する大域変数¶ ³
変数名 初期値 取り得る値 概要

listarith true [true,false] 算術演算子によるリスト評価を制御

inflag false [true,false] 内部表現への作用を制御µ ´
大域変数 listarithは trueであれば算術演算子によるリスト評価を実行します. 大域変数 inflag

は trueの場合,成分を取り出す函数は与えられた式の内部表現に対して処理を行います.簡易化は
式の再並び換えを行う事に注意が必要になります. その為,first(x+y)は inflagが trueであれば x
となりますが,大域変数 inflagが falseならば yとなります.
尚,大域変数 inflagに影響を受ける函数に以下のものがあります.

inflagの影響を受ける函数¶ ³
函数 概要

part 式の成分を取出す函数

substpart 式の成分の代入を行う函数

first リストの先頭を取出す函数

rest リストの残りを取出す函数

last リストの末尾を取出す函数

length リストの長さを返す函数

for-in構文 リストを用いるループ文

map リストに函数を作用させる函数

fullmap リストに函数を作用させる函数

maplist リストに函数を作用させる函数

reveal 式の置換えを行う函数

pickapart 成分を取出す函数µ ´
2.6.3 リスト処理に関連する主な函数

リストに関連する述語函数¶ ³
函数 trueを返す条件
atom(〈変数 〉 ) アトムの場合

listp(〈変数 〉 ) リストの場合

member(〈式1〉 ,〈式2〉 ) 〈式1〉が 〈式2〉に含まれている場合µ ´
〈変数 〉 がアトムかリストであるかは atom函数と listp函数で調べる事が出来ます.atom函数は

引数がアトムであれば true,それ以外は falseを返します. listp函数は引数がリストであれば true,
そうでなければ falseを返します.
尚,ここでの判定では内部表現は無関係です.その為,sin (x) や x+yの様な式では各変数に値が束

縛されていなければアトムでもリストにもなりません.
member函数は二つの引数を取り,〈式1〉 が 〈式2〉 に含まれていれば true,それ以外は falseを返

します.

(%i34) member(sin(x),cos(x)+sin(x)+x^2);

(%o34) true

(%i35) member(sin(x),[cos(x)+sin(x)+x^2]);

(%o35) false

(%i36) member(sin(x),[cos(x),sin(x),x^2]);

(%o36) true

(%i37) member(sin(x),[cos(x),sin(x)+x^2]);

(%o37) false

(%i38) member(sin(x),f(cos(x),sin(x),x^2));

(%o38) true

(%i39) member(sin(x),f(cos(x),sin(x)+x^2));

(%o39) false

リストの基本処理を行う函数¶ ³
length(〈リスト 〉 ) 〈リスト 〉 の長さを返却
copylist(〈リスト 〉 ) 〈リスト 〉の複製
reverse(〈リスト 〉 ) 〈リスト 〉の並びを逆にしたリストを返却
append(〈リスト1〉 ,〈リスト2〉 ,· · ·) 複数のリストの結合を実行

econs(〈式1〉 ,〈式2〉 ) 〈式2〉 を 〈式1〉 に追加
endcons(〈式 〉 ,〈リスト 〉) 〈式 〉を 〈リスト 〉の後に結合
delete(〈式1〉 ,〈式2〉 ,〈n〉 ) 〈式1〉を 〈式2〉の先頭から 〈n〉 個削除µ ´
リストの長さは lengthを用いて調べられます.この lengthは LISPの同名の函数と同じ動作を

しています.但し,Maximaの任意の式は内部表現では LISPのリスト構造を持ち,length函数はその
内部表現で,先頭の式の識別子を除いたリストの長さを返す函数です.

(%i22) length([1,2,3]);

(%o22) 3

(%i23) length([1,2,[1,2,[3,4,5]]]);

(%o23) 3

(%i24) length(x+y+z);

(%o24) 3

(%i25) length(x+y*z);

(%o25) 2

この例で,式 x+y+zの内部表現は ((MPLUS SIMP) x y z)となり,lengthは先頭の (MPLUS
SIMP)を除いたリストの長さを返しています.

reverse函数は与えられたリストの並びを逆にしたリストを返却します. 例えば,リスト [a1,a2,a3]
に対して reverse([a1,a2,a3]); を入力すれば [a3,a2,a1]を返却します.但し,各成分 aiはそのま
まです.
この函数もMaximaのリストに限定されず,式に対しても操作可能です. 特に,a > bの様な中置

演算子を持つ式の場合,b < aの様に演算子を挟んで左右が変換されます.この場合でも,内部表現
の先頭の演算子に対して逆向きになるので, reverse(a*c>b*d) の結果は (b*d>a*c)となります.

append函数は複数のリストの結合を行います.Maximaの append函数は LISPの append と同
様の事をします.

endcons函数は append函数と似た函数ですが,appendが先頭に追加するのに対し, endconsは
後に追加します.

即ち,econs(〈式1〉 ,〈リスト1〉 )で 〈式1〉 を 〈リスト1〉 の後に追加します. 但し,Maximaの式は
内部的にはリストの為, econs(〈式1〉 ,〈式2〉 ) とすると 〈式2〉 に 〈式1〉 が追加されます.

(%i43) endcons(x+y,[1,2,3,4]);

(%o43) [1, 2, 3, 4, y + x]

(%i44) endcons(x+y,sin(x)+cos(y));

(%o44) cos(y) + y + sin(x) + x

(%i45) endcons(x+y,sin(x)/cos(y));

sin(x) (y + x)

(%o45) --------------

cos(y)

(%i46) endcons(x+y,sin(x)*cos(y));

(%o46) sin(x) (y + x) cos(y)

(%i47) endcons(x+y,sin(x)-cos(y));

(%o47) - cos(y) + y + sin(x) + x

この様に endcons函数はリストではなく,式に追加する場合はその式の内部表現に依存します.
delete函数は 〈式2〉 に含まれる 〈式1〉 を式の先頭から 〈n〉 個削除します.但し, 〈式1〉 が函数,或

いは単項式でなければ除去出来ません.
尚,〈n〉 はオプションで,指定しない場合と 〈式2〉 に含まれる 〈式1〉 が 〈n〉 個よりも少ない場合,

〈式1〉 を全て削除します.

指定した部分式を取出す函数¶ ³
first(〈式 〉 )
last(〈式 〉 )
rest(〈式1〉 ,〈n〉 )
sublist(〈リスト 〉 ,〈函数 〉 )
substpart(〈x〉 ,〈式 〉 ,〈n1〉 ,· · ·,〈nk〉 )µ ´

first函数は 〈式 〉 の最初の成分を返します.
これに対し,last函数は 〈式 〉 の最後の成分を返します.
rest函数,first函数と last函数は,Maximaの大域変数 inflagによって,与式の内部表現に対する操作

に変更出来ます.inflagはデフォルト値が falseの為, 内部表現に対して操作したければ, inflag:true;
を実行し,inflagの値を trueに変更する必要があります.

rest函数は,〈n〉 が正整数であれば,〈式1〉の先頭から n個の成分を除いた式を返します.
nが負整数であれば,〈式1〉 の後から 〈n〉個の成分を除いた式を返します.

(%i52) rest(x+y+z,2);

(%o52) x

(%i53) rest(x+y+z,-2);

(%o53) z

(%i54) rest([x+y+z,sin(x)+cos(x),exp(x)],-2);

(%o54) [z + y + x]

(%i55) rest([x+y+z,sin(x)+cos(x),exp(x)],2);

x

(%o55) [%e ]

sublist函数は 〈述語函数 〉が trueを返す 〈リスト 〉に含まれる成分を抜出したリストを返します.
例えば, sublist([1,2,3,4],evenp); は [2,4]を返します.
substpart函数は 〈式 〉 の 〈n1〉 ,· · ·,〈nk〉 で指定した成分を 〈x〉 で置換えます. 〈x〉 は式,アトム,

演算子が指定可能です.
〈n1〉 ,· · ·,〈nk〉 の指定方法は, 〈式 〉 がリストで,第m番目の成分であればmを指定します. 更に,

リストのm番目がリストで,その n番目の成分を入れ替えたければ, 列m,nで指定します.

(%i13) substpart(x,[1,2,3,4],2);

(%o13) [1, x, 3, 4]

(%i14) substpart(x,[1,[2,3],4],2);

(%o14) [1, x, 4]

(%i15) substpart(x,[1,[2,3],4],2,2);

(%o15) [1, [2, x], 4]

最初の例ではリスト [1,2,3,4]の第二成分 2を xで置換える為に,2を指定しています. 複合リスト
[1,[2,3],4]の場合,第二成分はリスト [2,3]なので,第二成分を xで置換すれば [1,x,4]になります.第
二成分に含まれる 3を xで置換えたければ,第二成分のリストの第二成分を指定すれば良い事にな
ります.
この事はMaximaの一般の式に対しても適応が可能です.但し,Maximaの場合は入力した式の順

番と,Maximaに入力された後の順番が異なる事がある為, 注意が必要になります.

(%i16) expr:(x+1)/(x^2+x+1)+exp(x);

x x + 1

(%o16) %e + ----------

2

x + x + 1

(%i17) substpart(sin(x),expr,1);

x + 1

(%o17) sin(x) + ----------

2

x + x + 1

(%i18) substpart(sin(x),expr,2);

x

(%o18) sin(x) + %e

(%i19) substpart(sin(x),expr,2,2);

x + 1 x

(%o19) ------ + %e

sin(x)

(%i20) substpart(sin(x),expr,2,1);

sin(x) x

(%o20) ---------- + %e

2

x + x + 1

(%i21) substpart("+",expr,2,0);

x 2

(%o21) %e + x + 2 x + 2

この例では式 expr の指定した成分を sin(x) で置換える操作を行っています. ここで第一成分
は入力した順序とは異なって%eˆxとなっている事に注意して下さい.次に,第二成分は有理式全
体となりますが,この第二成分の構造は (割算,x+1,xˆ2+x+1)となっています.Maximaの内部表
現では式はリストで表現され, 演算子が先頭の第 0成分となり,以下にその引数が続く構造となっ
ています. その為,2,1で有理式の分子,2,2で有理式の分母,最後の 2,0が演算子となります. そこ
で,substpart(”+”,expr,2,0)を実行すると割算が和に置換えられてしまい, 有理式が xˆ2+2*x+2で
置換えられてしまいます.

2.6.4 map函数族

map 函数は LISP ではお馴染の函数で, 基本的に函数をリストに作用させるものです. これは
MathematicaやMapleでも採用されており,非常に便利な函数です.Maximaでは, 式は内部的にリ
スト構造を持っていますが,表にはそれが現われていない為,map函数の動作が判り難い側面もあ
ります.
ここでは,内部形式と絡めて,map函数のお仲間について解説します.
Maximaでのmap函数には,map,maplist,scanmapの 3個あります.これらの函数は全てMaxima

の式に函数を作用させる働きがあります.この中で,scanmapのみが一つの函数を一つの式に作用さ
せますが,mapとmaplistは n個の引数を持つ函数に n個の式に作用させる事が出来ます.
又,map函数とmaplist函数はリストや式の第一層に含まれる部分式に作用させる事が出来ます.

更に,scanmap函数は式の全階層の部分式に作用させる事が可能です.
先ず,map函数の例を示します.

(%i34) map(sin,x*y);

(%o34) sin(x) sin(y)

(%i35) map(sin,x*y+y);

(%o35) sin(x y) + sin(y)

(%i36) map(sin,factor(x*y+y));

(%o36) sin(x + 1) sin(y)

(%i37) map(lambda([x,y],x*y),x+y,w+z);

(%o37) y z + w x

最初の x*yの場合,主演算子は*,第一層には被演算子の xと yがある為,sinは式 x*yの第一層の
部分式 xと yに作用しますが,演算子の置換を行わない為に sin(x)+sin(y)が得られます.x*y+yの
場合,この式の第一層には二つの部分式 x*yと wがあるので今度は sin(x*y)+sin(y)となります.
ところが同値な式でも内部表現が異なると異った結果になります.次の例では factor(x*y+y)の

結果にmap函数で sinを作用させていますが,factor(x*y+y)が (x+1)*yとなる為,この式の第一層
には部分式 x+1と yがあり,主演算子が*なので, sin(x+1)*sin(y)が得られます.
次のmaplist函数は基本的にmap函数と同様ですが,こちらは主演算子をリストに置換します.

(%i15) map(sin,factor(x*y+y));

(%o15) sin(x + 1) sin(y)

(%i16) maplist(sin,factor(x*y+y));

(%o16) [sin(x + 1), sin(y)]

(%i17) :lisp %o15;

((MTIMES SIMP) ((%SIN SIMP) ((MPLUS SIMP) 1 X)) ((%SIN SIMP) Y))

(%i17) :lisp %o16;

((MLIST SIMP) ((%SIN SIMP) ((MPLUS SIMP) 1 X)) ((%SIN SIMP) Y))

上の例で示す様に,内部表現でMTIMESがMLISTの変化している事に注目して下さい.

2.6.5 map函数族に関連する大域変数

大域変数maperror¶ ³
変数名 初期値値 概要

maperror true mapやmaplist函数を制御しますµ ´
大域変数maperrorはmap函数とmaplist函数の動作に影響します.先ず,map函数とmaplist函

数の引数は 〈函数 〉,〈式1〉,· · ·,〈式n〉 で,〈函数 〉は n変数の函数です.ここで,大域変数maperrorの
値が true の場合,各 〈式i〉の主演算子は基本的に同じもので,成分の個数も同じ個数でなければな
りませんが,大域変数maperrorが falseの場合,それ以外の引数でも適用可能になり,以下の動作と
なります.

1. 全ての 〈式i〉が同じ長さでなければ,最短の 〈式j〉 を終えた時点で停止します.

2. 〈式i〉の主演算子が全て同じものでなければ, [〈式1〉,· · ·,〈式n〉] に 〈函数 〉を作用させ,apply
と同じ動作になります.

大域変数maperrorが trueの場合,上の二つの状況であればエラーメッセージが出力されます.

(%i40) maperror:false;

(%o40) false

(%i41) map(lambda([x,y],x*y),x+y+a,w+z);

‘map’ is truncating.

(%o41) y z + w x

(%i42) map(lambda([x,y],x*y),x+y+a,w*z);

‘map’ is doing an ‘apply’.

(%o42) w (y + x + a) z

(%i43) maperror:true;

(%o43) true

(%i44) map(lambda([x,y],x*y),x+y+a,w*z);

Arguments to ‘mapl’ not uniform - cannot map.

-- an error. Quitting. To debug this try debugmode(true);

2.6.6 map函数いろいろ

map函数に関連する述語函数¶ ³
mapatom(〈式 〉)µ ´

mapatom函数は,〈式 〉がmap函数によってアトムとして扱われる時, trueを返します

map函数¶ ³
map(〈函数 〉,〈式1〉, · · ·,〈式n〉)
maplist(〈函数 〉,〈式1〉, 〈式2〉,· · ·)µ ´

ma函数は n個の式の列 〈式1〉,· · ·,〈式n〉から n個の成分を取出し,n個の引数を取る 〈函数 〉を
作用させた結果を返します.

maplist函数はmap函数に似ています.〈式i〉の各成分に 〈函数 〉を作用させたリストを生成しま
す.〈函数 〉は函数の名前や lambda式となります.

maplistがmap函数と異なる点は,mapの場合,〈式i〉の主演算子で式を纏めたものが出力される
のに対して,maplistでは,主演算子の代りにMaximaのリスト演算子が置かれる事です.

(%i27) maplist(sin,x+y);

E(%o27) [sin(y), sin(x)]

(%i28) map(sin,x+y);

(%o28) sin(y) + sin(x)

(%i29) maplist(lambda([x,y],x*y),x+y,w+z);

(%o29) [y z, w x]

(%i30) map(lambda([x,y],x*y),x+y,w+z);

(%o302) y z + w x

scanmap函数¶ ³
scanmap (〈函数 〉,〈式 〉)
scanmap(〈函数 〉,〈式 〉,bottomup)µ ´

scanmap(〈函数 〉,〈式 〉)の場合,再帰的に 〈函数 〉を 〈式 〉の内部表現の頂上から適用して行き
ます.これは徹底した因子分解が望ましい時には特に便利です.
例えば, (a2 + 2a + 1)y + x2 を factorで因子分解しても,そのまま元の式が返却されるだけです

が,scanmap函数を使って factor函数を式に作用させると, 与式の部分式 a2 + 2a + 1 が因子分解
された結果が返されます.

(%i3) exp:(a^2+2*a+1)*y + x^2

(%i4) factor(exp);

2 2

(%o4) a y + 2 a y + y + x

(%i5) scanmap(factor,exp);

2 2

(%o5) (a + 1) y + x

scanmap函数による結果は式の内部表現に依存します.上の例の式の内部表現を以下に示します.

((MPLUS SIMP)

((MEXPT SIMP) X 2)

((MTIMES SIMP)

((MPLUS SIMP) 1 ((MTIMES SIMP) 2 A)

((MEXPT SIMP) A 2)) Y))

この内部表現に対して,scanmap函数は factor函数を上の階層から各部分式に対して作用させま
す.この式の場合,最初に x2 と (a2 + 2a + 1)y に factor函数を作用させ,その次に x2 の x と 2 ,
a2 + 2a + 1 と y 等々と下の階層の成分に作用します.その結果, a2 + 2a + 1(=(+ 1 (* 2 a) (ˆ a
2)))の展開以外はそのままとなる為,上記の結果を得ています.この作用の様子は factor函数の代
りに’fを与えると判り易くなります.

(%i18) scanmap(’f,exp);

f(2) f(2)

(%o18) (f(f(f(f(a) ) + f(f(2) f(a)) + f(1)) f(y)) + f(f(x) ))

この性質がある為に式を全て展開してしまうと,各項に factor函数を作用させる為に,入力値が
そのまま返却されてしまいます.

(%i16) expand(exp);

2 2

(%o16) a y + 2 a y + y + x

(%i17) scanmap(factor,expand(exp));

2 2

(%o18) a y + 2 a y + y + x

scanmap(〈函数 〉,〈式 〉,bottomup)は scanmap(〈函数 〉,〈式 〉)とは逆に, 内部表現の最下層から
〈函数 〉を作用させます.

2.7 配列

2.7.1 Maximaの配列について

Maximaはリストの他に配列が扱えます.Maximaで配列を生成する場合,幾つかの方法がありま
す.先ず,array函数で配列を宣言して具体的な値を割当てて行く方法と, 配列を宣言せずに直接値
を割当てて行く方法,make art q函数やmake array 函数を使う方法があります.

(%i1) a1[1,2]:10;

(%o1) 10

(%i2) a1[0,3]:1;

(%o2) 1

(%i3) a2:make_art_q(10);

(%o3) {Array: #(NIL NIL NIL NIL NIL NIL NIL NIL NIL NIL)}

(%i4) array(a3,fixnum,5);

(%o4) a3

(%i5) make_array(hashed,5);

(%o5) {Array: #(NIL NIL $HASHED NIL NIL G13202)}

最初に示しているのが配列を無宣言で生成する方法で,この場合は配列に直接式や文字列等を入
力する事が可能です.又,Maximaの配列は Cの配列と同様に 0から開始します.配列への値の代入
は,この例の様に直接指定する事で行います.他に, fillarray函数等を用いて他の配列やリストから
生成する事も可能です.
配列を函数を用いて生成する場合,目的に応じて三種類の函数が使えます. 先ず,make art q函数

で生成する配列は LISPのmake-array函数を直接用いて配列を生成するものです.その為,Maxima
の配列の中では最も原始的なものとなります.

array函数で配列を生成する場合,引数に配列名を与える必要があります. そして,最後のmake array
函数は array函数の様に配列名を与える必要はありません.これらの函数では共に,配列の型を指定
する事が可能です.

Maximaには生成した配列を調べる函数として listarray函数と arrayinfo函数の二つがあります.

配列の情報を表示する函数¶ ³
listarray(〈配列名 〉)
arrayinfo(〈配列名 〉)µ ´
最初の listarray函数は配列データを表示し,arrayinfo函数は配列の型や大きさを返す函数です.
では,先程の例から,listarray函数と arrayinfo函数で配列の情報を調べてみましょう.

(%i6) listarray(a1);

(%o6) [1, 10]

(%i7) arrayinfo(a1);

(%o7) [hashed, 2, [0, 3], [1, 2]]

(%i8) arrayinfo(a2);

(%o8) [declared, 1, [9]]

(%i9) arrayinfo(a3);

(%o9) [complete, 1, [5]]

先ず,listarry(a1)で,配列 a1に 1と 10が設定されている事が判ります. arrayinfo(a1)で返却さ
れる hashedが配列 a1の型になります.その次に配列の次元があり,最後にデータが設定されてい
る個所が表示されています.
ここで表示されている配列の型を詳しく述べる為に,配列を生成する函数について詳細を説明し

ましょう.
配列の生成¶ ³

make art q(〈整数 〉)
array(〈配列名 〉,〈整数1〉,· · ·, 〈整数n〉)
array(〈配列名 〉,〈型 〉,· · ·, 〈整数1〉,〈整数n〉)
array([〈配列名1〉,· · ·,〈配列名2〉], 〈整数1〉,· · ·,〈整数n〉)
make array(〈型 〉,〈整数1〉, · · ·,〈整数n〉)
make array(functional,〈函数 〉, 〈型 〉,〈整数1〉, · · ·,〈整数n〉)µ ´

make art q函数は LISPのmake-array函数を呼出して配列を生成するだけの函数で,Maximaの
配列を生成する函数の中では最も原始的な配列を生成する函数です.

array函数は引数に配列名と次元を指定し,〈整数 〉次の配列を生成します.ここで 〈整数 〉 は 5
以下の整数でなければなりません. 即ち,array函数で生成可能な配列は 5次以下の配列です.例え
ば,array(a,2,3,4,5,6) は問題ありませんが,array(b,2,3,4,5,6,7)はエラーになります.
この array函数は大域変数 use fast arraysの影響を受けます.

use fast arrays¶ ³
変数名 初期値 概要

use fast arrays false 配列の種類を制限µ ´
先ず,大域変数 use fast arraysが trueの場合,make array函数を用いて any型の配列を生成しま

す.尚,make array函数で any型の配列を生成する場合, LISPの make-array函数がそのまま使わ
れて初期値が NILの LISPの配列を生成します. 初期値が設定される事を除くと,make art q函数
で生成される配列との違いはありません.
次に大域変数 user fast arraysがデフォルトの falseであれば,array函数は指定した 〈型 〉 を反映

した配列を構成します.ここで設定可能な型には,数値配列の場合,浮動小数点型と整数型配列,そ
の他に complete型があります.

array函数で指定可能な型¶ ³
型 引数 概要

flonum [flonum,float] 浮動小数点データ.array-mode属性に floatを設定
fixnum [fixnum,integer] 整数データ.array-mode属性に fixnumを設定
function function 函数型

complete complete その他µ ´
〈型 〉で flonum型が指定された場合,初期値として 0.0が設定され, 配列の array-mode属性に

floatが設定されます.
〈 型 〉 で fixnum 型が指定された場合, 初期値として 0 が設定され, 配列の array-mode 属性に

fixnumが設定されます.
〈型 〉に functionと completeが指定された場合と 〈型 〉 が無指定の場合,Maxima内部では NIL

が設定されます.この NILを listarray函数は####で表示します.

(%i1) array(a1,flonum,5);

(%o1) a1

(%i2) listarray(a1);

(%o2) [0.0, 0.0, 0.0, 0.0, 0.0, 0.0]

(%i3) array(a2,fixnum,3);

(%o3) a2

(%i4) listarray(a2);

(%o4) [0, 0, 0, 0]

(%i5) array(a3,3);

(%o5) a3

(%i6) listarray(a3);

(%o6) [#####, #####, #####, #####]

この例では最初に配列 a1を flonum型のデータ配列として生成し,listarray函数で中身を見ていま
す.配列の成分が 0.0で初期化されていますね.以降,fixnum型と型の指定を行わずに生成し,listarray
函数を用いて中身を見ています.

make array函数は array函数よりも複雑な配列が生成可能です. このmake array函数で指定可
能な 〈型 〉 には any型,flonum型, fixnum型,hashed型と functional型があります.尚,functional型
の場合は第一引数のみに設定します.

(%i45) make_array(’any,5);

(%o45) {Array: #(NIL NIL NIL NIL NIL)}

(%i46) make_array(’fixnum,5);

(%o46) {Array: #(0 0 0 0 0)}

(%i47) make_array(’flonum,5);

(%o47) {Array: #(0.0 0.0 0.0 0.0 0.0)}

(%i48) make_array(’hashed,5);

(%o48) {Array: #(NIL NIL HASHED NIL NIL G13873)}

(%i49) make_array(functional,’sin,flonum,3);

(%o49) {Array: #(NIL NIL $FUNCTIONAL NOTEXIST %SIN #(0.0 0.0 0.0))}

flonum 型と fixnum 型を指定した場合,make-array 函数を用いた配列のみを生成します. その
為,flonum型の場合は初期値が 0.0,fixnum型の場合,初期値が 0の配列となります.尚,内部的には
LISPの通常の配列となります.

flonum型と fixnum型以外の型を指定した場合,内部的に構造体mgenarray型のデータが生成さ
れます.このデータ型の場合,配列データ本体はmgenarrayの content に設定されます.
先ず,any型の場合,LISPのmake-array函数を用いて初期値 NILの配列が生成されます. 尚,lis-

tarray函数で表示を行うと####で初期値が表示されます.
hashed型と functional型に関しては,contentに配列本体が設定される仕様の筈ですが,通常の配

列成分の割当を行うと,配列データが壊れる為,これらの型は事実上使えません.
array函数とmake array函数で生成した配列は大域変数 arraysに登録されます. make array函

数で,hashed型と functional型を指定した場合,gensym函数で生成したシンボルがこのリストに登
録されます.

配列が登録されるリスト¶ ³
変数名 初期値 概要

arrays [] 生成した配列名が登録されるリストµ ´
大域変数 arraysに登録される配列は,array函数で生成した全ての配列と, make array函数で型

が hashedで生成した配列です.尚,この場合は LISPの gensym函数で生成されたシンボルが表示
されます.

(%i1) array(a1,fixnum,5)$

(%i2) array(a2,flonum,5)$

(%i3) array(a3,5)$

(%i4) make_array(hashed,5)$

(%i5) arrays;

(%o5) [a1, a2, a3, g13158]

但し,listarray函数や arrayinfo函数でこのシンボルは使えません. その為,array函数で生成した
配列名のみが登録されたリストと考えた方が実用上問題がありません.

2.7.2 配列操作に関連する函数

リストや配列の値を配列に割当てる函数¶ ³
fillarray(〈配列 〉,〈リスト 〉)
fillarray(〈配列 〉,〈配列 〉)µ ´
第一引数の 〈配列 〉に第二引数に指定した 〈リスト 〉か 〈配列 〉の値を入れます. 〈配列 〉が浮動

小数点数 (整数)配列ならば,第二引数は浮動小数点 (整数)のリストか浮動小数点 (整数)の配列の
どちらかでなければなりません.
第一引数の配列には第二引数の内容が先頭から順番に入れられますが, もしも,第一引数の配列

が第二引数よりも大きければ,第二引数の最後部の元で第一引数の配列の残りの個所を埋めてしま
います.

配列から指定したデータを取出す函数¶ ³
arrayapply(〈配列 〉,[〈添字1〉,· · · , 〈添字k〉])µ ´

arrayapplyは第一引数に 〈配列 〉を取り,その後に配列の添字リストを指定します.返却値は指定
した添字に対応する配列の値です.

配列の大きさを変更する函数¶ ³
rearray(〈配列 〉,〈次元1〉, · · ·,〈次元n〉)µ ´

rearray函数で配列の大きさの変更を行います.この場合,新しい配列に古い配列の元は番号順に
代入されて行きます.新しい配列が古い配列よりも大きなものであれば, 残りは 0か 0.0の何れかで
埋められます.

配列を削除する函数¶ ³
remarray(〈配列1〉,〈配列2〉,· · · )
remarray(all)µ ´

remarray函数は函数を除去し,占領されていた保存領域を解放します.引数が all であれば全て
の配列を除去します.
ここで,〈式i〉の内部表現での先頭にある演算子 (主演算子と呼びます) は全て同じもので,map函

数を作用させた結果は 〈函数 〉を作用させた各成分を主演算子で繋げたものとなります.

2.8 行列

2.8.1 行列について

Maximaでは行列を扱う事が出来ます.但し,MATLABの様な数値行列処理ソフトの様にリスト
の一種類として直接設定出来るものではありません.
一般的には,matrix函数を用いて行列を生成します.このmatrix函数はMaximaの複数のリスト

から行列を生成する函数です.構文は以下の様になっています.
行列の定義を行うmatrix函数¶ ³

matrix(〈リスト1〉 ,· · ·,〈リストn〉 )µ ´
ここで, 〈リストi〉 は行列の i 行に対応します. Maximaのリストは [1,2,3]の様に,コンマで区

切った式を大括弧で括ったものです.行列はMaximaの内部では,先頭がmatrixで開始する S式と
なっており, substpart等の函数を用いて,matrixを”[”に変更すれば,リストに置換出来ます. 又,逆
にリストから行列に置換する事も可能です,
以下に行列の生成と行列からリスト,リストから行列への変換の例を示します.

(%i3) a:matrix([1,2,3],[4,5,6]);

[ 1 2 3 ]

(%o3) [ ]

[ 4 5 6 ]

(%i4) :lisp a;

((MATRIX SIMP) ((MLIST SIMP) 1 2 3) ((MLIST SIMP) 4 5 6))

(%i4) ?car(a);

(%o4) (matrix, simp)

(%i5) b:substpart("[",a,0);

(%o5) [[1, 2, 3], [4, 5, 6]]

(%i6) c:substpart(matrix,b,0);

[ 1 2 3 ]

(%o6) [ ]

[ 4 5 6 ]

行列の和と差は演算子 + と演算子 − を各々用います.尚,可換積演算子 ∗ と商演算子/は行列の
成分同士の積と商になります.通常の行列積を行いたい場合には, 非可換積演算子を用います.
以下にmatrix函数による行列の生成と,演算子 +,−, ∗, / の結果を示します.

(%i10) a:matrix([1,0,0],[0,2,0],[0,0,3]);

[ 1 0 0 ]

[ ]

(%o10) [ 0 2 0 ]

[ ]

[ 0 0 3 ]

(%i11) b:matrix([1,2,3],[1,3,5],[9,7,5]);

[ 1 2 3 ]

[ ]

(%o11) [ 1 3 5 ]

[ ]

[ 9 7 5 ]

(%i12) a+b;

[ 2 2 3 ]

[ ]

(%o12) [ 1 5 5 ]

[ ]

[ 9 7 8 ]

(%i13) a-b;

[ 0 - 2 - 3 ]

[ ]

(%o13) [ - 1 - 1 - 5 ]

[ ]

[ - 9 - 7 - 2 ]

(%i14) b*b;

[ 1 4 9 ]

[ ]

(%o14) [ 1 9 25 ]

[ ]

[ 81 49 25 ]

(%i15) b/b;

[ 1 1 1 ]

[ ]

(%o15) [ 1 1 1 ]

[ ]

[ 1 1 1 ]

この様に,演算子 + と − は通常の行列の和と差になりますが,演算子 ∗ は勝手が違います.何故
なら,この積演算子 ∗ は可換性を持っている為, 可換性を持たない行列の積は表現出来ないからで
す.その為,通常の行列の積演算子は非可換積の演算子.を用います.この演算子.は見落し易いので,
ここでは,非可換積, 或いは非可換積と表記します.この非可換積による演算は行列 A と B に対し

て, A . B と記述します.この非可換積とその他の演算子に関しては大域変数で, その分配律や結合
律が制御出来ます.デフォルトでは,分配律と結合律を満す様になっています.次に,非可換積に対
応する行列の冪乗は記号^^で記述します. 特に,行列 A の逆行列が存在する場合, A^ ^ (-1)で行

列 A の逆行列を表現します.尚,非可換積の詳細に関しては,1.6.5小節を参照して下さい.

2.8.2 行列を生成する函数

Maximaにはmatrix函数の他にも行列を生成する函数を持っています.
行列の生成を行う函数¶ ³

entermatrix(〈整数1〉 ,〈整数2〉 )
ident(〈整数 〉 )
zeromatrix(〈整数1〉 ,〈整数2〉 )
genmatrix(〈配列 〉, 〈整数1〉 ,〈整数2〉 , 〈整数3〉 ,〈整数4〉)
ematrix(〈整数1〉 ,〈整数2〉 ,〈x〉 , 〈整数3〉 ,〈整数4〉 )
diagmatrix(〈整数1〉 ,〈整数2〉 )
coefmatrix([〈方程式1〉 ,· · ·], [〈変数1〉 ,· · ·])
augcoefmatrix([〈方程式1〉 ,· · ·], [〈変数1〉 ,· · ·])
echelon(〈行列 〉 )µ ´

entermatrixを実行すると,Maximaの要求に従って 〈整数1〉 × 〈整数2〉 個の成分を入力して 〈整
数1〉 行 ×〈整数2〉 列の行列を生成する函数です.

(%i1) entermatrix(3,3);

is the matrix 1. diagonal 2. symmetric 3. antisymmetric

4. general

answer 1, 2, 3 or 4

1;

row 1 column 1: a;

row 2 column 2: b;

row 3 column 3: c;

matrix entered.

[ a 0 0 ]

[ ]

(%o1) [ 0 b 0 ]

[ ]

[ 0 0 c ]

この様に対話的に行列を生成する事が出来ますが,実用上,用途は限られるかもしれません.

ident函数は, 〈整数 〉 次の単位正方行列,即ち,対角成分が全て 1で, 他が全て 0となる正方行列
を生成します.

(%i6) ident(3);

[ 1 0 0 ]

[ ]

(%o6) [ 0 1 0 ]

[ ]

[ 0 0 1 ]

zeromatrix函数は, 〈整数1〉 行 〈整数2〉 列の零行列を返します.

(%i7) zeromatrix(3,2);

[ 0 0 ]

[ ]

(%o7) [ 0 0 ]

[ ]

[ 0 0 ]

genmatrix函数を用いると,与えられた二次元配列から行列が生成出来ます. この genmatrxi函
数は二次元配列から行列を抜き出す形になります. 抜き出し方は,〈整数1〉,〈整数2〉 で配列の大きさ
を指定します.それから,〈整数3〉,〈整数4〉で行列の 1行,1列目の成分を指定します.すると,〈整数1〉
から 〈整数3〉 迄の配列の第一成分を行とし, と 〈整数2〉 から 〈整数4〉 迄の第二成分を列とする領
域が 〈整数1〉 − 〈整数3〉行, 〈整数2〉 − 〈整数4〉 列の行列として抜き出されます. ここで,実際の配
列が生成する行列よりも小さかった場合,即ち,i行 j列の成分が未定の場合,〈配列 〉ij の形式で行列
の成分が表示されます.又,以下に示す様に配列を函数で与える事も可能です.

(%i3) h[i,j]:=1/(i*j+1)

(%i4) genmatrix(h,3,3);

[ 1 1 ]

[ 1 - - ]

[ 2 3 ]

[ ]

[ 1 1 1 ]

(%o4) [ - - - ]

[ 2 3 4 ]

[ ]

[ 1 1 1 ]

[ - - - ]

[ 3 4 5 ]

ematrix函数は 〈整数1〉 行 〈整数2〉 列の行列で, (〈整数3〉,〈整数4〉)成分のみが 〈x〉 となり,他が
全て零となる行列を生成します.

(%i9) ematrix(4,3,1,1,1);

[ 1 0 0 ]

[ ]

[ 0 0 0 ]

(%o9) [ ]

[ 0 0 0 ]

[ ]

[ 0 0 0 ]

diagmatrix函数は n行 n列で,対角成分が xの対角行列を返します. ここで対角成分は全て 〈行
列 〉 のものです. 尚, diagmatrix(n,1)は ident(n)と同じ n次元の単位行列を生成します.

(%i19) diagmatrix(3,2);

[ 2 0 0 ]

[ ]

(%o19) [ 0 2 0 ]

[ ]

[ 0 0 2 ]

coeffmatrixは連立一次方程式の変数リストに対応する係数行列を返します. 尚,連立一次方程式
は,演算子=の右辺,或いは左辺の何れかが 0の場合,式だけでも構いません.

(%i22) coefmatrix([2*x-3*y-1=0,3*x+3*y+10=0],[x,y]);

[ 2 - 3 ]

(%o22) [ ]

[ 3 3 ]

(%i23) coefmatrix([2*x-3*y-z,3*x+3*y+10*z],[x,y]);

[ 2 - 3 ]

(%o23) [ ]

[ 3 3 ]

augcoefmatrix函数は,与えられた方程式と指定した変数のリストから係数行列を生成します.行
列は, 〈方程式1〉,· · · から構成される線形方程式系に含まれる 〈変数1〉 ,· · · の係数から構築されま
す.coefmatrix函数との違いは,生成する係数行列に各方程式の定数項が列として付加される点です.

(%i25) augcoefmatrix([2*x-3*y-z,3*x+3*y+10*z],[x,y]);

[ 2 - 3 - z ]

(%o25) [ ]

[ 3 3 10 z ]

(%i26) augcoefmatrix([2*x-3*y=1,3*x+3*y=10],[x,y]);

[ 2 - 3 - 1 ]

(%o26) [ ]

[ 3 3 - 10 ]

(%i27)

echelon函数は, 〈行列 〉の echelon形式を生成します. これは初等的な行操作で各々の行の最初
の非零元を 1,その元を含む列に対しては, その元を含む行よりも下の成分を全て零となる様に変形
するものです (上三角行列).

[2 1 - a -5 b ]

(%o2) [ ]

[a b c ]

(%i3) echelon(d2);

[ a - 1 5 b ]

[1 - ----- - --- ]

[ 2 2 ]

(%o3) [ ]

[ 2 c + 5 a b ]

[0 1 ------------]

[ 2 ]

[ 2 b + a - a]

行列処理を指定する大域変数¶ ³
変数名 初期値 概要

assumescalar true 引数がスカラー値であると仮定.
detout false 余因子行列を用いた逆行列計算で,行列式の処理を制御
ratmx false 行列成分の CRE表現の表示を制御
scalarmatrixp true 1行 1列行列のスカラへの自動変換を制御
sparse false 疎行列を計算するかどうかを指定µ ´

assumescalarが trueの場合,値が割当てられていない変数と行列の可換積は, 行列の各成分との
可換積で自動的に置換されます.falseの場合,変数は各成分に分配されません.



162 第 2章 Maximaのデータとその操作

detoutが trueであれば,逆行列を計算した時に行列式の割算がそのまま行列の外に残されます.
この大域変数が効力を持つ為には doallmxopsと doscmxopsが false でなければなりません.この
設定をその他の二つが設定される evで与える事も出来ます.

ratmxが falseであれば,行列式や行列の和,差,積が行列の表示形式で行われ, 逆行列の結果も一
般の表示となります.trueであれば,これらの演算はCRE表現で実行され,逆行列の結果もCRE表
現となります.これは成分が往々にして望みもしない展開 (ratfacの設定に依存するものの)の原因
になります.

scalarmatrixpが trueであれば,二つの行列の非可換積の計算で得られた 1行 1列の行列はスカ
ラーに変換されます.もし,allに設定されていれば,この変換は, 1行 1列の行列は何時でもスカラ
に変換されます.但し,falseであれば, この変換は実行されません.

sparseが trueで ratmx:trueであれば,determinantは疎行列式を計算する為のルーチンを利用し
ます.

do一家¶ ³
変数名 初期値 概要

doallmxops true 行列演算子の評価に関連

domxexpt true 行列に対する指数函数の処理

domxmxops true 対行列,対リスト間の演算を管理
domxnctimes false 非可換積の実行に関連

doscmxops false スカラと行列間の演算を実行

doscmxplus false スカラ+行列の処理に関連µ ´
doallmxopsが trueであれば,全ての行列演算子が評価されます. falseであれば,演算子を支配す

る個々の dot大域変数の設定が実行されます.
domxexptが trueの場合,%eˆmatrix([1,2],[3,4])はmatrix([%e,%eˆ 2], [%eˆ 3,%eˆ 4]) となりま

す.一般的に,この変換は 〈基底 〉〈 次数 〉 の形式の変換に影響します.尚,〈基底 〉 はスカラか定数の
式であり,〈次数 〉 はリストか行列です. この大域変数が falseであれば,この変換は実行されません.

domxmxopsが trueであれば,行列と行列間の演算子や行列とリストの間の演算子が実行されま
す.この大域変数が falseなら,これらの演算は実行されません. 尚,この大域変数はスカラーと行列
との間の演算には影響を与えません.

domxnctimesが falseであれば行列の非可換積が実行されます.
doscmxopsが trueであればスカラと行列間の演算子が実行されます.
doscmxplusが trueであれば,スカラ+行列が行列値となります.この大域変数は doallmxopsと

独立した変数です.
行列の括弧を設定する大域変数¶ ³

変数名 初期値 概要

lmxchar [ 行列の右側の括弧で用いる文字を設定

rmxchar ] 行列の左側の括弧で用いる文字を設定µ ´
lmxcharは行列の (左)の括弧として表示する文字を設定します. 右側は rmxcharで指定します.

rmxcharは行列の (右)の括弧として表示する文字を設定します. 左側は lmxcharで指定します.

matrix element変数一族¶ ³
変数名 初期値 概要

matrix element add + 行列の和の演算子を指定
matrix element mult * 行列の成分間の積の演算子を指

定

matrix element transpose false 転置の際に作用させる函数を設

定µ ´
matrix element addは行列同士の和を計算する際に用いる演算子を設定します.函数名や lambda

式であっても構いません.
matrix element multは行列の成分同士の積を計算する際に用いる演算子を設定します.函数名

や lambda式であっても構いません.
matrix element transposeは上記の大域変数と同様に,転置行列を計算する際に,作用させる函数

や lambda式をを指定します.

2.8.3 行列に関連する函数

行列に関連する述語函数¶ ³
構文 trueを返す条件
matrixp(〈式 〉 ) 〈式 〉 が行列の場合
diagmatrixp(〈式 〉 ) 〈式 〉 が対角行列の場合
nonscalarp(〈式 〉) 〈式 〉 がスカラでない場合
scalarp(〈式 〉 ) 〈式 〉 がスカラの場合µ ´

matrixp函数は 〈式 〉 が行列であれば true,そうでなければ false を返します.
diagmatrixp函数は 〈式 〉 が対角行列の場合に trueを返します.
nonscalarp函数は 〈式 〉 が非スカラ,つまり,アトムを含み, 非スカラと宣言されたリストや行列

であれば trueを返しますが, スカラの場合は falseを返します.
scalarp函数が trueとなるのは,〈式 〉 が数,定数やスカラとして宣言された変数,数,定数,そして

その様な変数の合成で行列やリストを含まない場合です.

行列の成分に函数を作用させる函数¶ ³
matrixmap(〈函数 〉 ,〈行列 〉)µ ´

matrixmap函数は,行列 〈m〉 の各成分に 〈函数 〉 を作用させる函数です.リストに対する map
函数の行列版です.

(%i11) A:ident(3)*3;

[ 3 0 0 ]

[ ]

(%o11) [ 0 3 0 ]

[ ]

[ 0 0 3 ]

(%i12) matrixmap(lambda([x],cos(x)*exp(-x)),A);

[ - 3 ]

[ %e cos(3) 1 1 ]

[ ]

(%o12) [ - 3 ]

[ 1 %e cos(3) 1 ]

[ ]

[ - 3 ]

[ 1 1 %e cos(3) ]

行や列の取り出しと追加を行う函数¶ ³
col(〈行列 〉 ,〈i〉 )
row(〈行列 〉 ,〈i〉 )
addcol(〈行列 〉 ,〈リスト1〉, 〈リスト2〉 ,· · ·,〈リストn〉 )
addrow(〈行列 〉 ,〈リスト1〉 , 〈リスト2〉 ,· · ·,〈リストn〉 )
copymatrix(〈行列 〉 )
setelmx(〈x〉 ,〈i〉 ,〈j〉 , 〈行列 〉)µ ´

col函数と row函数は, 〈行列 〉 に対し, 〈i〉 で指定される列と行を各々行列の形式で返す函数です.

(%i17) m1;

[ x 0 - 3 x 0 0 ]

[ ]

[ 0 x 0 - 3 x 0 ]

[ ]

[ 0 0 x 0 - 3 x ]

(%o17) [ ]

[ y - 3 0 y 0 0 ]

[ ]

[ 0 y - 3 0 y 0 ]

[ ]

[ 0 0 y - 3 0 y ]

(%i18) row(m1,1);

(%o18) [ x 0 - 3 x 0 0 ]

(%i19) col(m1,1);

[ x ]

[ ]

[ 0 ]

[ ]

[ 0 ]

(%o19) [ ]

[ y ]

[ ]

[ 0 ]

[ ]

[ 0 ]

addcol函数は列として,addrow函数は行として,複数の複数のリストや行列を 〈行列 〉 に追加し
ます.尚,追加するリストや行列は,行列の大きさに矛盾しないものでなければなりません.

copymatrix函数は 〈行列 〉 の複製を行います. この命令は 〈行列 〉 を成分毎に再生成する時だ
けに使います.

setelmx函数は 〈行列 〉 の (〈i〉,〈j〉) 成分を 〈x〉 で置換します.

(%i13) A;

[ 4 0 0 0 ]

[ ]

[ 0 4 0 0 ]

(%o13) [ ]

[ 0 0 4 0 ]

[ ]

[ 0 0 0 4 ]

(%i14) setelmx(4,2,3,A);

[ 4 0 0 0 ]

[ ]

[ 0 4 4 0 ]

(%o14) [ ]

[ 0 0 4 0 ]

[ ]

[ 0 0 0 4 ]

尚,setelmx函数を使わなくても,行列成分を直接指定して置換える事も可能です. この場合, A[i,j]:x
で行列 Aの (i,j)成分を xで置換します.但し,この場合の返却値は xになります.

小行列を生成する函数¶ ³
minor(〈正方行列 〉 ,〈i〉 ,〈j〉)
submatrix(〈行1〉, · · · , 〈行m〉, 〈行列 〉, 〈列1〉, · · · , 〈列n〉 )
submatrix(〈行1〉, · · · , 〈行m〉, 〈行列 〉)
submatrix(〈行列 〉, 〈列1〉, · · · , 〈列n〉 )µ ´

minor函数は与えられた 〈正方行列 〉 の 〈i〉, 〈j〉 成分の小行列,つまり, 〈正方行列 〉 から 〈i〉 行
と 〈j〉 列を抜いた行列を返します.

submatrix函数は 〈行i〉 行と 〈列j〉 列が削除された新しい行列を生成します.submatrix函数は
minor函数よりも多くの行と列が削除出来ます.

転置,上三角,共役な行列を計算する函数¶ ³
transpose(〈行列 〉 )
triangularize(〈行列 〉 )µ ´

transpose函数は 〈行列 〉 の転置行列を生成します.

(%i9) A:matrix([1,2,3],[4,3,1]);

[ 1 2 3 ]

(%o9) [ ]

[ 4 3 1 ]

(%i10) transpose(A);

[ 1 4 ]

[ ]

(%o10) [ 2 3 ]

[ ]

[ 3 1 ]

triangularize函数は 〈行列 〉 の上三角行列形式を生成します. 尚,行列が正方行列である必要は
ありません.

(%i6) A:matrix([1,2,3,4],[3,4,5,1],[2,3,1,5]);

[ 1 2 3 4 ]

[ ]

(%o6) [ 3 4 5 1 ]

[ ]

[ 2 3 1 5 ]

(%i7) triangularize(A);

[ 1 2 3 4 ]

[ ]

(%o7) [ 0 - 2 - 4 - 11 ]

[ ]

[ 0 0 6 - 5 ]

階数と対角和を計算する函数¶ ³
rank(〈行列 〉 )
mattrace(〈行列 〉 )µ ´

rank函数は 〈行列 〉 の階数を求めます.行列の階数は行列から求めた小行列式で,零にならない
小行列式で,最大のものの大きさです. 尚,rankは行列成分の値が非常に零に近い場合には誤った答
を返す事があります.

mattrace函数は,〈行列 〉 が正方行列の場合,対角和,即ち,行列の主対角成分の総和を計算しま
す.この函数は,ncharpolyで利用されています. ncharpolyはMaximaの charpolyの代りに使える
函数です.尚,mattrace函数を利用する為には,予め load(”nchrpl”); を実行する必要があります.

(%i14) load(nchrpl)$

(%i15) A:matrix([1,2,3],[4,3,1],[-2,0,-2]);

[ 1 2 3 ]

[ ]

(%o15) [ 4 3 1 ]

[ ]

[ - 2 0 - 2 ]

(%i16) mattrace(A);

(%o16) 2

余因子行列の計算に関連する函数¶ ³
adjoint(〈正方行列 〉)
invert(〈正方行列 〉)µ ´

adjoint函数は 〈正方行列 〉の余因子行列を計算します.
invert函数は逆行列を余因子行列を用いた方法で計算します. これは bfloat値成分や浮動小数点

を係数とする多項式を成分とする行列の逆行列をCRE形式に変換せずに計算出来ます.determinant
命令は余因子の計算で利用されるので,ratmxが falseならば,その逆行列は成分表現を変更せずに
計算されます. 現行の実装は高い次数の行列に対して効率的なものではありません. 尚,大域変数
detoutが trueの場合,行列式の部分は逆行列の外側に出されたままとなります.

invertが返した結果は展開されていません.最初から多項式成分を持つ行列の場合,
expand(invert(mat)) ,detout; で生成された出力は見栄えが良くなります.

(%i28) A:matrix([t,1,t-2],[1,t,0],[t+1,1,t-1]);

[ t 1 t - 2 ]

[ ]

(%o28) [ 1 t 0 ]

[ ]

[ t + 1 1 t - 1 ]

(%i29) adjoint(A);

[ (t - 1) t - 1 - (t - 2) t ]

[ ]

(%o29) [ 1 - t (t - 1) t - (t - 2) (t + 1) t - 2 ]

[ ]

[ 2 ]

[ 1 - t (t + 1) 1 t - 1 ]

(%i30) invert(A),expand,detout;

[ 2 2 ]

[ t - t - 1 2 t - t ]

[ ]

[ 1 - t 2 t - 2 ]

[ ]

[ 2 2 ]

[ - t - t + 1 1 t - 1 ]

(%o30) -------------------------------

2 t - 1

尚,行列式を行列の中に入れる場合, adjoint(〈行列 〉)/determinant(〈行列 〉)を計算した結果を
expandで展開したり,有理数係数の多項式が行列の成分に現われる場合は,ratsimpを用いると良
いでしょう.

(%i35) adjoint(A)/determinant(A),expand;

[ 2 2 ]

[ t t 1 2 t t ]

[ ------- - ------- - ------- ------- - ------- ]

[ 2 t - 1 2 t - 1 2 t - 1 2 t - 1 2 t - 1 ]

[ ]

[ 1 t 2 t 2 ]

(%o35) [ ------- - ------- ------- ------- - ------- ]

[ 2 t - 1 2 t - 1 2 t - 1 2 t - 1 2 t - 1 ]

[ ]

[ 2 2 ]

[ t t 1 1 t 1 ]

[ - ------- - ------- + ------- ------- ------- - ------- ]

[ 2 t - 1 2 t - 1 2 t - 1 2 t - 1 2 t - 1 2 t - 1 ]

(%i36) adjoint(A)/determinant(A),ratsimp;

[ 2 2 ]

[ t - t 1 t - 2 t ]

[ ------- - ------- - -------- ]

[ 2 t - 1 2 t - 1 2 t - 1 ]

[ ]

[ t - 1 2 t - 2 ]

(%o36) [ - ------- ------- ------- ]

[ 2 t - 1 2 t - 1 2 t - 1 ]

[ ]

[ 2 2 ]

[ t + t - 1 1 t - 1 ]

[ - ---------- ------- ------- ]

[ 2 t - 1 2 t - 1 2 t - 1 ]

行列式を計算する函数¶ ³
determinant(〈行列 〉)
newdet(〈行列 〉)
newdet(〈配列 〉 ,〈整数 〉)
permanent(〈行列 〉 ,〈整数 〉)µ ´

determinant函数は,Gaussの消去法と似た方法で 〈行列 〉 の行列式を計算します.計算結果の書
式は大域変数 ratmxの設定に依存します.
疎行列の行列式を計算する特別な方法もあり,ratmx:trueと sparse:trueに設定した場合に使え

ます.
newdet函数は 〈行列 〉 や 〈配列 〉 の行列式を計算します.この際に,Johnson-Gentleman tree

minorアルゴリズムを用います. 尚 〈整数 〉 を指定した場合,1行から 〈整数 〉 行と 1列から 〈整数 〉
列の正方行列を取出し,その行列式を計算します. この整数値が無指定の場合,〈行列 〉 が正方行列
であればそのまま行列式を計算し,〈行列 〉 が正方行列で無ければ, 余分な末尾の行,或いは列を削
除した正方行列の行列式を返します.

(%i23) A;

[ 1 2 3 4 ]

[ ]

(%o23) [ 3 4 5 1 ]

[ ]

[ 2 3 1 5 ]

(%i24) newdet(A,3);

(%o24)/R/ 6

permanent函数は,〈行列 〉 の permanentを計算します. 尚 〈整数 〉 を指定した場合,1行から 〈整
数 〉 行と 1列から 〈整数 〉 列の正方行列を取出し,その行列式を計算します. ここで,permanentは
行列式に似ていますが,符号の変化のないものです.

特性多項式の生成に関連する函数¶ ³
charpoly(〈行列 〉 ,〈変数 〉 )
ncharpoly(〈行列 〉 ,〈変数 〉 )µ ´

charpoly函数は 〈行列 〉 の特性多項式 det (<変数 > I − 〈行列 〉) を計算します.
determinant (〈行列 〉 − diagmatrix (length (〈行列 〉) , 〈変数 〉))と同じ結果を返します.
ncharpoly函数は 〈変数 〉に対する 〈行列 〉の特性多項式を計算します.これはMaximaの charpoly

とは別物です.
ncharpolyでは与えられた行列の羃乗の対角和を計算しますが,対角和は特性多項式の根の羃乗

の総和に等しいものです.これらの諸量から根の対称式の計算が可能ですが,それらは特性多項式
の係数です.charpoly は var*ident[n]-aの行列式を計算している. そんな訳で ncharpoly は優れて
いる. 例えば,整数成分の非常に大きな行列の場合は算術的に多項式の計算を避ける為です.予め
load(”nchrpl”); で読込む必要があります.

2.8.4 eigenパッケージ

Maximaに標準で附属する eigenパッケージには,固有値計算に関連する函数と, 基底の直交化に
関連する函数が含まれれています.
ここで説明する函数を利用する為には, load(eigen); を予め実行します. この load(eigen)を実行

する事で,函数が読込まれ,以下の大域変数が定義されます.

eigenパッケージで定義される大域変数¶ ³
変数名 初期値 概要

hermitianmatrix false Hermit行列かどうかを指定
nondiagonalizable false 非対角化行列かどうかを指定

knowneigvals false 固有値を既知として扱うかどうかを指定

knowneigvects false 固有ベクトルを既知として扱うかどうかを指定

listeigvects [] 固有ベクトルのリスト

listeigvals [] 固有値のリスト

rightmatrix [] 左行列

leftmatrix [] 右行列µ ´
大域変数 hermitianmatrixが trueの場合,与えられた行列が Hermite行列であると仮定します.

その為,大域変数 leftmatrixは rightmatrixの転置行列と複素共役な行列になります.又,rightmatrix
は 〈行列 〉 の正規化した固有ベクトルを列とする行列になります.
大域変数 nondiagonalizableが falseの場合,大域変数 leftmatrixと rightmatrix に行列が設定さ

れ,leftmatrix . 〈行列 〉 . rightmatrixが対角行列で,〈行列 〉 の固有値がその対角成分に現れるもの



172 第 2章 Maximaのデータとその操作

となります.但し,nondiagonalizableが trueであれば,これらの行列は生成されません.
knowneigvalsが trueの場合,行列の固有値は既知で,大域変数の listeigvalsに保存されていると

eigenパッケージの函数は仮定します.その為,knowneigvalsが trueの場合は listeigvalsに eigenvalues
函数の出力と同じリストが設定されていなければなりません

knowneigvlects が true の場合, 行列の固有値に対応する固有ベクトルが既知であり, 大域変数
の listeigvectsに保存されていると eigenパッケージの函数が仮定します. その為,knowneigvects
が trueの場合は listeigvectsに eigenvects函数の出力と同じリストが設定されていなければなりま
せん

内積函数¶ ³
innerproduct(〈x〉 ,〈y〉 )
inprod(〈x〉 ,〈y〉 )(innerproduct函数の別名)µ ´

innerproduct函数は内積を表現します.短縮名は inprodです. リスト 〈x〉 と 〈y〉 を引数として
取り, 〈x〉 の複素共役 . 〈y〉 で定義されています.ここで,非可換積は通常のベクトルの内積演算子
と同じものです.

eigenパッケージには以下の行列操作の函数が含まれています.

行列操作の函数¶ ³
columnvector(〈リスト 〉 )
conjugate(〈リスト 〉 )
conj(〈リスト 〉 )(conjugate函数の別名)µ ´

columnvector函数は与えられたリストから列ベクトルを生成します.

(%i6) columnvector([1,2,3]);

[ 1 ]

[ ]

(%o6) [ 2 ]

[ ]

[ 3 ]

conjugateは引数の複素共役を返します.

(%i3) A:matrix([1,2*%i],[1-%i,4]);

[ 1 2 %i ]

(%o3) [ ]

[ 1 - %i 4 ]

(%i4) conjugate(A);

[ 1 - 2 %i ]

(%o4) [ ]

[ %i + 1 4 ]

固有値の計算に関連する eigenパッケージ函数¶ ³
eigenvalues(〈行列 〉)
eivals(〈行列 〉)(eigenvalues函数の別名)
eigenvectors(〈行列 〉 )
eivects(〈行列 〉 )(eigenvectors函数の別名)
similaritytransform (〈行列 〉)
simtran (〈行列 〉)(similaritytransform函数の別名)µ ´

eigenvalues函数 (短縮名 eivals)は引数に一つの行列を取り,固有値と固有ベクトルを含むリスト
を返します.返却されるリストの第一成分が固有値リスト,第二成分が固有値リストに対応する重
複度のリストとなります.

eigenvalues函数では charpoly函数で特性多項式を計算し,solve函数を使ってその特性多項式の
根を求めています.solve函数は厳密解を求める函数の為, 高次多項式に対しては,その根を見付け
損なう事があります.更に,不正確な答を返す事もあります.しかし eigenパッケージに含まれてい
る conjugate函数, innerproduct函数,univector函数,columnvector函数と gramschmidt函数を必
要としません.

eigenvectors函数は引数として一つの行列を取り,リストを返します.このリストに含まれる最初
の副リストには eigenvalues函数の出力,他の副リストには行列の各々の固有値に対応する固有ベ
クトルが含まれています.
尚,algsys函数が固有ベクトルの計算で使われています.尚,固有値が不正確な場合, algsysが解を

生成する事が出来ない事があります.この場合,eigenvalues函数を使って最初に見つけた固有値の
簡易化を行う事を勧めます.

similarityransform函数は 〈行列 〉を引数とし,uniteigenvectors函数の出力結果リストを返します.

ベクトルの正規化に関連する函数¶ ³
gramschmidt([〈リスト1〉 ,· · ·,〈リストn〉])
gschmidt([〈リスト1〉 ,· · ·,〈リストn〉]) (gramschmidt函数の別名)
unitvector(〈リスト 〉 )
uvect(〈リスト 〉 )(unitvector函数の別名)
uniteigenvectors(〈行列 〉)
ueivects(〈行列 〉)(uniteigenvectors函数の別名)µ ´

gramschmidt函数はGram-Schmidtによる直交ベクトルを求めます. この函数は eigenパッケー
ジに含まれる函数なので,予め load(eigen) を実行して利用します.gramschmidtは引数にリスト
の列で構成されるリストを取ります. ここで 〈リストi〉 は全て長さが等しくなければなりません
が, 各リストが直交している必要はありません.

gramschmidtは互いに直交したリストで構成されたリストを返します. 尚、返ってきた結果には
因子分解された整数が含まれる事があります. これはMaximaの factor函数が gram-schmidtの処
理の過程で使われた為です. こうする事で式が複雑なものになる事を回避し,生成される変数の大
きさを減らす助けにもなっています.

unitvectorは 〈リスト 〉 の大きさを 1にしたリストを返します. uniteigenvaluesは与えられた
〈行列 〉 の固有値と固有ベクトルで構成されたリストを返します.出力リストの第一成分のリスト
には eigenvalues 函数の出力があり,第二成分の副リストには正規化した固有ベクトルが,第一成分
のリストの固有値に対応する順番で並んでいます.

uniteigenvectors函数は与えられた行列から長さを 1にした固有ベクトルを返します.

第3章 プログラム

この章で解説する事:

• Maximaでプログラム

• 函数の定義

• データの入出力

3.1 Maximaでプログラム

Maximaには制御文として,if文,反復処理に doループや goといった,原始的な構文があります.
又,block文を用いる事で局所変数を利用する事も出来ます.

Maximaの処理言語は Cや PASCALの様な手続型の言語に見えます. しかし,LISP上で動作す
る為に,データの内部表現を利用した方が効率的なプログラムが記述し易い側面もあります.

Maximaには compile函数等で,Maximaの処理言語で記述した函数をコンパイルする手段もあ
り,それによってある程度の速度向上も見込めます,それでも,Maximaの処理言語で記述したプロ
グラムを LISPで解釈して実行する手間もある為,直接, LISPで記述する方が速度的には有利です.

3.1.1 block文

block文¶ ³
block(〈変数リスト 〉, 〈式1〉, 〈式2〉, · · · , 〈式n〉)µ ´

Maximaの blockは FORTRANの subroutine,PASCALの procedureに似たものです. block文
は文の集合体ですが,block内部の文にラベル付けが行え,局所変数も扱えます.局所変数は block
文外部にある同名の大域変数との名前の衝突を避ける事に使えます.同名の大域変数が存在した場
合,block文の実行中, その変数はスタックに保存されるので,その間は参照出来ません.block文が
終了した時点で,スタックに保存さていた値がもとに戻されますが,block文の局所変数の値は破棄
されてしまいます.
尚,局所変数を用いない場合には,局所変数のリストを省略したり, 空のリスト []で置換えても構

いません.更に,局所変数に初期値を設定する事も可能です.この場合,[a:1,b:2]の様に局所変数に対
して:による通常の割当を行います.
ここで,block内部で用いられているものの,block文の局所変数リストに含まれていない変数は

block文の外部で用いられている変数と同様に大域変数として扱われます.その為,値は block文の
終了後も保持されます.

block の値は, 最後の文の値か block から return 函数に渡された引数の値となります. 函数
go は制御を go の引数でラベル付けされた block 内の文に移動する為に使えます. 文のラベル
付けは,block 内部で, アトムを文の前に置く事で対処します. 例えば,block( 〈 式1〉, · · · , 〈 式n〉,
loop,〈 式 〉, · · ·,go(loop),· · · ) の様に用います. go 函数の引数は go 函数を含む block 内部に現わ
れるラベル名でなければなりません. goを含まない block文中のラベルに goで移動する事は出来
ません.

return 函数は block 文の変数リスト以外の何処にでも置けます.return を置かなかった場合に
は,block文の末尾の式の値が返却値となります.
以下に block文を用いた函数の例を示します.この函数では局所変数として, a,kを用いています.

(%i67) a:10;

(%o67) 10

(%i68) neko(x):=block([a:2,k],k:sin(x)*a,return(k));

(%o68) neko(x) := block([a : 2, k], k : sin(x) a, return(k))

(%i69) neko(10);

(%o69) 2 sin(10)

(%i70) a;k;

(%o70) 10

(%i71) k;

(%o71) k

この例で示す様に,局所変数は block文内部でのみ利用され,同名の変数には影響を与えていま
せん.

3.1.2 if文

if文は条件分岐で用います.その構文は C等の言語と違いはありません.
if文の構文¶ ³

if 〈述語 〉 then 〈式1〉 else 〈式2〉µ ´
if文は 〈述語 〉 を評価して,trueであれば, 〈式1〉 を実行し,falseであれば, 〈式2〉 を実行します.
〈述語 〉 はその値が true,又は falseであるかが評価可能な式で, 述語函数や演算子等で構成され

たものです.
if文で利用可能な演算子¶ ³

演算子 記号 種類

等しい =, equal 中置式演算子 (infix)
等しくない # 中置式演算子 (infix)
大きい > 中置式演算子 (infix)
小さい < 中置式演算子 (infix)
以上 >= 中置式演算子 (infix)
以下 <= 中置式演算子 (infix)
and and 中置式演算子 (infix)
または or 中置式演算子 (infix)
否定 not 前置式演算子 (prefix)µ ´
尚,前置式演算子 (prefix)と中置子演算子 (infix)は,演算子を引数の置き方で区分したもです.詳

細は演算子の節を参照して下さい.
これに対し,〈式1〉 と 〈式2〉 は任意のMaxima の式 (勿論,if文を含んで入れ子になっても構いま

せん)が利用可能です.

3.1.3 do文による反復処理

Maximaでは,do文を用いて反復処理を行います.

do文の基本構造¶ ³
for 〈制御変数処理 〉 〈終了条件 〉 do 〈本体 〉µ ´

do文には,do文内部のみで用いられる局所変数があり,これを制御変数と呼びます. この制御変数
は do文の中だけで効力を持ちます.それから,終了条件の判定を行い,終了条件を満さなければ,do
文の本体を実行し,それから制御変数処理で, 制御変数に変更を加え,再度,終了条件の判定を行う
事を反復します.
最初に 〈制御変数処理 〉 の個所について述べます.
制御変数処理では,最初に制御変数に初期値を割当て,次に,反復処理で再度回って来ると,制御

変数に新しい値を割当てます.
ここで,最初の初期値の割当てでは,次の二つの同値な書き方があります.

制御変数の初期値の割当¶ ³
〈変数 〉 : 〈初期値 〉

〈変数 〉 from 〈初期値 〉µ ´
この制御変数の初期値の割当はどちらを用いても構いません.尚,初期値が 1の場合,: 〈初期値 〉

や from 〈初期値 〉 は省略可能です.
次に,制御変数を一定の値で増加させたり,減少させたければ,do文の内部に, step 〈増分 〉 を追

加します.この時, 〈増分 〉 を正実数とすれば,通常の増分となり,負の実数とすれば,減少分になり
ます. 尚, 〈増分 〉 が 1の場合は step 1を省略する事が可能です
制御変数に何等かの函数を割当てたければ,next 〈制御変数の式 〉 とすれば,iの値に 〈制御変数

の式 〉 で計算した値が制御変数に割当てられます.
例えば,以下の二つの反復処理は同値になります.

制御変数の割当方法¶ ³
do i from 1 step 2 · · ·
do i:1 next i+2 · · ·µ ´

これらは,全て制御変数 iの初期値を 1,増分を 2とする反復処理を実行します. 最初の二つでは
stepを用いて増分を定めています.最後の nextを用いたものでは, 函数 i+2で制御変数の値を定め
ています.
これらの方法に加えて,リストを用いた制御変数の値の割当て方もあります.

リストを用いた制御変数の割当¶ ³
for i in 〈リスト 〉 · · ·µ ´
この場合,リストの成分には数値以外の函数や式を用いる事も可能です.

(%i10) for i in [1,2,3,4,5,6,7,8,9,10] do print(i);

1

2

3

4

5

6

7

8

9

10

(%o10) done

(%i11)for i in [sin,cos,tan] do print(subst(i,f,f(%pi/4)));

sqrt(2)

-------

2

sqrt(2)

-------

2

1

(%o17) done

この例では, 最初にリスト [1,2,3,4,5,6,7,8,9,10] の元を表示し, 最後の例では, f(%pi/4) の f に
sin,cos,tanを順番に代入した結果を表示させています.

do文の終了条件の与え方には,次の三種類があります.

do文の終了条件の与え方¶ ³
thru 制御変数の境界値に達した時点で反復処理を終える.
unless 条件を満した時点で反復処理を終える.
while 条件を満さなくなった時点で反復を終える.µ ´
ここで,終了条件を与える式は,MAXIMAの述語,即ち,trueか falseかが判別可能な式です.unless

を用いた do文は,C等の repeat-until文,whileを用いた do文は while文に相当します.
thru,unlessと whileを用いた例を以下に示します.尚,三種類ともに全て同じ反復処理,即ち,1か

ら 10迄の数を表示して終了,を実行するのです.

終了条件の例¶ ³
for i:1 thru 10 do print(i);
for i:1 while i <= 10 do print(i);
for i:1 unless i > 10 do print(i);µ ´

尚,通常の do文によって返される値はアトムの doneです.函数 returnを用いると, 本体の中で
doから早目に抜けて必要な値を与える事に使えます. blockにある do文中の returnは do文を出
るだけで,block全体から出る訳ではありません.同様に go函数も block中の do文から抜ける為に
使ってはなりません.

プログラムに関連する大域変数¶ ³
変数名 初期値 概要

backtrace [] 函数リスト

dispflag true block文中の函数出力を制御
prederror true if文や is函数のエラーメッセージ出力を制御
errorfun false エラー発生時に起動させたい函数名µ ´

backtraceは debugmode:allの時に,入力された函数全てのリストを値として持ちます.
dispflagが falseならば,block文の中で呼ばれた函数の出力表示を禁止します. 記号$のある block

文の末尾では dispflagを falseに設定します.
prederrorが trueであれば,if文や is函数で,trueか falseであるか述語の評価に失敗すると,何時

でもエラーメッセージが表示されます.
falseであれば,unknownが代りに返されます.
errorfunに引数を持たない函数名が設定されていれば,エラー発生時に, その函数が実行されま

す.この設定は batchファイルで,エラーが生じた場合にMaximaを終了したり,端末からログアウ
トしたい時に使えます.

block文内部で利用する函数¶ ³
go(〈ラベル 〉 )
return(〈式 〉 )
break(〈引数 〉 ,· · ·)
catch(〈式1〉 ,· · ·,〈式n〉 )
throw(〈式 〉 )µ ´

go函数は block内部で指定した block文中のラベルに移動する事に用いられます. ラベルとして
アトムを用い,このアトムは文の前に置きます.

block([x],

x:1,

loop,x+1,

...,

go(loop),

...)

goの引数は同じ blockの中で現われるラベルでなければなりません. goを含む別の blockにあ
るラベルに goを用いて移動する事は出来ません.

return函数は block文から引数を伴って抜ける時に用います.場所は block文の何処に置いても
構いません.

break 函数は,〈 引数 〉 の評価と表示を行い, (maxima-break) にて利用者がその環境を調べ, 変
更する事が出来る様にします. (maxma-break) からは exit; を入力すれば計算が再開されます.
尚,Ctrl+a(ˆa)で maxima-breakに何時もで対話的に入る事が出来ます.Ctrl+xは maxima-break
内部で,本体側の処理を終了せずに, 局所的に計算を止める事に用いても構いません.

catch函数は throwと対で用います.これは非局所的回帰 (non-local return)で用いる函数で,最
も近い throwに対応する catchに行きます.その為,throwに対応する catchが必ず必要で,そうで
なければエラーになります. 〈式i〉 の評価が何らの throwの評価に至らなかった場合, catchの値は
最後の引数 〈式n〉 の値となります.

(%i51) g(l):=catch(map(lambda([x],

if x<0 then throw(x) else f(x)),l));

(%o51) g(l) := catch(map(lambda([x], if x < 0 then

throw(x) else f(x)), l))

(%i52) g([1,2,3,7]);

(%o52) [f(1), f(2), f(3), f(7)]

(%i53) g([1,2,-3,7]);

(%o53) - 3

函数 gは,lが非負の数のみであれば lの各要素に対する fのリストを返します. それ以外では,g
は lの最初の負の要素を捉えて,それを放します.

throw函数は 〈式 〉 を評価し,近くにある catchに値を投げ返します. throwは catchと一緒に使
われます.

エラー処理を行う函数¶ ³
errcatch(〈式1〉 ,· · ·,〈式n〉 )
error(〈引数1〉 ,· · ·,〈引数n〉)
errormsg()µ ´

errcatch函数は引数を一つづつ評価して,エラーが生じなければ最後の値のリストを返します.,
任意の引数の評価でエラーが生じた場合,errcatchはエラーを捉えて即座に [](空のリスト)を返しま
す.この函数はエラーが生じていると疑われる batchファイルで,エラーを捉えなければ怱ち batch
を終了させる様にすると便利です.

error函数は引数の評価と表示を行い,Maximaのトップレベルか,errcatchにエラーを返します.
エラー条件を検知した場合や,Ctrl+ˆが入力出来ない場所なら何処でも函数を中断させられるので
便利です.
大域変数 errorにはエラーを記述するリストが設定されており,最初のものは文字列で,残りの対

象は問題を起しているものです.
errormsg函数は最新のエラーメッセージを再表示します.変数 errorにはエラーを記したものの

リストが設定されており,最初は文字列で,残りは問題の対象です.

ttyintfun:lambda([],errormsg(),print(””)) で利用者中断文字 (ˆ u)をメッセージの再表示を行う
様に設定します.

3.2 Maximaで函数の定義

3.2.1 函数定義

Maximaでは,予めシステムが持っている函数の他に,利用者定義の函数が扱えます.
Maximaの函数定義では演算子:=を用いる方法と define函数を用いる方法があります.

函数の定義方法¶ ³
define ( 〈函数名 〉, 〈, (〈引数1〉, · · · , 〈引数n〉)), 〈式 〉 )
〈函数名 〉(〈引数1〉, · · · , 〈引数n〉) := 〈函数本体 〉µ ´
基本的に defineで定義する函数は block文,if文や do文を含まない,一つの式のみので構成され

た函数を定義します.
一般的な函数の定義は演算子:=を用いた函数定義で函数を定義します. ここで,演算子:=を用い

る函数定義で,その函数本体は式をコンマで区切ったものになります.
函数の定義を行うと,大域変数 fuctionsに定義した函数が追加されます.

大域変数 functions¶ ³
変数名 初期値 概要

functions [] 利用者定義函数のリストµ ´
大域変数 functionsは利用者が定義した全ての函数名を含むリストです. 利用者が函数定義を行

うと自動的に functionsのリストに函数名が追加されます. functionsに追加されるのは演算子:=の
左辺となります.

(%i1) functions;

(%o1) []

(%i2) f(x):=sin(x);

(%o2) f(x) := sin(x)

(%i3) f(10);

(%o3) sin(10)

(%i4) functions;

(%o4) [f(x)]

この例では,函数 f(x)を定義すると,立ち上げ時には空リストであった functionsに演算子:=の左
辺が登録される為,f(x)が追加されています.
函数定義の演算子:=を用いた函数で,最も簡単なものは,函数本体が式1,式2, · · · ,式nの様に複数の

式を並べたものです.この場合,函数の返却値は単純に最後の式の結果だけです.例えば, f(x):=(1-x,2+x,2*x)
で函数 fを定義した場合,この函数の結果は最後の式 2*xを計算した値になります.
但し,この函数では割当が実行されていない為に,変数の内容の書換えは生じません.
では, f(x):=(y:x,z:2+y,2*z) と定義した函数を実行した場合はどうなるでしょうか？

(%i1) f(x):=(y:x,z:2+y,2*z);

(%o1) f(x) := (y : x, z : 2 + y, 2 z)

(%i2) f(x);

(%o2) 2 (x + 2)

(%i3) y;

(%o3) x

(%i4) z;

(%o4) x + 2

(%i5) f(2);

(%o5) 8

(%i6) x;

(%o6) x

(%i7) y;

(%o7) 2

(%i8) z;

(%o8) 4

この様に,先頭の式から順番に処理されて行き,最後の式の結果だけが返却されています.更に,こ
の函数の場合,内部で用いた変数 yと zの値が書換えられている事に注意して下さい.
この様に,Maxima内部で用いた変数は,通常は大域変数として扱われます. その為,変数の値の

書換が生じます.この事を避ける為に,Maximaでは函数内部のみで有効な局所変数が扱えます.こ
の局所変数は,後述の block文の中で定義可能です.

Maximaの lambda式を用いると無名の函数が構築出来ます.この lambda函数の構文は最初に
函数の変数を宣言し,その後に函数本体が続きます.函数本体は基本的にMaximaの式をコンマで
区切ったものになります.

lambda函数¶ ³
lambda([〈変数1〉, · · · , 〈変数n〉], 〈函数本体 〉 )µ ´
構文自体は LISPの lambda式と同様の構文となっています.
次の例では lambda([i],2*i+1)で引数 iに 1を加える無名函数を構成し,その函数に map函数で

リスト [1,2,3]に作用させた結果と,lambda式を利用した函数 neko を示しています.

(%i58) map(lambda([i],2*i+1),[1,2,3]);

(%o58) [3, 5, 7]

(%i59) neko(x):=map(lambda([i],sin(2*i+1)),x);

(%o59) neko(x) := map(lambda([i], sin(2 i + 1)), x)

(%i60) neko([1,2,3,4,5]);

(%o60) [sin(3), sin(5), sin(7), sin(9), sin(11)]

(%i61) i;

(%o61) i



3.2. Maximaで函数の定義 185

ここで,lambda函数内部で用いた疑似変数 iが lambda函数内部のみで値を持つ事に注意して下
さい.

Maximaでは引数の個数が可変な函数も定義出来ます.この場合,最後の引数を特別な引数リス
トとして割当てて,函数を定義します.尚,引数が少ない場合, 安易な処理を行っていると少し問題
になるかもしれません.

(%i41) f([u]):=u;

(%o41) f([u]) := u

(%i42) f(1,2,3,4,5);

(%o42) [1, 2, 3, 4, 5]

(%i43) f(a,b,[u]):=[a,b,u];

(%o43) f(a, b, [u]) := [a, b, u]

(%i44) f(1,2,3,4,5,6);

(%o44) [1, 2, [3, 4, 5, 6]]

(%i45) f(1,2);

(%o45) [1, 2, []]

単純に式を並べる方式では,函数の返却値は最後に処理された式の値となります. そうではなく,
函数本体に含まれる式の返却値が必要な場合には,block文と return文を組合せます.
定義した函数の内容は函数 dispfunや fundefで参照する事が出来ます.

定義した函数の内容を表示する函数¶ ³
dispfun(〈函数名1〉, · · · , 〈函数名n〉 )
dispfun(all)
fundef( 〈関数名 〉 )µ ´
利用者定義の函数 〈函数名1〉, · · · , 〈函数名n〉 の内容を表示します.この函数の表示では,函数を

定義した時点での函数や定数等がそのまま表示されます.
尚,引数に allを設定すると,大域変数 functionsと arraysで与えられる函数を全て表示します.
fundef 函数は 〈 関数名 〉 に対応する函数の定義を返します. fundef は dispfun に似ています

が,fundefでは display函数を呼出さない点で異なります.

(%i9) neko(x):=sin(x)*exp(x);

(%o9) neko(x) := sin(x) exp(x)

(%i10) dispfun(neko);

(%t10) neko(x) := sin(x) exp(x)

(%o10) done

(%i11) fundef(neko);

(%o11) neko(x) := sin(x) exp(x)

この例で示す様に,dispfunを実行すると結果は%tラベルに表示されていますが, fundefの方は
通常の%oラベルに表示されています.それ以外で違いはありません.

利用者定義函数を削除する函数¶ ³
remfunction (〈函数1〉, 〈函数2〉, · · ·)
remfunction (all)µ ´

remfunction函数は利用者定義函数をMaximaから削除します.利用者定義函数は大域変数 func-
tionsにその名前が保存されており,remfunctionは functionsに含まれている函数名の削除を行い
ます.尚,引数として allが与えられた場合, 大域変数 functionsに含まれる全ての利用者定義の函数
が削除されます.

3.2.2 マクロの定義

Maximaではマクロの定義も出来ます.マクロの定義では演算子::=を用います. Maximaのマク
ロは最終的に LISPの S式に展開し,それから評価が行われます.
複雑なマクロを生成する為に,block文に似た buildq函数を用います.

buildq函数¶ ³
buildq( 〈変数リスト 〉, 〈式 〉 )µ ´

buildqの引数は,実際の式の代入が実行される迄,Maximaの解釈で勝手に変換される事を防ぐ必
要があります.その為に単引用符’を用います.

buildqの応用に漸化式の計算があります.ここで,幾つかの数列を定義してみましょう.

フィボナッチ数

フィボナッチ数は,次の漸化式を満す数です.

フィボナッチ数¶ ³
F0 = 0
F1 = 1

Fn+1 = Fn + Fn−1µ ´
では,Maximaのマクロを利用してこの数列を定義してみましょう.

(%i10) fb:F[n-1]+F[n-2];

(%o10) F + F

n - 1 n - 2

(%i11) define(F[n],buildq([u:fb],u));

(%o11) F := F + F

n n - 1 n - 2

(%i12) F[0]:0;F[1]:1;F[2]:1;

(%o12) 0

(%o13) 1

(%o14) 1

(%i15) F[10];

(%o15) 55

(%i16) F[140];

(%o16) 81055900096023504197206408605

漸化式が単純な場合は buildqを使わずに, define(F[n],F[n-1]+F[n-2]) でも問題はありません.

ルジャンドルの多項式
ルジャンドルの多項式¶ ³

P0(z) = 1
P1(z) = z

Pn+1(z) = (2n − 1)zPn − (n − 1)Pn−1µ ´
(%i10) pnz:((2*n-1)*z*Pz[n-1]-(n-1)*Pz[n-2])/n;

(2 n - 1) Pz z - (n - 1) Pz

n - 1 n - 2

(%o10) -------------------------------------

n

(%i11) define(Pz[n],buildq([v:pnz],expand(v)));

(2 n - 1) Pz z - (n - 1) Pz

n - 1 n - 2

(%o11) Pz := expand(-------------------------------------)

n n

(%i12) Pz[0]:1;Pz[1]:z;

(%o12) 1

(%o13) z

(%i14) Pz[10];

10 8 6 4 2

46189 z 109395 z 45045 z 15015 z 3465 z 63

(%o14) --------- - --------- + -------- - -------- + ------- - ---

256 256 128 128 256 256

この例では,expandを用いて式を展開しています.この式の展開を省くと,結果が複雑になるので,
注意が必要になります.次の例では同じ計算で,expandを省略したものの計算結果を示しています.

(%i17) kill(Pz);

(%o17) done

(%i18) Pz[n]:=((2*n-1)*z*Pz[n-1]-(n-1)*Pz[n-2])/n;

(2 n - 1) z Pz - (n - 1) Pz

n - 1 n - 2

(%o18) Pz := -------------------------------------

n n

(%i19) Pz[0]:1;Pz[1]:z;

(%o19) 1

(%o20) z

(%i21) Pz[4];

2

5 z (3 z - 1)

7 z (-------------- - 2 z) 2

2 3 (3 z - 1)

-------------------------- - ------------

3 2

(%o21) -----------------------------------------

4

この例で示す様に, 式の簡易化が行われていない為に,n=4 でも必要以上に複雑な結果となって
います.
このマクロに関連する大域変数として,macroexpansionがあります.

大域変数macroexpansion¶ ³
変数名 初期値 概要

macroexpansion false マクロ展開の制御で利用µ ´
大域変数macroexpansionはマクロ展開を制御する大域変数です.Maximaのマクロは最初に呼出

された時点でマクロ展開が実行されます.ここで,展開されたマクロを保持していれば,次に同じマ
クロの呼出しがあった場合,展開する手間が不要となるので,その分,効率的な処理が行えます.一
方で,展開したマクロを保持する為には, それなりのメモリが必要となります.

• false
マクロが呼出される度にマクロの展開を行います

• expand
マクロ呼出で評価されると,以後,マクロの展開をしなくても良い様に展開内容を記憶します.

マクロの呼出では通常 grindと displayも呼出します.その分,全ての展開を記憶する為に特
別にメモリーを必要とします.

• displace
最初の呼出でマクロ展開が実行されます.expandの場合と比較して,幾らか少ない保存領域を
必要とする割に処理速度は同程度ですが,マクロ呼出が記憶されない欠点を持っています.展
開は displayか grindが呼出されていれば参照出来ます.

3.2.3 apply函数

Maximaで重要な函数の一つに apply函数があります.この apply函数はMapleやMathematica
にもある函数です.

apply函数¶ ³
apply( 〈函数 〉, 〈リスト 〉 )µ ´

apply函数は 〈函数 〉 を 〈リスト 〉 に適用した結果を与えます.
例えば, apply(min,[1,5,-10.2,4,3]) は -10.2 となります.
函数の呼出しで,それらの引数が評価されておらず,それらの評価を希望する場合も applyは便利

です.例えば, filespecがリスト [test,case]であれば, apply(closefile,filespec) は closefile(test,case)
と同値です.
一般的に,applyで評価させる場合,単引用符’を函数の先頭に置いて,函数を名詞型として,apply

に評価させます.幾つかのアトムの変数が,とある函数と同じ名前を持っていれば,函数としてでは
なく,その変数値が利用される事になりす.何故なら,applyそれ自身が第二の引数と同様に第一の
引数をも評価するからです.

(%i30) neko(x):=2*x;

(%o30) neko(x) := 2 x

(%i31) neko:-128;

(%o31) - 128

(%i32) neko(10);

(%o32) 20

(%i33) apply(’neko,[20]);

(%o33) 40

(%i34) appply(neko,[20]);

(%o34) appply(- 128, [20])

この例では,nekoを函数として定義していますが,同時に,変数としての nekoには-128が束縛さ
れています.その為,applyでは単引用符’がなければ変数として評価されてしまいます.

3.2.4 最適化

Maximaは LISPで記述されています.Maximaのプログラムも含めた全てのデータは内部表現と
して LISPの S式になっており,LISPがこの内部表現を解釈して処理を実行しています.その為,函
数やデータを LISPの函数やデータに変換してしまえば, 内部表現の解釈の手間が省ける為,処理速
度向上が見込める事になります.

Maximaには利用者定義函数を LISPの函数に変換する函数を幾つか持っています.
LISP函数に変換する函数¶ ³

translate( 〈函数1〉, · · · , 〈函数n〉 )
translate(functions)
translate(all)
translate file( 〈ファイル 〉 )
translate file( 〈ファイル 〉, 〈LISP ファイル 〉 )
tr warnings get()µ ´

translate函数はMaximaの処理言語で記述した利用者定義の函数を LISPの函数に変換する函
数です.Maxima言語で記述された函数は,裏の LISPで解釈されて実行されます. これを LISPの
函数にすれば,解釈する手間は省ける分,処理の高速化が望めます.
引数は 〈函数1〉, · · · , 〈函数n〉 の様に利用者定義函数を直接指定する方法に加えて,引数に allや

functionsを指定して,利用者定義函数を一度に変換する事も出来ます.
変換される函数が内部で局所変数を利用する場合,mode declare函数を用いて,その局所変数の

型を宣言する必要があります. この場合,block文で局所変数を宣言した直後に mode declare函数
による局所変数の型の宣言を入れます.この宣言の方法を以下に示しておきましょう.

mode declareの使い方¶ ³
f( x1, x2, · · · ) := block([ 局所変数1,局所変数2, · · · ],

mode declare( 局所変数1,型1,局所変数2,型2, · · · ),
〈函数本体 〉 )µ ´

このmode declare函数の詳細に関しては,1.5.4を参照して下さい.
次に,函数を translate函数で変換すると,大域変数 savedefが falseの場合,変換された函数の名

前は大域変数 functionsに割当てられた函数名リストから削除されて, 今度は大域変数 propsに割
当てられたリストに函数名が追加されます.
当然の事ですが,函数は虫取りが完遂されるまで変換すべきではありません. 更に,translate函数

は変換する函数内部の式が簡易化されていると仮定しています. そうでなければ最適化されていな
い LISP函数が生成されてしまうので,変換する意味が半減するかもしれません.
その為,大域変数の simpを falseに設定して変換式の簡易化を禁じる事をしてはいけません.
注意すべき事に,translate函数を用いて LISPの函数に変換しても,Maximaと LISP の整合性の

問題から以前と同じ動作をする保証はありません.
translate file函数はMaxima言語で記述したプログラムを含むファイルを LISP函数のファイル

に変換する函数です.translate fileはMaximaのファイル名,LISPのファイル名と translate fileが
評価した引数の情報を含むファイル名を成分とするリストを返します.
最初の引数はMaximaファイルの名前で,オプションの第二の引数は生成すべき LISPファイル

名です.第二の引数は第一引数に,trispのデフォルト値の tr output fileの値を第二ファイル名のデ
フォルト値として与えます. 例えば, translate file(”test.mc”)) でファイル test.mcを LISPファイ
ルの test.lispに変換します.
更に,生成されるものには translate函数が出力した様々な重要性の度合を持った警告メッセージ

のファイルがあります.第二ファイル名は常に UNLISPです. このファイルは変数を含み,それに
は変換されたコードでのバグ追跡の為の情報が含まれています.
変換に関連する大域変数は他の函数と比較しても多く,その上,名前も長いものが多い為,名前と

綴を憶えるのは大変ですが, apropos(tr ) を実行すれば, tr で開始するMaximaの大域変数等のリ
ストが出力されるので,このリストを出して名前を確認すると良いでしょう.

コンパイルを行う函数¶ ³
compile ( 〈函数1〉, · · · , 〈函数n〉 )
compile (functions)
compile (all)
compile file( 〈ファイル 〉, 〈コンパイルされたファイル 〉, 〈LISP のファイル名 〉 )
compile file( 〈ファイル 〉, 〈コンパイルされたファイル 〉 )
compile file( 〈ファイル 〉 )
compfile( 〈ファイル 〉, 〈函数1〉, · · · , 〈函数n〉 )µ ´

compile 函数は指定した Maxima の処理言語で記述した函数を LISP の函数に変換し, それを
LISPの函数 compileを用いてコンパイルします.尚,compile函数は函数名リストを返します.尚,
引数に functionsや allを指定すると利用者定義函数を全てコンパイルします.
これに対し,compile file 函数は指定したファイルのコンパイルを行います. 先ず, 指定された

〈ファイル 〉 にはMaximaのプログラムが含まれており,compile file函数は,これを LISP函数に
translate函数で変換し, その結果を compile函数でコンパイルします.変換とコンパイルに成功す
ると, 今度は結果をMaximaに読込みます.

compile fileは四個のファイル名のリストを返します.このリストに含まれるファイル名は,元の
Maximaプログラムファイル, LISPへの変換ファイル,変換に関する註釈ファイルと compileでコ
ンパイルされたプログラムのファイルです.尚,コンパイルに失敗すると,返却されるリストの第四
成分は falseになります.

compfile函数は 〈函数1〉, · · · , 〈函数n〉 を LISPの函数に変換し, 〈ファイル 〉 に書込みます.

(%i28) neko(x):=sin(x);

(%o28) neko(x) := sin(x)

(%i29) compfile("mike",neko);

Translating neko

(%o29) /home/yokota/mike

(PROGN (DEFPROP NEKO T TRANSLATED) (ADD2LNC ’NEKO PROPS)

(DEFMTRFUN (NEKO ANY MDEFINE NIL NIL) (X) (DECLARE (SPECIAL X))

(SIMPLIFY (LIST ’(%SIN) X))))

translate函数と compile函数によるシステムに関連する大域変数を纏めておきましょう.

translate函数と compile函数のシステムに関連する大域変数¶ ³
変数名 初期値 概要

compgrind false compile函数による出力制御
savedef true translate函数による変換後に元の函数を残す
translate false translate函数による自動変換を制御
transrun true 変換前の函数の実行

undeclaredwarn compfile 未定義変数に対する警告を制御µ ´
大域変数 compgrindが trueであれば,compileによる函数定義の出力が整形表示されます.
大域変数 savedefが trueであれば,利用者函数を translate函数で変換しても, 元のMaximaの

プログラムを残します.その為,大域変数 functionsに割当てられた利用者函数名リストから,変換
した函数名を削除せずに残します. 又,dispfun函数で函数の定義が表示可能で,函数の編集も出来
ます.

falseの場合は,functionsに割当てた利用者函数名リストから該当函数名を削除します.
大域変数 translateが trueであれば,利用者定義函数が自動的にLISP函数に変換されます.尚,Max-

imaと LISPの整合性の問題から,変換される前と同じ動作をするとは限らない事に注意して下さい.
変数がmode declareされた CRE表現の場合,rat函数を一つ以上の引数で用いたり,ratvars函数

を使ってはなりません.又,prederror:falseは変換しません.
大域変数 transrunが falseであれば,translate函数で変換されたものではなく, 元のMaximaの

函数 (それらが存在していれば)が実行されます
大域変数 undeclaredwarnには四種類の設定項目があります.

undeclearewarnの設定項目¶ ³
設定 動作

false 警告を表示しません

compfile compfileであれば警告します
translate translateや translate:trueであれば警告します
all compfileや translateであれば警告しますµ ´

mode declare( 〈変数 〉 ,any)を実行して 〈変数 〉 が一般のMaximaの変数である事を宣言しま
す.即ち,float,又は fixnumである事に限定されません.compile函数でコンパイルされる利用者定義
函数中の変数を宣言する特別な動作は全て無効にしなければなりません.

次に,translate函数に影響を与えるた大域変数は非常に多くあります. 最初に,大域変数 tr state vars
に割当てられたリストに含まれる大域変数を以下に纏めておきます.

tr state varsに纏められた変数¶ ³
変数名 初期値 概要

define variable []
transcompile false compile函数に必要な宣言を自動生成
translate fast arrays true 配列変換を制御

tr array as ref true 変換函数の配列評価を指定

tr function call default general 函数変換を制御

tr numer false 数値変数の型を制御

tr semicompile false 機械コードへの翻訳を制御

tr warn fexpr compfile fexpr型に対する警告制御
tr warn meval compfile meval型函数に対する警告制御
tr warn mode all 変数型に対する警告を制御

tr warn undeclared compile 未宣言変数に対する警告を制御

tr warn undefined variable all 未定義変数に対する警告を制御µ ´
大域変数 transcompileが trueであれば,translate函数は可能な compile函数に必要な宣言を生

成します.compile函数は transcompile:trueを用います.
大域変数 translate fast arraysが trueの場合,配列の変換を行います. 大域変数 tr array as ref

は,translate fast arraysが falseの場合のみに translate file函数で変換されたプログラムの配列参
照に影響を与えます.大域変数 tr array as refが trueであれば,変換された函数は配列を評価しま
すが,falseであれば,変換されたプログラム中の単なるシンボルとして配列が現れます.
大域変数 tr function call defaultは apply,expr,generalと falseを取り, 初期値は generalとなり

ます.

• falseの場合:Maxima内部函数mevalを呼出して式を評価する事を意味します.

• exprの場合:引数固定の LISP函数と仮定します.

• applyの場合:apply函数を用いて函数を引数に作用させて変換します.

• generalの場合:内部表現がmexprsとmlexprsに対しては良いコードを与えますが,macrosに
対しては駄目です.generalの場合,コンパイルされた函数中で変数の割当が正確である事を保
証します.例えば,函数 f(x)を変換する際に,ここで fが値を割当てられた変数であったとす
ると,警告を出して,apply(f,[x])の事として函数を変換します.

尚,デフォルト設定で何等の警告メッセージが無ければ,translate函数で変換し, compile函数で
コンパイルした利用者定義函数には元のmaxima函数と完全な互換性がある事を意味します.

tr numerが trueであれば,数の属性をそのまま LISPの変数にも継承させます.
tr optimize max loopは考えられる形式で translate函数でのマクロ展開と最適化工程ループの

最大回数を定めます.これマクロ展開エラーを捉える為で, 非中断の最適化属性です.

tr semicompileが trueであれば,translate file函数と compile函数の出力形式は拡張されたマク
ロになりますが,LISPコンパイラで機械コードに翻訳されたものにはなりません.

tr warn fexprは,内部形式が fexpr型のものが与えられると警告します. fexpr型は通常変換さ
れたプログラム内の出力であってはならず,全ての文法的に正しい特殊なプログラム書式に変換さ
れます.

tr warn mevalは函数mevalが呼び出されると警告します.mevalが呼出されると,変換の問題点
を指定します.

tr warn modeは,変数が指定した型に対して適切でない値が指定されていれば警告します.
tr warn undeclaredが未宣言の変数に関する警告を ttyに送るべき時を決めます.
tr warn undefined variableは未宣言の大域変数が存在する場合に警告します.
以下に,大域変数 tr state varsにも含まれない translate函数に関連する大域変数を纏めておき

ます.
LISP函数への変換に関連する大域変数¶ ³

変数名 初期値 概要

tr bound function applyp true 変換函数の割当てに対し警告

tr file tty messagesp false メッセージ出力制御

tr float can branch complex true 逆三角函数の複素数値の制御

tr optimize max loop 100 マクロループの回数

tr warn bad function calls true 変換時の警告を制御µ ´
tr bound function applypが trueの場合,函数の引数として割当てられた変数が,函数として用

いられている場合に警告します.例えば, g(f,x):=f(x+1)の様な場合です.
tr file tty messagespは,translate fileがファイルの変換を行う間に生成されたメッセージを tty

に送るかどうかを決めます. falseであれば,ファイルの translateによる変換に関するメッセージ
は UNLISPファイルのみに挿入されます.trueであれば,メッセージは ttyに送られ, UNLISPファ
イルにも挿入されます.

tr float can branch complexは逆三角函数が複素数値を返しても良いかどうかを宣言します.逆
三角函数は sqrt,log, acos等です.trueの場合,xが float(浮動小数点型)であったとしても,acos(x)は
any型となります.falseにしている時は,xが float型で,その時に限って,acos(x)は float型となり
ます.

tr warn bad function callsは,変換時に不適切な宣言が行われた為に,函数の呼出しが生じた場
合に警告します.

3.2.5 函数定義に関連する函数

funmake函数¶ ³
funmake( 〈関数 〉, [〈引数1〉, · · · , 〈引数n〉] )µ ´

funmake函数は 〈関数 〉 で指定した函数を呼出して評価を行う事はしません.単純に,〈関数 〉 (
〈引数1〉, · · · , 〈引数n〉 )を返すだけです.

(%i2) funmake(f,[x,y,z]);

(%o2) f(x, y, z)

(%i3) funmake(neko,[x,y,z]);

(%o3) neko(x, y, z)

(%i4) funmake(expand,[128,"うちのタマ知りませんか？"]);

(%o4) expand(128, うちのタマ知りませんか？)

(%i5) funmake(a,[1,2,3]);

(%o5) a(1, 2, 3)

(%i6) a:10;

(%o6) 10

(%i7) funmake(a,[1,2,3]);

Bad first argument to ‘funmake’: 10

-- an error. Quitting. To debug this try debugmode(true);

(%i8) funmake(’a,[1,2,3]);

(%o8) a(1, 2, 3)

関数に指定したアトムに値が束縛されている場合,そのアトムは内部で評価される為,エラーに
なります.この場合は,単引用符’を先頭に付けて名詞型とすれば問題ありません.

tr warnings get函数は translate函数による変換中に translate函数が出力する警告のリストを
表示します.

declare translated函数¶ ³
declare translated( 〈函数1〉, · · · , 〈函数n〉 )µ ´

declare translated函数は,引数の函数が既に変換されている事を宣言する函数です.Maximaの
プログラムファイルを LISPに変換する際に,そのファイル中のどの函数が translate函数で変換
された函数,或いは compile函数でコンパイルされた函数として呼出されるべきか,そして,どれが
Maximaの函数で, 又,未定義のものであるかを知る事は translate函数にとって重要な事です.
ファイルの先頭にこの宣言を置くと,ある記号がたとえ LISP函数の値を持っていなかったとし

ても,呼出された時にそれを持つ事を教えます. (mfunction-call fn arg1 arg2, · · · )が生成されるの
は, 〈函数n〉 が LISP函数に変換されるべきものであるかを translate函数が知らない時です.

局所変数を定義する函数¶ ³
local( 〈局所変数1〉, · · · , 〈局所変数n〉 )µ ´

local函数は,この函数が利用される文中で, 〈局所変数1〉, · · · , 〈局所変数n〉 を全ての属性に対し
て局所的なものにします.localは block文,函数定義の本体,lambda式,又は ev函数でのみ一度だ
け使えます.この local函数は文脈からも独立しています.

3.3 データ入出力について

3.3.1 データの入出力

この節ではMaximaのデータ入力と出力について述べます. Maximaは Common LISPで記述
されている為,基本的に入出力は LISPの入出力函数を用いたものです.特にMaximaは表示され
ている形式と内部形式が別物の為,結果や式を保存したり,逆に保存したものを読込む際には,注意
が必要になります.
ファイルに画面と同じ出力を行いたければ,writefileを用います. このwritefileは LISPの dribble

函数を用いたもので,入力と出力をそのまま指定したファイルに保存します.但し,writefileでは指
定したファイルを新規に生成するので,単純に既存のファイルに記録したければ,appendfile函数を
用います.writefileと appendfileで開いたファイルを閉じる場合は closefile()で開いたファイルを
閉じます.
但し,これらのファイルは実質的に記録ファイルであって,Maximaでそのまま再利用は出来ませ

ん.再利用可能なファイルを生成するのは,saveと stringout です.ここで,saveは Maximaの内部
表現を保存する函数で,loadや loadfileを用いてMaximaに読込みます.これに対して,stringoutや
grindは内部形式ではなく, Maximaの入力に対応する通常の形式でデータの保存を行います.

MaximaでCの scan命令に似たものに,readと readonlyがあります. これらの函数は,引数として
与えた文字列を全て同一行に表示し,キーボードからの入力を待ちます.利用者は通常のMaximaの
入力と同様に式を入力します. ここで,行末には;か$を付けます.すると,readの場合は式をMaxima
で評価し, readonlyの場合は評価せずにそのまま受け取ります.
この様に,単純なファイル操作に限定されるとは言え,必要なものは一応揃っており,足りない部

分は LISPで補う事になります.

ファイル処理に関連する大域変数¶ ³
変数名 初期値 概要

batchkill false 以前のバッチファイルの影響を無効にする

batcount 0 式番号を記録

file search Maxima .macファイル読込ディレクトリ
file search LISP .lispファイル読込ディレクトリ
file search demo .demファイル読込ディレクトリ
file string print true ファイル名の表示を決める

loadprint false 読込に伴うメッセージ表示を制御

packagefile false パッケージ作成時の情報を制御µ ´
batchkillが trueの場合,バッチファイルを読み込む際に kill(all)と reset()が自動的に実行される

ので,以前のバッチファイルの影響が全て無効になります. batchkillが他のアトムであれば,batchkill
の値で killが実行されます.

batcountにはファイルからのバッチ処理された最後の式の番号を設定します. batcon(batchcount-
1)は以前の処理から,最新の batch処理された式の処理結果を保存します.

file search maxima,file search lispと file search demo は loadや他の函数で,ファイルの読込を
行う際に検索されるディレクトリのリストです.

file search maximaはMaximaのプログラム (末尾が,macとmc)向け, file search lispはLISPの
プログラム (末尾が,fas,lisp,lsp)向け, file search demoがLISPのプログラム (末尾が,dem,dm1,dm2,
dm3,dmt) 向けとなっており, 各々が Maxima のリスト形式となっています. 尚, これらの値は,
src/init-cl.lispで設定されています.

file string print は true であれば, ファイル名は文字列, false であれば, リストとして出力され
ます.

loadprint はファイル読込に伴うメッセージ表示を制御する大域変数です. loadprint が取る値
は,true,loadfile,autoload,falseの四種類あり, 各々,対応が異なります.

• trueであれば,メッセージが常に表示されます.

• loadfileの場合は,loadfile命令が用いられた時のみに表示されます.

• autoloadの場合は,ファイルが自動的に読込まれた時のみに表示されます,

• falseの場合,メッセージが決して表示されません.

packagefileは saveや translateを用いてパッケージ (ファイル)を作成する時に, packagefile:true
と設定すると,ファイルが読み込まれる時点で必要な場所を除いた情報がMaximaの大域変数,例
えば values,functionsに追加される事が避けられます.
この方法でパッケージに含まれる物は,利用者のデータを付け加えた時点で, 利用者の側では得

られません.これは名前の衝突の問題を解決するものでは無い事に注意して下さい.この大域変数
は単にパッケージファイルへの出力に影響を与える事に注意して下さい.尚,この変数の値を true
に設定すると, Maximaの初期化ファイルの生成でも便利です.

3.3.2 ファイル処理に関連する函数

バッチ処理に関連する函数¶ ³
batch(〈ファイル名 〉 )
batchload(〈ファイル名 〉 )
batcon(〈引数 〉 )µ ´

batch函数は指定されたファイルに含まれるMaximaの命令行を逐次評価します. ファイルはパ
スを含まない場合,file search Maximaに含まれるディレクトリ上を検索し,存在した場合には読込
みと実行をします.
ファイルの内容は基本的にMaximaでの入力行と同じもので,行末には;か$を置きます.又,%と

%thを用いて入力と出力を指定する事も出来ます. 尚,空行,Tabや改行コードは無視されます.
batch処理ファイルは通常のテキストエディタで編集する事も出来ますし, Maximaの stringout

函数で出力したものも使えます.
深刻なエラーが生じた場合,ファイル末端に達した場合にのみ,利用者に制御が戻されます.但し,

利用者はどの時点でも Cntrl-gを押せば,この処理を止められます.
batchload函数は指定されたファイルのバッチ処理を行います.batchとの違いは, batchloadでは

ファイルに記述された式の入力や出力表示等を行わない事です.
batcon函数は中断されたファイルのバッチ処理を再開します.

読込を行う函数¶ ³
load(〈ファイル名 〉 )
loadfile(〈ファイル名 〉 )
read(〈文字列1, · · ·〉 )
readonly (〈文字列1〉 ,· · ·)µ ´

load函数は文字列や,リストで表現されたファイル名の読込みを行います. ディレクトリが指定
されていなければ,最初にカレントディレクトリ, それから file search Maxima,file search lispや
file search demoといった大域変数に保存されているディレクトリを検索し,指定されたファイルを
読込もうとします.

load函数はファイルが batch処理に対応している事を見付けると,batchloadを用います (これは,
黙って端末に出力やラベルを出力せずにファイルの batch処理を実行する事を意味しています).
他のファイルの読込を行うMaxima命令に,loadfile,batchと demo があります.loadfileは saveで

書込んだファイルに対して動作し,batchと demo は strignoutで書込まれたり,テキストエディタ
で命令のリストとして生成されたファイル向けです.

loadfile函数は指定されたファイルを読込みます.この函数は以前のMaximaの処理で save函数
で保存した値をMaximaに戻す事に使えます.
ここで,パスの指定はオペレーティングシステムのパスの指定に方法に従います. 例えば,unixの

場合は/home/userディレクトリにある foo.mcファイルを読込むのであれば,”/home/user/foo.mc”
となります.

尚.save函数で保存したファイルを loadfileで読込むと,Maximaは初期化されるので注意が必要
です.

read函数は画面上に全ての引数を表示して入力を待ちます. 利用者が式を入力すると,入力した
式はMaximaに引渡されて評価が行なわれます.

(%i45) a:read("mikeneko");

mikeneko

diff(x^2+1,x);

(%o45) 2 x

この例では,mikenekoと表示された後に、diff(x^2+1,x);を入力しています.ここでの入力で
も通常の入力と同様に行末に;か$が必要です.この例では,入力した値がMaximaに評価されて,結
局,aに 2*xが割当てられています.

readonly函数は引数を全て表示し,それから式を読込みます.基本的には readと同様ですが,read
と違うのは,読込んだ式を評価しない事です.

(%i46) a:readonly("mikeneko");

mikeneko

diff(x^2+1,x);

2

(%o46) diff(x + 1, x)

stringout函数¶ ³
stringout(〈ファイル名 〉 ,〈式1〉 , 〈式2〉 ,· · ·)
stringout(〈ファイル名 〉 , [〈m〉 ,〈n〉 ])
stringout(〈ファイル名 〉,input)
stringout(〈ファイル名 〉 ,functions)
stringout(〈ファイル名 〉 ,values)µ ´

stringout函数は指定したファイルに Maximaが読込める書式で出力します. 直接,〈 式1〉 ,〈式
2〉, · · · と式を並べると,各式が順番にファイルに書込まれます.
引数に,[〈m〉 ,〈n〉 ]と指定すると入力行のm行から n行がファイルの書込まれます.
これらに対し,inputを指定すると入力行全てが書込まれます. functionsを指定すると大域変数

functionsに記載された利用者定義の関数が全て保存されます.
同様に valuesを指定すると,大域変数 valuesに記載された利用者定義の変数の値が書込まれます.
尚,この stringout函数は writefileを実行中に利用する事も可能です.
大域変数の grindが trueであれば,stringoutは文字列では無く,grindと同じ書式で出力します.

書込みを行う函数¶ ³
appendfile(〈ファイル名 〉 )
writefile(〈ファイル名 〉 )
with stdout(〈ファイル名 〉,〈式1〉,· · ·, 〈式n〉)
closefile()µ ´

appendfile函数は指定したファイルにMaximaの入出力の追加を行います. writefile函数との違
いは,同名のファイルが存在した場合,writefile函数は上書きをしてしまいますが,appendfile函数は
既存のファイルの末尾にMaximaの入出力を追加する事が違います.尚,writefile函数と同様に指定
したファイルは closefile()函数で閉じます.

writefile函数は書込み用にファイルを新規に開きます. writefile函数を実行すると,それ以降の
Maximaへの入力と出力処理は全て指定したファイルに記録されます.その為,このファイルをその
ままMaximaに再度読込ませたりする事は出来ません.
ファイル名の指定は文字列で行います.文字列ではなく,ABCDの様に二重引用符無しで指定す

ると,Maximaはアトムの内部表現で用いる$を頭に付けたファイル名,即ち,この例では$ABCDで
ファイルを生成します. 尚,この writefileの実体は LISPの dribble函数です.
ファイルを閉じる場合は closefile()を用います.
以下に簡単な例を示します.

(%i1) writefile("test1");

(%o1) #<OUTPUT BUFFERED FILE-STREAM CHARACTER test1>

(%i2) 1+2+3;

(%o2) 6

(%i3) diff(sin(x)*x+2,x);

(%o3) sin(x) + x cos(x)

(%i4) closefile();

(%o4) #<CLOSED OUTPUT BUFFERED FILE-STREAM CHARACTER test1>

上記の writefileで生成したファイル test1の内容を以下に示します.

;; Dribble of #<IO TERMINAL-STREAM> started 2005-11-17 06:31:16

(%o1) #<OUTPUT BUFFERED FILE-STREAM CHARACTER test1>

(%i2) 1+2+3;

(%o2) 6

(%i3) diff(sin(x)*x+2,x);

(%o3) sin(x) + x cos(x)

(%i4) closefile();

;; Dribble of #<IO TERMINAL-STREAM> finished 2005-11-17 06:31:40

この様に,Maximaの画面入出力そのままが保存されています. この writefile函数を記録ファイ
ルの生成に利用すれば,下手なフロントエンドも不要になります.

with stdout函数は,指定されたファイルを開き,〈式1〉,· · ·, 〈式n〉の評価と書込みを行います.各式
の評価の標準出力への任意の出力は端末の代りに指定したファイルに送られ,端末側には常に false
が返されます.

closefile函数は closefile()で appendfileや writefileで開かれたファイルを閉じます.closefile函数
は LISPの close函数を使った函数です.この close函数は開かれたストリームを閉じる函数です.

save函数¶ ³
save(〈ファイル名 〉 ,〈引数1〉 , 〈引数2〉 ,· · ·)
save(〈ファイル名 〉 , 〈名称1〉 = 〈式1〉 , 〈名称2〉 = 〈式2〉 ,· · ·)
save(〈ファイル名 〉 ,[〈m〉 ,〈n〉 ])
save(〈ファイル名 〉 ,values,functions,labels, · · ·)
save(〈ファイル名 〉 ,all)µ ´

save函数は指定したファイルに,指定した式や関数等の値を書込みます. 更に,保存した値は削除
されずにMaxima本体にも残っています.
〈引数1〉 ,〈引数2〉 ,· · ·で,各引数の値を保存します.
[〈m〉 ,〈n〉]で m 番目の入力行から n 番目の入力行の内容を保存します.
引数に大域変数 values,functions,labels等を指定する事も可能です. values,functionsで利用者が

設定した変数値や定義関数を全て保存します. 又,labelsを指定すると,入出力行の内容が全て保存
します.
最後に,引数に allを指定すれば,Maximaの内容をファイルに保存しますこの場合は,入力や計算

結果だけではなく,Maximaの様々な設定も一緒に保存されるので,処理した内容以上にファイルが
大きなものとなるので注意が必要になります. save函数の返却値は保存先のファイル名となります.

(%i1) 1+2+3;

(%o1) 6

(%i2) a1:x^2+y^2+1;

2 2

(%o2) y + x + 1

(%i3) resultant(x-t,y-t^2,t);

2

(%o3) y - x

(%i4) save("test",all);

(%o4) test

save函数で保存したファイルは loadfile函数でMaximaに再び読込む事が出来ます. 但し,loadfile
を実行すると,以前のMaxima自体を初期化し,save函数を実行した時点にまで戻してしまう効果
があるので,注意が必要になります.

以下の例では最初に loadfileでファイル testを読込んでいますが, 行ラベルは上の saveで保存す
る場合と同じものになっている事と,二度目に loadfileを実行するとラベルが (%i8)から (%i5)に
戻っている事に注意して下さい.

(%i1) loadfile("test");

(%o4) test

(%i5) %i2;

2 2

(%o5) a1 : y + x + 1

(%i6) %i1;

(%o6) 6

(%i7) %o3;

2

(%o7) y - x

(%i8) loadfile("test");

(%o4) test

(%i5)

尚,saveファイルの内容は,LISPの S式そのものとなります.要するに,LISP上で動くMaximaの
為のデータファイルになります.ファイルの先頭側に実行内容の内部形式が記述されますが,その
後にはMaximaの諸設定が保存されます.その為,以下に示す例は 1+2+3から 4行の入力だけでし
たが,save函数で保存したファイル (test)は 256行に及びます.これは内部形式で記述するとどうし
ても長くなる側面もありますが, 実際は,入出力以外に様々な設定 (大域変数の値等)も保存されて
いる為です.

;;; -*- Mode: LISP; package:Maxima; syntax:common-lisp; -*-

(in-package "MAXIMA")

(DSKSETQ $%I1 ’((MPLUS) 1 2 3))

(ADDLABEL ’$%I1)

(DSKSETQ $%O1 6)

(ADDLABEL ’$%O1)

(DSKSETQ $%I2 ’((MSETQ) $A1 ((MPLUS) ((MEXPT) $X 2) ((MEXPT) $Y 2) 1)))

(ADDLABEL ’$%I2)

(DSKSETQ $%O2 ’((MPLUS SIMP) 1 ((MEXPT SIMP) $X 2) ((MEXPT SIMP) $Y 2)))

(ADDLABEL ’$%O2)

(DSKSETQ $%I3

’($(RESULTANT) ((MPLUS) $X ((MMINUS) $T))

((MPLUS) $Y ((MMINUS) ((MEXPT) $T 2))) $T))

(ADDLABEL ’$%I3)

(DSKSETQ $%O3

’((MPLUS SIMP) ((MTIMES SIMP) -1 ((MEXPT SIMP RATSIMP) $X 2)) $Y))

(ADDLABEL ’$%O3)

(DSKSETQ $%I4 ’($(SAVE) &TEST $ALL))

(ADDLABEL ’$%I4)

(DSKSETQ $A1 ’((MPLUS SIMP) 1 ((MEXPT SIMP) $X 2) ((MEXPT SIMP) $Y 2)))

(ADD2LNC ’$A1 $VALUES)

以下略

その為,Maxima内部でデータがどの様に処理されているかを見る場合には重宝するでしょう.と
は言え,作業を一旦中断し,中断した個所から再度処理を行う必要がなければ,save以外の命令,例
えば,stringoutや grindを用いた方が総合的な使い勝手自体は良いでしょう.

その他の函数¶ ³
filename merge(〈文字列1〉 ,〈文字列2〉 )
file search(〈ファイル名 〉 )
file type(〈ファイル名 〉 )µ ´

filename merge 函数は二つの 〈 文字列1〉 と 〈 文字列2〉 の結合を行います. 内部的には, 先頭に
#Pを́文字列の先頭に付けた対象を生成しますが,Maxima上では,単純に文字列を繋ぎ合せた様に
しか見えません.
基本的にはMaximaの各種命令でファイルの検索を行う際にパス指定のあるファイル名を生成

する際に用いられる函数です.
file search函数は指定したファイルを file search lisp,file search maximaと file search demoに

含まれるディレクトリ上で検索し, ファイルが存在すればファイル名を返し,存在しなければ,false
を返します.

file type函数は指定したファイルの属性を返します.但し,ファイル名の末尾で判断する函数なの
で,返却する値も,object,lispやmaximaを返します. ここで,objectはコンパイルされた LISPファ
イル,lispは LISPのテキストファイル, maximaはMaximaの処理言語で記述されたファイルにな
ります.

第4章 Maximaのシステム回りの函数

この章で解説する事:

• システムの初期化

• 処理の中断

• ラベルの参照

• システムの状態を調べる

• 外部プログラムの起動

• 式の変換を行う函数

• システムに関連する大域変数

• Maximaの終了

4.1 システムの初期化

4.1.1 maxima-init.macファイル

Maximaは起動時にカレントディレクトリ上にあるファイル maxima-init.macを自動的に読込
みます.

maxima-init.mac内部にはMaximaの函数や大域変数の設定が記述可能です. 更に,setup autoload
函数や load函数を用いて,必要なパッケージの読込も自動的に行えます.
以下に非常に簡単な例を示します.

/*-*-MAXIMA-*-*/

showtime:all;

put(surfg, d_chain_bisection,root_finder)$

put(surfg, 0.0000000001,epsilon)$

put(surfg, 20000,iterations)$

put(surfg, 500,width)$

put(surfg, 500,height)$

put(surf, yes,do_background)$

put(surf, 5,background_red)$

put(surf, 5,background_green)$

put(surf, 5,background_blue)$

put(surf, 0.14,rot_x)$

put(surf, -0.3,rot_y)$

setup_autoload("surfplot.mc",surfplot)$

load("fox.mc")$

Maximaが読込むファイルにて,註釈は Cと同様に/* */の中に記述します. この例では頭に註
釈行として/*-*-MAXIMA-*-*/と置いてMaxima言語のファイルである事を示しています.その後
に,大域変数の設定や,属性の設定等を行っています.
ここで,setup autoload函数や load函数を用いたファイルの読込について簡単に解説しておきま

しょう.この setup autoload函数の構文を以下に示しておきます.

setup autoload函数¶ ³
setup autoload (〈ファイル 〉, 〈函数1〉, · · · , 〈函数n〉 )µ ´
この setup autoload函数は,指定した函数が呼出された時点で函数が未定義の場合, 指定したファ

イルの読込を実行します.ここで,setup autoload函数は引数として配列函数を扱えない事に注意し
て下さい.
この例では,ファイル surfplot.mcに含まれている surfplot函数が呼出された時に, 未定義であれ

ば,ファイル surfplot.mcを読込む設定となっています.



4.1. システムの初期化 207

load函数の場合は予め指定されたファイルの内容を全てMaximaに読込みます. この場合,ファ
イルの定義,未定義とは無関係に読込を実行します.
このmaxima-init.macファイルを利用する事で,Maximaの起動時の環境が容易に変更出来ます.

4.1.2 セッションの初期化

Maximaでは,reset函数を用いてMaxima起動時の初期状態に戻す事や kill函数を用いて,処理
で蓄えたラベルや大域変数等を破棄する事が可能です.

reset函数¶ ³
reset()µ ´
この rest函数は引数無しで, reset(); と入力します.reset函数を用いると, 入出力の基数は 10に,

係数環はZに,ラベルと大域変数 fpprecはMaximaの初期状態に戻されます.但し,完全にMaxima
を起動した状態に戻る訳ではなく, 利用者が定義した大域変数とその値,以前のラベルに保存され
ている値を削除しません.尚,ラベルのカウンタが初期化されている為,ラベルに保存されている値
は, 処理を進める内に上書きされて消去されます.

kill函数¶ ³
kill(〈引数1〉, 〈引数2〉, · · ·)
kill(〈整数 〉)
kill([〈整数1〉,〈整数2〉])
kill(all)
kill(allbut(〈引数1〉, 〈引数2〉, · · ·))µ ´

kill函数は reset函数とは異なり,必ず引数を取ります.kill函数は引数によって削除する対象が異
なります.

• 〈引数i〉 が変数 (単配列要素を含みます),函数, 又は配列の場合,指定された項目は,その属性
の全てと一緒にMaximaから消去されます.

• 〈引数i〉 が labelsであれば,古くなった全ての入出力行,中間行が消去されます.

• 〈引数i〉 が clabelsであれば,入力行だけが消去されます.

• 〈引数i〉 が elabelsであれば,中間行のみが消去されます.

• 〈引数i〉 が dlabelsであれば,出力行のみが消去されます.

• 〈引数i〉 が任意の他の情報のリスト,即ち,大域変数 infolistsに含まれる名前であれば,そのク
ラスとその属性全ての項目が削除されます.

infolistsに含まれる valuesや variablesに対応して,kill(values)や kill(variables) で同じ式を指定
するラベルが消去される迄,メモリの占有領域を解放しない事に注意して下さい.例えば,大きな式

が%i10行の変数 xに対して割り当てられていた場合, 占有された保存領域を解放する為には,kill(x)
と kill(%o10)も実行しなければなりません.

kill(all)で入出力行,中間全てを削除します. これに対し,kill(allbut(· · ·)は,allbutで指定したも
のを除外して kill(all) を実行します.但し,allbutで指定する引数は,functions等の大域変数 infolists
に含まれる項目ではなく,より具体的なものです.

kill 函数は与えられた引数から全ての属性を削除する為,kill(values) は大域変数 values に割当
てられたリストの全ての項目に関連する属性を削除しますが, それに対して remove函数群 (rem-
value,remfunction,remarray,remrule)は指定した属性のみを削除します.ここで,remove函数群は
リスト名か指定した引数が存在しなければ falseを出力しますが,これに対して,kill函数は指定した
項目が存在しない場合でも,常に doneを出力します.

collapse函数¶ ³
collapse (〈式 〉)
collapse ([〈式1〉, · · · , 〈式n〉])
collapse (listarray(′〈配列 〉))µ ´
全ての共通 (つまり,等しい)部分式を共有する (つまり,同じセルを用いる)事で引数を潰し,領

域を節約します.因に,collapse函数は optimize函数でも用いられています. saveで保存したファイ
ルを読込んだ後に collapse函数を呼出すと良いでしょう. 又,配列 aに対し,collapse(listarray(’a))
とする事で,配列の成分を潰す事も出来ます.

4.2 処理の中断

4.2.1 中断の制御文字

Maximaの計算が異常に長い場合,間違って計算を実行させた場合,Maximaを一旦中断する必要
が生じます.Maximaの計算を中断したければ,通常は制御文字ˆ c(Ctrl+C) を使います.Maximaは
ˆ z(Ctrl+Z)が入力されても中断しますが,この場合はMaxima を出て,UNIXの shellレベルに戻
るので,通常は Ctrl+Cを用います.

(%i11) factor(2137498127943870982374);

Maxima encountered a Lisp error:

EXT:GC: User break

Automatically continuing.

To reenable the Lisp debugger set *debugger-hook* to nil.

(%i12)

この例では,Ctrl+Cで因数分解を中断させています.

4.2.2 break函数による中断

Maximaでは break函数を用いてプログラムを意図的に中断させる事も可能です.
break函数¶ ³

break(〈式1〉,· · ·,〈式n〉)µ ´
break函数の返却値は 〈式n〉 の処理結果となります. break函数は引数の全ての式を処理した後

にMaxima-breakに入ります. Maxima-breakではプロンプトとして大域変数 prompt に設定した
文字列が表示されます. このMaxima-breakでは通常の函数が利用可能ですが,ev函数の省略型や,
ラベルを用いた処理が行えません.Maxima-breakから抜ける為には exit; と入力します.

(%i17) break(integrate(sin(x)*x,x),factor(x^2-1));

sin(x) - x cos(x) (x - 1) (x + 1)

Entering a Maxima break point. Type exit; to resume

_diff(sin(x)-x*cos(x),x);

x sin(x)

_expand((x-1)*(x+1));

2

x - 1

_exit;

(%o17) (x - 1) (x + 1)

(%i18)

この例では,
∫

xsin(x)dxとx2−1の因子分解を実行した後にMaxima-breakに入っています.Maxima-
break で出力されるプロンプトはデフォルトで となっています.Maxima-break から抜ける為に
は, exit; と入力します. すると,break函数は最後に処理した式の結果を返却値として返します.

4.3 ラベルの参照

Maximaは入力を入力ラベル%iに, 計算結果を出力ラベル%oに各々保存しています. これらの
値の参照は,入力の場合は%i〈ラベル番号 〉 で, 出力の場合は%o〈ラベル番号 〉で行います.
最新の結果は%で参照出来ます.これに対し,最新の入力は で参照します.
出力ラベル%oに割当てられた値を参照する函数に%thがあります.この%thは%th(6)の様に用

い,6個前の結果を参照します.更に,%i7や%o8とすれば,(%i7) 行で入力した式や,(%o8)に表示さ
れた結果を参照する事が出来ます.但し, はその様な使い方が出来ません.
入力と出力ラベルは kill(labels)で全て削除する事が可能です.これを実行すると,入力と出力ラ

ベルに割当てられた値は消去され, 各ラベルのカウンタも 1に戻されます,その為,入力は (%i1),出
力も (%o1)から開始する事になります.

(%i101) 1+2;

(%o101) 3

(%i102) resultant(x-t,y-t^2,t);

2

(%o102) y - x

(%i103) algsys([2*x+3*y=1],[x,y]);

2 %r2 - 1

(%o103) [[x = %r2, y = - ---------]]

3

(%i104) %;

2 %r2 - 1

(%o104) [[x = %r2, y = - ---------]]

3

(%i105) _;

(%o105) %

(%i106) %i101;

(%o106) 3

(%i107) %o101;

(%o107) 3

(%i108) %i102;

2

(%o108) resultant(x - t, y - t , t)

(%i109) kill(labels);

(%o0) done

(%i1)

この例では,様々な処理を行い,それらを%や で確認し,最後に kill(labels)でラベル (%iや%o)の
内容を全て消去しています.ラベルの消去を行った為に, kill(labels)を入力した (%i109)から,(%o1)
を経て (%i1)に初期化されている事に注目して下さい.
この他にも,ラベルの処理に関連する大域変数をMaximaは持っています.

ラベルに関連する大域変数¶ ³
変数名 初期値 概要

% 最新の処理結果

%% maxima-breakの間に処理された最新の値
inchar %i 入力ラベルで用いる文字を指定

outchar %o 出力ラベルで用いる文字を指定

linechar %t 中間表示の際に用いられる文字を指定

linenum 入力番号が設定されている

prompt _ demo函数のプロンプトを指定
nolabels false 入力値と計算結果をラベルへの束縛を制御µ ´
大域変数%%は大域変数%と同じ意味を持ちます.但し,大域変数%%は大域変数%が使えないblock

文内部のみで利用可能です.

(%i1) block(integrate(x^2,x,0,3),%*3);

(%o1) 3 %

(%i2) block(integrate(x^2,x,0,3),%*3);

(%o2) 9 %

(%i3) block(integrate(x^2,x,0,3),%%*3);

(%o3) 27

(%i4) %%;

(%o4) %%

(%i5) block(integrate(x^2,x),print(%%),%%*3,print(%%),

diff(%%,x),print(%%));

3

x

--

3

3

x

2

3 x

2

(%o5) 3 x

この例は大域変数%と大域変数%%の違いを説明するものです.最初に block文内部の積分を実
行して大域変数%に保存された式を 3倍にしますが,block文内部の大域変数%は block文を実行す
る直前の結果が設定されているので,block文内部で書換えられる事はありません.これに対し,大
域変数%%は block文内部の局所変数で, その値も block文内部のみで更新されます.その為,block
文が終了すると,値は消去されます.

大域変数 incharはMaximaの入力行ラベルで用いられる文字です.デフォルトでは入力行ラベ
ルは (%i1)の様に大域変数 incharで指定した%iの後に番号が続きます.
大域変数 outcharは出力ラベルで用いられる文字です.大域変数 incharと同じ使い方をします.
大域変数 linecharには中間表示に於ける式の前に置かれる文字を指定します.
大域変数 linenumにはその時点での入力行番号が割当てられています.

(%i17) linenum;

(%o17) 17

大域変数 promptは demo函数のプロンプト記号を指定する大域変数です. この大域変数の値は
playback(slow)や break函数でMaxima-breakに入った時点で表示されます.

(%i1) integrate(x^2-1,x);

3

x - 3 x

(%o1) --------

3

(%i2) prompt:"(;_;)";

(%o2) (;_;)

(%i3) playback(slow);

(%i1) integrate(%,x);

(;_;)

3

x - 3 x

(%o1) --------

3

(;_;)

(%i2) prompt:"(\;_\;)";

(;_;)

(%o2) (;_;)

(%o3) done

(%i4) break(x-1);

x - 1

Entering a Maxima break point. Type exit; to resume

(;_;)exit;

(%o4) x - 1

この例では大域変数 promptにMaximaの文字列 (;_;)を指定しています. playback(slow)を
実行すると,過去の入力と出力を一纏めにして出力し, 大域変数 promptを出力して Enterキーの
入力待ちとなります.
大域変数 nolabelsが trueの場合,入力値と計算結果をラベルに束縛しません. 即ち,%等で結果

や入力を参照出来なくします.この様にする事で,batch処理の空き領域を増す為に,kill(labels)を実
行しなくて済みます.

Maximaにはラベルの処理に関連する函数として,%th0函数と labels函数を持っています.

ラベル処理に関連する函数¶ ³
%th(〈正整数 〉)
labels(〈シンボル 〉)µ ´

%th函数は 〈正整数 〉番前の計算結果を取出します. 即ち,%th〈i〉を含む式の入力ラベルが%〈j〉
であれば,%th〈i〉は%i〈j − i〉の結果, 即ち,%o〈j − i〉の値となります.
この%thは batchファイルでは非常に便利です.これは%oラベルの値が batch ファイルをどの

時点で処理するかで異なるのに対し,%thはその函数を実行する時点を中心として指定した整数程
前の結果を返す事になるからです.

label函数は,シンボルを引数として取り,そのシンボルに対応するリストを返します.ここで,シ
ンボルとしては入力ラベル (デフォルトで%i),出力ラベル (%o)等のラベルが一般的ですが,labels
函数は引数として任意のシンボルを取ります.入出力等のラベルを設定する大域変数 inchar,outchar
や linecharを再設定すれば,labels函数はそのラベルに対応するリストを返しますが, ラベルの最初
の文字は labelsに与えた引数の最初の文字に適合します.

4.4 結果の表示

Maximaでは入力行に;を付けると,Maximaが評価した値が表示されます.数値の四則演算はデ
フォルトの場合,直ちに処理されて結果が表示されます.更にMaximaは結果を二次元的に表示す
るのがデフォルトとなっています.

(%i43) 2/5;

2

(%o43) -

5

(%i44) integrate(f(x),x,a,b);

b

/

[

(%o44) I f(x) dx

]

/

a

(%i45) expand((x+1)*(x-1));

2

(%o45) x - 1

この表示は式が小さなものであれば良いものですが,式が長くなると非常に判り難いものになり
ます.そこで,表示を 1行で済む様に指定が行える大域変数 display2d があります.この変数の値が
デフォルトの trueであれば,二次元的な表示を行い,falseを指定すると,結果を入力可能な書式で表
示します.

(%i4) display2d;

(%o4) true

(%i5) ’integrate(f(x),x);

/

[

(%o5) I f(x) dx

]

/

(%i6) expand((x+1)^3);

3 2

(%o6) x + 3 x + 3 x + 1

(%i7) display2d:false;

(%o7) false

(%i8) ’integrate(f(x),x);



4.4. 結果の表示 215

(%o8) ’integrate(f(x),x)

(%i9) expand((x+1)^3);

(%o9) x^3+3*x^2+3*x+1

Maximaにはこの他にも特殊な表示を行う函数があります.基本的にMaximaはキャラクター端
末しかなかった昔の手順に対応したシステムの為,積分記号の様に文字を使って数式を表示すると
いった,ある意味では涙ぐましい努力の跡があります. しかし,現在のWindowシステムから見る
と非常に古臭く感じるものが多いのが現状です.

4.4.1 表示に関連する大域変数

表示に関連する大域変数¶ ³
変数名 初期値 概要

ibase 10 入力数値の基数を指定

obase 10 出力数値の基数を指定

absboxchar ! 絶対値を描く際に用いる文字を指定

display2d true 結果の二次元表示の有無

display format internal false 内部表現に対応した表示切替

%edispflag false %eの羃の表示を指定
exptdispflag true 負の羃を分数で表示

pfeformat false 有理数の表示書式を決定

stardisp false 可換積演算子の表示を指定

linel 79 一行に表示する文字数

ttyoff false 出力の停止制御µ ´
大域変数 ibaseは入力数値の基数を指定します.値は十進数で 2から 35が指定可能です.10より

大きな値を設定した場合,Aから開始する大文字のアルファベットを用います.尚,大域変数 obase
は表示の際に用いる数値の基数を指定する大域変数となります.
大域変数 absboxcharは絶対値を描く際に用いる文字を指定します. 尚,絶対値は精々一行の高さ

しかありません.
大域変数 display2dが falseであれば,結果表示が二次元的書式ではなく,一行に収まる様に表示

します.長い式や複雑な式,表示に余裕が無い場合は有効です.
大域変数 display format internalが trueであれば,内部の数学的表現を隠した表示ではなく,内

部表現を反映した表示に切替ります.但し,内部表現そのもので表示する訳ではありません.ここで
の出力は part函数に対応するののではなく, inpart函数に準じたものになるとも言えます.

内部書式との比較¶ ³
入力 part inpart
a − b; a − b a + (−1)b
a/b; a

b ab−1

sqrt(x); sqrt(x) x1/2

x ∗ 4/3; 4x
3

4
3xµ ´

%edispflagは自然底%eの羃表示を定める大域変数です.デフォルトの falseの場合, %eの負の羃
は負の羃のまま羃表示されます.例えば,exp(-x)は%eの負の羃%eˆ (-x))で表示されます.trueの場
合,%eの負の羃は%eの羃の商の形式で表示されます.即ち,%eˆ (-x)は 1/%eˆ xの形式で表示され
ます.

(%i2) exp(-x);

- x

(%o2) %e

(%i3) %edispflag:true;

(%o3) true

(%i4) exp(-x);

1

(%o4) ---

x

%e

大域変数 exptdispflagが trueであれば,Maximaは負の羃を持った項を分数式で表示します.例
えば,xˆ (-1)は 1/xとなります.
大域変数 pfeformatが trueであれば,有理数は行の中で表示され,整数の分母は有理数の積とし

て表示されます.例えば,入力が b/4であれば,1/4*bと表示されます. 但し,a/bの様に,a,bの両方
が不定元であれば,通常の表示になります.
大域変数 stardispが falseの場合,可換積演算子は出力では省略されています. trueであれば,可

換積演算子を表示します.
大域変数 linelは一行に表示される文字数を設定します.変更は何時でも可能です.
大域変数 ttyoffが trueであれば入力行の表示のみを行い,通常の出力を止めます.但し, エラー表

示は行います.尚,writefile函数で開いたファイルに対しても,同じ出力になります.

4.4.2 式の表示を行う函数

式の表示を行う函数¶ ³
display(〈式1〉, 〈式2〉, · · ·)
disp(〈式1〉, 〈式2〉, · · ·)
ldisplay(〈式1〉, 〈式2〉, · · ·
ldisp(〈式1〉, 〈式2〉, · · ·
print(〈式1〉, 〈式2〉, · · ·)
grind(〈式 〉)µ ´

display函数式を表示します.その左側が未評価の 〈式i〉 で, その右側の行の中心がその式の値と
なります.この函数は blockや do文で, 中途結果の表示を行うのに便利です.displayの引数は通常,
アトム,添字された変数や函数呼出しになります.

disp函数は display函数にに似ていますが引数の値のみを表示します.
ldisplay函数と ldisp函数は各々,display函数と disp函数に似ていますが, 中間ラベルを生成す

る点が異なります.

(%i18) a1[1,2]:128;

(%o18) 128

(%i19) display(a1[1,2]);

a1 = 128

1, 2

(%o19) done

(%i20) disp(a1[1,2]);

128

(%o20) done

(%i21) ldisplay(a1[1,2]);

(%t21) a1 = 128

1, 2

(%o21) [%t22]

(%i22) ldisp(a1[1,2]);

(%t22) 128

(%o22) [%t23]

print函数は,〈式1〉がら順番に評価を実行して,その結果を表示します.ここで,〈式i〉に含まれる
アトムや函数の前に単引用符’が置かれていたり,文字列 (全体を二重引用符で括ったもの)の場合

は, 評価を行わずに,そのまま表示を行います.
grind函数は,〈式 〉 をMaximaの入力に適した形式で表示します.grind函数の返却値は常に done

です.
式の内部表現に関連して表示する函数¶ ³

dispterms (〈式 〉)
reveal (〈式 〉, 〈深度 〉)
tcl output (〈リスト 〉, 〈添字 〉, 〈飛幅 〉)
tcl output (〈リスト 〉, 〈添字 〉)µ ´

dispterms函数は与式を内部表現に合せて,第一層の部分式を順次表示します.
reveal函数は 〈整数 〉 で指定された各々の成分の長さで 〈式 〉 を表示します.和は sum(n),積は

product(n)として表示されます.
ここで nは和や積の成分の数になります.指数函数は exptで表現されます.

(%i10) aa:integrate(1/(x^3+2),x)$

(%i11) aa;

1/3

2 x - 2

atan(------------)

2 1/3 2/3 1/3 1/3

log(x - 2 x + 2 ) 2 sqrt(3) log(x + 2 )

(%o11) - ----------------------- + ------------------ + -------------

2/3 2/3 2/3

6 2 2 sqrt(3) 3 2

(%i12) reveal(aa,1);

(%o12) sum(3)

(%i13) reveal(aa,2);

(%o13) negterm + quotient + quotient

(%i14) reveal(aa,3);

atan log

(%o14) - quotient + ---------- + ----------

product(2) product(2)

(%i15) reveal(aa,4);

log atan(quotient) log(sum(2))

(%o15) - ---------- + -------------- + -----------

product(2) expt sqrt 3 expt

(%i16) reveal(aa,5);

sum(2)

atan(----------)

log(sum(3)) product(2) log(x + expt)

(%o16) - ----------- + ---------------- + -------------

6 expt 2/3 2/3

2 sqrt(3) 3 2

tcl output函数は,〈添字 〉 を展開した 〈リスト 〉に対応する tclのリストを表示します. ここで,
飛幅の初期値は 2で,引数がリストで構成されたリストではなく, 数値リスト形式の場合,飛幅から
外れた全ての要素が表示されます.

4.4.3 後戻し表示を行う函数

xmaximaの様なグラフィカルなGUI端末を持たなかった時代では特に重宝されたと思われる機
能に,以前処理した結果の表示を再度表示する機能があります. これは playback函数を用いて指定
した入力とその処理結果を再表示させるもので,その際に再計算を行うものではありません.

後戻し表示の函数 playback¶ ³
playback (〈整数 〉 )
playback ([〈整数1〉, 〈整数2〉])
playback ([〈整数 〉])
playback ()
playback(input)
playback(slow)
playback(time)
playback(grind)µ ´

playback函数は入力と出力行の後戻し表示を行います.引数が整数 nであれば,最近の n個の式
(入力行%,出力行%oと中間行%eを各々1個として数えます)を再表示します.
ここで,引数がリスト [〈整数1〉, 〈整数2〉]の場合, 〈整数1〉 から 〈整数2〉 迄の全ての行を再実行し

ます.〈整数1 =整数2〉 であれば, 引数として [〈整数 〉]を指定します.
引数が指定されない場合には,全ての行が再表示されます.
その他の引数に,input,slow,time,grindがあります.
先ず,inputの場合は入力行を再表示行します.
slowの場合は example函数によるデモの処理と同様に一つの入力とそれに対応する処理結果を

表示するとMaxima-breakに入り,Enterキーの入力待ちになります. 又,Enterキー以外のキーが入
力されると playback函数を終了します.尚, Maxima-breakに入った時点で表示されるプロンプト
は大域変数 prompt で設定されています.

timeの場合には計算時間が式と同様に表示されます.

更に,gctimeか totaltimeであれば,showtime:all;を用いたのと同様に,計算時間の詳細が表示さ
れます.

stringであれば,全ての入力行の再表示を文字列として返します.
grindの場合は,Maximaで再入力が可能な式を出力する grindモードで式の再表示を行います.
尚,playback函数による行の再表示は, playback([2,5],10,time,grind) の様に複数の引数を組合せ

ても構いません.

4.4.4 エラー表示

エラー表示に関連する大域変数¶ ³
変数名 初期値 概要

error size 10 エラーメッセージの大きさを制御

error syms [errexp1,errexp2,errexp3] エラーメッセージµ ´
大域変数 error sizeはエラーメッセージの大きさを制御します. error sizeよりも大きな式は文字

列に置換され,文字列には式が設定されています.文字列は利用者が設定可能なリストから取られ
ます.
大域変数 error symsはエラーメッセージで,error sizeよりも大きな式は文字列に置換され,その文

字列には式が設定されています. 文字列は error symsリストから取られて初期値は errexp1,errexp2,
errexp3等々となっています.エラーメッセージが表示された後,例えば, ”the function foo doesn’t
like errexp1 as input.”であれば, errexp1; と利用者が入力すると,その式を見る事が出来ます.

error symsは,必要なら別の文字列を設定しても構いません.

4.5 記号?

Maximaの記号?は変った性質を持っています.これは函数としては,オンラインマニュアルを表
示する演算子の様に振舞いますが,特殊な記号としては, 裏の LISPに?の直後の文字列を流し込む
働きを行います.

?函数¶ ³
? ? 〈事項 〉 〈事項 〉に関連するオンラインマニュアルを表示
? ?〈LISPに評価させる函数 〉 S式を評価µ ´
函数?はMaximaで二つの意味を持ちます.一つは,ヘルプの表示で用います. この場合,?との間

に空行や tabを入れて調べたい 〈事項 〉 を入力します. 〈事項 〉 に関連する項目が複数存在する場
合,Maximaは一覧表を表示して事項に対応する数値か noneが入力されるのを待ちます. 数値が入
力されると該当するヘルプを表示して終了し,noneの場合は何もせずにそのまま終了します.

(%i5) ? prefix;

0: (maxima.info)prefix.

1: optimprefix :Definitions for Expressions.

Enter space-separated numbers, ‘all’ or ‘none’: 0

Info from file /usr/local/info/maxima.info:

5.5 prefix

==========

A ‘prefix’ operator is one which signifies a function of one argument,

which argument immediately follows an occurrence of the operator.

‘prefix("x")’ is a syntax extension function to declare x to be a

‘prefix’ operator.

See also ‘syntax’.

(%o5) false

もう一つの?の意味は, 引数の間に空白を空けずに利用した場合,?の直後の引数を LISP の函数
として評価して結果を返そうとします.この場合の?引数は,LISPに評価させる函数を記述します
が,LISPの S式ではなく.Maximaの函数風に表記しなければなりません.更に,その LISPの函数
に与える引数も,Maximaの表の表記にする必要があります.例えば,Maximaの変数 xは,内部では
$xと表記されています. その xに割当てられた式の carを取りたい場合, ?car(x); と入力し, ?(car
$x)とはしません.LISPの函数で式の評価をする場合,:lisp函数もありますが,こちらは LISPの S
式を引数に取りますが,その返却値は函数?と違ってラベルには保存されません.

(%i14) a1:x^2+y+z;

2

(%o14) z + y + x

(%i15) ?car(a1);

(%o15) ("+", simp)

(%i16) :lisp (car $a1)

(MPLUS SIMP)

4.6 システムの状態を調べる

4.6.1 大域変数 infolists

大域変数 infolists¶ ³
変数名 概要

infolists Maximaの情報リストの全ての名前を含むリストµ ´
Maximaには大域変数 infolistsがあります.この infolistsには以下の大域変数が含まれています.

• labels
代入された全ての i,oと t等のラベル付けられた行

• values
代入された全てのアトム.利用者変数でMaximaのものではない.

• オプションや大域変数 (:,::,や関数による割当が設定されたもの)

• functions
全ての利用者定義関数 ( f(x) := · · · で設定されたもの)

• arrays
宣言されたものと非宣言の配列 (:,::,や::=で設定されたもの)

• macros
利用者定義された任意のマクロ

• myoptions
利用者によって再設定された全てのオプション (それらがデフォルト値に再設定されていよ
うが無関係に)

• rules
利用者定義の並び照合と簡易化の規則 (tellsimp, tellsimpafter,defmatchや defruleで設定さ
れたもの)

• aliases
利用者定義の別名を持つアトム (alias,ordergreat, orderless関数や declareで名詞型 (noun)
として宣言されたもの)

• dependencies
関数の依存関係を持つアトム (dependsや gradef関数で設定されたもの)

• gradefs
利用者定義の微分を持つ関数 (gradef関数で設定されたもの)

• props
上で言及された他の任意の属性,例えば,atvalues,matchdeclares等の declare関数で指定され
た属性を持つアトム.

• let rule packages
全ての利用者定義のリストで,規則パッケージに特別な default let rule packageを追加する
もの. (default let rule packageは規則パッケージの名前で, 利用者によって明示的に設定さ
れなかった時に用いられます)

この infolistsに含まれる項目を参照して,情報の蒐集やデータや属性の変更や削除を行う函数が
多くあります.又,利用者もこの変数名を直接入力する事で,Maximaに含まれる対象を調べる事が
出来ます.

4.6.2 status函数

status函数¶ ³
status (〈引数 〉 )µ ´

status函数は利用中のMaximaに関する様々な情報を与えられた 〈引数 〉 に従って返却します.

• time
計算で消費した時間.

• day
その週の日

• date
年,月,日々のリスト

• daytime
時間,分,秒のリスト

• runtime
現時点のMaximaで集計された cpu時間に原子”milliseconds”をかけたもの.

• realtime
利用者がMaximaを立ち上げてから経過した実際の時間 (秒単位).

• gctime
計算で費やしたゴミ集め (garbage collection)時間

• totalgctime
Maximaでゴミ集めに費やした総時間.

• freecore
アドレス空間を使い果す前に拡張可能な Maximaのコアブロックの数 (１ブロックは 1024
語).250*blocks(mcにて得られる最大値)から値を減じ, Maximaが使い上げた中心核のブロッ
ク数を知せます (固定ファイル無しのMaximaは大体 191ブロックで開始します).

• feature
システム機能のリストを返す.現在,mcに対するリストを以下に示します.
Maxima,noldmsg,maclisp,pdp10,bignum,fasload,hunk,funarg, roman,newio,sfa,paging,mcと
its.
これらの任意の機能は status(feature,· · ·)の第二引数として与えても構いません.

4.6.3 room函数

room函数¶ ³
room ()
room (true)
room (false)µ ´

room函数は保存領域の状況を詳細な記述で出力します.この room函数は LISPの room函数を
利用したものです.

4.6.4 処理時間に関連する函数

処理時間表示の函数¶ ³
time (〈%o1〉, 〈%o2〉, · · ·)µ ´

time函数は 〈%oi〉 の計算で費した計算時間をミリ秒単位で表記したリストを返します.尚,大域
変数数 showtimeを trueにすると,各%o-行 (結果表示行)で計算時間が表示されます).

(%i17) showtime;

(%o17) false

(%i18) integrate(sin(x)*exp(-x),x,0,inf);

1

(%o18) -

2

(%i19) showtime:true;

Evaluation took 0.00 seconds (0.00 elapsed) using 72 bytes.

(%o19) true

(%i20) integrate(sin(x)*exp(-x),x,0,inf);

Evaluation took 0.02 seconds (0.02 elapsed) using 109.234 KB.

1

(%o20) -

2

(%i21) time(%o18,%o20);

Time:Evaluation took 0.00 seconds (0.00 elapsed) using 96 bytes.

(%o21) [0.015998, 0.015998]

時間に関連する大域変数¶ ³
変数名 初期値 概要

lasttime 直前に入力した計算時間

showtime false 処理時間の表示の有無µ ´
大域変数 lasttimeは直前に入力した式の計算時間をミリ秒単位で timeと gctimeの組合せを成

分とするリストです.
大域変数 showtimeが trueであれば,出力式と共に計算時間の自動表示を行います. showtime:all;

とすれば,CPU時間も含めてMaximaは計算処理でのゴミ収集 (gc) に費した時間を零でなければ
表示します.この時間は time=の時間表示に含まれています.尚,time=には計算時間のみが含まれ,
中間表示時間やファイルを読込む時間は含まれてません.そこで,gcへの反応性に分けて認識する
事が難しい為, 表示する gctimeには計算の実行中に費やした全ての gctimeを含んでいます. その
為に,稀に time=よりも gctimeの方が大きくなる事があります.

4.7 外部プログラムの起動

MaximaではMaxima外部のプログラムを起動する事が可能です. この場合,system函数を用い
ます.

system函数¶ ³
system(〈命令 〉)µ ´

Maxima外部の命令を実行します.オプションを持つ命令を実行したければ, 命令全体を文字列
として system函数に引渡します.例えば, ls -a を実行したければ, system(”ls -a”); と入力します.
尚,UNIX環境で外部プログラムを長く利用し,その間にMaximaも利用したければ, 命令の末尾

に&を入れた文字列を system函数に引き渡します.例えば,surfを用いたければ, system(”surf&”);
の様に&を使います. もし,&が無ければ,system函数が外部プログラムが終了するまで実行され続
ける為, Maximaによる処理は停止したままです.

4.8 式の変換を行う函数

Maximaでは出力式を TeX,FORTRANの書式に変換出来ます.

tex函数¶ ³
tex(〈式 〉)
tex(〈式 〉, 〈ファイル名 〉)
tex(〈ラベル行 〉, 〈ファイル名 〉)µ ´

tex函数は,与えられた 〈式 〉や 〈ラベル行 〉を TEXの書式に変換します.〈ファイル名 〉を指定
すると,出力結果は指定ファイルに保存されます.尚,指定ファイルが既存すれば,その結果はファ
イル末尾に追加されます.尚,ラベル行を変換する場合,式のラベル番号も生成されます.

fortran函数¶ ³
fortran(〈式 〉)µ ´

fortran函数は与式を FORTRANの構文に変換します.一行が長くなり過ぎると, 継続文字を用
います.この文字は 1から 9までが順番に振られ足りなくなると, :,;,<,=,>,?,@が順番で振られ,そ
れから再度 1から 9を繰返します.
尚,fortran函数はMaximaの式を FORTRANの書式に合せます.例えば,式中の演算子^は**で

置換し,複素数 a + b%iは (a,b)に置換えます.
与式は方程式であって構いません.この場合,方程式の右辺を左辺に割当てる形で出力します.こ

こで,右辺が行列名であれば,fortran函数は,行列の各成分に割当を行う様に変換します.
尚,与式に fortran函数で解釈出来ないものがあった場合,grind函数を用いて式の表示を行います.
大域変数 fortindentは左側の空白を制御します.デフォルトの 0で,通常の空白行 (6空白行)と

なります.
大域変数 fortspacesが trueであれば,fortranは 80列で出力を行います. 尚,fortran函数は常に

返却値は doneです.

(%i56) expr:(a+b+1)^5;

5

(%o56) (b + a + 1)

(%i57) fortran(expand(expr));

b**5+5*a*b**4+5*b**4+10*a**2*b**3+20*a*b**3+10*b**3+10*a**3*b**2+3

1 0*a**2*b**2+30*a*b**2+10*b**2+5*a**4*b+20*a**3*b+30*a**2*b+20*a

2 *b+5*b+a**5+5*a**4+10*a**3+10*a**2+5*a+1

(%o57) done



4.9. システムに関連する大域変数 227

便利な函数¶ ³
alias(〈新名称1〉, 〈旧名称1〉, 〈新名称2〉, 〈旧名称〉, · · ·)
apropos(〈文字列 〉)µ ´

alias函数は (利用者,又はシステム)函数,変数,配列等に別名を与えます. 引数は新名称と旧名称
の一組となるので,偶数個の引数が必要になります.

apropos函数は 〈文字列 〉を含むMaximaの函数,大域変数のリストを返します,例えば, apropos(exp);
の結果は,expand,exp, exponentialize等の名前の一部に文字列 expを含む全ての大域変数や函数の
リストになります.
この函数を使えば,大域変数や函数の名前の一部だけさえ覚えていれば, 残りの名前を探す事が

可能になります.

4.9 システムに関連する大域変数

システムに関連する大域変数¶ ³
変数名 初期値 概要

aliases [] 別名リスト

debugmode false break loopに入るかどうかを設定
myoptions [] オプションを蓄えるリスト

optionset false オプション設定時にメッセージの有無µ ´
大域変数 alieasesは alias,ordergreat,orderless函数やアトムを名詞型と宣言する事によって設定

された別名を持つアトムのリストです.
debugmodeが trueの場合,エラーが生じた時や falseで中断モードに入った時は何時でもMaxima

の break loopに入ります.allが設定されていれば,実行中の函数のリストに対して backtraceを調
べる事が出来ます.

myoptionsは利用者が設定した全てのオプションを蓄えるリストです.
optionsetが trueであれば,Maximaはオプションが再設定された時点でメッセージを表示しま

す.これはオプションの綴りが不確かな場合,割当てた値が本当にオプションの値となっているかを
確認したい時に便利です.

4.10 Maximaの終了

quit函数¶ ³
quit()µ ´

Maximaの終了は quit函数を用います.この函数は引数を必要としませんが, 必ず quit() と入力
し,後の小括弧 ()を忘れないで下さい. 尚,Maximaを一時的に停止させるのであれば,Ctrl+C(ˆC)
を入力します.

to lisp函数を用いて LISP環境に入った場合に,Maximaを全て終了させたければ, ($quit) と入
力します.こちらの表記がMaximaの quit函数の LISP上に於ける表記になります.

この章で解説する事:

• 三角函数

• 指数函数と対数函数

• 代数方程式

• 極限

• 微分

• 積分

• 常微分方程式

5.1 三角函数

5.1.1 三角函数一覧

Maximaは沢山の三角函数を持っています. 三角函数の恒等式,即ち,cos2 (x) + sin2 (x) = 1 や
cos (2x) = 2cos2 (x)− 1 の様なものは予めMaximaに組込まれていますが,多くの恒等式を規則と
して利用者が付加する事が出来ます.

Maximaで予め定義された三角函数は下記のものがあります.
三角函数と双曲函数¶ ³

cos 余弦函数

cosh 双曲線余弦函数

cot 余接函数

coth 双曲線余接函数

csc 余割函数

csch 双曲線余割函数

sec 正割函数

sech 双曲線正割函数

sin 正弦函数

sinh 双曲線正弦函数

tan 正接函数

tanh 双曲線正接函数µ ´
逆三角函数と逆双曲函数¶ ³

函数名 概要

acos 逆余弦函数

acosh 逆双曲線余弦函数

acot 逆余接函数

acoth 逆双曲線余接函数

acsc 逆余割函数

acsch 逆双曲線余割函数

asec 逆正割函数

asech 逆双曲線正割函数

asin 逆正弦函数

asinh 逆双曲線正弦函数

atan 逆正接函数

atan2 逆正接函数

atanh 逆双曲線正接函数µ ´
尚,逆正接函数には atan函数 と atan2函数の二種類がありますが,atan2函数は, atan2(y,x)の

様に引数を二つ必要とする函数で,区間 (−π, π) の間で atan(y/x) を計算します.

三角函数は,倍角公式や角の和の公式等の様々な公式を持っています. これらを自動的に入力し
た式に適応する事も可能です.この場合, 大域変数 trigexpandを trueにした状態で,角の整数倍に
は, 大域変数 trigexpandtimes,角の和に対しては大域変数 trigexpandplusを各々trueに設定する
と,角の和と積の公式を用いて, 与式の自動展開を行います.

(%i43) x+sin(5*x)/cos(x),trigexpand=true,expand;

5

sin (x) 3 3

(%o43) ------- - 10 cos(x) sin (x) + 5 cos (x) sin(x) + x

cos(x)

(%i44) trigexpand(cos(3*x+2*y));

(%o44) cos(3 x) cos(2 y) - sin(3 x) sin(2 y)

(%i45) trigexpand:true;

(%o45) true

(%i46) trigexpandtimes:true;

(%o46) true

(%i47) trigexpandplus:true;

(%o47) true

(%i48) cos(3*x+2*y);

3 2 2 2

(%o48) (cos (x) - 3 cos(x) sin (x)) (cos (y) - sin (y))

2 3

- 2 (3 cos (x) sin(x) - sin (x)) cos(y) sin(y)

この例で,最初の x+sin(5 *x)/cos(x),trigexpand=true,expand は ev函数による評価の書式の一

つで,与式 x + sin(5x)
cos(x) を大域変数 trigexpandを trueにした状態で展開を行う事を意味します.こ

の表記はMaximaのトップレベルだけで利用可能です.尚,ev函数の詳細に関しては,評価の節を参
照して下さい.
三角函数の半角公式に関しては,大域変数 halfanglesを trueにすると式の自動展開を実行しま

す.この変数は大域変数 trigexpandの影響は受けません.
これらの変数とば別に,三角函数の引数の declare函数による宣言に従って, 三角函数を含む式の

自動簡易化も行えます.

(%i2) declare(i,integer,a,even,b,odd);

(%o2) done

(%i3) sin(x+(a+1/2)*%pi);

(%o3) cos(x)

(%i4) sin(x+(b+1/2)*%pi);

(%o4) - cos(x)

(%i5) cos(x+b*2*i*%pi);

(%o5) cos(x)

この例では,最初に変数 iを整数,aを偶数,bを奇数として宣言しています. すると,Maximaは三
角函数の引数を評価し,式を自動的に変換しています.
三角函数に付随する函数は,大域変数の trigexpand,trigreduceと trigsignを参照して下さい.二

つの shareパッケージはMaximaに組み込みの簡易化の規則の trigと atrigを拡張します.

三角函数に関連する大域変数¶ ³
変数名 初期値 概要

halfangles false 半角公式の自動適用を制御

trigexpandplus true 和公式の自動適用を制御

trigexpandtimes true 角度の積による展開を制御

triginverses all 逆関数との合成による簡易化を制御

trigsign true 負の引数の簡易化を制御µ ´
大域変数 halfanglesが trueの場合,半角 θ

2 に対して簡易化が実行されます.この変数は大域変数
trigexpandの影響は受けません.
大域変数 trigexpandplusは和の規則を制御する大域変数です. つまり,大域変数 trigexpandに

trueが設定されている時に, 引数の和を含む sin (x + y) の様な三角関数の自動展開が, 大域変数
trigexpandplusが trueの場合に限って実行されます.
大域変数 trigexpandtimesは trigexpand函数の積規則を制御します. 大域変数 trigexpandが true

に設定されている時に,三角函数の引数が整数, 或いは有理数倍の式に対し,三角函数の自動展開が
実行されます.

triginversesは三角函数,双曲函数とその逆函数との合成の簡易化を制御します.

• all
両方,例えば,atan (tan (x)) と tan (atan (x)) の両方が x に簡易化されます.

• true
arcfunction(function(x))の簡易化が切り捨てられます.

• false
arcfunc(func))と fun(arcfun(x))の簡易化が切り捨てられます.

trigsignが trueであれば三角函数に対し負の引数の自動簡易化を行います. 例えば,trigsinが true
の時に限り,sin(-x)は-sin(x)に変換されます.

5.1.2 三角函数に関連する函数

三角函数の展開と簡易化に関連する函数¶ ³
trigexpand(〈式 〉 )
trigreduce(〈式 〉 ,〈変数 〉 )
trigsimp(〈式 〉 )
trigrat(〈三角函数を含む式 〉 )µ ´

trigexpand函数は 〈式 〉に含まれる三角函数や双曲函数に対し, 倍角公式等を適用して式の展開
を実行します.最良の結果を得る為に, 予め expand函数等で ∠式 〉を展開しておきましょう.
簡易化の利用者制御を拡張する為,この函数は一度に一つのレベルのみの角の和と角の積の展開

を行います.sinと cosの全体の自動展開が必要ならば, 大域変数 trigexpandを trueに設定してお
きます.

trigreduce函数は 〈変数 〉 の積を持つ三角函数と双曲 sin函数と cos函数の積と羃乗を結合しま
す.分母で現われたこれらの函数を消去する事も試みます.尚, 〈変数 〉 が省略されると, 〈式 〉 の全
ての変数が利用されます.

(%i1) trigreduce(-sin(x)^2+3*cos(x)^2+x);

cos(2 x) cos(2 x) 1 1

(%o1) -------- + 3 (-------- + -) + x - -

2 2 2 2

trigsimp 函数は tan,sec 等を含む 〈 式 〉 の簡易化の為に, 恒等式 sin2 (x) + cos2 (x) = 1 と
cosh2 (x) − sinh2 (x) = 1 を使って, sin,cos,sinh,cosh へと変換します.trigreduceと組合せる事で,
より良い簡易化が得られます.

(%i6) trigreduce(trigsimp(-sin(x)^2+3*cos(x)^2+x));

cos(2 x) 1

(%o6) 4 (-------- + -) + x - 1

2 2

(%i7) trigsimp(trigreduce(-sin(x)^2+3*cos(x)^2+x));

(%o7) 2 cos(2 x) + x + 1

trigrat函数は sin,cos,tan 等の三角函数による有理式の正規簡易化を与えます.与式に含まれる
これらの三角函数の引数は, 変数,有理数と%piによる線形結合とします.trigratの計算結果は簡易
化された sin と cos を含む有理式となります.

(%i10) trigexpand((cos(2*x+%pi*4/3*y+%pi/2)+sin(2*y+x))/(cos(x)+sin(y)));

4 %pi y 4 %pi y

(%o10) (- cos(2 x) sin(-------) - sin(2 x) cos(-------) + cos(x) sin(2 y)

3 3

+ sin(x) cos(2 y))/(sin(y) + cos(x))

5.1.3 atrig1パッケージ

atrig1パッケージには逆三角函数に対する幾つかの追加の簡易化規則が含まれています.Maxima
で既知の規則と共に,次の角が実装されています.¶ ³

0, %pi/6, %pi/4, %pi/3,%pi/2.µ ´
他の3つの象限に於ける角度でも利用可能です.尚,このパッケージを利用する為には予め load(atrig1);

を実行する必要があります.

5.2 指数函数と対数函数

5.2.1 指数函数と対数函数の概要

Maximaには指数函数 expと対数函数 logがあります.指数函数 expはMaxima内部では,自然
底%eの羃として表現されています.
これらの函数は次の大域変数の設定によって,Maxima上で自動的に簡易化が実行されます.

対数函数に関連する大域変数¶ ³
変数名 初期値 概要

%e to numlog false 指数に対数を持つ冪乗の簡易化

logabs false logを含む不定積分の結果を制御
logarc false 逆三角函数や逆双曲函数を対数関数で表現

logconcoeffp false logcontractで潰される係数を制御
logexpand true 対数関数の積や羃の自動変換

lognegint false 対数関数の引数が負の場合の処理を制御

lognumer false 対数函数の浮動小数点引数の制御

logsimp true logを含む指数関数の冪乗の自動化を制御µ ´
先ず,%e to numlogが trueであれば,rを有理数,xを式とすると, erlogx が xr に簡易化されます.

尚,radcan函数も, この変換を行います.
大域変数 logabsが trueの場合,integrate(1/x,x)の様に計算結果に log函数が結果に含まれる場

合, log函数が log(abs( · · · ))で置換えられます. 但し,logabsが falseであれば log( · · · )の項を持
つものになります. 尚,定積分の場合は,logabs:trueの設定が利用されます.これは,不定積分の両端
点での評価が必要となる事が多い為です.
大域変数 logarcが trueであれば,逆三角函数,逆双曲函数を対数函数の書式に変換します. logarc(〈式 〉)

はこの大域変数の設定なしで, 特定の式に対し,evを用いた式の再評価を行います.
大域変数 logcoeffpは logcontractを用いた時に潰される係数の制御に利用する述語函数名を一

つ設定します.函数名は評価を行わない様に,名詞型にします,log函数から sqrtを生成したい場合,
係数が 1/2の様な有理数でなければなりません. 従って,述語函数は,log函数の係数が整数である
か, 有理数の何れかとすれば良いでしょう.これを具体的に記述すれば,以下の様に設定すれば十分
です.

logconcoeffp:’logconfun

logconfun(m):=featurep(m,integer) or ratnump(m)

ここで, logcontract(1/2*log(x)); と入力されると, logcontract 函数は, 評価を実行する前に,
logconcoeffpに設定された評価函数に与式に含まれる log函数の係数 1

2 を引き渡します.尚,falseの
場合は featurep(1/2,integer)で評価します.述語函数の結果が trueであれば, この例では, log

(
x

1
2

)
を評価します. falseの場合はそのままの式を返します.
大域変数 logexpandが trueの場合,自動的に log(ab) を blog(a) に変換します.ここで,allの場

合,自動的に log(ab) は log(a) + log(b) に変換されます.superの場合, a = 1 でない有理数 a/b に

対し, log(a/b) は log(a)− log(b) に変換されます.尚,整数 b に対し, log(1/b) は logexpandとは無
関係に,常に簡易化されます.falseの場合はこれらの簡易化は全て実行されません.
大域変数 lognegintが trueの場合,正整数 nに対し, log (−n) を log (n) + iπ で置換える規則が

内部的に設定されます.
大域変数 lognumerが trueの場合,logの負の浮動小数引数は logに渡される前に, 常にその絶対

値に変換されます.
大域変数 logsimpが falseの場合,logを含む%eの羃乗の自動簡易化は実行されません.

5.2.2 対数函数に関連する函数

対数函数を潰す函数¶ ³
logcontract(〈式 〉 )µ ´

logcontract函数は logを含む式を簡易化で潰すものです.即ち,log函数の係数を引数に取込み,
その結果によって,log から sqrt 等の函数に変換します. 具体的には,〈 式 〉 を再帰的に調べて,
a1*log(b1)+a2*log(b2)+cの形の部分式を log(ratsimp(b1ˆ a1 * b2ˆ a2))+ c に変換します.

(%i8) 2*log(x)+4*log(y)+8;

(%o8) 4 log(y) + 2 log(x) + 8

(%i9) logcontract(%);

2 4

(%o9) log(x y ) + 8

(%i10) declare(n,integer);

(%o10) done

(%i11) logcontract(2*log(x)+4*n*log(y)+8);

2 4 n

(%o11) log(x y ) + 8

(%i12) logcontract(2*log(x)+4*m*log(y)+8);

4 2

(%o12) m log(y ) + log(x ) + 8

この logcontract函数は log函数の整数係数に対して影響を与える函数です. 例えば, declare(n,integer);

を実行して, logcontract(2*log(x)+4*n*log(y)+8); を実行すれば,第二式の 4+n+log(y)の係数
4*nは述語函数 featurep(coeff,integer)を満す為, log

(
x2y4n

)
+8 に簡易化されています.ところが,

logcontract(2*log(x)+4*m*log(y)+8); の場合,mは integerとして未宣言の為,簡易化が途中で停
止しています.

分枝を定める函数¶ ³
plog(〈x〉 )µ ´

plog函数は複素数値の自然対数函数の主分枝を −π < carg(x) ≤ π とします.

極座標形式に変換する函数¶ ³
polarform(〈式 〉 )µ ´

polarform函数は与えられた 〈式 〉 を r*%eˆ (%i*theta) の形に変換します.尚,多項式が実数係
数多項式の場合は入力のままで返され, 函数を含む場合には,その函数が負であれば %e%i%pi をか

けたものが返されます.

(%i31) polarform((1+%i)^3);

3 %i %pi

--------

4

(%o31) 2 sqrt(2) %e

(%i32) polarform(x^2+1);

ppppp 2

(%o32) x + 1

(%i33) polarform((x+1)^2);

Is x + 1 zero or nonzero?

pos;

2

(%o33) x + 2 x + 1

(%i34) polarform(sin(x+1));

Is sin(x + 1) positive or negative?

neg;

%i %pi

(%o34) - %e sin(x + 1)

5.3 代数方程式

5.3.1 Maximaの方程式

Maximaの方程式は演算子=の両側に演算子=を持たない式を配置した式です.即ち, xˆ 2+2*x+1=0
の様な形式になります. 但し,右辺が 0の場合は,演算子 = を省略しても構いません.
尚,ここで演算子 = の左右の式は lhs函数と rhs函数を使って取り出す事が出来ます.

lhsと rhs¶ ³
lhs(〈方程式 〉 )
rhs(〈方程式 〉 )µ ´

(%i17) eq1:x^2+2*x+1=y^2;

2 2

(%o17) x + 2 x + 1 = y

(%i18) lhs(eq1);

2

(%o18) x + 2 x + 1

(%i19) rhs(eq1);

2

(%o19) y

この例では方程式として x2 +2 ∗x+1 = y2 を eq1に割当てており, lhs(eq1) で方程式の左側の

x^2+2*x+1, rhs(eq1) で方程式右側の y^2を各々取り出しています.尚,これらの lhs函数と rhs函
数は演算子=に対してのみ使える函数です.他の二項演算子 (例えば, > 等)に対しては使えません.

Maximaで扱える方程式としては,1変数多項式で構成された方程式,多変数多項式で構成された
連立方程式, cos や log 等の初等函数を含むより一般的な方程式があります.他に,微分’diffを含む
方程式や積分’integrateを含む方程式もあります.尚,微分を含む方程式は,常微分方程式の節で詳
細を述べる為, ここでは,微分と積分を含まない代数方程式を中心に述べます.

Maximaで連立方程式を扱う場合, [eq1, · · · , eqn] の様に, 複数の方程式で構成したリストで連
立方程式を表現します.

(%i25) eq2:[2*x^2-5*y=1,x+y*x+y^2=4];

2 2

(%o25) [2 x - 5 y = 1, y + x y + x = 4]

(%i26) eq2[1];eq2[2];

2

(%o26) 2 x - 5 y = 1

(%i27)

2

(%o27) y + x y + x = 4

この例では,二つの方程式 2*x^2-5*y=1と x+y*x+y^2=4から構成される方程式のリストを eq2
に割当てています.連立方程式をリストで表現する為,一つの方程式を取り出す場合は,リストの成
分の取り出しと同じ方式で行えます.

Maximaには与えられた方程式の近似解を数値的に解く函数に allroots函数と realroots函数が
あります.又,厳密解を求める函数として,linsolve函数と solve函数, 厳密解が計算出来る場合には
厳密解を計算し,厳密解の計算に失敗した場合に近似解を計算する algsys函数があります.
これらの函数は,与えられた方程式が 1変数の多項式で構成される場合,線形連立方程式の場合,

多変数多項式で構成される方程式系の場合,そして,より一般的な初等関数を含む方程式の場合に
区分出来ます.
先ず,方程式系が一つの 1変数多項式で構成される場合,近似解を計算する allroots函数と realroots

函数が使えます.線形連立方程式の場合は,linsolve函数を使って厳密解を計算出来ます.そして,多
変数多項式で構成された方程式系に対しては, algsys函数を用いて可能であれば厳密解,近似解が
計算可能な場合は,近似解を計算出来ます.最後に,より一般的な方程式に対しては solve函数を用
いて厳密解の計算が行えます.

Maximaで計算した結果を自動的に変数に代入したり,重複リストを生成させる為には,以下の
大域変数を調整する必要があります.

方程式に関連する大域変数¶ ³
変数名 初期値 概要

backsubst true 三角関数化した方程式の代入を抑制

globalsolve false 解の値の自動代入を制御

multiplicities not set yet 重複度リスト

programmode true allroots,linsolve,solve等の出力を制御µ ´
大域変数 backsubstは三角関数化した方程式に対して代入の制御を行います. backsubstが false

の場合,方程式を三角関数化した後で,代入を防ぎます. これは,後代入でとてつもなく大きな式が
生成される様な問題で必要となります.
大域変数 globalsolveはMaximaで方程式を解いた時に,方程式の変数に求めた解を自動代入す

るかどうかを制御する変数です.globalsolveが trueの場合, 解かれた変数に解が実際に割当てられ
ます.

(c101) globalsolve:true;

(d101) true

(c102) solve([xx*2+yy*3-1=0,xx+yy=10],[xx,yy]);

(d102) [[xx : 29, yy : - 19]]

(c103) xx;

(d103) 29

(c104) yy;

(d104) - 19

(c105) globalsolve:false;

(d105) false

(c106) solve([mm*2+nn*3-1=0,mm+nn=10],[mm,nn]);

(d106) [[mm = 29, nn = - 19]]

(c107) mm;nn;

(d107) mm

(c107)

(d107) nn

globalsolve:trueとした状態で,ある方程式を解いた後に同じ変数の方程式を解こうとすると次の
エラーが出るので注意します.例えば,上記の例の (c106)行の方程式以下の行で置き換えた場合に
は次の様になります.

(c106) solve([xx*2+yy*3-1=0,xx+yy=10],[xx,yy]);

a number was found where a variable was expected -solve

-- an error. quitting. to debug this try debugmode(true);)

(c107)

尚,重複度が 2以上の変数が存在する場合,自動代入は実行されない事に注意して下さい.
大域変数multiplicitiesは,solveや realrootsで返される個々の解に対応する重複度のリストが設

定されます.

(%i2) multiplicities;

(%o2) not_set_yet

(%i3) solve(x^2-4*x+4,x);

(%o3) [x = 2]

(%i4) multiplicities;

(%o4) [2]

(%i5) realroots(x^4+2*x^3-3*x^2-4*x+4);

(%o5) [x = - 2, x = 1]

(%i6) multiplicities;

(%o6) [2, 2]

(%i7) solve(x^5+x^4-2*x^3-2*x^2+x+1,x);

(%o7) [x = 1, x = - 1]

(%i8) multiplicities;

(%o8) [2, 3]

(%i9) factor(x^5+x^4-2*x^3-2*x^2+x+1);

2 3

(%o9) (x - 1) (x + 1)

この例では,最初に方程式 x2−4x+4 = 0を solve函数を用いて解き,次に,realroots函数を使って,
方程式 x4+2x3−3x2−4x+4 = 0を解いています. そして最後には, x5+x4−2x3−2x2+x+1 = 0

を解いています. 最初の例では,重複度が 2の為,multiplicitiesにはリスト [2]が割当てられていま
す.次の例では,重複度が各々2である事が判ります.最後の例では, x = 1 の重複度が 2, x = −1 の
重複度が 3である事が判ります.実際, 与式を factor函数で因子分解すれば確認出来ます.
大域変数 programmodeは,返却値に中間行ラベルを付けて出力するかどうかを制御します.pro-

grammodeが falseの場合,solve,realroots,allrootsと linsolve函数は%tラベル (中間行ラベル)に答
をラベル付けして出力します.

trueの場合,これらの函数はリストの要素として答えを返します (programmode:falseも使われ
ます.但し,backsubstが falseに設定された時を除きます).

(%i4) programmode:false;

(%o4) false

(%i5) solve(x^2+1,x);

Solution:

(%t5) x = - %i

(%t6) x = %i

(%o6) [%t5, %t6]

(%i6) programmode:true;

(%o6) true

(%i7) solve(x^2+1,x);

(%o7) [x = - %i, x = %i]

5.3.2 1変数多項式方程式の場合

最初に,方程式が多項式で構成された場合について述べます.方程式系が一つの 1変数多項式の
みで構成されている場合,その近似解を allroots函数と realroots函数を用いて計算出来ます.

数値解を求める函数¶ ³
allroots(〈方程式 〉 )
realroots(〈多項式 〉 ,〈許容範囲 〉 )
realroots(〈多項式 〉 )µ ´
最初の allroots函数は単変数の実数係数多項式の実数解と複素解全てを計算します. allroots函

数は,多項式が実係数で,大域変数 polyfactorが trueの場合に実数上で因子分解を行いますが,係
数に%iが含まれていれば複素数上で因子分解を行います.

(%i14) allroots(%i*x^2+1=0);

(%o14) [x = .7071067811865475 %i + .7071067811865475,

x = - .7071067811865475 %i - .7071067811865475]

(%i15) polyfactor:true;

(%o15) true

(%i16) allroots(x^2+1=0);

2

(%o16) x + 1.0

(%i17) allroots(%i*x^2+1=0);

(%o17) %i (x - .7071067811865475 %i - .7071067811865475)

(x + .7071067811865475 %i + .7071067811865475)

allrootsは重複解を持つ時に不正確な結果を返す事があります. この場合は与式に %i をかけた
ものを計算すれば解決するかもしれません.

allrootsは多項式方程式以外には使えません.rat命令を実行した後に, 方程式の分子が多項式で,
分母が高々複素数でなければなりません. polyfactorが trueであれば,allrootsの結果として常に同
値な式 (但し,因子分解されたもの)が返されます.

realroots函数は与えられた実単変数多項式 〈多項式 〉 の全ての実根を 〈許容範囲 〉 で指定する
許容範囲内で求めます. 尚, 〈許容範囲 〉 が 1よりも小さければ,全ての整数根を厳密に求めます.
〈許容範囲 〉 は必要であれば,任意の小さな数を設定しても構いません. 〈許容範囲 〉 を省略した場
合は, 大域変数 rootsepsilonの値が使われます.

(%i34) realroots(x^2-2=0,1.0e-5);

370727 370727

(%o34) [x = - ------, x = ------]

262144 262144

(%i35) float(sqrt(2)-rhs(%o34[2]));

(%o35) 2.289179735770474E-6

この例では方程式 x2 − 2 = 0 の解を精度 10−5 以内で求めています. 解は浮動小数点ではなく,
有理数で返されます.

realrootsは解の大域変数 multiplicitiesに重複度の情報をリストの形式で追加します.大域変数
multiplicitiesに解の重複度リストを設定する函数は,他に solve函数があります.ここで,重複度リ
ストは,求めた解に対応する形で, 整数のリストとして表現されています.

allroots函数に影響を与える大域変数¶ ³
変数名 初期値 概要

polyfactor false 因子分解の有無

rootsepsilon 1.0E-7 根を含む区間µ ´
polyfactorは allrootsで利用される大域変数です.polyfactorが trueであれば, polyfactorに与え

られた多項式が実係数多項式なら実数上で因子分解し, 係数に%Iが含まれていれば複素数上で因
子分解を行った結果を返します.

rootsepsilonは realroots函数によって見つけられた根を含む確実な区間を設定する際に使う実
数です.

5.3.3 区間内の根の個数

Maximaでは指定した半開区間に存在する根の個数を計算する函数 nroots函数があります.この
nroots函数は 1変数多項式に対して利用可能です.

区間内の根の個数を返す函数¶ ³
nroots(〈多項式 〉 ,〈下限 〉 ,〈上限 〉)µ ´

nroots函数は 〈上限 〉 と 〈下限 〉 で指定された半開区間 (〈下限 〉 ,〈上限 〉 ] 内部に,幾つの 1変
数多項式 〈多項式 〉 の根があるかを返します.
ここで,区間の終点は負の無限大と正の無限大に各々対応する minf,infでも構いません.このア

ルゴリズムには Sturm級数による手法が適用されています.

(%i18) nroots(x^2+2,-2,2);

(%o18) 0

(%i19) nroots(x^2-2,-2,2);

(%o19) 2

(%i20) nroots(x^2-5,-2,2);

(%o20) 0

(%i21) nroots(x^2-1,-2,2);

(%o21) 2

5.3.4 多項式方程式の場合

allroots函数と realroots函数は,1変数多項式の近似解を計算する函数です, これに対し,方程式
の厳密解を計算出来る函数として,linsolve函数,algsys函数と solve函数あります.
ここで,多変数の線形多項式で構成された方程式系に対しては linsolve函数が使えます.

線形連立方程式を解く函数¶ ³
linsolve([〈方程式1〉 ,〈方程式2〉 ,· · ·], [〈変数1〉, 〈変数2〉 ,· · ·])µ ´

linsolve函数は与えられた線形連立方程式を変数リストに対して解きます. 各方程式は各々与え
られた変数リストの多項式でなければなりません.

(%i68) linsolve([x+y-2=0,y-x+1=0],[x,y]);

3 1

(%o68) [x = -, y = -]

2 2

linsolveに影響を与える大域変数¶ ³
変数名 初期値 概要

linsolve params true 解に助変数を導入

linsolvewarn true linsolveの警告を抑制µ ´
linsolve paramsが trueであれば,linsolveはまた記号%riを生成し, algsysに記載された任意の助

変数を表現する為に用いられます.
falseであれば,以前の様に linsolveが動作します.即ち,不定方程式型に対し, 他の項の幾つかの

引数に対して解きます.
linsolvewarnが falseであれば,dependent equations eliminated(従属方程式が消去された)とい

うメッセージ出力が抑制されます.
多変数多項式で構成された連立方程式は,solve函数や algsys函数で解く事が出来ます. 但し,solve

函数がより一般的な方程式の厳密解の計算が可能ですが,algsys函数は, 多変数多項式で構成された
方程式系の厳密解の計算に失敗すると,今度は近似解を計算する点で,多項式方程式の計算では使
い易いでしょう.

algsys函数¶ ³
algsys([〈方程式1〉 ,〈方程式2〉 ,· · ·], [〈変数1〉 ,〈変数2〉 ,· · ·])µ ´

algsys函数では,方程式と変数はリストで与えます.返却される解に%r1や%r2といった記号が
含まれる事がありますが, これらは助変数を表示する為に用いられるもので, 助変数は大域変数
%rnum listに蓄えられています.

(%i1) algsys([2*x+3*y=1],[x,y]);

2 %r1 - 1

(%o1) [[x = %r1, y = - ---------]]

3

(%i2) %rnum_list;

(%o2) [%r1]

この例の様に,(連立)方程式の階数が足りない場合には,助変数を補って与えられた方程式を解き
ます.尚,%rnum listには algsysで使われた助変数が逐次追加されて行きます.

algsysは以下の手順で方程式を解き,必要であれば再帰的に処理を行います.

1. 最初に方程式を factorで因子分解し,各因子から構成される部分系 systemi, 即ち,方程式の
集合を構築します.

2. 部分系 systemi から,方程式 eqn を取出し,それから変数 var を選択します.この変数の選択
は,方程式 eqn に含まれる変数の中から, 最小次数のものを選出します.それから方程式 eqn

と systemi の部分系 systemi \ {eqn} に含まれる方程式 eqj を変数 var を主変数とする多

項式と看倣して終結式を計算します.この操作によって,新しい部分系 systemi+1 は systemi

よりも少ない変数で生成されます.それから 1の処理に戻ります.

3. 一つの方程式で構成される部分系が最終的に得られると,その方程式が多変数で, 係数が浮動
小数でなければ,厳密解を求める為に solveを呼出します.更に,方程式が単変数で線型,二次,
或いは四次の多項式であれば,solveを再び呼出します.

係数が浮動小数点で近似されている場合,方程式が単変数で線型,二次又は四次の何れでも
なく,大域変数 realonlyが trueの場合, 実数値解を見付ける為に函数 realrootsを呼出しま
す.realonlyが falseの場合,解を求める為に函数 allrootsを呼出します.
尚,algsysが要求以下の精度解を生成した場合、大域変数 algepsilonの値をより小さな値に変
更しても構いません.大域変数 algexactが trueであれば,solve函数を呼出します.

4. 3の段階で得られた解を以前の段階に代入して,解の計算過程の 1に戻ります. 尚,浮動小数
を近似した多変数方程式に対しては,次のメッセージを表示します:¶ ³

”algsys cannot solve - system too complicated.” (意味：”algsysでは解けません - 系が
あまりにも複雑です.”)µ ´

radcan函数を使えば,大きくて複雑な式が出来ます.この場合,pickapartや reveal を解の計
算に用います.

ここで終結式は二つの一変数多項式に対して定義されるもので, 例えば,多項式 f と g の根を

αi, βj とすると, res(f, g, x) = an
mbm

n

∏
0≤i≤m,0≤i≤n(αi − βj) となる事が知られています.

この事は f と g に共通の零点が存在する場合には終結式が零になる事を意味します.従って, f

と g が多変数の場合, f と g を f と g の共通の変数 x1 の多項式と看倣してその終結式を計算する

と, f と g の共通の根が存在する場合,終結式は零でなければなりません.こうする事で,変数 x を

含まない新しい多項式 res(f, g, x) が得られます.この操作を方程式系に対して行う事で,1変数の
多項式が得られると,その多項式を解いて,一段前の方程式系に代入し,解を求めて行く方式となっ
ています.
話を簡単にする為に,一次の連立方程式で簡単に説明しましょう.
最初に次の線形方程式が与えられたとします.

f : ax + by + p = 0g : cx + dy + q = 0

この連立方程式を構成する多項式 f と g の終結式は次の行列式を計算すると得られます.

res(f, g, x) = det
(

a by + p

c dy + q

)
この行列式は (ad− bc)y − ap + cp です.ここで, f と g の終結式は f と g が共通の解を持つ場

合に 0となります.この事から, 方程式 (ad− bc)y − ap + cp = 0 が得られます.この終結式では,前

の方程式系から変数 x と方程式が減り,変数 y の方程式となります.そこで,終結式から得られた
方程式を解けば y = ap−cp

ab−bc が得られます. それから,一段前の方程式系に戻って変数 y に値を代入

すれば,変数 x の方程式が得られ,その方程式を解けば最終的に x = bq−dp
ad−bc が得られます.

algsys函数は,方程式系を構成する各多項式を因子分解し,次数を落した各因子で構成される方
程式の集合に対して終結式を計算し,新しい方程式系から変数を一つ消去します.この処理を繰返
す事で,最終的に一変数の多項式方程式が得られると, そこから一つの変数の解を定められます.こ
の計算で,4次以下の方程式であれば, 公式を用いた根の計算が可能ですが,それ以外の場合で厳密
解が計算出来なければ, allrootsを用いて近似数値解を求め,それから,処理を逆に遡る事で,全ての
解を求める事が出来ます.algsysはこの様なアルゴリズムを採用しているのです.

algsys函数に影響を与える大域変数¶ ³
変数名 初期値 概要

%rnum list [] algsys函数で解に導入された変数リスト
algexact false algsys函数が solve函数を呼出すかどうかを制御
algepsilon 108 algsysで用いられる変数
realonly false trueの場合,algsys函数は実数解のみを返却µ ´

%rnum listは algsys函数で方程式の解を求めた時に導入された変数%rが生成順で追加されるリ
ストです.これは後に解に代入する時に便利です.

algexact は algsys 函数に影響を与える大域変数の一つです. true であれば,algsys は solve を
呼出し,realroots を常に利用します. false であれば, 終結式が単変数でない場合と quadratic か
biquadraticな場合のみ solveの呼出しを行います.algexact:trueとすると,厳密解のみを保証する
ものではなく,algsysが最初に厳密解を計算しようと試み, 結局,allか失敗した時に近似解のみを生
成します.

algepsilonは algsys函数で利用される定数です.
realonlyが trueであれば,algsysは実数解,即ち%iを持たない解のみを返します.

5.3.5 一般的な方程式

一般的な方程式に対しては solve函数が利用出来ます.この solve函数は,与えられた方程式系の
厳密解を返す函数です.方程式には,sin 等の三角函数, 指数函数や対数函数を含んだ方程式が扱え
ます.但し,algsys函数と違い,厳密解の計算に失敗した場合に数値近似解を計算する様な事は行い
ません.

solve函数¶ ³
solve(〈式 〉 )
solve(〈式 〉 ,〈変数 〉 )
solve([〈方程式 〉 ,· · ·,〈方程式n〉 ])
solve([〈方程式 〉 ,· · ·,〈方程式n〉 ], [〈変数1〉 ,· · ·,〈方程式n〉 ])µ ´

solve函数は代数方程式 〈式 〉 を 〈変数 〉 に対して解き,解のリストを返します.

〈式 〉が方程式でなければ, 〈式 〉が零に等しいと設定されていると仮定します.即ち,式 x^2+2*x+1

が 〈式 〉 であれば,solveは方程式 x^2+2*x+1=0が与えられたと solve函数は解釈します.
〈変数 〉 は和や積を除く函数の様なアトムでない式でも構いません. 尚, 〈式 〉 が函数 f(x)の多項

式であれば,最初に f(x)に対して解き, その結果が cであれば,方程式 f(x)=cを解く事で対処出来
ます.
具体的には以下の処理を行います.

(%i26) solve(log(x)^2-2*log(x)+1,log(x));

(%o26) [log(x) = 1]

(%i27) solve(%o25[1],x);

(%o27) [x = %e]

〈式 〉 が 1変数のみの場合は 〈変数 〉 を省略出来ます.更に, 〈式 〉 は有理式でも良く,その上,三
角関数,指数関数等を含んでいても構いません.

solveは与えられた方程式が単変数の場合は次の手順で解の計算を行います.

• 方程式が変数 var の線形結合であれば, var に対して自明に解けます.

• 方程式が a · varn + b の形式ならば,解は (−b/a)1/n に 1の n 乗根を掛けたもので得られ

ます.

• 方程式が変数 var の線形結合ではなく,方程式に含まれる変数 var の各次数の gcd(nとしま
す)が次数を割切る場合,大域変数 multiplicitiesに n が追加されます.そして,solveは varn

で方程式を割った結果に対して再び呼出されます.

• 方程式が因子分解されている場合,各因子に対して solveが呼出されます.

• 方程式二次,三次,又は四次の多項式方程式の場合,解の公式を必要があれば用います.

solve([〈方程式1〉 ,· · ·,〈方程式n〉 ], [〈変数1〉 ,· · ·,〈変数n〉 ])の場合, 多項式の方程式系を linsolve,
或いは algsysを用いて解き,その変数で解のリストを返します.ここで,linsolveを用いる場合,第一
引数のリスト [〈方程式i〉 ,i=1,· · ·,n]は解くべき方程式を表現し,第二の引数リストは求めるべき未
知変数のリストになりますが,方程式中の変数の総数が方程式数と等しい場合,第二の引数リスト
は省略しても構いません.
与えられた方程式が十分でない場合,inconsistentと云うメッセージを表示します. これは大域変

数 solve inconsistent errorで制御出来ます. 単一解が存在しない場合は singularと表示されます.

solve函数の挙動に影響する大域変数¶ ³
変数名 初期値 概要

solvedecomposes true polydecompを用いるかどうかを制御
solveexplicit false 陰的な解を許可するかどうかを制御

solvefactors true 因子分解の実行の有無

solvenullwarn true 空リストの警告の有無

solveradcan false radcanを用いるかどうかを指定
solvetrigwarn true 方程式を解く際に逆三角函数を利用するかを指定

solve inconsistent error true 階数が不十分な連立方程式に対するエラー表示の有無

breakup true 解の表示を制御µ ´
solvedecomposesが trueであれば,多項式を解く際に,solveに polydecompを導入します.
solveexplicitが trueであれば solveに陰的な解,即ち,f(x)=0の形式で返す事を禁止します.
solvefactorsが falseであれば,solveは式の因子分解を実行しません.
solvenullwarnが trueであれば,空の方程式リストや空の変数リストで solveを呼んだ場合に警告

が出ます.例えば, solve([],[]) と入力すると警告と一緒に空リスト []が返されます.
solveradcanが trueであれば solveは radcanを用います.radcanを使うと solveは遅くなります

が,指数や対数関数を含む問題に対処出来ます.
solvetrigwarnが falseであれば,solveは方程式を解く為に逆三角関数を利用し, それによって解

が失なわれる事を警告しません.
solve inconsistent errorが trueであれば,solveと linsolveは, [a+b=1,a+b=2]の様に階数が不十

分な線形連立方程式に遭遇すればエラーを表示します.尚,falseであれば,空リスト []を返します.
breakupが falseであれば,solveはデフォルト値の幾つかの共通部分式で構成されたものとして

はなく,一つの式として三次又は二次の方程式の解の表示を行います.但し,breakupが trueとなる
のは programmodeが falseの時だけです.

5.3.6 漸化式の場合

Maximaでは,漸化式を扱う事が可能です.但し,機能的にはまだ不十分です.

漸化式を解く函数¶ ³
funcsolve(〈方程式 〉 ,〈g(t)〉 )µ ´

funcsolve函数は 〈方程式 〉 を満たす有理函数 〈g(t)〉 が存在すれば有理函数のリストを返し,存
在しない場合は空リスト []を返します.
但し,方程式は 〈g(t)〉 と 〈g(t + 1)〉 の一次線形多項式でなければなりません.即ち,funcsolveは一

次線形結合の漸化式に対して利用可能です.

(%i28) funcsolve((n+1)*foo(n)-(n+3)*foo(n+1)/(n+1) =

(n-1)/(n+2),foo(n));

dependent equations eliminated: (4 3)

n

(%o28) foo(n) = ---------------

(n + 1) (n + 2)

5.4 極限

5.4.1 極限について

Maximaでは limitが使えます.一見すると limitは代入と似たものに見えるかもしれませんが,
実際は全く異った操作です.
例えば, sin(x)

x の原点での値は何になるでしょうか？ 安易な代入では,00 となってしまって判りま
せんね. 因に,00 や

∞
∞ は不定形と呼ばれるものです. 字面は同じでも,安易に割ってしまっては意味

がありません.
さて, sin(x)

x に話を戻しましょう. この場合は sin (x)の原点周りの級数展開を考えると判り易く
なります.

sin (x) =
∑∞

i=0(−1)i x2i+1

(2i+1)! となり,これを x で割ってしまうと, sin(x)
x = 1 + x · (xの羃級数)と

なります.以上から,x を 0 に近づけると 1 になる事が判ります.
Maxima で試してみましょう.Maxima には limit 函数があり, この函数は limit(〈 函数 〉, 〈 変

数 〉, 〈値 〉) で極限の計算が行えます.

(%i39) limit(sin(x)/x,x,0);

(%o39) 1

(%i40) plot2d(sin(x)/x,[x,-50,50]);

極限はこの様な計算を行いますが,この近付けるという操作には方向を考えなければなりません.
例えば, 1

x はどうでしょうか.この函数は x > 0なら正で,x < 0で負になっており,x = 0が不連続点
になっています.

limit函数を使って,Maximaで計算させてみましょう.猶,Maximaでは数直線の右側 (プラス側)
から近付ける場合,左側 (マイナス側)から近付ける場合を各々オプションの plusやminusを用い
て指定する事が可能です.

(%i73) limit(1/x,x,0);

(%o73) und

(%i74) limit(1/x,x,0,plus);

(%o74) inf

(%i75) limit(1/x,x,0,minus);

(%o75) minf

この様に右側から近付けた場合には正の無限大,左側から近付けた場合には負の無限大となって
いますね.この様に左右の極限が異なるので, limit(1/x,x,0)の結果は undとなっています.

limit函数は正の無限大を inf,負の無限大をminf,複素数での無限大 (無限遠)を infinity, 左右の
極限が異なる場合には und,未定でも有界なものには indといった表記で結果を返します.

(%i89) limit(1/(x^2-1),x,1,plus);

(%o89) inf

(%i90) limit(1/(x^2-1),x,1,minus);

(%o90) minf

(%i91) limit(1/(x^2-1),x,1);

(%o91) und

(%i92) limit(sin(1/x),x,0);

(%o92) ind

(%i93) limit(1/(x^2+1),x,%i);

(%o93) infinity

極限に関連する大域変数¶ ³
変数名 初期値 概要

lhospitallim 4 limit函数で用いられる l’Hospital則の適用階数の上限
limsubst false limitの代入を制御
tlimswitch false taylor展開を利用するかどうかµ ´

lhospitallimは limitで用いられる l’Hospital則の適用回数の最大値.これは limit(cot(x)/csc(x),x,0)
の様な場合に無限ループに陥いる事を防ぐ為のものです.

limsubstは limitが未知の形式に代入を行う事を防ぎます.これは limit(f(n)/f(n+1),n,inf)の様
な式が 1となるバグを避ける為です. limsubstが trueであれば,この様な代入が許容されます.

tlimswitchが trueであれば,極限パッケージは可能な時に taylor展開を利用します.

limit函数¶ ³
limit(〈式 〉 ,〈変数 〉 , 〈値 〉 ,〈方向 〉 )
limit(〈式 〉 ,〈変数 〉 ,〈値 〉 )
limit(〈式 〉)µ ´

limit函数は与えられた 〈式 〉 の極限を計算します. この際,変数が近づく方向を指定する事も可
能です

.この場合, 〈変数 〉 が 〈値 〉 に 〈方向 〉 で指定したから接近する場合の 〈式 〉 の極限を計算しま
す.ここで,方向は右極限なら plus,左極限なら minusとします. 尚,方向を省略した場合は両側極
限が計算されます.
ここで,原点の極限計算であれば特別に zeroaや zerobも使えます. この場合は zeroaが原点の左

側 (-側),zerobが原点の右側 (+側)から近付ける事を意味します.この場合は,minusや plusの様な
方向を指定する必要はありません.
計算手法は,Wang,p.の”Evaluation of definite integrals by symbolic manipulation”- ph.d. thesis

- Mac tr-92 October 1971.を参照して下さい.
limitは特別な記号として inf(正の無限大)とminf(負の無限大)を用います. 出力では und(未定

義),ind(不定だが有界)と infinity(複素無限大)が使われる場合があります.尚,-infとminf, -minfと
infはMaximaでは意味が異なるので注意して下さい.この様に極限を含む式の計算で limit函数が

利用可能です即ち,上記の-infや-minfに加え,inf-1様な定数式に対しては,limit函数は式のみで評
価を行います.

(%i51) inf - 1;

(%o51) inf - 1

(%i52) limit(%);

(%o52) inf

(%i53) limit(-inf);

(%o53) minf

(%i54) limit(x^2+inf*x);

Is x positive, negative, or zero?

pos;

(%o54) inf

この例で示す様に,inf-1の様な式はMaximaでは自動的に評価されませんが, limit函数を用いて値
を評価する事が出来ます.limit(xˆ 2+inf*x)の様な式も同様です.

tlimit函数¶ ³
tlimit(〈式 〉 ,〈変数 〉 ,〈値 〉 , 〈方向 〉 )
tlimit(〈式 〉 ,〈変数 〉 ,〈値 〉 )
tlimit(〈式 〉)µ ´

tlimswitchを trueにした limit関数です.この tlimit函数は 〈式 〉 のテイラー展開を行い.その展
開式に対して極限計算を行います.

5.5 微分

5.5.1 微分に関係する函数

Maximaの式の微分では diff函数を用います.
diff函数¶ ³

diff(〈式 〉 ,〈変数1〉 ,〈階数1〉 , · · ·,〈変数n〉 ,〈階数n〉 )
diff(〈式 〉 ,〈変数 〉)
diff(〈式 〉)µ ´

1階微分の場合のみ,diff(〈式 〉 ,〈変数 〉 )の様に階数を省略しても構いませんが,通常は 〈変数i〉
と 〈階数i〉 の一組を指定して 〈式 〉 の微分を行います.

diff(〈式 〉) は全微分を与えます.即ち,〈式 〉の各変数に対する微分と,各変数の函数 del との積の
和になります.

(%i41) diff(f(x*y));

d d

(%o41) (-- (f(x y))) del(y) + (-- (f(x y))) del(x)

dy dx

(%i42) diff(g(x+y+z));

d d

(%o42) (-- (g(z + y + x))) del(z) + (-- (g(z + y + x))) del(y)

dz dy

d

+ (-- (g(z + y + x))) del(x)

dx

微分方程式を記述する場合,函数の名詞型’diffを用います.ここで微分の名詞型の表示はデフォル
トで二次元的書式になりますが,大域変数 display2dを falseにすれば入力と同等の式を一行で返し
ます.
この大域変数 display2dは微分の名詞型に限らず,Maximaの計算結果の表示で数学式の表示を制

御する大域変数ですが,名詞型の微分の表示のみを制御する大域変数として大域変数 derivabbrev
があります.

微分の表示を制御する大域変数¶ ³
変数名 デフォルト値 概要

derivabbrev false 微分の表示を制御µ ´
この大域変数 derivabbrevが trueの場合,名詞型の微分は添字で表示されます.

(%i30) ’diff(f(x),x);

d

(%o30) -- (f(x))

dx

(%i31) derivabbrev:true$

(%i32) ’diff(f(x),x);

(%o32) f(x)

x

(%i33) display2d:false$

(%i34) ’diff(f(x),x);

(%o34) ’diff(f(x),x,1)

更に,大域変数 derivsubstで名詞型の微分項の代入制御が行えます.

名詞型の微分の代入を制御する大域変数¶ ³
変数名 デフォルト値 概要

derivsubst false 名詞型の微分を含む項の代入を制御µ ´
大域変数 derivsubstは微分を含む項の代入の制御を行います.例えば, d2y

dt2 は y の t による二階

微分ですが, dy
dt を x と置くと, d2y

dt2 は
dx
dt で置換えられます.大域変数 derivsubstは名詞形の微分

を含む式に対し,この様な置換を行うかどうかを制御するものです.
大域変数 derivsubstが falseの場合,susbst函数による微分項の置換が出来ませんが, trueの場合

は置換が行えます.

(%i33) derivsubst;

(%o33) false

(%i34) subst(x,’diff(y,t),’diff(y,t,2));

2

d y

(%o34) ---

2

dt

(%i35) derivsubst:true;

(%o35) true

(%i36) subst(x,’diff(y,t),’diff(y,t,2));

dx

(%o36) --

dt

(%i37) subst(x,’diff(y,t),2*t+t^2*’diff(y,t,2));

2 dx

(%o37) t -- + 2 t

dt

次に,微分の名詞型を含む式に対し,その式に含まれる 名詞型微分の最大の階数を返す函数があ
ります.

derivdegree函数¶ ³
derivdegree(〈式 〉 ,〈従属変数 〉 , 〈独立変数 〉 )µ ´

derivdegree函数は 名詞型の微分を含む式で,〈独立変数 〉 に対する 〈従属変数 〉 の微分で最も高
い階数を見付けます. この函数は多項式の次数を返す函数 hipowと似ています.

(%i16) derivdegree(’diff(y,x,3)*x^4+’diff(y,x,2)*’diff(y,x),y,x);

(%o16) 3

(%i17) ’diff(’diff(y,x,2),x,3)+’diff(y,x,2);

5 2

d y d y

(%o18) --- + ---

5 2

dx dx

(%i19) derivdegree(%,y,x);

(%o19) 5

但し,derivdegree函数は与式の展開等を行わないので,場合によっては間違った答を返す事もあ
るので注意が必要です.

(%i26) ’diff(’diff(y,x,2),x,3)*(x^2-1)+’diff(y,x,2)

-’diff(y,x,5)*(x-1)*(x+1);

5 5 2

2 d y d y d y

(%o26) (x - 1) --- - (x - 1) (x + 1) --- + ---

5 5 2

dx dx dx

(%i27) derivdegree(%,y,x);

(%o27) 5

(%i28) expand(’diff(’diff(y,x,2),x,3)*(x^2-1)+’diff(y,x,2)

-’diff(y,x,5)*(x-1)*(x+1));

2

d y

(%o28) ---

2

dx

(%i29) derivdegree(%,y,x);

(%o29) 2

この例では,与式の 5階の微分は式を展開する事で消去されるものですが, derivdegree函数は与
式を展開する事を行わずに,安易に式に含まれる yの xによる微分で階数の最も高い項を求め,そ
の階数を返却しています.

5.5.2 vectパッケージ

vectパッケージはデフォルトでMaximaに含まれているパッケージで,grad,div,curlや laplace等
の微分演算子や,それらの式を簡易化する函数が含まれています.

vectパッケージでは非可換積.を内積演算子として再定義します.その為,非可換積が可換化され
てしまうので注意が必要です.更に,非可換積の結合律と対応する羃への簡易化を勝手に実行しない
様にする為,関連する大域変数 dotsassoc と dotexptsimpが false に設定されます.その一方で,ス
カラーに対して,大域変数 dotscrules を trueにする事で, スカラーとの非可換積が可換に設定され
ます.

vectパッケージには以下の演算子の宣言とそれに付随する函数が収録されています.
vectパッケージに含まれる主な函数¶ ³

express(〈expression〉)
potential(〈grad〉
scalefactors(〈座標変換 〉)
vectorsimp(〈ベクトル式 〉)µ ´

vectパッケージに含まれる演算子¶ ³
演算子 演算子の属性 左束縛力 右束縛力

~ 内挿式演算子 134 133
grad 前置式演算子 142
div 前置式演算子 142
curl 前置式演算子 142
laplacian 前置式演算子 142µ ´

vect.macに含まれているこれらの演算子は宣言された演算子であり,演算子の実体は,share/vector/vector.mac
に含まれています.
これらの演算子の簡易化は expandflags変数に割当てられたリストに含まれる変数で制御されま

す.但し,これらの変数を変更しても,直ちに上記の演算子が自動的に簡易化を行うのではありませ
ん.vectパッケージの函数 vectsimp函数を用いて簡易化を行います.但し,vectorsimp函数自体は内
部で演算子の属性を大域変数 expandflagのリストに登録された大域変数から設定し,vsimp函数で
実際の処理を実行させています.

expandflags¶ ³
変数名 初期値 概要

expandall false 全ての演算子を展開

expandplus false 被演算子に含まれる和を展開

expanddot false 和と内積.を展開
expanddotplus false 和と内積.を展開
expandgrad false grad演算子を展開
expandgradplus false 被演算子の和で展開

expanddiv false div演算子を展開
expanddivplus false 被演算子の和で展開

expandcurl false curl演算子を展開
expandcurlplus false 被演算子の和で展開

expandlaplacian false laplace演算子を展開
expandlaplacianplus false 被演算子の和で展開

expandprod false 積に対して展開

expandgradprod false grad演算子にて積を展開
expanddivprod false div演算子にて積を展開
expandlaplacianprod false laplace演算子にて積を展開
expandcurlcurl false curl curlを処理
expandlaplaciantodivgrad false laplace演算子を dvi gradに展開
expandcross false 外積を展開

expandcrosscross false 外積を外積に対して展開

expandcrossplus false 外積を和に対して展開

firstcrossscalar false 外積とスカラーの処理µ ´
全てのこれらの大域変数はデフォルト値として falseを持ちます.変数名の後に plusの付く大域

変数は加法性と被演算子の分配性に関連します.同様に,後に prod の付く大域変数は可換積や内積
(通常は非可換積)等の積演算に対する被演算子への分配性に関連するものです.

expandcrosscrossは p~(q~r)を (p,r)*q-(p.q)*rで置換するかどうかを決めます.
expandcurlcurlは curl curl p を grad div p + div grad p で置換するかどうかを決定します.
expandcrossが trueの場合,expandcrossplusと expandcrosscrossが trueと同じ効果があります.
二つの大域変数 expandplusと expandprodは似た名前の大域変数に trueに設定した場合と同効

果です.これらが,trueであれば,他の大域変数, expandlaplaciantodivgradは laplace演算子を div
gradで置換えます.
尚,簡便の為にこれら全ての大域変数は,load(vect)で vectパッケージの読込を行った時点で evflag

として宣言されます.

(%i1) load(vect)$

(%i2) expandall:true$

(%i3) laplacian(a*V+b*W);

(%o3) laplacian (b W + a V)

(%i4) vectorsimp(laplacian(a*V+b*W));

(%o4) laplacian (b W) + laplacian (a V)

(%i5) vectorsimp(grad(a*V+b*W));

(%o5) grad (b W) + grad (a V)

(%i6) vectorsimp(grad(a*b));

(%o6) grad (grad a . b) + grad (a . grad b)

(%i7) (V1+V2).(W1+W2);

(%o7) (V2 + V1) . (W2 + W1)

(%i8) vectorsimp((V1+V2).(W1+W2));

(%o8) V2 . W2 + V2 . W1 + V1 . W2 + V1 . W1

express函数¶ ³
express(〈式 〉 )µ ´

express函数は名詞型の微分を展開します.express函数はvectパッケージで定義されるgrad,div,curl,
laplacianと外積~を認識します. 但し,express函数は微分を名詞型で返す為,実際の微分の計算は
ev函数に’diffオプションを付けて行います.

(%i1) load(vect);

(%o1) /usr/local/share/maxima/5.9.2/share/vector/vect.mac

(%i2) e1:laplacian(x^2*y^2*z^2);

2 2 2

(%o2) laplacian (x y z )

(%i3) express(e1);

2 2 2

d 2 2 2 d 2 2 2 d 2 2 2

(%o3) --- (x y z ) + --- (x y z ) + --- (x y z )

2 2 2

dz dy dx

(%i4) ev(%,’diff);

2 2 2 2 2 2

(%o4) 2 y z + 2 x z + 2 x y

(%i5) v1:[x1,x2,x3]~[y1,y2,y3];

(%o5) [x1, x2, x3] ~ [y1, y2, y3]

(%i6) express(v1);

(%o6) [x2 y3 - x3 y2, x3 y1 - x1 y3, x1 y2 - x2 y1]

5.6 積分

5.6.1 記号積分について

Maximaは記号積分,数値積分の両方が行えます.積分に関しては,Rischの積分も不完全ながら
実装されています.その為,単純に公式を当て嵌めるだけで積分の計算を行う様なシステムよりも
一段と優れた処理が行えます.
尚,Maximaは初等函数 (有理式,三角函数,対数,指数)と多少の拡張 (error函数, dilogarithm)で

可積分なものに限定しているので,g(x)や h(x)の様な未知函数の積分は扱いません.

記号積分を行う函数¶ ³
integrate( 〈式 〉, 〈変数 〉 )
integrate( 〈方程式 〉, 〈変数 〉 )
integrate( [〈式1〉, · · · , 〈式n〉], 〈変数 〉 )
integrate( [〈方程式1〉, · · · , 〈方程式n〉], 〈変数 〉 )
integrate( 〈行列 〉, 〈変数 〉, 〈下限 〉, 〈上限 〉 )
risch( 〈式 〉, 〈変数 〉 )µ ´

Maximaで記号積分を行う函数に,integrate函数と risch函数の二種類があります. integrate函数
から,risch函数の本体を呼出す事も,ev函数を利用する事で出来るので,記号積分に関しては integrate
函数だけで済ませられます.
積分の計算で式が非常に繁雑になる場合や,積分記号の文字による表示が不要な場合, 大域変数

display2dを false にすると,一行で結果が表示されます.複雑な式の計算を行う場合には非常に便
利です.

integrate函数は,通常の式,方程式 (演算子=を含む式)や,これらのリストや行列の積分も行えま
す.それに対し,risch函数は比較の演算子 (=, >,<)を含まない通常の式の積分に限定されます.
ここで,integrate函数を使って方程式の不定積分を行うと,積分定数が結果の式に導入されます.

この積分定数は順番が付いており,この順番が大域変数 integration constant counterに記録されて
います.

積分定数を定める大域変数¶ ³
変数名 初期値 概要

integration constant counter 0 積分定数のカウンターµ ´
実際に,その動作を見てみましょう.

(%i1) integration_constant_counter;

(%o1) 0

(%i2) integrate(x^2,x);

3

x

(%o2) --

3

(%i3) integrate(x^2=0,x);

3 /

x [

(%o3) -- = integrationconstant1 + I 0 dx

3 ]

/

(%i4) integration_constant_counter;

(%o4) 1

(%i5) integrate(x^3=0,x);

4 /

x [

(%o5) -- = integrationconstant2 + I 0 dx

4 ]

/

(%i6) integration_constant_counter;

(%o6) 2

この様に,演算子=を含まない式の積分では,結果に積分定数 integrationconstantが含まれませ
んが,演算子=を含む式の積分では integrationconstantが出現します. これは,integration函数が呼
出す sinint函数内部でその様に設定されている為で, この設定は,他では行われていません.

integrate函数から risch函数を用いて Risch積分を実行させる事が可能です. この場合は ev函
数を用います.

(%i21) ev(integrate(3^log(x),x),’risch);

log(3) log(x)

x %e

(%o21) -----------------

log(3) + 1

(%i22) trigsimp(%);

log(x)

x 3

(%o22) ----------

log(3) + 1

(%i23)

尚,integrate函数は depends函数で設定される大域変数 dependenciesの影響を受けません.
integrate函数は定積分の計算も出来ます.この場合は defint函数と同じ引数を取ります.尚,実際

に定積分を実行するのは defint函数です. 尚,defint函数による定積分は,数値計算を主体としたも
のではなく,記号積分を主体としたものです.数値積分が必要な場合,Maximaには romberg函数や
quanc8函数 を用います.
広義の定積分では,正の無限大には inf, 負の無限大には minf,複素無限大には infinity が使えま

す.この infとminfに関しては, -infや-minfが各々minfや infと同値なものではない事に注意して
下さい. Maximaの広義の積分や,その他の代入操作で infやminfは普通に使えますが, -infや-minf
を用いると全く無意味な結果を得る事があるので注意が必要です.その為,-infや-minfが現れる式
は limit函数で一度評価しましょう.
積分形式 (例えば,幾つかの助変数に関してある数値を代入するまで計算出来ない積分)が必要で

あれば,名詞型の’integrateを利用します.
risch函数は Rischの積分アルゴリズムから,Transcendential caseを用いて式の積分を行います.

尚,Maximaでは Rischアルゴリズムで代数的な場合は実装されていません.
risch函数は,integrateの主要部が処理出来ない入れ子状態の指数函数と対数函数の場合の処理が

行えます.integrateは,これらの場合,自動的に risch函数を適用します.

(%i24) risch(x^2*erf(x),x);

2

3 2 - x

%pi x erf(x) + (sqrt(%pi) x + sqrt(%pi)) %e

(%o24) -------------------------------------------------

3 %pi

(%i25) diff(%,x),ratsimp;

2

(%o25) x erf(x)

risch函数の動作を制御する大域変数に erfflagがあります.

大域変数 erfflag¶ ³
変数名 デフォルト値 概要

erfflag true risch函数による erf函数の挿入を制御µ ´
この大域変数 erfflagが falseであれば,erf函数が被積分函数の中に含まれていない場合,risch函

数が答に erf函数を入れる事を抑制します.
ここで,erf函数は次の error函数の事です.

erf函数¶ ³
erf(x)µ ´
この函数の微分 d

dx (erf(x)) は 2e−x2

√
π
となります.

5.6.2 変数の変換について

Maximaでは被積分函数の変数を新しい変数で置換えて積分する事が可能です. この為に,changevar
函数を用います.

変数変換を行う函数¶ ³
changevar(〈式 〉, 〈f(x, y)〉, 〈y〉, 〈x〉)µ ´

changevar函数は 〈式 〉 に現われる全ての 〈x〉 に対する積分で 〈f(x, y)〉 =0を満す 〈y〉 を新し
い変数とする変数変換を行います.

(%i17) assume(z>0)$

(%i18) I1:’integrate(%e^sqrt(1-y),y,0,1);

1

/

[ sqrt(1 - y)

(%o18) I %e dy

]

/

0

(%i19) changevar(I1,y+z^2-1,z,y);

0

/

[ z

(%o19) - 2 I z %e dz

]

/

- 1

(%i20) ev(%,risch);

- 1

(%o20) - 2 (2 %e - 1)

changevarは総和 (
∑

)や積 (
∏

)の添字の変更にも使えます. この場合,添字の変更は単純な
シフトだけで,次数の高い函数には出来ません.

(%i3) sum(a[i]*exp(i-5),i,0,inf);

inf

====

\ i - 5

(%o3) > %e a

/ i

====

i = 0

(%i4) changevar(%,i-5-n,n,i);

inf

====

\ n

(%o4) > %e a

/ n + 5

====

n = - 5

5.6.3 有理式の記号積分

多項式 f(x),g(x)の有理式 f(x)
g(x) の積分は比較的容易に行えます. 但し, 1

x3−x2−x+4 の様に式の因

数分解が容易に出来ない式の積分は, そのままでは上手く計算出来ません.

(%i3) integrate(1/(x^3-x^2-x+4),x);

/

[ 1

(%o3) I --------------- dx

] 3 2

/ x - x - x + 4

この様な有理式の記号積分に対しては,大域変数 integrate use rootsofを true に設定する事で,
計算を行う事が可能です.

代数的数の利用を制御する大域変数¶ ³
変数名 デフォルト値 概要

integrate use rootsof false 代数的数を用いた integrate函数による積分を実行µ ´
先程の例に対し,大域変数 integrate use rootsofを trueにして,integrate函数を使って再度積分

を行ってみましょう.

(%i4) integrate_use_rootsof:true;

(%o4) true

(%i5) integrate(1/(x^3-x^2-x+4),x);

====

\ log(x - %r1)

(%o5) > ------------------

/ 2

==== 3 %r1 - 2 %r1 - 1

3 2

%r1 in rootsof(x - x - x + 4)

今度は積分の計算が出来ました.大域変数 integrate use rootsofが trueの場合,integrate函数は
有理式の積分で,分母の多項式から定義される代数的数を導入して積分を実行します.この意味を
上の例で説明しましょう. 先ず,integrate use rootsofが trueに設定されると,Maximaの integrate
函数は与式の分母 x3 − x2 − x + 4 から,方程式 x3 − x2 − x + 4 = 0 の根を決めます. Maximaで
は単純に%r1等の%rで開始する変数を導入します.しかし,このままでは判り難いので,この例で
はもう少し細かく説明します.この例の場合,三次方程式なので根は 3個あり,それらを α,β,γ とし
ます.すると,与式はこれらの根を用いると, 1

(x−α)(x−β)(x−γ) に変換されます. この様に分母が一次

式の有理式の積になってしまえば,後は簡単に計算する事が可能です.実際の処理では,各根に記号
を割当てる必要は無く,方程式 xxxの解,%r2といった括り方で十分です.
但し,この方法の気持ちの悪さは方程式の根%r1が不明瞭ながら式に存在する事です. 又,返却さ

れる式は名詞型の為に,安易に微分が出来ません.何故なら,Maximaによるこの処理は,単純に公式
に当て嵌めているだけだからです.

5.6.4 記号積分の結果検証

記号積分の計算は,記号微分と比べると格段に難しい問題の為,比較的単純な式の計算でも思わ
ぬ結果を得る事があります.その為,記号積分の結果は何等かの形で検算を行う事を強く薦めます.
最も簡単な方法は積分した結果を微分して同じ結果が得られるかどうかを確認する事です.

Maximaの integrate函数や risch函数を使っても上手く計算出来ない簡単な式の例として,
√

x + 1
x − 2

を挙げておきます.この式は
√

(x+1)2

x に変形出来ますが,この式の形の違いで結果が大きく異なり
ます.

(%i44) integrate(sqrt(x+1/x-2),x);

3/2

2 x - 6 sqrt(x)

(%o44) ------------------

3

(%i45) integrate(sqrt(factor(x+1/x-2)),x);

/

[ abs(x - 1)

(%o45) I ---------- dx

] sqrt(x)

/

(%i46) assume(x<1 and x>0);

(%o46) [x < 1, x > 0]

(%i47) integrate(sqrt(x+1/x-2),x);

3/2

2 x - 6 sqrt(x)

(%o47) ------------------

3

(%i48) integrate(sqrt(factor(x+1/x-2)),x);

3/2

2 x - 6 sqrt(x)

(%o48) - ------------------

3

(%i49) diff(%,x);

3

3 sqrt(x) - -------

sqrt(x)

(%o49) - -------------------

3

(%i50) ratsimp(%);

x - 1

(%o50) - -------

sqrt(x)

この例では,同じ integrateでも
√

(x+1)2

x の場合は名詞型を返しており,何も考えずに x−1√
x
の計

算を行ってはいません. 又,assume函数を使って,条件 0 < x < 1 を追加した場合でも,
√

x + 1
x − 2

の integrateの計算は間違っています.
この様にMaximaの積分は正しい答を返すとは限りませんが,内部処理を適切に行う事で正しい

答を得る事も可能な場合もあります.これはMaximaに限った話ではなく, 数式処理一般でも言え
る事です.更に,記号積分の結果は面倒でも確認した方が安全な事は強調しておきます.
何故,式の形の違いで計算結果に違いが出るのでしょうか？これは結局,式の並びの照合等によっ

て処理を行っている為,式を変形していれば.並びも勿論異なる為に照合の結果も異なり,それによっ
て処理の流れが違う為です.数値計算で積分を行う場合, 式の並びは無関係で,函数の値を主に見る
為,この様な現象は累積誤差の事を除くとまず生じません.
並び照合の結果が違っていても,正しく処理が出来ていれば,最終的に一致する筈ですが,この例

の様に何処かの処理を間違えると当然結果が異なります. 積分の計算の場合は特に,expand函数で
式を展開したり,factor函数で因子分解する等の処理を実行して積分したものと結果を照合する事
は,正しい答を得る為の有効な手段です.
しかし,実際はこれでも不十分な事があります.例えば, 3

5−4cos(x) の積分です.

(%i13) integrate(3/(5-4*cos(x)),x);

3 sin(x)

(%o13) 2 atan(----------)

cos(x) + 1

(%i14) trigsimp(diff(%,x));

3

(%o14) - ------------

4 cos(x) - 5

積分した結果を微分すると,元の式が出ているので,正しい結果が出ている様に思えます.しかし,
残念ながら計算は間違っています.何故なら被積分函数は滑かな連続函数ですが,結果の方は不連
続函数になっています.この様な代物はグラフを利用して積分した函数を描くと間違いが一目で判
ります.

Maximaは積分処理の為の函数を幾つか持っています.その中でも,integrate函数は最もよく使わ
れるものです.この integrate函数は不定積分も定積分も処理出来ます. 定積分だけであれば,defint
函数もあります.更に,極限操作が必要な場合には, ldefint函数もあります.但し.この函数は非常に
曲者で,integrate函数で記号積分した結果に境界値を代入する代物です.従って,区間内に極が存在
する場合,その極を検出せずに安易に計算を行う為,注意が必要です.
参考迄に,defint函数の処理を簡単に説明しておきます.通常は integrate函数は, LISP内部では

$integrateです.この函数は内部で sinint函数を呼出し,その結果に上限と下限の値を代入していま
す.sinintでも risch積分を行う rischintを呼出す事も出来ますが,ev函数を用いて rischintを用いる
様に指定する事が出来ません. この点を修正する為には defint.lispの antiderivの修正が最低でも
必要となります.

defint函数や integrate函数による定積分の計算で,下限や上限に記号や式が含まれている場合,そ
の正負を尋ねてくる事があります.この場合,正であれば pos; , 負であれば neg; ,零であれば zero;
と入力します. 更に,assumeを用いて正負の指定を予め行っていれば,Maximaはこの様な質問を行
ないません.

(%i24) integrate(sqrt(2*x-x^2),x,0,a);

Is a positive, negative, or zero?

pos;

2

asin(a - 1) + (a - 1) sqrt(2 a - a ) %pi

(%o24) ------------------------------------ + ---

2 4

(%i25) assume(a>0 and a<1);

(%o25) [a > 0, a < 1]

(%i26) integrate(sqrt(2*x-x^2),x,0,a);

2

(a - 1) sqrt(2 a - a ) - asin(1 - a) %pi

(%o26) ------------------------------------ + ---

2 4

留数を計算する函数¶ ³
residue( 〈式 〉, 〈変数 〉, 〈点 〉 )µ ´

residue函数は 〈点 〉 回りの 〈式 〉 の複素平面上での留数を計算します. 尚,留数は式の laurent
級数展開を行った時の (〈変数 〉 − 〈点 〉)−1 の項の係数になります.

(%i5) residue(x/(x^2+1),x,%i);

1

(%o5) -

2

(%i6) residue(sin(x)/x^4,x,0);

1

(%o6) - -

6

ラプラス変換に関連する函数¶ ³
laplace( 〈式 〉, 〈旧変数 〉, 〈新変数 〉 )
ilt( 〈式 〉, 〈旧変数 〉, 〈新変数 〉 )
delta (t)µ ´

laplace函数は 〈旧変数 〉 と 〈新変数 〉 に対する 〈式 〉 の Laplace変換を計算します.逆 laplace
変換は iltです.
〈式 〉 は多項式に函数 exp,log, sin,cos,sinh,coshと erf 函数を含むもの,atvalueの従属変数が使

われている定数係数の線形微分方程式でも構いません.
初期条件は零で指定されていなければならないので,他の一般解の何処かに押込める境界条件が

あれば,その境界条件に対して一般解を求めて値を代入して定数消去が出来ます.
〈式 〉 に畳込み (convolution integral)を含んでいても構いません. laplace函数が適切に動作す

る為に,函数の従属性をはっきりと表示していなければなりません.つまり,函数 fが変数 xと yに
従属しているのであれば, laplace(’diff(f(x,y),x),x,s) の様に函数 fが現れる場合は何時でも f(x,y)
と記述する必要があります.
尚,laplace函数は depends函数で設定される大域変数 dependenciesの影響を受けません.

(%i106) laplace(%e^(2*t+a)*sin(t)*t,t,s);

a

%e (2 s - 4)

(%o106) ---------------

2 2

(s - 4 s + 5)

(%i107) ilt(%,s,x);

2 x + a

(%o107) x %e sin(x)

ilt函数は 〈新変数 〉 と 〈旧変数 〉 に対する 〈式 〉 の逆 Laplace変換を計算します. 〈式 〉 は分母
が一次と二次の因子を持った有理式でなければなりません.ilt函数による処理を効率的に行う為に
は有理式の展開を予め実行しておくと良いでしょう.

laplaceと iltの両方を solveや linsolveと使うと,単変数の微分方程式か畳込み積分方程式を解く
事が出来ます.

(%i11) ’integrate(sinh(a*x)*f(t-x),x,0,t)+b*f(t)=t^2;

t

/

[ 2

(%o11) I f(t - x) sinh(a x) dx + b f(t) = t

]

/

0

(%i12) laplace(%,t,s);

a laplace(f(t), t, s) 2

(%o12) b laplace(f(t), t, s) + --------------------- = --

2 2 3

s - a s

(%i13) linsolve([%],[’laplace(f(t),t,s)]);

2 2

2 s - 2 a

(%o13) [laplace(f(t), t, s) = --------------------]

5 2 3

b s + (a - a b) s

(%i14) ilt(rhs(first(%)),s,t);

Is a b (a b - 1) positive, negative, or zero?

pos;

sqrt(a b (a b - 1)) t

2 cosh(---------------------) 2

b a t 2

(%o14) - ----------------------------- + ------- + ------------------

3 2 2 a b - 1 3 2 2

a b - 2 a b + a a b - 2 a b + a

delta函数は Diracの δ 函数です.尚,laplace函数のみが δ 函数を認識しています.

(%i38) laplace(delta(t-a)*sin(b*t),t,s);

Is a positive, negative, or zero?

pos;

- a s

(%o38) sin(a b) %e

この例では変数 aに関して何らの仮定や割り当てが無い為に, ”Is a positive, negative, or zero?”
とMaximaが尋ねています. assumeを用いて a > 0と仮定した場合を以下に示します.

(%i39) assume(a<0);

(%o39) [a < 0]

(%i40) laplace(delta(t-a)*sin(b*t),t,s);

(%o40) 0

定積分を行う函数¶ ³
defint( 〈式 〉, 〈変数 〉, 〈下限 〉, 〈上限 〉 )
ldefint( 〈式 〉, 〈変数 〉, 〈下限 〉, 〈上限 〉 )
tldefint( 〈式 〉, 〈変数 〉, 〈下限 〉, 〈上限 〉 )µ ´

defint函数は定積分を実行する函数です.内部的に integrateを利用しており, 原始函数を求める
と,単純に 〈上限 〉 と 〈下限 〉 の値を代入したものの差を取るものです.但し,後述の ldefintと比較
して, 区間 ( 〈下限 〉, 〈上限 〉 )に於ける 〈式 〉 の極の判定を行っているので,ldefintよりは安全で
す. 但し,romberg等の数値成分による手法の方が間違いが少ないので,その点は注意して使う必要
があります.

ldefint函数は 〈変数 〉 の 〈上限 〉 と 〈下限 〉 に関する 〈式 〉 の不定積分に対し, limit函数を用
いた評価を行い,〈式 〉 の定積分を計算します. 尚,上限と下限にminfと infといった値を与える場
合には注意が必要です. 又,-infや-minfの様に符号を付けて用いると,無意味な結果を得る事があ
るので注意が必要になります.

(%i20) ldefint(exp(-x)*sin(x),x,0,-minf);

minf minf

%e sin(- minf) %e cos(- minf) 1

(%o20) - ------------------ - ------------------ + -

2 2 2

(%i21) ldefint(exp(-x)*sin(x),x,0,inf);

1

(%o21) -

2

尚,ldefintは内部的で函数の極の判別を一切行わずに,integrateで安易に記号積分した結果に,上
限と下限を limit函数で代入するだけの函数です.
一応,右極限と左極限を下限と上限で取る様にしていますが,単純に上限に対しては’minus,下限

に対しては’plusを内部的に付けるだけなので,上限や下限で不連続になる函数の場合,上限と下限
の大小関係を逆にして ldefintを計算すれば無意味になる可能性もあります.
ここで,defint 函数や integrate 函数で定積分を行う場合は区間内での極の判別も行っています

が,ldefintでは内部的に sinint函数を用いるだけです.ところで, この sinint函数は極の判別を一切
行わずに機械的な記号積分を行います.その為, 函数をいきなり ldefint函数を用いて積分したり,結
果の検証を省く事は薦められません.
次の例をよく吟味して下さい.

(%i15) ldefint(1/x^2,x,0,1);

1

(%o15) (limit -) - 1

x -> 0 x

(%i16) ldefint(1/x^2,x,-1,0);

1

(%o16) - (limit -) - 1

x -> 0 x

(%i17) %o15+%o16;

(%o17) - 2

(%i18) defint(1/x^2,x,-1,1);

Integral is divergent

-- an error. Quitting. To debug this try debugmode(true);

(%i19) ldefint(1/x^2,x,-1,1);

(%o19) - 2

この例で示す様に,%i19の ldefintの結果は-2となっています.これはMaximaで 1
x2 を安易に記

号積分し,区間の上限と下限の極限を取っている為, この様な現象が生じています.これに対し,defint
と integrateでは極が存在する為にエラーを返しています.
尚,ldefintには zeroa,zerobが使えます.zeroaが 0の右極限,zerobが 0の左極限を表現します.

(%i7) ldefint(1/x^2,x,zeroa,1);

1

(%o7) (limit -) - 1

x -> 0+ x

(%i8) ldefint(1/x^2,x,zerob,-1);

1

(%o8) (limit -) + 1

x -> 0- x

但し,1+’zeroaの様な使い方は出来ないので注意します.
尚,ldefintは defintと同様に積分で Risch積分を用いる事が出来ません.
tldefint函数は大域変数 tlimswitch が trueに設定された ldefint函数に相当します.尚,大域変数

tlimsiwtchが trueの場合,極限の計算で Taylor展開を利用する事を意味します.

5.6.5 数値積分について

Maximaには記号積分だけではなく,数値計算による定積分の計算も行えます. この処理を行う
函数には quanc8函数と romberg函数があります.

数値積分を行う函数¶ ³
quanc8( ′〈被積分函数 〉,実数1,実数2 )
quanc8( 〈被積分函数 〉,変数,実数1,実数2 )
romberg( 〈被積分函数 〉, 〈積分変数 〉, 〈実数1〉, 〈実数2〉 )
romberg( 〈被積分函数 〉, 〈実数1〉, 〈実数2〉 )µ ´
先ず,quanc8函数は load(”qq”)で予め読込む必要があります. quanc8( ′〈被積分函数 〉,実数1,実

数2 )で,最初の引数で指定された函数の定積分を区間 実数1 から 実数2 で計算します.第一引数に
函数名を用いる場合,函数名に単引用符’を付けなければなりません

quanc8( 〈被積分函数 〉,変数,実数1,実数2 )で,函数か式 (最初の引数)の変数 (二番目の引数)で
区間 実数1 から 実数2 で定積分を計算したものになります.
使われる手法は Newton-Cotesの 8次多項式による求積法で,ルーチンは適応型です.
romberg函数は romberg積分を行う函数です.romberg函数は quanc8函数とは違い, Maximaに

自動的に読込まれます.
最初の引数は translate 函数で変換された函数か compile 函数でコンパイルされた函数でなけ

ればなりません.尚,compile函数でコンパイルされた函数の場合は, 浮動小数点型,即ち,flonum型
の函数として宣言されていなければなりません. 函数が translate函数で変換されたものでなけれ
ば,romberg函数は translate函数で変換せずにエラーを返します.積分の精度は大域変数 rombergtol
と rombergitで制御されます.
この romberg函数は再帰的に呼び出されていても構わないので,二重,三重積分が実行可能です.
尚,romberg函数の 〈実数1〉 と 〈実数2〉 は, 内部で倍精度の浮動小数点を用いている為,多倍長精

度 (bigfloat型)に変換した数値は扱えません.
相対誤差が大域変数 rombergtolよりも小さければ romberg函数は計算結果を返します. 諦める

前に大域変数 romgergit倍の刻幅を半分にして試みます.romberg函数は反復と函数評価の大域変
数 romgergabsと rombergminで制御されます.
一般的に,quanc8函数の方が計算速度,精度と安定性で romberg函数に勝っている事が多いので

すが,romberg函数の方が精度が良好な場合もあります. 以下の
√

2x − x2 の積分の例では romberg
函数の方が精度で勝り, e−xsin (x) の積分では計算速度で romberg函数が勝っています.

(%i31) showtime:all;

Evaluation took 0.00 seconds (0.00 elapsed) using 80 bytes.

(%o31) all

(%i32) romberg(sqrt(2*x-x^2),x,0,1);

Evaluation took 0.18 seconds (0.18 elapsed) using 316.305 KB.

(%o32) .7853897937007632

(%i33) quanc8(sqrt(2*x-x^2),x,0,1);

Evaluation took 0.02 seconds (0.02 elapsed) using 35.602 KB.

(%o33) .7849358178522697

(%i34) integrate(sqrt(2*x-x^2),x,0,1);

Evaluation took 0.12 seconds (0.12 elapsed) using 212.039 KB.

%pi

(%o34) ---

4

(%i35) bfloat(%);

Evaluation took 0.00 seconds (0.00 elapsed) using 808 bytes.

(%o35) 7.853981633974483B-1

(%i36) bfloat(integrate(sqrt(2*x-x^2),x,0,1));

Evaluation took 0.12 seconds (0.12 elapsed) using 212.875 KB.

(%o36) 7.853981633974483B-1

(%i37) romberg(exp(-x)*sin(x),x,0.,1.);

Evaluation took 0.01 seconds (0.00 elapsed) using 28.227 KB.

(%o37) .2458370426035679

(%i38) bfloat(integrate(exp(-x)*sin(x),x,0.,1.));

Evaluation took 0.13 seconds (0.13 elapsed) using 298.562 KB.

(%o38) 2.458370070002374B-1

数値積分の方が上記の極を検出し易い事もあり,両者の結果を比較するのも検算の方法としては
良いでしょう.

romberg函数に影響を与える大域変数¶ ³
変数名 初期値 概要

rombergabs 0.0 romberg函数の終了条件を与える変数
rombergit 11 刻幅を設定

rombergtol 1.0E-4 romberg函数の精度
rombergmin 0 函数評価の最小回数µ ´
大域変数 rombergabsは romberg函数の終了条件を与える大域変数の一つです. romberg函数内

部の反復処理で生成された値の列を y[0],y[1],y[2],· · ·とする時,n回目の反復処理で,(abs(y[n]-y[n-1])
≤ rombergabsか abs(y[n]-y[n-1])/(y[n]=0.0ならば 1.0,それ以外は y[n]) ≥ rombergtolを満した
時点で,romberg函数は答を返します.その為,rombergabsが 0.0であれば相対誤差の検証が出来ま
す.この追加変数は小さな値域で積分計算を行いたい時に便利です. そこで,小さな主要な値域で最
初に積分する事で相対的精度検証を用い,後に続く残りの値域上の積分は絶対的精度の検証で用い
ます.

romberg積分命令の精度は大域変数の rombergtolと rombergitで支配されます. rombergは,隣り
合った近似解での相対差が rombergtolよりも小さければ値を返します.諦める前に刻幅の romgergit
を半分にして試行します.

大域変数 rombergminは romberg函数による函数評価の最小数を制御します. romberg函数は
その第一引数を少なくとも 2rombergmin+2 + 1 回評価します.これは通常の収束テストが時々悪い通
り方をする周期的函数の積分に有効です.

二重積分を行う函数¶ ³
dblint( ’〈F 〉,′ 〈R〉,′ 〈S〉, 〈浮動小数点1〉, 〈浮動小数点2〉 )µ ´

dbint函数は二重積分の数値計算を行う函数で,Maximaの処理言語で記述されています. この函数
を利用する為には予め load(dblint) で dbint函数を読込んでおく必要があります.
先ず,dblint(’F,’R,’S,浮動小数点1,浮動小数点2)で以下の二重積分を計算します∫ 浮動小数点2

浮動小数点1

∫ S(x)

R(x)

F (x, y)dydx

第一引数の 〈F 〉は x,yの 2変数の函数,第二引数と第三引数の 〈R〉と 〈S〉は各々変数 xの函数
で,dblint函数には函数名のみを名詞型で引渡します.更に,これらの函数は全て,translate函数で
LISPの函数に変換されたものか,compile函数でコンパイルされたものでなければなりません.そ
の為,これらの函数は全て flonum型の函数でなければならず,多倍長精度は扱えません.

dblint函数は 〈R〉と 〈S〉の両方に Simpson則を用います.そして,dblint函数は二つの大域変数
dblint x, dblint yを持ち,各々の大域変数が xと yの区間の分割数を定めます. 尚,収束性を改善
する為に大域変数 dblint xと dblint yを大きくした場合は計算時間が増大します.

dblint函数は X軸を大域変数 dblint xの値で分割し,X軸上の各値に対して最初に R(x)と S(x)
を計算します.それから,R(x)と S(x)の間の Y軸を大域変数 dblint yの値で分割し,Y軸に沿って
積分を Simpson則を用いて計算します.
それから,X軸に沿った積分を函数の値を Yの積分で Simpson則を用いて計算します. この手順

は色々な理由で数値的に不安定であるが,それなりに速いものです.
但し,高周波成分を持った函数や特異点 (領域に極や分岐点)を持つ函数に対する適用は避けて下

さい.
Y軸方向の積分は R(x)と S(x)がどれだけ離れているかに依存し,距離 S(X)-R(X)が Xで急速

に変化すれば,Y軸方向の積分で大きな誤差が発生するかもしれません.
函数値は保存されない為,その函数の計算に時間がかかるものであれば,何かを変更する度に再

計算する羽目になるので,その分時間がかかります.

dblint函数に影響を与える大域変数¶ ³
変数名 初期値 概要

dblint x 10 X軸方向の分割数
dblint y 10 Y軸方向の分割数µ ´
二重積分を行う函数 dbiint函数は二つの大域変数 dblint x,dblint yを持っています.それらは

X,Yの区間の分割数を定め,2*dblint x+1点が X方向に計算され,Y方向は 2*dblint y+1点となり
ます.これらの変数は勿論, 互いに独立して変更する事か可能です.

5.6.6 antidパッケージ

antidパッケージには antid,antidiffと nonzeroandfreeofといった函数が含まれたパッケージです.

antidiffと antid¶ ³
antid(〈式 〉 ,〈x〉 ,〈u(x)〉 )
antidiff(〈式 〉 ,〈x〉 ,〈u(x)〉 )
nonzeroandfreeof(〈x〉,〈y〉)µ ´

antid函数は与式の不定積分を行います.antidの返却値は二成分のリストで, このリスト Lとす
る時,L[1]+’integrate(L[2],x)が与式の不定積分となります. ここで,函数 〈u(x)〉は未知函数でも構
いません.

(%i8) load(antid);

(%o8) /usr/local/share/maxima/5.9.2/share/integration/antid.mac

(%i9) antid(sin(3*x+1),x,3*x+1);

cos(3 x + 1)

(%o9) [- ------------, 0]

3

(%i10) antid(sin(u(x)+1),x,u(x));

(%o10) [0, sin(u(x) + 1)]

antidiffは内部でantid函数を用いた函数で,antid函数で不定積分した結果をL[1]+’integrate(L[2],x)
の形式に変換て表示します.

(%i11) antidiff(sin(3*x+1),x,3*x+1);

cos(3 x + 1)

(%o11) - ------------

3

(%i12) antidiff(sin(u(x)+1),x,u(x));

/

[

(%o12) I sin(u(x) + 1) dx

]

/

函数 nonzeroandfreeofは述語函数で,〈y〉が零でなく, 〈x〉が 〈y〉に含まれない場合に trueを返す
函数です.この述語函数を用いて,antidや antidiffの規則を定めています.

5.7 常微分方程式

5.7.1 常微分方程式の扱い

Maximaでの常微分方程式の記述は通常の方程式と同様に,演算子=を挟んで左右に式が記述さ
れる形となりますが,式中に名詞型の微分’diffが含まれるものです.

Maximaでは常微分方程式の一般解を計算する函数に ode2函数と desolve函数があります.ここ
で,desolve函数で扱える常微分方程式は,未知函数がどの変数に依存するものかを明示的に記述し
たものでなければなりません.
例えば,次の書式は desolve函数にとっては正確な書式ではありません.何故なら, 函数 fと函数

gが何の変数であるかが明確でないからです.

’diff(f,x,2)=sin(x)+’diff(g,x);

’diff(f,x)+x^2-f=2*’diff(g,x,2);

desolve函数の場合,常微分方程式の未知函数が何の函数であるかを下記の様に明確に記述しな
ければなりません.

’diff(f(x),x,2)=sin(x)+’diff(g(x),x);

’diff(f(x),x)+x^2-f(x)=2*’diff(g(x),x,2);

desolve函数に対し,ode2函数は xˆ 2*’diff(y,x)+3*y*x=sin(x)/x の様に函数と変数との関係を明
記しなくても構いません.
微分方程式の初期値問題を解く場合,desolve函数を用いる場合と ode2函数を用いる場合で解き

方の手順が異なります.
最初に,desolve函数を用いて初期値問題を解く場合,atvalue函数を用いて初期値を函数の属性と

して与えます.ここで atvalue函数による初期値設定は desolve函数で微分方程式を処理する前に
行わなければ意味がありません.

atvalue函数は函数 fが一変数の場合, atvalue(f(x),x=pt,val) の様に函数 f(x) に対する x=ptで
の値 valを設定します.

fが多変数の場合, atvalue(f(x,y,· · ·),[x1=p1,x2=p2,· · ·],[val1,val2,· · ·]) の様に境界と対応する
値をリストで与えます.
では,簡単な例を解いてみましょう.

(%i66) eq1:’diff(y(x),x,2)+2*x=sin(x);

2

d

(%o66) --- (y(x)) + 2 x = sin(x)

2

dx

(%i67) atvalue(’diff(y(x),x),x=0,0);

(%o67) 0

(%i68) atvalue(y(x),x=0,0);

(%o68) 0

(%i69) desolve([eq1],[y(x)]);

3

x

(%o69) y(x) = - sin(x) - -- + x

3

この例では,常微分方程式 dy(x)
dx + 2x = sin (x) を初期条件 [y(0) = 0, dy(x)

dx |0 = 0] で解いてい
ます.
次に,ode2函数を用いる場合では,与えられた一般解と初期条件から特殊解を計算する bc函数,ic1

函数や ic2函数を併用します.

(%i14) x^2*’diff(y(x),x)+3*y(x)*x=sin(x)/x;

2 d sin(x)

(%o14) x (-- (y(x))) + 3 x y(x) = ------

dx x

(%i15) ode2(%,y(x),x);

%c - cos(x)

(%o15) y(x) = -----------

3

x

(%i16) ic1(%o15,x=%pi,y(%pi)=0);

cos(x) + 1

(%o16) y(x) = - ----------

3

x

この例では,微分方程式 x2 dy
dx + 3xy = sin(x)

x を ode2函数を用いてその一般解を求め, ic1函数か
ら, x = π の場合に y が 0となる条件で特殊解を求めています.

(%i91) ’diff(y,x,2) + y*’diff(y,x)^3 = 0;

2

d y dy 3

(%o91) --- + y (--) = 0

2 dx

dx

(%i92) ode2(%,y,x);

3

y + 6 %k1 y

(%o92) ------------ = x + %k2

6

(%i93) ratsimp(ic2(%o92,x=0,y=0,’diff(y,x)=2));

3

2 y - 3 y

(%o93) - ---------- = x

6

(%i94) bc2(%o92,x=0,y=1,x=1,y=3);

3

y - 10 y 3

(%o94) --------- = x - -

6 2

一般解を求める函数¶ ³
desolve([〈方程式1〉 ,· · ·,〈方程式n〉], [〈変数1〉 ,· · ·,〈変数n〉 ])
ode2(〈常微分方程式 〉 ,〈従属変数 〉 , 〈独立変数 〉 )µ ´

desolve函数は,常微分方程式系を与えられた変数に対して解きます.
初期値は desolve函数を呼出す前に atvalue函数で与えなければなりません. desolveの引数とし

ては 〈方程式i〉で構成されるリストと従属変数 〈変数1〉 ,· · · 〈変数n〉のリストを指定します.desolve
函数では,求める函数と変数の関連性を明確に指定しなければなりません.
解はリストの形式で返却されますが,解を得られなかった場合,desolve函数は false を返します.

(%i1) ’diff(f(x),x)=’diff(g(x),x)+sin(x);

d d

(%o1) -- (f(x)) = -- (g(x)) + sin(x)

dx dx

(%i2) ’diff(g(x),x,2)=’diff(f(x),x)-cos(x);

2

d d

(%o2) --- (g(x)) = -- (f(x)) - cos(x)

2 dx

dx

(%i3) atvalue(’diff(g(x),x),x=0,a);

(%o3) a

(%i4) atvalue(f(x),x=0,1);

(%o4) 1

(%i5) desolve([%o1,%o2],[f(x),g(x)]);

x x

(%o5) [f(x) = a %e - a + 1, g(x) = cos(x) + a %e - a + g(0) - 1]

(%i6) [%o1,%o2],%o5,diff;

x x x x

(%o6) [a %e = a %e , a %e - cos(x) = a %e - cos(x)]

この例で,[%o1,%o2],%o5,diff;は ev函数による評価の一つの形態です. 詳細は 1.8.1の小節を参
照して下さい.この場合は式 [%o1,%o2]の f(x)と g(x)を代入し,%o1と%o2に含まれる微分の名詞
型を評価させ,desolveによる結果の検証を行っています.この評価の結果%o6で=の両辺の式が等
しい事から解が正しく求められている事が判ります.

ode2函数は 3個の引数を取ります.最初の 〈常微分方程式 〉 は一階, 又は二階の常微分方程式を
与えます.尚,常微分方程式の右側 (rhs(exp))が 0ならば, 左側のみを与えるだけでも構いません.
第二引数には 〈従属変数 〉, 最後の引数が 〈独立変数 〉 となります.
求解に成功すると,従属変数に対する陽的な解,或いは陰的な解の何れかを返します.ここで,%c

は一階の方程式の定数,%k1と%k2は二階の方程式の定数を表記する為に用いられます.
ode2函数が何らかの理由で解が得られなかった場合,エラーメッセージの表示等の後に falseを

返します.
現在,一階常微分方程式向けに実装され,検証されている解法は,線型,分離法, 厳密 (積分因子が

多分要求されます),同次,bernoulli方程式,そして一般化同次法があります.
二階の常微分方程式に対しては,定数係数,厳密,定数係数に変換可能な非定数係数を持つ線型同

次方程式,Euler又は同次元方程式,仮想変位法, そして変形分離で解ける二つの独立な一階の線型
な方程式に縮約可能となる方程式を含まないものがあります.
常微分方程式を解く手順では,幾つかの変数は純粋に情報的な目的,methodが記述する解法の集

合です.例えば,linear,intfactorが記述する積分因子を用い, odeindexは Bernoulli法や一般化同次
法の添字を記述し,ypは仮想変位による特殊な解法を記述しています..

境界値問題を解く函数¶ ³
bc2(〈一般解 〉 ,〈xの値1〉, 〈yの値1〉, 〈xの値2〉 ,〈yの値2〉 )
ic1(〈一般解 〉 ,〈xの値 〉 , 〈yの値 〉)
ic2(〈一般解 〉 ,〈xの値 〉 , 〈yの値 〉,〈yの微分値 〉)µ ´

bc2函数は,二階の微分方程式の境界条件問題を解きます. ここで 〈一般解 〉 は ode2函数等で計
算した微分方程式の一般解です. 二階の方程式の一般解では定数が二つ現われる為,特殊解を求め
る為, 異なった二点での連立方程式を解く必要があります.
その為, 〈xの値1〉 と 〈yの値1〉 が一つの点での値 〈xの値2〉 と 〈yの値2〉 がもう一つの別の点で

の値を定めます.ここで値の与え方は一般解の変数を xと yとすると, 〈xの値1〉 と 〈y の値1〉 を
x=x0や y=y0の様に 〈対応する変数 〉 =〈境界値 〉 の書式で記述します.

ic1函数は初期値問題 (ivp)と境界値問題 (bvp)を解く為の ode2パッケージに含まれるプログラ
ムです.〈一般解 〉 は ode2等で計算した微分方程式の一般解です.そして,後の二つが境界条件を与
えます.一般解の変数を x, y とすると, 〈xの値1〉 と 〈yの値1〉 は x = x0 や y = y0 の様に 〈対応
する変数 〉 =〈境界値 〉 の書式になります.

ic2函数は二階の常微分方程式の境界値問題を解く函数です. 〈一般解 〉 は ode函数等で計算し
た微分方程式の一般解となり, 後の二つがその境界条件を与えます.
一般解の変数を x , y とすると, 〈xの値 〉 と 〈yの値 〉 は, x が 〈xの値 〉 の場合, y の値が 〈yの

値 〉 で y を x で微分した函数の, x = 〈xの値 〉 での値が 〈yの微分値 〉 となります.

演算子

A
and, 47

D
do, 49

E
else, 49
elseif, 49

F
for, 49
from, 49

I
if, 49
if文で利用可能な演算子, 177
in, 178

N
next, 49, 178
not, 47

O
or, 47

S
step, 49, 178

T
then, 49
thru, 49, 179

U
unless, 49, 179

W
while, 49, 179

記号

<, 47
<=, 47
>, 47
>=, 47

*, 44
**, 44
+, 44
-, 44
., 44
/, 44
:, 48
::, 48
::=, 48
:=, 48, 183
=, 47, 238
#, 47
ˆ, 44
ˆˆ, 44

型

A
any, 25, 27, 39
any check, 25

B
big, 25
bigfloat, 74
bignum, 25, 74
bool, 25
boolean, 25

C
clause, 39
complex, 25

E
expr, 39

F
fixnum, 25, 74
fixp, 25
float, 25, 74
floatnum, 25

flonum, 25
I

integer, 25
L

list, 25
listp, 25

N
none, 25
number, 25

R
rat, 25
rational, 25
real, 25

函数

:lisp, 8, 71
?round, 78
?truncate, 78
A

acos, 230
acosh, 230
acot, 230
acoth, 230
acsc, 230
acsch, 230
activate, 13
addcol, 164
addrow, 164
adjoint, 168
algsys, 244
alias, 227
allroots, 241
and, 12
antid, 276
antidiff, 276
append, 143
appendfile, 200
apply, 189
apply1, 52
apply2, 52
approps, 227

args, 113
array, 152
arrayapply, 155
arrayinfo, 151
asec, 230
asech, 230
asin, 230
asinh, 230
askinteger, 138
asksin, 139
assume, 12, 14
at, 29
atan, 230
atan2, 230
atanh, 230
atom, 142
atomgrad, 31
atvalue, 29
augcoefmatrix, 158

B
batch, 198
batchload, 198
batcon, 198
bc2, 281
bezout, 98
bfloat, 76
bfloatp, 76
bigfloat, 74
block, 49, 61, 176
bothcoef, 94
break, 180, 209
buildq, 186

C
cabs, 76
carg, 76
catch, 180
changevar, 262
charpoly, 171
closefile, 200
coeff, 94

coefmatrix, 158
col, 164
collapse, 208
columvector, 172
combine, 97
compfile, 191
compile, 191
compile file, 191
conj, 172
conjugate, 172
constantp, 76
content, 100
copylist, 143
copymatrix, 164
cos, 230
cosh, 230
cot, 230
coth, 230
csc, 230
csch, 230

D
dblint, 275
deactivate, 13
declare, 6, 15, 18, 62
declare translated, 195
define, 183
define variable, 22, 25
define variable函数の処理手順, 27
defint, 270
defmatch, 56
defrule, 53
delete, 143
delta, 268
demoivre, 134
denom, 97
depends, 31, 261
derivdegree, 255
derivlist(ev函数の引数), 62, 66
desolve, 279
determinant, 170

detout(ev函数の引数), 62
diagmatrix, 158
diagmatrixp, 163
diff, 253
diff(ev函数の引数), 62
dipslay, 217
disp, 217
dispform, 44, 110
dispfun, 185
displate, 115
disprule, 50
dispterms, 218
distrib, 135
divide, 104
do, 49, 178
ldisplay, 217

E
derivlist, 66
echelon, 158
econs, 143
eigenvalues, 173
eigenvectors, 173
eivals, 173
eivects, 173
eliminate, 98
ematrix, 158
endcons, 143
entermatrix, 158
entier, 76
equal, 68
erf, 261
errcatch, 181
error, 181
errormsg, 181
ev, 21, 60
eval, 69
eval(ev函数の引数), 62
evenp, 76
ev函数で利用可能な引数, 62
expand, 64, 134

expand(ev函数の引数), 62, 64
exponentialize, 134
express, 258
ezgcd, 100

F
factcomb, 97
facts, 12, 19
facttimes, 97
feature, 19
featurep, 15, 20
Ffortran, 226
file search, 203
file type, 203
filename, 203
fillarray, 155
first, 144
fix, 76
floatnump, 76
forget, 12
freeof, 112
fullratsimp, 105
fullratsubst, 128
funcsolve, 248
fundef, 185
funmake, 194

G
Γ函数, 46
gcd, 100
gcde, 100
gcdex, 100
gcfactor, 100
genmatrix, 158
get, 24
gfactor, 100
gfactorsum, 100
go, 180
gradef, 31
gramschmidt, 174
grind, 217
gschmidt, 174

H
hipow, 95, 255
horner, 93

I
ic1, 281
ic2, 281
ident, 158
if, 49, 177
ilt, 268
imagpart, 76
infix, 41
inflagの影響を受ける函数, 142
innerproduct, 172
inpart, 118, 127, 129
inprod, 172
integerp, 76
integrate, 23, 259
intosum, 137
invert, 168
is, 68
isolate, 115
isqrt, 76

K
kill, 207, 209
killcontext, 13

L
labels, 213
lambda, 61, 184
laplace, 268
last, 144
ldefint, 270
ldisp, 217
length, 143
let, 55
letrat, 55
letrules, 50
letsimp, 55
lfreeof, 112
lhs, 238
limit, 23, 251, 261

linsolve, 243
listarray, 151
listofvars, 113
listp, 142
load, 198
loadfile, 198
local, 195
local(ev函数の引数), 62, 66
logcontrcat, 236
lopow, 95
lratsubst, 128

M
kill, 42
mainvar, 92
make array, 152
make art q, 152
map, 148
mapatom, 148
matchdeclare, 54
matchfix, 41
matrix, 156
matrixmap, 164
matrixp, 163
mattrace, 167
max, 76
member, 142
mfuncall, 72
min, 76
minor, 166
mod, 104
mode declare, 25, 190
mode identity, 25
multthru, 135

N
nary, 41
ncharpoly, 171
newcontext, 13
newdet, 170
noeval(ev函数の引数), 62
nofix, 41

nonscalarp, 163
nonzeroandfreeof, 276
not, 12
nouns(ev函数の引数), 62
nroots, 243
nterms, 95
num, 97
numberp, 76
numer(ev函数の引数), 62
numerval, 140

O
oddp, 76
ode2, 279
optimize, 125
or, 12
ordergrear, 8
ordergreatp, 9
orderless, 8
orderlessp, 9
print, 217

P
parmanent, 170
part, 118, 127, 129
partition, 117
pickapart, 118
playback, 219
ploarform, 237
plog, 237
postfix, 41
powers, 95
prefix, 41
printprops, 35, 54
product, 23, 120
properties, 24, 33, 62
propvars, 33
put, 24

Q
qput, 24, 27
quanc8, 261, 273
quit, 228

quotient, 104
R

radcan, 137
random, 76
rank, 167
rat, 82
ratcoef, 94
ratdenom, 97
ratdiff, 97
ratdisrep, 82
ratexpand, 105
ratnumer, 97
ratnump, 94
ratp, 94
ratsimp, 68, 105
ratsubst, 128
ratvars, 82, 96
ratweight, 96
read, 198
readonly, 198
realpart, 76
realroots, 241
rearray, 155
rectform, 117
rem, 33
remainder, 104
remarray, 155
remfunction, 186
remlet, 58
remove, 33, 42
remrule, 58
remvalue, 113
reset, 207
residue, 267
rest, 144
resultant, 98
return, 180
reveal, 218
reverse, 143
rhs, 238

risch, 259
risch(ev函数の引数), 62, 66
romberg, 261, 273
room, 224
rootscontract, 97
row, 164

S
save, 201
scalarp, 163
scanmap, 149
scsimp, 138
sec, 230
sech, 230
setelmx, 164
setup autoload, 206
showvars, 96
similaritytransform, 173
simtran, 173
sin, 230
sinh, 230
solve, 246
sqrt, 76
status, 223
stringout, 199
sublis, 129
sublist, 144
submatrix, 166
subst, 127
substinpart, 130
substpart, 130, 144
sum, 23, 120
sumcontract, 136
supcontext, 13
system, 225

T
tan, 230
tanh, 230
tcl output, 218
tellrat, 89
tellsimp, 56

tex, 226
throw, 180
tldefint, 270
tlimit, 252
to-maxima, 70
to lisp, 70
totaldisrep, 82
translate, 33
transplate, 190
transpose, 167
triangularize, 167
trigexpand, 233
trigrat, 233
trigreduce, 233
trigsimp, 233

U
ueivects, 174
uniteigenvectors, 174
unitvector, 174
unknown, 134
unorder, 8
untrllrat, 91
use fast arrays, 152
uvect, 174

W
with stdout, 200
writefile, 200

Z
zeromatrix, 158

記号

%ith, 213
記号

！ , 46
！！, 46
>m, 6
Am, 6
’, 69
”, 69
?, 71, 220
%i, 209

%o, 209
属性

A
additive, 23
algebraic, 39
alphabetic, 5, 20, 112
analitic, 24
argpos, 39
assign, 27, 28
assign-mode-check, 27
atomgrad, 35
atvalue, 35

C
commutative, 23
complex, 22
constant, 20

D
decreasing, 24
dependency, 31

E
english, 39
even, 22
evenfun, 24
evflag, 24, 62
evflag属性をデフォルトで持つ大域変
数, 62

evfun, 24, 61–63
evfun函数の作用の順番, 64
evfun属性を持つ函数, 63

G
gradef, 31, 35

I
imaginary, 22
increasing, 24
integer, 22
irrational, 22

L
lassociative, 23
linear, 23
logical, 39

lpos, 39
M

mainvar, 20
matchdeclare, 35, 54
mode, 25
mode check errorp, 26
mode check warnp, 26
mode checkp, 26
multiplicative, 23

N
noninteger, 22
nonscalar, 20
noun, 24

O
odd, 22
oddfun, 24
outative, 23

P
pos, 39
posfun, 24

R
rassociative, 23
rational, 22
real, 22
rpos, 39

S
scalar, 20
special, 20, 27
symmetric, 23

T
transfun, 33

U
untyped, 39

V
value check, 27

大域変数

%, 211
%e to numlog, 235
%edispflag, 215
%num list, 246

%%, 211
%emode, 132
%enumer, 65, 132
A

absboxchar, 215
algebraic, 88
algepsilon, 246
algexact, 246
aliases, 227
arrays, 154
assume pos, 16
assumescalar, 161

B
backsubst, 239
backtrace, 180
batchkill, 197
batcount, 197
berlefact, 104
breakup, 248

C
cauchysum, 121
compgrind, 192
context, 13
contexts, 13
current let rule , 58

D
dblint x, 275
dblint y, 275
debugmode, 227
demoivre, 132
dependencies, 31, 32, 261
derivabbrev, 253
derivsubst, 254
detout, 161
dispflag, 180
display format internal, 215
display2d, 214, 215, 259
doallmxops, 162
domain, 74
domxexpt, 162



索 引 293

domxmxops, 162
domxnct, 162
dontfactor, 104
doscmxops, 162
doscmxplus, 162
dot0nscsimp, 45
dot0simp, 45
dot1simp, 45
dotassoc, 45
dotconstrules, 45
dotdistrib, 45
dotexptsimp, 45, 256
dotident, 45
dotsassoc, 256
dotscrules, 45, 256

E
erfflag, 261
error size, 220
error syms, 220
errorfun, 180
expon, 64, 132
expop, 64, 132
exptdispflag, 215
exptisolate, 116
exptsubst, 127

F
facexpand, 104
factorflag, 104
file search demo, 197
file search LISP, 197
file search Maxima, 197
file string print, 197
float2bf, 74
fppintprec, 74
fpprec, 74
functions, 183

G
gcd, 100
genindex, 121
gensumnum, 121

globalsolve, 239
gradefs, 31, 32

H
halfangles, 232
hermitianmatrix, 171

I
ibase, 215
inchar, 211
inflag, 114, 130, 141
infolists, 222
integrate use rootsof, 264
integration constant counter, 259
intfaclim, 104
isolate, 116

K
keepfloat, 85
knowneigvals, 171
knowneigvects, 171

L
lasttime, 225
leftmatrix, 171
let rule packages, 58
letrat, 58
lhospitallim, 251
ligarc, 235
limsubst, 251
linechar, 211
linel, 215
linenum, 211
linsolve params, 244
linsolvewarn, 244
listarith, 141
listconstvars, 114
listdummyvars, 114
listeigvals, 171
listeigvects, 171
lmxchar, 162
loadprint, 197
logabs, 235
logconcoeffp, 235

logexpand, 235
lognegint, 235
lognumer, 235
logsimp, 235

M
m1pbranch, 74
macroexpansion, 188
maperror, 147
matrix element add, 163
matrix element mult, 163
matrix element transpose, 163
maxapplydepth, 58
maxapplyheight, 58
maxnegex, 64, 134
maxpogex, 134
maxposex, 64
modulus, 88
multiplicities, 239
myoptions, 227

N
newfac, 104
nolabels, 211
nondiagonalizable, 171

O
obase, 215
opsubst, 127
optimprefix, 126
optionset, 227
outchar, 211

P
packagefile, 197
partswitch, 119
pfeformat, 215
piece, 119
polyfactor, 242
prederror, 69, 180
prodhack, 120
prompt, 209, 211, 219
props, 33

R

radexpand, 74
radsubstflag, 128
rataigdenom, 85
ratdenomdivide, 87
ratepsilon, 85
ratexpand, 87
ratfac, 87
ratmx, 161
ratprint, 85
ratsimpexpons, 87
ratweights, 87
ratwtlvl, 87
readonly, 246
resultant, 100
rightmatrix, 171
rmxchar, 162
rombergabs, 274
rombergit, 274
rombergmin, 274
rombergtol, 274
rootsconmode, 87
rootsepsilon, 242
rpgrammode, 239

S
savedef, 192
savefactors, 104
scalarmatrix, 161
showtime, 225
simpsum, 121
solve inconsistent error, 248
solvedecomposes, 248
solveexplicit, 248
solvefactors, 248
solvenullwarn, 248
solveradcan, 248
solvetrigwarn, 248
sparse, 161
stardisp, 215
sublis apply lambda, 129
sumexpand, 121

sumhack, 121
T

tlimswitch, 251, 272
tr array as ref, 193
tr bound function applyp, 194
tr file tty messagesp, 194
tr float can branch complex, 194
tr function call default, 193
tr numer, 193
tr optimize max loop, 194
tr semicompile, 193
tr warn bad function calls, 194
tr warn fexpr, 193
tr warn meval, 193
tr warn mode, 193
tr warn undeclared, 193
tr warn undefined variable, 193
transcomile, 193
translate, 192
translate fast arrays, 193
transrun, 192
trigexpandplus, 232
trigexpandtimes, 232
triginverses, 232
trigsign, 232
ttyoff, 215

U
undeclaredwarn, 192
undeclearewarnの設定項目, 192

V
values, 112

定数

%e, 65, 75
%gamma, 75
%phi, 75
%pi, 75
false, 75
inf, 75, 261
infinity, 75, 261
minf, 75

true, 75
zeroa, 75
zerob, 75

文脈

global, 12
initial, 12

え

演算子

前置表現の～, 36
内挿表現の～, 36
～の型, 39
～の属性, 36
～の束縛力, 37
～の束縛力 (bp), 37
～の左束縛力 (lbp), 37
～の右束縛力 (rbp), 37
後置表現の～, 36
無引数の～, 36

か

解の自動代入, 239
き

規則, 50
～の削除, 58
～の定義, 53
～の表示, 50

こ

項順序, 7
し

辞書式順序, 7
(Maximaの)次数リスト, 7
述語, 12
主変数, 20
主変数 (mainvar), 7

そ

属性, 18
～値, 18
～値の削除, 33
～を追加, 19
アトムの属性, 20
函数の属性, 24

数値属性, 22
ち

重複度のリスト, 240
な

並び方 (パターン), 52
並びの照合 (パターンマッチング), 52

ふ

文脈, 12
～の切替, 15
～の切替, 13
～の生成, 13
～を無効にする, 13
～を有効にする, 13
親文脈, 13
～の生成, 13
～の削除, 13

へ

変数, 5
～順序, 6

ほ

方程式, 238
も

文字, 5
れ

連立方程式, 238

え

S式, 70

か

可換積, 44

し

CRE表現, 81
順序

～の定義, 2
順序, 2
～集合, 2
逆辞書式順序, 3
項順序, 2
斉次逆辞書式順序, 4

斉次辞書式順序, 3
辞書式順序, 3
全順序, 2
全順序集合, 2

せ

正準表現, 4
多変数多項式の正準表現, 5

全微分, 253

た

多項式と単項式の一般表現, 80
単変数多項式の CRE表現, 81

ひ

非可換積, 44
む

無平方, 103
無平方因子分解, 103


